
 Ú vod 
 

 Skriptá  Zá klady programovania – Turbo Pascal sú  sumarizá ciou poznatkov z klasického kurzu 

programovania. Nekladú  si za cieľ podrobne vyložiť všetky pojmy. Tiež vý klad a poradie tém má  klasické 

usporiadanie, od najjednoduchšieho k zložitejšiemu, od zá kladný ch pojmov k zložitejším štruktú ram. Vý klad 

a príklady sa nevyhý bajú  ú lohá m s matematický m zá kladom, ako sa o to pokú šajú  modernejšie učebnice 

programovania. Domnievam sa, že ak má te odmietavý  vzťah k matematike, nemôže vá s programovanie 

a algoritmizá cia zaujať. Musíte byť trochu matematik, aby ste presne a jednoznačne vedeli formulovať 

a zapísať svoje myšlienky a trochu umelec, aby ste doká zali abstraktne myslieť a vytvoriť si svoje originá lne 

riešenie daného problému. Zá kladné „ťahy štetcom“ sa možno naučíte prá ve z tý chto skrípt. 

 Obsahom vý kladu sú  zá kladné algoritmické konštrukcie sekvencia, vetvenie a cyklus, a príkazy 

Turbo Pascalu, umožňujú ce zrealizovať tieto konštrukcie v programe. Preto sa v príkladoch používajú  len 

štandardné ú dajové typy integer, real, boolean a char a jediný  štruktú rovaný  typ string. Neklá dol som si za 

cieľ o každom zavedenom pojme uviesť všetko, skôr uká zať jeho najtypickejšie použitie, preto sa miestami 

vedome dopú šťam nepresností alebo skôr neú plností, mali by však byť skutočne nepodstatné; možno sa 

dopú šťam aj chý b, ale tý ch určite nevedome. Tiež si nekladiem ná roky na autorstvo použitý ch príkladov. 

Vybral som ich z veľkého množstva príkladov, s ktorý mi som sa stretol za viac ako desať rokov praxe. 

O originá lnosti riešenia ťažko hovoriť, keďže ide o „triviá lne“ príklady. Tý mto spĺňam aj požiadavku 

študentky Maji S., ktorá  ma raz poprosila o vydanie Triviá lností pá na procesora Pirošku (premenovanie je 

tiež dielom študenta, tentoraz Tomá ša Z.). 

 Skriptá , okrem vý kladu zá kladný ch štruktú r a príkazov, obsahujú  aj veľké množstvo riešený ch a 

neriešený ch ú loh. Najprv by ste sa sami mali potrá piť nad ú lohou, a až potom siahnuť po riešení v skriptá ch. 

Riešenia si treba ozrejmiť, tý m sa naučíte minimá lne „čítať“ programy a možno aj nové riešenie daného 

problému. Ideá lne by bolo, keby ste si pri štú diu programu klá dli a najmä odpovedali na otá zky charakteru: 

Počíta program sprá vne aj pre takéto hodnoty?, Musí to byť zapísané prá ve takto?, Počiatočná  hodnota 

premennej musí byť prá ve taká to?, Čo sa stane, keď to napíšem takto?, Čo keď zamením poradie tý chto 

príkazov? A ak tento príkaz nahradím tý mto? atď. Svoje tvrdenia si vždy overte priamo v TP odladením 

zmeneného programu.  

 Na zá ver vá m prajem veľa originá lnych riešení, najprv sa však treba naučiť „v pote tvá re zopá r 

zá kladný ch ťahov štetcom“. 

Algoritmus, vlastnosti algoritmu, formy zápisu algoritmu, algoritmické 
konš trukcie 

Ak sa začnete zaujímať o počítače a najmä programovanie, určite narazíte na slovo algoritmus. 

Algoritmus je ná vod, postup, ako vyriešiť určitý  problém. Tý m problémom môže byť vý mena pneumatiky 

pri defekte, zostavenie mlynčeka na mletie mäsa tak, aby to, čo vznikne zomletím, bolo ešte ďalej 

použiteľné, bezpečné prejdenie na druhú  stranu cesty, „vzorné“ umytie zubov atď. Z tý chto príkladov by 

malo byť zrejmé, že algoritmy ná s učia od narodenia a na každom kroku („toto sprav takto a toto zas 

takto...“). Sná ď v každej domá cnosti sa nachá dza zbierka algoritmov na prípravu jedá l – kuchá rska kniha. 

Každý  ná vod na obsluhu, ktorý  by ste mali dostať pri kú pe nového vý robku, je vlastne algoritmom. 

Vo všetký ch tý chto príkladoch je vykoná vateľom algoritmu človek. Pri zá pise algoritmu určeného pre 

človeka si môžeme dovoliť použiť spojenia „krá tko povariť“, „pritiahnuť až na doraz“, „vyladiť stanicu“ 

a pod. Ak je však vykoná vateľom algoritmu stroj, napr. počítač, sotva by porozumel uvedený m príkazom. 

Preto aj my prejdeme na odbornejšiu terminológiu (chceme sa predsa venovať počítačom a nie vareniu). 

Algoritmus je konečná  postupnosť elementá rnych príkazov, vykonanie ktorý ch, pre prípustné vstupné ú daje, 

po konečnom počte krokov (akcií) vedie mechanicky k získaniu korektný ch vý stupný ch ú dajov. Strašná  veta, 

skú sme ju „rozpitvať“. Konečná  postupnosť elementá rnych príkazov znamená , že ide o príkazy, ktoré sú  

zapísané v presne stanovenom poradí, je ich konečný  počet a pre vykoná vateľa algoritmu sú  elementá rne, t.j. 

každý  z príkazov pozná  a vie ho vykonať. Vykoná vanie algoritmu je proces, do ktorého vo všeobecnosti 

vstupujú  na začiatku prípustné vstupné ú daje a ich spracovaním získavame nové – vý stupné ú daje (na čo by 

sme iná č „trá pili“ počítač). Celý  proces samozrejme nesmie trvať nekonečne dlho (to by sme sa 

nedopracovali k žiadnym vý sledkom) a, keďže počítač je síce usilovný  ale hlupá k1, celý  proces musí 

prebiehať mechanicky - bez rozmý šľania. Pod korektný mi vý stupný mi ú dajmi rozumieme zmysluplné 

vý sledky adekvá tne vstupný m ú dajom; ak napríklad opakovane použijeme rovnaké vstupné ú daje, musíme 

dostať tie isté vý sledky.  

Z tejto analý zy vyplý vajú  aj vlastnosti algoritmu, z ktorý ch by sme uviedli len niekoľko: 

1. hromadnosť – algoritmus slú ži na riešenie celej skupiny ú loh určitého typy2; aby sme dodržali tú to 

vlastnosť, musí byť algoritmus napísaný  všeobecne (univerzá lne). Z matematiky vieme, že zovšeobecniť 

zá pis ná m umožňujú  premenné. Už na zá kladnej škole ná s učili, že napríklad obsah obdĺžnika sa 

vypočíta S = a.b, kde a a b sú  strany obdĺžnika a nie S = 5.7 pre nejaký  konkrétny obdĺžnik. Preto prvý  

zá pis môže byť príkazom algoritmu, druhý  nie. Zaujímavé je, že algoritmus musí byť napísaný  

všeobecne ale jeho realizá cia vždy prebieha s konkrétnymi ú dajmi; premenné nadobú dajú  konkrétne 

hodnoty v zá vislosti od zadaný ch vstupný ch hodnôt. Ak budú  vstupné ú daje - strany obdĺžnika a = 5 

a b = 7, tak konkrétny vý počet bude S = 5.7 a po ňom premenná  S nadobudne hodnotu 35. Hromadnosť 

vlastne znamená , že algoritmus slú ži na riešenie určitej skupiny konkrétnych ú loh, ktoré sa líšia len 

vstupný mi hodnotami. 

                                                 
1 Jeden mú dry pá n raz povedal: „Počítač je usilovný  hlupá k na triedenie jednotiek a nú l“. Dú fam, že s ním sú hlasíte. 
2 My predsa nevieme vymeniť len tú  konkrétnu pneumatiku na aute, na ktorom ná s to učili, ale u daného typu auta 

kedykoľvek ktorú koľvek pneumatiku; alebo prejsť cez cestu by sme mali vedieť bezpečne cez ktorú koľvek cestu, 

podobnú  tej, na ktorej ná s to učili; zuby si tiež vieme umyť, či ich má me 32 alebo menej, atď.  

2. deterministickosť (jednoznačnosť) – jednotlivé kroky (akcie) a ich poradie je presne a jednoznačne 

určené. Deterministickosť zaručuje, že pri opätovnom použití algoritmu pre tie isté vstupné ú daje 

dostaneme rovnaký  vý sledok. Umožňuje tiež realizá ciu algoritmu počítačom (počítač sa nemusí sá m 

rozhodovať). 

Ď alšie vlastnosti sú  konečnosť, rezultatívnosť, elementá rnosť atď. 

Formy zá pisu algoritmu: 

• zá pis v jazyku vý vojový ch diagramov (JVD) používa štá tnou normou stanovené značky, prehľadne 

(graficky) vyjadruje tok riadenia a dá t; ide o staršiu formu zá pisu algoritmu. 

Príklad: Algoritmus na ná jdenie najväčšieho z troch čísel vo forme vý vojového diagramu. 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Algoritmus sa vykoná va zhora nadol. Vetvou „+“ sa pôjde, keď bude podmienka splnená  a vetvou „ -“, 

keď podmienka nebude splnená . Symbol „:=“ čítame „priraď“. 

• slovný  zá pis používa tzv. kľú čové (vyhradené) slová , napr. ak – tak – inak, čítaj, začiatok a pod.  

Príklad: Slovný  zá pis algoritmu na ná jdenie najväčšieho z troch čísel. 

 začiatok 

 čítaj ( A , B , C ); 

 ak A > B 

        tak   MAX := A 

      inak MAX := B; 

 ak C > MAX 

      tak   MAX :=C; 

 píš ( MAX ); 

 koniec. 

začiatok 

vstup: 
A, B, C 

A > B 

MAX := A MAX := B 

C > MAX 

MAX := C 

vý stup: 
MAX 

koniec 

• zá pis štruktú rogramom1 – najnovšia grafická  forma zá pisu algoritmu. 

Príklad: Algoritmus na ná jdenie najväčšieho z troch čísel zapísaný  N-S diagramom. 

 

 

 

 

 

 

 

 

 

 

Položme si otá zku: Je každá  ú loha algoritmicky riešiteľná , t.j. existuje na každý  problém algoritmus, 

pomocou ktorého ho vieme vyriešiť? Intuitívne cítime, že asi nie. Na algoritmicky neriešiteľné ú lohy síce 

nenarazíme „na každom kroku“, ale predsa existujú . Podarilo sa napríklad doká zať tvrdenie: Neexistuje 

algoritmus, ktorý  by pre aký koľvek daný  program rozhodol (na zá klade zdrojového tvaru programu 

a vstupný ch ú dajov), či tento program skončí.2 Zatiaľ sme neobjasnili slovo program. Veľmi jednoducho: 

program je algoritmus ú lohy zapísaný  v jazyku, ktorému rozumie počítač, v tzv. programovacom jazyku. 

Onedlho sa jedný m z programovacích jazykov začneme podrobnejšie zaoberať. 

Ak je ú loha algoritmicky riešiteľná , jej algoritmus možno vytvoriť kombiná ciou len troch 

algoritmický ch konštrukcií (Na prvý  pohľad aké jednoduché!). Algoritmické  konš trukcie sú : sekvencia, 

vetvenie a cyklus.  

Teraz prerušíme vý klad algoritmický ch konštrukcií a povieme si niečo o programovaní 

a programovacom jazyku Pascal.  

 

Programovanie, programovací jazyk Turbo Pascal 
 Tvorbu algoritmov nazý vame algoritmizá cia a tvorbu programov programovanie. Keďže program 

je „len“ algoritmus zapísaný  v programovacom jazyku, prioritnou činnosťou je algoritmizá cia. Má lokto však 

vydrží najprv „vyrobiť“ nedozerné množstvo algoritmov (naučiť sa algoritmizá ciu) a až potom začať 

programovať. Je to pochopiteľné, pretože až počítač pomocou programu mu dokonale „zhmotní“ to, čo 

vymyslel na algoritmizá cii. Preto aj my, aby sme si hneď overili, čo sme vymysleli, spojíme algoritmizá ciu 

a programovanie. Pri preberaní algoritmický ch konštrukcií si hneď uvedieme aj príkazy programovacieho 

jazyka, ktoré ná m umožňujú  danú  konštrukciu zrealizovať v programe. Budeme sa zaoberať zá kladmi 

programovacieho jazyka Turbo Pascal, implementá ciou jazyka Pascal, ktorý  navrhol v roku 1971 profesor 

                                                 
1 Tento ná zov sme zvolili pre podobné formy zá pisu algoritmu, ako sú  napríklad kopenogramy (autori Kofrá nek, 

Pecinovský  a Nová k) alebo N-S diagramy (autori Nassi a Schneiderman). 
2 Toto tvrdenie napríklad vedie k všeobecnejšiemu tvrdeniu: Neexistuje algoritmus, ktorý  rozhodne, či v programe je 

alebo nie je chyba. Už vieme, prečo je „programá torský  chlebík“ taký  ťažký . 

vstup: A , B , C 

A > B 
  
  

MAX := A    MAX := B 
 

C > MAX 

 
 MAX := C 
 

vý stup: MAX 

á no nie 

á no nie 

informatiky Niklaus Wirth1. Pascal bol vytvorený  na vyučovanie programovania a má  vlastnosti a kontrolné 

mechanizmy nú tiace, alebo aspoň umožňujú ce, tvoriť zrozumiteľné a prehľadné programy s čo najmenším 

počtom začiatočníckych chý b. 

 Program nie je len mechanický m prepisom algoritmu do programovacieho jazyka. Ký m v algoritme 

sa snažíme vystihnú ť podstatu riešenia problému a nezaoberá me sa podrobnosťami komuniká cie 

s užívateľom, program sa snažíme spraviť užívateľsky „prítulný “ (budeme hovoriť o užívateľskom 

komforte). Na to slú žia ďalšie, rozširujú ce príkazy. Tieto a ďalšie vlastnosti programovacieho jazyka Turbo 

Pascal si budeme postupne ozrejmovať pri konkrétnych programoch. 

 Pri prá ci v Turbo Pascale odporú čame: 

• v Options – Environment – v položke Preferences... v skupine Auto save zapnú ť voľbu Editor files 

a zapamätať ju položkou Save \...\ TURBO.TP; tá to voľba spôsobí, že pred prekladom programu sa tento 

automaticky najprv uloží na disk (ak sme sú bor s programom ešte nepomenovali, vyzve ná s vložiť ná zov 

sú boru), čo pri prípadnom použití tlačidla RESET neznamená  stratu programu (najprv však skú ste 

Ctrl+Break alebo v prostredí Windows 9x Ctrl+Alt+Del) 

• v Options – Compiler... zapnú ť v skupine Runtime errors všetky kontroly na chyby počas behu programu 

a zapamätať ich položkou Save \...\ TURBO.TP (jednotlivé kontroly preberieme podľa aktuá lnosti) 

• aj písanie nový ch programov začínať cez F3 Open a hneď vložiť ná zov sú boru, do ktorého má  byť nový  

program ukladaný  

• čo najviac používať kontextový  help: napísať kľú čové slovo, presunú ť pod neho kurzor a stlačiť Ctrl+F1 

• používať po označení textu pomocou myši alebo Shift a „kurzorový ch šípok“ na kopírovanie kombiná ciu 

Ctrl+C a K, premiestňovanie Ctrl+C a V a mazanie Ctrl+C a Y (alebo položky z ponuky Edit) 

• programy spú šťať pomocou Ctrl+F9. Ak sme program prá ve napísali alebo v ňom spravili zmenu, dôjde 

najprv k prekladu programu, pred ktorý m sa vykoná  aj kontrola sprá vnosti zá pisu (odhalia sa skomolené 

vyhradené slová , chý bajú ce bodkočiarky a pod.) a ú plnosti programu (odhalia sa nedeklarované 

premenné, nedovolené priradenia a pod.). Tá to kontrola neodhalí tzv. logické chyby (chyby v algoritme) 

alebo chyby, ktoré môžu nastať počas behu programu (zle zadaná  vstupná  hodnota, delenie nulou2 

a pod.) 

Pri písaní programu si programá tor volí ná zov programu, mená  použitý ch premenný ch, konštá nt, 

typov a iný ch objektov. Keďže norma Pascalu dovoľuje používať len písmená  anglickej abecedy, 

vo zvolený ch ná zvoch nesmú  byť použité diakritické znamienka (dĺžne, mäkčene a pod.). Tieto ná zvy sú  tzv. 

identifiká tory. Identifiká tor je postupnosť písmen (anglickej abecedy) a číslic začínajú ca písmenom; medzi 

písmená  je dodefinovaný  aj podčiarnik „_“. Identifiká tory sú  napríklad: CISLO, SUCET, obvod_obdlznika, 

ObsahStvorca, Premenna1, MENO, S, X, y, A, a, N5, z_13, ZNAK, text, OBSAHUJE, ... Identifiká tory nie 

sú  napríklad: 1A (nezačína písmenom), SUCET CISEL (použitá  medzera), premenná 1 (nedovolené 

písmeno á ), suma$ (nedovolený  znak $). Nerozlišujú  sa malé a veľké písmená , zá pisy sucet, Sucet, SuCeT, 

                                                 
1 Narodil sa v roku 1934, predná šal na Vysokej škole technickej a univerzite v Zürichu. 
2 Ak spú šťate Turbo Pascal na najnovších počítačoch Pentium, pri použití unitu Crt môže byť po spustení programu 

zahlá sená  chyba Error 200: Division by zero. Odstr á niť ju možno použitím unitu FDelay pred Crt (FDelay sa 

nenachá dza v štandardnej knižnici UNITS). 



 Zá klady programovania  - Turbo Pascal   7 

SUCET pomenú vajú  v Pascale tú  istú  premennú . Ná zov objektu je dobré voliť si tak, aby ná m napovedal 

o jeho ú lohe, veľmi dlhé ná zvy však nie sú  praktické. 

Iná  situá cia je pri znakový ch reťazcoch alebo skrá tene len reťazcoch. Reťazec je text uzavretý  

v apostrofoch. Napríklad: ’Sú čet čísel = ’, ’Zadaj tri celé čísla: ’, ’Najväčšie číslo je ’, ’Vlož svoje meno’. 

V reťazcoch sa rozlišujú  veľké a malé písmená , dovolené sú  diakritické znamienka, medzery, interpunkčné 

znamienka. Príkazom write možno reťazce zobraziť na obrazovke v tvare, ako sú  zapísané, uľahčujú  najmä 

komuniká ciu medzi užívateľom a programom. 

S identifiká tormi a reťazcami sa budeme v programoch „stretá vať na každom kroku“. Identifiká tory 

budeme, aspoň zo začiatku, písať veľký mi písmenami, aby sa čitateľ ľahšie zorientoval. A teraz jednoduchý  

program na zozná menie sa so štruktú rou, spú šťaním a ú pravou programov. 

Príklad S-1: Vytvorte algoritmus a program, ktorý  pozdraví vypísaním slova Ahoj! na obrazovke. 
algoritmus POZDRAV_1;  program POZDRAV_1; 
začiatok    begin 
píš(’Ahoj’);   write(’Ahoj!’); 
koniec.    end.       
 
V ľavom stĺpci je slovný  zá pis algoritmu a v pravom program – prakticky preklad do angličtiny. Po spustení 

Turbo Pascalu (TP) pomocou sú boru turbo.exe v adresá ri BIN odporú čame stlačiť F3 Open (alebo F10 – File 

– Open...) a hneď zadať meno sú boru, do ktorého chceme program uložiť. Je dobré si uvedomiť rozdiel 

medzi menom programu a menom sú boru, v ktorom je uložený  daný  program. Meno programu je 

identifiká tor, ktorý  môže mať aj viac ako osem znakov. Meno sú boru môže mať najviac osem dovolený ch 

znakov (sú bor s programom POZDRAV_1 môžete nazvať napríklad s-1 alebo pozdrav1; príponu .pas pridá  

Turbo Pascal sá m). Po  zadaní mena sú boru a kliknutí na tlačidlo Open alebo stlačení klá vesu Enter Turbo 

Pascal automaticky prejde do editora („modré“ okno), slú žiaceho na písanie a upravovanie programov (ak sa 

v editore zobrazil text, znamená  to, že so zadaný m ná zvom už sú bor v adresá ri existuje a prá ve sa otvoril – 

pre nový  program musíme zvoliť iné meno sú boru). Po napísaní programu sa môžeme pokú siť spustiť 

program stlačením klá ves Ctrl+F9 alebo kliknutím na Run – Run (horné menu TP možno zaktívniť stlačením 

klá vesu F10). Po odstrá není prípadný ch chý b sa vykonajú  príkazy programu a automaticky sa TP prepne do 

editora. Vý stupy na obrazovku, v našom prípade pozdrav Ahoj!, je na užívateľskej obrazovke, do ktorej 

prepneme stlačením Alt+F5 alebo F10 - Debug - User screen. Po prezretí vý sledku prá ce programu stlačením 

ľubovoľného klá vesu sa vrá time do prostredia TP, presnejšie editora. 

 Ak chceme ukončiť prá cu v prostredí TP, môžeme použiť položky horného menu F10 – File – Exit 

alebo klá vesovú  skratku Alt+X. Pri neuložený ch programoch sa ná s TP pred ukončením opý ta na uloženie. 

Uzavrieť sú bor (program) bez ukončenia TP možno stlačením Alt+F3 alebo kliknutím na zeleno-žltý  

štvorček v ľavom hornom rohu okna (v editore). 

Ď alšia verzia programu POZDRAV demonštruje použitie príkazu ClrScr (Clear Screen), slú žiaceho na 

zmazanie obrazovky. Príkaz clrscr nepatrí do štandardu TP, nachá dza sa v jednotke (v unite) Crt, ktorá  

obsahuje rozširujú ce príkazy pre prá cu s obrazovkou, klá vesnicou, kurzorom, zvukom atď. Jednotlivé 

príkazy a funkcie unitu crt si môžeme prezrieť po vyvolaní kontextového helpu (po stlačení Ctrl+F1 

s kurzorom v slove crt) a vý bere položky Crt – Procedures. Dôležité je vedieť, že pri použití príkazov z unitu 

musíme hneď za menom programu uviesť uses (použi) a meno unitu, v našom prípade crt. Do programu sme 

vložili aj pozná mky, ktoré sa v TP vkladajú  medzi množinové zá tvorky – prekladač všetko medzi { a } 

ignoruje1. Rovnakú  funkciu plnia aj okrú hle zá tvorky s hviezdičkami, t.j. zá pisy (* a *). 
program POZDRAV_2; 
uses crt; { unit-jednotka obsahujuca prikazy rozsirujuce standard Pascalu } 
begin 
clrscr;   { prikaz na zmazanie obrazovky, nachadza sa v jednotke crt } 
write(’Ahoj!’); 
end. 
 
Tretia verzia programu odstraňuje nepohodlné prepínanie medzi užívateľskou obrazovkou a editorom. Príkaz 

vstupu readln (podrobnejšie sa ním budeme zaoberať neskôr) zastaví beh programu, môžeme si pozrieť 

vý stup na užívateľskej obrazovke, a až po stlačení klá vesu Enter program pokračuje resp. skončí, keďže 

nasleduje end s bodkou (bodka signalizuje prekladaču: koniec programu). 

 
program POZDRAV_3; 
uses crt; 
begin 
clrscr; 
write(’Ahoj!’); 
readln    { pocitac caka na stlacenie klavesu Enter, 
            pri verzii TP7 mozete pouzit aj prikaz readkey } 
end. 
 
V ďalšej verzii programu POZDRAV chceme, aby sa ná s počítač opý tal na meno a slušne ná s pozdravil: 

„Ahoj,  naše meno !“. Nemôžeme zvoliť konkrétne meno, lebo sa všetci nevolá me rovnako. Preto musíme 

použiť premennú , ktorá  bude nadobú dať hodnoty raz Jano, potom Miša, Peter, Alexandra,... Počítač musí 

mať v pamäti vyhradené miesto pre hodnotu každej premennej (ako by si ju iná č pamätal?). Aby vedel, že 

má  rezervovať pamäťové miesto a aké veľké, musíme mu to povedať. Robí sa to tzv. deklará ciou 

premenný ch, ktorá  začína slovkom var (variable – premenná ). Cez meno premennej sa odvolá vame na 

priradené pamäťové miesto a jeho veľkosť určujeme typom premennej. Ak je hodnotou premennej celé číslo, 

hovoríme o type integer, ak reá lne číslo, hovoríme o type real. O ú dajový ch typoch si povieme neskôr. 

Hodnotou našej premennej MENO je reťazec znakov, anglicky string. Pri použití typu string sa v pamäti 

vyhradí miesto až pre 255 znakov (má  niekto dlhšie krstné meno, asi nie). 
program POZDRAV_4; 
uses crt; 
var  MENO:string; 
begin 
clrscr; 
writeln(’Ako sa volas ?’); 
readln(MENO); 
writeln(’Ahoj, ’,MENO,’!’); 
readln 
end. 
 
Posledná  verzia programu POZDRAV je určená  pre tý ch, ktorí si chcú  vyskú šať príkaz GotoXY z unitu Crt. 

Potrebujete vedieť, že obrazovka má  štandardne 80 znakov v riadku (stĺpcov v smere x) a 25 riadkov na 

strane (v smere -y).  
 
program POZDRAV_5; 
uses crt; 
var  MENO:string; 

                                                 
1 Dopustili sme sa malej nepresnosti, lebo keby prvý m znakom za zá tvorkou bol $, prekladač by očaká val tzv. direktívu. 

begin 
clrscr; 
writeln(’Ako sa volas ?’); 
readln(MENO); 
clrscr; 
gotoxy(35,12); 
writeln(’Ahoj, ’,MENO,’!’); 
readln 
end. 
 

To bola odbočka do prostredia Turbo Pascalu a k prvému programu a teraz späť k algoritmický m 

konštrukciá m. 

 

SEKVENCIA 
Sekvencia je najjednoduchšou algoritmickou konštrukciou. Použijeme ju, ak sa majú  príkazy vykonať za 

sebou, v poradí, ako sú  zapísané. 

Má  tvar:   

 v slovnom zá pise a v TP: 

  p1; 

  p2; 

  ... 

 ...  pn; 

 

  kde p1, p2, ... , pn sú  príkazy 

 Príkazy od seba oddeľujeme bodkočiarkou! 

 

Vykonanie sekvencie: Príkazy p1, p2 až pn sa vykonajú  za sebou v poradí, ako sú  zapísané. 

Každá  z verzií programu POZDRAV (príklad S-1) obsahuje príkazy, ktoré sa po spustení programu 

vykonajú  postupne za sebou, ako sú  zapísané. Príklad S-1 je príkladom na sekvenciu. 

 

Príkaz priradenia 

Slú ži na priradenie hodnoty premennej. 

Má  tvar: p := v   kde p je premenná  a v vý raz; symbol „:=“ sa číta „priraď“ 

Vykonanie: vyhodnotí sa vý raz na pravej strane a jeho hodnota sa priradí premennej na ľavej strane príkazu 

priradenia. 

Príklady: A := 7; MENO := ’Peter’; I := 5; I := I + 1; OBSAH := A*A (hviezdička je symbol pre ná sobenie) 

Je dobré predstaviť si, čo sa v skutočnosti v počítači udeje. Napríklad pre príkaz I := I + 1. 

Premenná  I zaberá  v pamäti určité pamäťové miesto, zná zornime si ho takto:   I                      Tu je 

zapamätaná  hodnota premennej I. Po spustení programu tam môže byť ná hodné číslo, v Turbo Pascale 

vo verzii 7 tam je automaticky vložená  0. V našom prípade nech tam je hodnota 5, t.j. I                              

Príkaz I := I + 1 zoberie hodnotu premennej I, t.j. 5, zväčší ju o 1 na 6 a tú to hodnotu zapíše do premennej na 

ľavej strane príkazu priradenia, teda do I      . Predchá dzajú ca hodnota 5 je nená vratne stratená ! 

p1 

p2 

pn 

5 

6 

Aj keď to je asi najjednoduchší a najčastejší príkaz, dobre si ho premyslite. Zapamätajte si, že sa vykoná va 

sprava doľava a predchá dzajú ca hodnota premennej na ľavej strane sa prepíše novou hodnotou. 

Príkaz priradenia má  rovnaký  tvar v algoritmizá cii aj v Pascale, tu sa však vyžaduje, aby premenná  p 

a vý raz v boli rovnakého typu, čo je pochopiteľné, lebo ťažko by sa napríklad vý raz typu string (reťazec 

znakov) priraďoval premennej typu integer (celé čísla); dovolená  je len jedna vý nimka: celé číslo (integer) 

môže byť priradené premennej pre reá lne čísla (typu real). 

Príklad: Napíšte postupnosť (sekvenciu) príkazov priradenia, ktoré vymenia hodnoty dvoch premenný ch. 

Napríklad premenná  A nech má  hodnotu 5 a premenná  B hodnotu 7. Po vykonaní sekvencie nech je 

v A hodnota 7 a v B hodnota 5. 

Analý za:  A     B 

      ? 
   A    B 

 

Aj keď nechceme čitateľa podceňovať, sekvencia 

  A := B;  

B := A; 

nie je sprá vna (odsimulujte si, čo sa udeje v počítači). Ak si uvedomíte prečo, sprá vne riešenie ľahko zistíte. 

Vykonaním príkazu A := B sa prepíše pôvodná  hodnota premennej A (5), ktorú  však ešte potrebujeme. 

Riešením je odložiť si ju skôr, než dôjde k jej prepísaniu, niekde „bokom“, do pomocnej premennej napr. 

POM. Sprá vna sekvencia: POM := A; A :=B; B := POM;  

Pozná mky: 

1. Ú loha má  ešte jedno analogické riešenie (POM := B;...).  

2. Ak hodnoty premenný ch A a B sú  čísla, nemusíme použiť ani tretiu premennú  (napr.: A := A-B; 

B := B+A; A := B-A;). 

3. Ú loha vý meny hodnôt dvoch premenný ch je analogická  s ú lohou vý meny tekutín (napr. voda a mlieko) 

v dvoch pohá roch navzá jom. 

 

Príklad S-2: Vytvorte algoritmus a program na vý počet obsahu a obvodu obdĺžnika. 

Analý za: Algoritmus často rieši určitý  problém z nejakej vednej disciplíny, v tomto prípade z matematiky. 

Preto treba mať znalosti z daného vedného odboru, v našom prípade poznať vzorce na vý počet obsahu 

a obvodu obdĺžnika. Ď alej je dobré si uvedomiť, čo musíme poznať pre daný  vý počet (rozmery obdĺžnika - 

vstupné ú daje) a čo očaká vame, že vykonaním algoritmu získame (obsah a obvod - vý stupné ú daje).  Keďže 

zá kladnou vlastnosťou algoritmu je hromadnosť, vstupné aj vý stupné ú daje zadá vame cez premenné. 

Pre rozmery obdĺžnika sú  to bežne písmená  A a B, pre obsah a obvod si zvolíme ná zvy premenný ch OBSAH 

a OBVOD. Pri zostavovaní algoritmu uvažujte, ako by ste daný  problém riešili bez počítača („manuá lne“) 

a tento postup zapíšte vo forme algoritmu. 

 

 

 

5 7 

7 5 

manuá lne algoritmus program 
Začínam 
potrebujem vedieť rozmery obdĺžnika 
použijem vzorec na vý počet obsahu S = a.b 
použijem vzorec na vý počet obvodu 
O = 2.(a+b) 
ozná mim vý sledok 
skončil som. 

začiatok 
čítaj (A, B); 
OBSAH := A*B; 
 
OBVOD := 2*(A+B); 
píš (OBSAH, OBVOD); 
koniec. 

begin 
readln(A, B); 
OBSAH := A*B; 
 
OBVOD := 2*(A+B); 
write(OBSAH, OBVOD); 
end. 

 

Program treba doplniť o deklará ciu premenný ch. V prvej alternatíve budeme počítať len s celý mi číslami 

(typ integer), v druhej s reá lnymi (typ real). Programy sú  doplnené o užívateľsky  komfort, pod ktorý m 

rozumieme predovšetký m pred každý  príkaz vstupu read dať príkaz vý stupu write, ktorý  vypíše na 

obrazovke, čo treba zadať a tiež jednoznačný  vý stup vý sledný ch hodnôt.  

Program očaká va vloženie dvoch celý ch čísel, ktoré môžeme vložiť naraz, oddelené medzerou (nie čiarkou!) 

alebo napíšeme jedno, stlačíme Enter a potom druhé a Enter. 
program OBDLZNIK_CELE; 
uses crt; 
var A, B, OBSAH, OBVOD: integer; 
begin 
clrscr; 
write(’Zadaj strany obdlznika (cele cisla): ’); 
readln(A,B); 
OBSAH:=A*B;  
OBVOD:=2*(A+B); 
writeln(’Obsah obdlznika = ’,OBSAH); 
writeln(’Obvod obdlznika = ’,OBVOD); 
readln 
end. 
 
Typickou chybou začiatočníkov je, že po odladení programu (odstrá není chý b, ktoré ná jde prekladač TP) už 

ďalej „slepo veria počítaču“ a program ďalej netestujú . Musíte sa prinú tiť po napísaní programu ho 

otestovať, t.j. zadať čo najviac taký ch vstupný ch hodnôt, pre ktoré pozná te vý sledky a tiež vložiť tzv. 

hraničné hodnoty, ktoré sú  na hranici dovolený ch vstupný ch hodnôt alebo pri ktorý ch sa mení charakter 

vý počtu. Program OBDLZNIK by sme mohli otestovať pre A a B rovné 0, 1 a napríklad 2 a 4 (2.4=8 a 

2.(2+4)=12) alebo iné čísla, pre ktoré si ľahko spamäti zistíme sprá vne vý sledky. Skú ste aj  A = 1000 

a B = 1000. Ak má te zapnuté kontroly chý b počas behu programu (ako sme to odporučili na strane 6), 

počítač ukončí vý počet chybový m hlá sením Error 215: Arithmetic overflow – preplnenie, vý sledkom 

vý počtu je číslo väčšie ako preň vyhradené pamäťové miesto. Prečo je tomu tak, si povieme pri ú dajový ch 

typoch na nasledujú cej strane. Ak kontrolu nemá te zapnutú , možno si zlý  vý sledok 16 960 (pre obsah) ani 

nevšimnete! Takže dôveruj ale preveruj! 
program OBDLZNIK_REALNE; 
uses crt; 
var A, B, OBSAH, OBVOD: real; 
begin 
clrscr; 
write(’Zadaj strany obdlznika: ’); 
readln(A,B); 
OBSAH:=A*B;  
OBVOD:=2*(A+B); 
writeln(’Obsah obdlznika s presnostou na stotiny = ’,OBSAH:4:2); 
writeln(’Obvod obdlznika s presnostou na stotiny = ’,OBVOD:4:2); 
readln 
end. 

Reá lne čísla vkladá me podobne ako celé, len ich zá pis môže obsahovať aj desatinnú  bodku. Znova všetkému 

nerozumiete, ale naraz sa všetko nedá . Tie tajomné čísla :4:2 objasníme pri príkaze write. 

 
Ú dajové typy 

Jednoduché  štandardné  údajové  typy: 

Meno typu: Integer (ordiná lny typ) 

Množina hodnôt: celé čísla z intervalu –32 768 po 32 767 (konštanta MaxInt) 

Dovolené operá cie: + (sčítanie), - (odčítanie), * (ná sobenie), div (celočíselné delenie), mod (zvyšok po 

celočíselnom delení) 

Funkcie: abs(x), sqr(x), odd(x), trunc(x), round(x) 

  succ(x), pred(x), ord(x) 

Relačné operá tory: <, <=, =, <>, >=, > 

Vysvetlivky: 

V TP sú  pre hodnoty typu integer vyhradené v pamäti 2 B, t.j. 16 bitov. V jednom bite sa zapamätá va 

znamienko (0+/1-) a do zvyšný ch pätná stich bitov sa ukladá  hodnota bez znamienka pomocou 0 a 1. 

Všetký ch možností je 215 = 32 768. Preto do daného pamäťového miesta možno uložiť všetky nezá porné celé 

čísla od 0 po 32 767 alebo zá porné celé čísla od –1 po – 32 768. Ak je vý sledkom operá cie s typom integer 

číslo mimo tento rozsah, nastane v príklade S-2 opísaná  chyba, tzv. preplnenie (v Options – Compiler... 

prepínač Overflow Checking – kontrola preplnenia). Ak ná m uvedený  rozsah nevyhovuje, môžeme použiť 

ďalšie celočíselné typy: Shortint, Longint, Byte a Word, ktorý ch rozsahy si môžete pozrieť pomocou 

kontextového helpu (Ctrl+F1 s kurzorom pod napr. integer). Najväčší rozsah má  typ longint, kde konštanta 

MaxLongInt má  hodnotu 2 147 483 647 (= 231 –1). 

U celočíselný ch typov nie je dovolené klasické delenie, lebo jeho vý sledkom nemusí byť celé číslo. 

Dovolené je tzv. celočíselné delenie (celočíselný  podiel) div a zvyšok po celočíselnom delení mod. Tieto 

funkcie si treba ozrejmiť, lebo sa v numerický ch algoritmoch a programoch často využívajú . Napríklad 

15 div 6 = 2 a 15 mod 6 = 3 lebo 15:6 = 2 zvyšok 3, alebo 7 div 10 = 0 a 7 mod 10 = 7 lebo 7:10 = 0 

zvyšok 7, alebo 21 div 3 = 7 a 21 mod 3 = 0 lebo 21:3 = 7 zvyšok 0. Všeobecne pre kladné celé čísla A a B 

platí: Nech A div B = C a A mod B = D potom B.C + D = A a 0<=D<B. Pre zá porné celé čísla funkcie div 

a mod v TP pracujú  iná č ako v matematike! 

Vý sledkom: abs(x) je absolú tna hodnota čísla x, napr. abs(8) = 8, abs(-5) = 5, abs(0) = 0 

sqr(x) je druhá  mocnina čísla x, napr. sqt(8) = 64, sqr(-5) = 25, sqr(0) = 0 

odd(x) je true (pravda), ak x je nepá rne číslo, inak false (nepravda), napr. odd(0) = false 

trunc(x) je celá  časť reá lneho čísla x, napr. trunc(8.9) = 8, trunc(-5.6) = -5, trunc(0.5) = 0 

round(x) je zaokrú hlené reá lne číslo x, napr. round(8.9) = 9, round(-5.6) = -6, round(0.5) = 1 

succ(x) je nasledovník x, napr. succ(5) = 6, succ(-1) = 0, succ(-5) = -4 

pred(x) je predchodca x, napr. pred(5) = 4, pred(-1) = -2, pred(1) = 0 

ord(x) je poradové číslo x, napr. ord(5) = 5, ord (0) = 0, ord (-12) = -12 

Meno typu: Real (nie je ordiná lny typ!) 

Množina hodnôt: niektoré reá lne čísla z intervalu približne –1,7.1038 po 1,7.1038 

Dovolené operá cie: +, -, *, / (delenie) 

Funkcie: abs(x), sqr(x), sqrt(x), trunc(x), round(x), int(x), frac(x), sin(x), ln(x), exp(x), arctan(x) 

Relačné operá tory: <, <=, =, <>, >=, > 

Vysvetlivky: 

Keďže reá lnych čísel v matematike je nekonečne veľa (aj medzi dvoma ľubovoľný mi reá lnymi číslami) 

a pamäť počítača je konečná , počítač si doká že zapamätať len niektoré reá lne čísla. 

Ď alšie ú dajové typy pre reá lne čísla sú  single, double, extended a comp, ich použitie je však podmienené 

prítomnosťou alebo emulá ciou matematického koprocesora (funkcie v F10 – Options - Compiler... – 

Numeric processing).  

Niektoré funkcie sú  popísané vyššie, u typu integer, a má lo používané funkcie nepopíšeme. 

Vý sledkom: sqrt(x) je druhá  odmocnina z nezá porného čísla x, napr. sqrt(9) = 3, sqrt(2) = 1,4142135624 

  int(x) je celá  časť (cifry pred desatinnou bodkou) z reá lneho čísla x, napr. int(123.456) = 123 

  frac(x) je desatinná  časť z reá lneho čísla x, napr. frac(123.456) = 0.456 

exp(x) je ex 

  ln(x) je prirodzený  logaritmus kladného čísla x1 

 

Meno typu: Boolean (ordiná lny typ) 

Množina hodnôt: {False, True} 

Dovolené operá cie: not (negá cia), and (logický  sú čin), or (logický  sú čet), xor (logický  xor) 

Funkcie: succ(x), pred(x), ord(x) 

Relačné operá tory: <, <=, =, <>, >=, >, in (je prvkom) 

Vysvetlivky: 

Ide o logické hodnoty False – nepravda, 0 a True – pravda, nie 0; false < true.  

pred(true) = false, succ(false) = true; ord(false) = 0, ord(true) = 1 

 

Meno typu: Char (ordiná lny typ) 

Množina hodnôt: znaky tabuľky ASCII 

Funkcie: upcase(z) 

               succ(z), pred(z), ord(z), chr(x) 

Relačné operá tory: <, <=, =, <>, >=, >, in 

Vysvetlivky: 
Znaky tabuľky ASCII aj s poradový mi číslami možno zobraziť pomocou programu: 
program ZNAKY_ASCII; 
uses crt; 
var I:byte; 
begin 
clrscr; 
for I:=32 to 255 do 
  write(I:6,chr(I):2);             { vypis v tabulke } 
                                                 
1 Posledné dve funkcie možno použiť na vý počet xy pomocou vý razu exp(y*ln(x)), x kladné reá lne a y reá lne číslo. 

readln; 
clrscr; 
for I:=1 to 255 do 
  begin 
  writeln(I:6,chr(I):4);  { vypis pod sebou } 
  if I mod 23 = 0 then readln  { zastavi vypis vzdy po 23 riadkoch } 
  end 
end. 
 
Prvý ch 32 znakov (0-31) je riadiacich. Znak s poradový m číslom 32 je medzera atď.  

Vý sledkom funkcie chr(x) je znak zodpovedajú ci poradovému číslu x. 

Napríklad succ(’A’) = B, pred(’A’) = @, ord(’A’) = 65 a chr(65) = A, chr(ord(z)) = z, ord(chr(x)) = x. 

Všimnite si, že ak sa jedná  o konkrétny znak, t.j. konštantu, musí byť v apostrofoch, podobne ako reťazcová  

konštanta. 

Funkcia UpCase(z) zmení malé písmeno na veľké, na iné znaky nemá  vplyv, napr. upcase (’b’) = B. 

 

Ú dajové typy, pre ktoré sú  definované funkcie succ (successor - nasledovník), pred (predecessor - 

predchodca) a ord (ordinal - poradové číslo) nazý vame ordiná lne (každá  hodnota má  presne určené miesto). 

Sú  to typy integer, boolean a char. V TP sú  pre ne zavedené ešte dva praktické príkazy inc (increase - 

zväčšiť) a dec (decrease - zmenšiť). Príkaz inc(p) zväčší hodnotu premennej p o jednu pozíciu vpravo, napr. 

inc(5) = 6, inc(’A’) = B, inc(false) = true a opačne dec(p) zmenší hodnotu premennej p o jednu pozíciu vľavo, 

napr. dec(5) = 4, dec(’B’) = A, dec(true) = false. Tieto príkazy pracujú  podobne ako funkcie succ a pred 

avšak môžu mať aj tvar inc(p,n) resp. dec(p,n), kde n je celé číslo (môže byť aj zá porné) udá vajú ce, o koľko 

treba posunú ť hodnotu premennej p. Napr. inc(5,10) = 15, inc(’a’,10) = k, dec(5,10) = -5, dec(’a’,5) =  W. 

Š truktúrovaný  údajový  typ: 

Meno typu: string alebo string[n] kde n je celé číslo od 1 po 255 – udá va maximá lnu dĺžku reťazca 

Množina hodnôt: reťazce zo znakov ASCII 

Dovolené operá cie:  + (spá ja reťazce) 

Funkcie: length, copy, delete, concat, insert, pos 

Relačné operá tory: <, <=, =, <>, >=, > 

Vysvetlivky: 

String vyhradí v pamäti miesto pre 255 znakový  reťazec (pre najviac 255 znakov), napríklad string[5] 

vyhradí v pamäti miesto pre päť znakový  reťazec (pre najviac 5 znakov, ostatné nebudú  zapamätané). 

Vý sledkom funkcie length(r) je celé číslo - dĺžka reťazca r. Prá zdny reťazec ’’ má  dĺžku 0. 

Pri relačný ch operá ciá ch sa najprv porovnajú  prvé znaky v porovná vaný ch reťazcoch, ak sú  zhodné, druhé 

atď. 

Príkazy vstupu, príkaz read 
Príkazy vstupu slú žia na vstup ú dajov z klá vesnice. 

Majú  tvar: 

čítaj (p1, p2, ... , pn) 

       kde p1 až pn sú  premenné 

 

V TP sú  to modifiká cie príkazu read: 

  read (p1, p2, ... , pn ) 
  readln (p1, p2, ... , pn )   kde p1 až pn sú  premenné 
  readln 

Vykonanie: zastaví sa beh programu a počítač čaká  na vloženie príslušného počtu hodnôt v stanovenom 

poradí a deklarovaný ch typov. 

Napríklad: readln(A,B,C); readln(TEXT); readln(X); readln(MENO); readln(ODPOVED); readln; 

Pozná mky: 

• príkazy ukončené ln (line) možno interpretovať: načítaj hodnoty premenný ch p1 až pn a nastav čítanie na 

nový  riadok 

• jednotlivé číselné hodnoty môžu byť oddelené medzerami alebo vkladané v samostatný ch riadkoch 

zadaním hodnoty a stlačením klá vesu Enter 

• „prá zdny“ príkaz readln neočaká va hodnotu, len stlačenie klá vesu Enter 

• príkaz read(p1, ... , pn) nebudeme predbežne používať 

• po zadaní väčšieho čísla, ako je dovolený  rozsah daného typu, sa zobrazí chybové hlá senie Error 201: 

Range check error - prekročenie rozsahu (v Options – Compiler... prepínač Range Checking – kontrola 

rozsahu hodnôt). Napríklad u typu integer po zadaní čísla väčšieho ako je maxint (32 767). Ak nedošlo 

k zobrazeniu chybového hlá senia, pozri str. 6 

• ak pri zadá vaní hodnôt zadaná  hodnota nezodpovedá  očaká vanému typu, napr. u číselného typu vložíme 

miesto číslice písmeno, zobrazí sa chybové hlá senie Error 106: Invalid numeric format – zlý  číselný  

formá t (v Options – Compiler... prepínač I/O Checking – kontrola vstupno/vý stupný ch operá cií) 

 
Príkazy vý stupu, príkaz write 
Príkazy vý stupu slú žia na zobrazenie vý stupný ch hodnôt na obrazovke monitora. 

Majú  tvary: 

       píš  (h1, h2, ... , hn) 

       kde h1 až hn sú  vý stupné hodnoty 

 

V TP sú  to modifiká cie príkazu write: 

  write (f1, f2, ... , fn ) 
  writeln (f1, f2, ... , fn )    kde f1 až fn sú  formá tovacie vý razy 
  writeln 

Vykonanie: na obrazovke monitora sa zobrazia vý stupné hodnoty v predpísaný ch formá toch (buď hodnoty 

premenný ch alebo text pôvodne uvedený  v apostrofoch). 

vstup: 
p1,p2,...,pn 

vý stup: 
h1,h2,...,hn 

Napríklad: writeln(’Dĺžka čiary je ’,DLZKA); write(’Zadaj kladné číslo: ’); writeln(A,’ ’,B); writeln; 

Pozná mky: 

• príkazy ukončené ln (line) možno interpretovať: zobraz vý stupné hodnoty a nastav vý stup (kurzor) 

na začiatok nového riadku 

• „prá zdny“ príkaz writeln nastaví vý stup na nový  riadok 

 

Formá tovacie výrazy: 

1) h   h je vý raz reprezentujú ci vý stupnú  hodnotu 

2) h : pzv  h ako vyššie, pzv je vý raz reprezentujú ci počet znakov vý stupu 

3) h : pzv : pdm h a pzv ako vyššie, pdm je vý raz reprezentujú ci počet desatinný ch miest vý stupu 

u typu real 

 

Napríklad: 

a) nech A je premenná  typu integer 

hodnota A  príkaz  na obrazovke  pozná mka 

12345  write(A) 12345   znamienko plus sa nezobrazuje 
-12345  write(A) -12345   znamienko mínus zaberá  jeden znak 
12345  write(A:10)      12345  zľava sa doplní päť medzier 
-12345  write(A:7)   -12345  zľava sa doplní jedna medzera 
12345  write(A:3) 12345   ignoruje počet znakov vý stupu 

b) nech X je premenná  typu real (matematický  koprocesor vypnutý ) 

hodnota X  príkaz  na obrazovke  pozná mka 

123,456789 write(X)  1.2345678900E+02 tzv. semilogaritmický  tvar, E – exponent sa 
číta „krá t desať na“, pre „+“ vynechaný  znak 

123,456789 write(X:10)  1.235E+02  neprehliadnite zaokrú hlenie mantisy 
123,456789 write(X:2)  1.2E+02  minimá lny vý pis s exponentom (8 znakov) 
123,456789 write(X:6:2) 123.46   6 znakov vý stupu, 2 za des. bodkou 
123,456789 write(X:4:2) 123.46   ignoruje počet znakov vý stupu 
123,456789 write(X:10:3)     123.457  zľava sa doplnia tri medzery 
123,456789 write(X:12:10) 123.4567890000 maximá lna presnosť zobrazenia 
123,456789 write(X:5:0)    123   zľava sa doplnia dve medzery  
0,0000987654321 write(X)   9.8765432100E-05 semilogaritmický  tvar 
0   write(X)   0.0000000000E+00  semilogaritmický  tvar 

c) nech S je premenná  typu string 

hodnota S  príkaz  na obrazovke  pozná mka 

Janosik  write(S) Janosik 
Janosik  write(S:3) Janosik   ignoruje počet znakov vý stupu 
Janosik  write(S:10)    Janosik  zľava sa doplnia tri medzery 

d) nech Q je premenná  typu boolean 

hodnota Q  príkaz  na obrazovke  pozná mka 

true  write(Q) TRUE 
false  write(Q:10)            FALSE  zľava sa doplní päť medzier 
true  write(Q:1) TRUE   ignoruje počet znakov vý stupu 

Pozná mky: 

• neprehliadnite zaokrú hľovanie reá lnych čísel na zadaný  počet desatinný ch miest 

• bez použitia formá tovacích vý razov napríklad v príkaze write(A,B), kde A a B sú  typu integer, neodlíšite 

od seba cifry kladný ch čísel! Okrem formá tovania je riešením aj vloženie medzery, t.j. write(A,’ ’,B). 

• formá tovacie vý razy sú  vý razy, t.j. môžu mať aj tvar napríklad writeln(A*B), write(X : PRESNOST+2 : 

PRESNOST), kde premenná  PRESNOST typu integer môže nadobú dať hodnoty 0 (jednotky), 

1 (desatiny), 2 (stotiny), ... , 10 

• ujasnite si vykonanie príkazu writeln(’Obvod = ’:40,2*(A+B):4:2) 

 

Vrá ťme sa však k sekvencii.  

Príklad S-3: Vytvorte program simulujú ci palubný  počítač auta. Zná my je objem palivovej ná drže a 

priemerná  spotreba paliva počas jazdy. Vždy pri natankovaní plnej ná drže sa vynuluje tzv. denné počítadlo 

kilometrov. Po zadaní prejdený ch kilometrov (z denného počítadla kilometrov) počítač vypočíta aktuá lny 

dosah auta. Po zadaní vzdialenosti do cieľa vypočíta priemernú  dobu jazdy. 

Analý za: Dosah auta pri plnej ná drži sa vypočíta ako podiel: objem palivovej ná drže / priemerná  spotreba 

paliva na 100 km, krá t 100 km. Keďže tento podiel nemusí byť celé číslo, zaokrú hlime ho na celé km, 

samozrejme nadol. Aktuá lny dosah sa zrejme rovná : dosah pri plnej ná drži – prejdené km. Priemerná  doba 

jazdy sa vypočíta ako podiel: vzdialenosť do cieľa / priemerná  rý chlosť  za hodinu (konštanta). Vý sledok je 

v hodiná ch a zobrazíme ho s presnosťou na desatiny hodiny. 

 
program POCITAC; 
uses  crt; 
const ObjemPalNadrze = 46;    { litrov } 
      PriemSpotPaliva = 6.5;  { litra na 100 km } 
      PriemRychlost = 90;     { km/hod. } 
var   PrejdeneKM, AktDosah, Dosah, VzdDoCiela : integer; 
                               PriemDobaJazdy : real; 
begin 
clrscr; 
write(’Prejdene kilometre (z denneho pocitadla km): ’); 
readln(PrejdeneKM); 
Dosah:=trunc(ObjemPalNadrze/PriemSpotPaliva*100); 
AktDosah:=Dosah-PrejdeneKM; 
writeln(’Aktualny dosah: ’,AktDosah,’km’); 
writeln; 
write(’Vzdialenost do ciela: ’); 
readln(VzdDoCiela); 
PriemDobaJazdy:=VzdDoCiela/PriemRychlost; 
writeln(’Priemerna doba jazdy: ’,PriemDobaJazdy:3:1,’ hod.’); 
readln 
end. 
 

Príklad S-4: Vytvorte program na „uhá dnutie“ celého čísla z intervalu 0 až 100, ak viete, že po zadaní 

zvyškov po delení 3 (zvysok3), 5 (zvysok5) a 7 (zvysok7) mysleného čísla sa myslene číslo rovná  zvyšku po 

celočíselnom delení vý razu (70*zvysok3+21*zvysok5+15*zvysok7) 105-timi. Strašné, ani len zadaniu sa 

nedá  rozumieť! 

Analý za: Treba zadať zvyšky po delení 3, 5 a 7 mysleného čísla – to sú  vstupné ú daje a my pre ne zvolíme 

vstupné premenné označené zvysok3, zvysok5 a zvysok7. Myslené číslo sa rovná  zvyšku po delení vý razu..., 

preto si tento vý raz vypočítajme. VYRAZ = 70*zvysok3+21*zvysok5+15*zvysok7. Myslené číslo sa rovná  

zvyšku po celočíselnom delení VYRAZu 105-timi. Zvyšok po celočíselnom delení ná m dá va funkcia mod. 

A tu je program: 
program UHADNEM_CISLO; 
uses crt; 
var zv3,zv5,zv7,myslene_cislo:byte; 
begin 
clrscr; 
writeln(’U H A D N E M   M Y S L E N E   C I S L O  D O  100 !’:60); 
write(’Zvysok po deleni 3: ’); readln(zv3); 
write(’Zvysok po deleni 5: ’); readln(zv5); 
write(’Zvysok po deleni 7: ’); readln(zv7); 
myslene_cislo:=(70*zv3+21*zv5+15*zv7) mod 105; 
writeln; 
writeln(’Myslel si si cislo ’,myslene_cislo); 
readln 
end. 
 

Príklad S-5: Vytvorte program, ktorý  prepočíta čas v sekundá ch na hodiny : minú ty : sekundy. 

Analý za: Vieme, že hodina má  3 600 sekú nd. Koľkokrá t sa 3 600 nachá dza v sekundá ch (bezo zvyšku), je 

hľadaný  počet hodín – to vie vypočítať funkcia div. Funkcia mod určí zvyšok sekú nd, čo zostal na minú ty 

a sekundy. Počet minú t zistíme opäť celočíselný m delením zvyšný ch sekú nd číslom 60. No a to čo zostane, 

sú  už len sekundy (< 60). Vý počet funkciou mod možno nahradiť aj príkazom CAS := CAS – 3600*H resp. 

CAS := CAS – 60*M. 
program CAS_V_SEKUNDACH; 
uses crt; 
var  H, M, CAS : longint; 
begin 
clrscr; 
write(’Zadaj cas v sekundach: ’); 
readln(CAS); 
H:=CAS div 3600; CAS:=CAS mod 3600; 
M:=CAS div 60;   CAS:=CAS mod 60; 
writeln(H,’ : ’,M,’ : ’,CAS); 
readln 
end. 
 
Ú loha: Doká žete upraviť predchá dzajú ci program tak, aby počítal iba s jednou premennou CAS? 
program CAS_V_SEKUNDACH_2; 
uses crt; 
var  CAS:longint; 
begin 
clrscr; 
write(’Zadaj cas v sekundach: ’); readln(CAS); 
gotoxy(34,12); 
writeln(CAS div 3600,’ : ’,CAS mod 3600 div 60,’ : ’,CAS mod 3600 mod 60); 
readln 
end. 

Nerieš ené  úlohy na sekvenciu: 

1. Vytvorte program na vý počet obsahu a obvodu štvorca s presnosťou na desatiny. 

Ná vod: Vzorec pre vý počet obsahu S = a2 a obvodu O = 4a. 

2. Vytvorte program na vý počet obsahu a obvodu kruhu s presnosťou na tisíciny. 

Ná vod: Vzorec pre vý počet obsahu S = π.r2 a obvodu O = 2π.r. TP pozná  konštantu PI = 3,1415... 

3. Vytvorte program na vý počet objemu, povrchu a telesovej uhlopriečky kvá dra s presnosťou na stotiny. 

Ná vod: Vzorec pre vý počet objemu V = a.b.c, povrchu S = 2(a.b+a.c+b.c), uhlopriečky u = a2+b2+c2 . 

4. Vytvorte program na vý počet objemu a povrchu gule s najväčšou možnou presnosťou. 

Ná vod: Vzorec pre vý počet objemu V = 4/3π.r3, povrchu S = 4π.r2. 

5. Vytvorte program na vý počet počtu 10, 5, 2 a 1 Sk mincí na vyplatenie zadanej sumy (použite funkcie 

div a mod ). 

Pozná mka: Pozri program CAS_V_SEKUNDACH. 

6. Vytvorte program na ná kup valú t za určitú  sumu korú n. Zmená reň si ú čtuje manipulačný  poplatok vo 

vý ške 1% z vloženej sumy.  Skutočne platená  čiastka sa zaokrú hľuje na celé koruny. Ná zov a kurz meny 

si zvoľte sami. 

7. Priemerná  denná  teplota sa počíta ako aritmetický  priemer teplôt nameraný ch o 6-tej rá no, o 12-tej 

a o 18-tej hodine,  pričom ú daj nameraný  o 18-tej hodine sa započítava dvakrá t. Vytvorte program, ktorý  

zo zadaný ch ú dajov vypočíta priemernú  dennú  teplotu. 

8. Sú  zadané odpory dvoch rezistorov v ohmoch. Určte hodnotu  vý sledného odporu pri ich sériovom 

a paralelnom zapojení. 

Ná vod: Pri sériovom zapojení R = R1+R2, pri paralelnom R = R1.R2 / (R1+R2). 

9. Vytvorte program na riešenie lineá rnej rovnice ax + b = 0    a, b reá lne, a ≠ 0. 

Ná vod: Koreň x = -b/a. 

10. Vytvorte program na prevod veľkosti uhla v radiá noch na stupne a minú ty. 

Ná vod: π radiá nov = 1800, 10=60’. TP pozná  konštantu PI. 

VETVENIE, ROZHODOVANIE 
Vetvenie použijeme, ak vykonanie príkazu (alebo skupiny príkazov) je podmienené splnením alebo 

nesplnením určitej podmienky. 

S rozhodovaním sa stretá vame v bežnom živote na každom kroku. Ak bude teplo, pôjdem na kú palisko, inak 

budem sedieť za počítačom. Vykonanie akcie „pôjdem na kú palisko“ je podmienené splnením podmienky 

„bude teplo“. Vykonanie akcie „budem sedieť za počítačom“ je podmienené nesplnením podmienky! 

Vetvenie pozná me biná rne (dve možnosti) a n-á rne (n možností, n≥ 2). 

Ú plné  biná rne vetvenie, podmienený  príkaz if 

Má  tvar: 

 

  ak podmienka 
 tak p1 
 inak p2 

 kde p1 a p2 sú  príkazy 

Vykonanie: Ak je podmienka splnená , vykoná  sa 

príkaz p1, ak nie je splnená , vykoná  sa príkaz p2. 

 

V TP ú plnému biná rnemu vetveniu zodpovedá  ú plný  príkaz if, ktorý  má  tvar: 

 if b 
 then p1 
 else p2 kde b je vý raz typu boolean, p1 a p2 sú  príkazy 

Vykonanie ú plného príkazu if: Ak vý raz b nadobudne hodnotu true, vykoná  sa príkaz p1, ak nadobudne 

hodnotu false, vykoná  sa príkaz p2. 

Napríklad: if X < 0 then writeln(’Číslo je zá porné’) else writeln(’Číslo je nezá porné’) 

  if PRIESTUPNY_ROK then DNI := 29 else DNI := 28 

  if (ZNAK >= ’A’) and (ZNAK <= ’Z’) then writeln(’veľké písmeno’) else writeln(’neviem’) 

  if odd(CISLO) then writeln(’nepá rne’) else writeln(’pá rne’) 

  if true then writeln(’vždy toto’) else writeln(’toto nikdy’) 

  if A mod B = 0 then writeln(A,’ je deliteľné ’,B) else writeln(A,’ nie je deliteľné ’,B) 

Neúplné  biná rne vetvenie, neúplný  príkaz if 

Má  tvar: 

 

  ak podmienka 
 tak p 

        kde p je príkaz 

Vykonanie: Ak je podmienka splnená , vykoná  sa 

príkaz p, inak je vetvenie bez ú činku. 

 

 

podmienka 

p1 p2 

+ – 

podmienka 

p 

+ 

– 

V TP neú plnému biná rnemu vetveniu zodpovedá  neú plný  podmienený  príkaz if, ktorý  má  tvar: 

 if b 
 then p   kde b je vý raz typu boolean a p je príkaz  

Vykonanie neú plného príkazu if: Ak vý raz b nadobudne hodnotu true, vykoná  sa príkaz p, ak nadobudne 

hodnotu false, príkaz if je bez ú činku. 

Napríklad: if MENO = ’Peter’ then writeln(’Ď alší Peter’) 

  if CISLO > NAJVACSIE then NAJVACSIE := CISLO 

  if (CISLO > MaxInt) or (CISLO <= -MaxInt) then writeln(’mimo rozsah integer’) 

  if ord(ZNAK) < 32 then writeln(’riadiaci znak’) 

  if X = HLADANE_CISLO then POCET := POCET + 1 

 

Príklad VIF-1: Vytvorte program na ná jdenie najväčšieho z troch celý ch čísel. 

Analý za: Znova odporú čame algoritmus tohto problému zostaviť obyčajný m „sedliackym“ rozumom. Čo 

spravíme prvé, keď sa dozvieme čísla, z ktorý ch treba ná jsť najväčšie? Asi porovná me prvé dve čísla 

a z nich vyberieme väčšie (ak sú  rovnaké, je jedno, ktoré vyberieme) a to ná sledne porovná me s tretím 

číslom. A má me najväčšie – maximum. Algoritmus je na strane 4.  
program MAXIMUM_1; 
uses crt; 
var A, B, C, MAX : integer; 
begin 
clrscr; 
write(’Zadaj tri cele cisla: ’); 
readln(A,B,C); 
if A>B 
  then MAX:=A 
  else MAX:=B; 
if C>MAX 
  then MAX:=C; 
writeln(’Z cisel ’,A,’, ’,B,’, ’,C,’ je najvacsie ’,MAX,’.’); 
readln 
end. 
 

Aby sme demonštrovali variabilnosť riešenia už tak jednoduchého problému, uvá dzame ďalšie programy 

riešiace rovnakú  ú lohu. 

Iná  možnosť (jej vý hodou je jednoduchosť a použiteľnosť pre ľubovoľný  počet čísel, treba len opakovať 

príkaz if ? > MAX then MAX := ?): 
program MAXIMUM_2; 
uses crt; 
var A, B, C, MAX : integer; 
begin 
clrscr; 
write(’Zadaj tri cele cisla: ’); 
readln(A,B,C); 
MAX:=A;  { priradenie pociatocnej hodnoty premennej MAX } 
if B>MAX 
  then MAX:=B; 
if C>MAX 
  then MAX:=C; 
writeln(’Z cisel ’,A,’, ’,B,’, ’,C,’ je najvacsie ’,MAX,’.’); 
readln 
end. 

V nasledujú com programe je logická  chyba, viete ju ná jsť? 
program MAXIMUM_3; 
uses crt; 
var A, B, C : integer; 
begin 
clrscr; 
write(’Zadaj tri cele cisla: ’); 
readln(A,B,C);   
write(’Z cisel ’,A,’, ’,B,’, ’,C,’ je najvacsie ’); 
if (A>B) and (A>C)     { preco sme presunuli vystup (v nom nie je chyba!)? } 
  then writeln(A); 
if (B>A) and (B>C) 
  then writeln(B); 
if (C>A) and (C>B) 
  then writeln(C); 
readln 
end. 
 

Takto to je už dobre? 
program MAXIMUM_4; 
uses crt; 
var A, B, C : integer; 
begin 
clrscr; 
write(’Zadaj tri cele cisla: ’); 
readln(A,B,C); 
write(’Z cisel ’,A,’, ’,B,’, ’,C,’ je najvacsie ’); 
if (A>B) and (A>C) 
  then writeln(A) 
  else if (B>A) and (B>C) 
         then writeln(B) 
         else writeln(C); 
readln 
end. 

Programá torský  štý l 

 Ak sa pozriete na zá pisy príkazov if aj celý ch programov, vidíte v nich určitú  štruktú ru, ktorou sa 

snažíme vyjadriť podriadenosť niektorý ch častí nad iný mi. Rovnocenné príkazy sú  na jednej ú rovni, 

podriadené časti sa snažíme posunú ť napríklad o tri medzery vpravo. Zá roveň taký to zá pis sprehľadňuje celý  

algoritmus alebo program. Veľmi vá m odporú čame používať určité logické schémy pri zá pisoch programov 

a algoritmov. Budú  pre všetký ch prehľadnejšie, zrozumiteľnejšie, pri tvorbe spravíte menej chý b a ľahšie sa 

v nich budú  hľadať chyby. Odporú čame dve schémy: 

1. zá pis  if   podmienka 

     then príkaz1 

     else príkaz2 

dobre graficky zná zorňuje, že ide o ú plné biná rne vetvenie s oboma vetvami na samostatný ch riadkoch; 

riadky s then a else sú  podriadené riadku if, kde príkaz začína, ďalší príkaz by bol už na ú rovni if; 

2. zá pis if    podmienka then príkaz1 

else príkaz2 

zdôrazňuje prítomnosť vetvy else, t.j. ú plného biná rneho vetvenia, vetva then je „schovaná “, keďže sa 

musí vyskytovať v každom príkaze if.  

Pri písaní programov v editore TP si všimnite, že podporuje štruktú rovaný  zá pis. 

 Prakticky prekladaču TP je jedno, ako je daný  program napísaný . Pre neho je rozhodujú ca len 

prítomnosť bodkočiarok medzi jednotlivý mi príkazmi. Celý  program by mohol byť napísaný  v jednom 

riadku (ak by sa tam zmestil). Sú časťou dobrého programá tora je aj dobrý  programá torský  štý l, aj čo sa tý ka 

zá pisu programov, ktorý  sa určite neprezentuje programom v jednom stĺpci alebo riadku! 

 A propos bodkočiarky. Platí: príkazy sa od seba oddeľujú  bodkočiarkami. Preto za begin, pred end 

a iný mi vyhradený mi slovami bodkočiarky nemusia byť, neoddeľovali by príkazy. V niektorý ch prípadoch 

sú  dokonca chybou (uvidíme pri cykloch). My v programoch na miestach, kde nemusia byť, nebudeme dá vať 

bodkočiarky, aj keď väčšinou by tam mohli byť. 

Zložený  príkaz 

 Ak si dobre všimnete príkaz if, za then aj else je dovolený  len jeden (!) príkaz. Ak potrebujeme na 

tý chto miestach použiť viacej príkazov, musíme vytvoriť tzv. zložený  príkaz.  

Zložený  príkaz použijeme na mieste, kde je dovolený  len jeden príkaz a my potrebujeme, aby došlo 

k vykonaniu viacerý ch príkazov. Zložený  príkaz vytvá ra zo skupiny príkazov jeden príkaz. 

Má  tvar:  začiatok  v TP:  begin 
  p1;     p1; 
  p2;     p2; 
  ...     ... 
  pn;     pn; 
  koniec     end  kde p1 až pn sú  príkazy 

Vykonanie: Z príkazov p1, p2 až pn je vytvorený  jeden (zložený ) príkaz. 

Napríklad: if A > B   if  X<=0 
 then begin      then writeln(’Pre X<=0 neviem vypočítať!’) 
 POM := A;     else  begin 
 A := B;   XnaN := exp( N*ln(X) ); 
 B := POM;   writeln( X, ’ na ’ , N , ’ = ’ , XnaN ) 
 end                 end 
 { po skončení A ≤ B } { vý počet mocniny xn } 

Poradie vyhodnocovania operá cií (priorita operá torov) 

 Všimnite si, že v programoch MAXIMUM 3 a 4 sme v podmienkach príkazov if použili zá tvorky. 

Otá zka znie: Kedy musíme použiť zá tvorky v zá pise vý razov a kedy nie? O tom rozhoduje priorita 

jednotlivý ch operá torov. Podobne ako v matematike má  aj v TP ná sobenie a delenie „prednosť“ pred 

sčítaním a odčítaním a pod. Všeobecne platí, že najprv sa vyhodnotia funkcie (abs, sqr, sqrt, odd,...).  

Priorita operá torov je nasledovná : 

1. uná rne mínus (zmena znamienka), not 

2. *, / , div, mod, and 

3. +, - , or, xor 

4. <, <=, =, <>, >=, >, in 

pri rovnosti operá torov sa postupuje zľava doprava; prvé sa vyhodnocujú  vý razy v zá tvorká ch, 

od vnú torný ch k vonkajším. 

Príklady: ( I = N ) or ( ODP = ’K’ ) Bez zá tvoriek by došlo ku chybe, lebo or má  vyššiu prioritu ako = 
     a preto by sa najprv vykonalo N or ODP pričom N je číselný  typ a 
 boolean boolean  ODP je typu char, čo nie sú  typy boolean očaká vané „okolo“ or. 

Vý raz  
a + b

c   musí byť zapísaný  pomocou zá tvoriek (A+B)/C, inak by došlo najprv k deleniu B/C (delenie 

má  vyššiu prioritu ako sčítanie) a až potom k sčítaniu, čo nezodpovedá  pôvodnému vý razu. 

Vý raz  
a

b.c  musí byť zapísaný  A/(B*C), iná č by nezodpovedal pôvodnému vý razu, pretože zá pis A/B*C  

(pravidlo „zľava doprava“) zodpovedá  vý razu  
a
b .c (čo nie je pôvodný  vý raz). Pôvodnému vý razu však 

zodpovedá  aj zá pis A/B/C! 

Vý raz  
a.b
c   môže byť zapísaný  bez zá tvoriek A*B/C (pravidlo „zľava doprava“; najprv sa vykoná  ná sobenie 

a potom delenie). 

 

Znova sme trochu mú drejší v teórii, vrá ťme sa však k príkladom využívajú cim vetvenie a príkaz if. 

Príklad VIF-2: Vytvorte program na usporiadanie troch celý ch čísel (zostupne). 

Analý za: Podľa potreby budeme vymieňať hodnoty premenný ch A, B, C, ký m nebude A<=B<=C. 
program USPORIADAJ; 
uses crt; 
var A, B, C, POM : integer; 
begin 
clrscr; 
write(’Zadaj tri cele cisla: ’); 
readln(A,B,C); 
if A>B 
  then begin POM:=A; A:=B; B:=POM end; 
if B>C 
  then begin POM:=B; B:=C; C:=POM end; 
if A>B 
  then begin POM:=A; A:=B; B:=POM end; 
writeln(A,’ <= ’,B,’ <= ’,C); 
readln 
end. 
 

Venujme sa ešte jednej „teoretickej“ otá zke. V príkaze if nič nebrá ni tomu, aby na mieste príkazu p1 

alebo p2 mohol byť znova, podľa potreby, použitý  príkaz if, ako to vidieť v programe MAXIMUM_4.  

Pri štruktú rach  if b1     if b1 
   then if b2       then  if b2 
              then p1       then p1 
else p2      else p2 
     else p3 

vzniká  otá zka, ku ktorému if patria vetvy else? 

Platí: Vetva else patrí vždy k najbližšiemu predchá dzajú cemu then.   

Vyjadrené štruktú rovane:  
  if b1     if b1 

   then if b2       then  if b2 
              then p1       then p1 

 else  p2       else  p2 
        else p3 

Ak chceme, aby else p2 v prvej štruktú re patrilo prvému then musíme to prekladaču povedať zá pisom: 

  if b1  
   then begin  
           if b2 
              then p1 
           end 
   else  p2   

 

Príklad VIF-3: Vytvorte program na vý počet vý sledného odporu dvoch rezistorov zapojený ch sériovo 

alebo paralelne s presnosťou na stotiny. 

Analý za: Z fyziky treba vedieť, že pri sériovom zapojení rezistorov vý sledný  odpor R=R1+R2, pri paralelnom  

 R1 . R2 
zapojení R =  ––––––––––– .  
 R1 + R2 

Doriešiť treba ešte zadanie spôsobu zapojenia. Rozhodli sme sa pre vloženie prvého písmena (typ char), 

presnejšie, pri vložení znaku p alebo P je zapojenie paralelné, inak sériové. 

 
program VYSLEDNY_ODPOR; 
uses crt; 
var R1, R2, R : real; 
        ZAPOJ : char; 
begin 
clrscr; 
write(’Zadaj hodnoty odporov R1 a R2: ’); readln(R1,R2); 
write(’Sposob zapojenia (P - paralelne, iny znak - seriovo): ’); 
readln(ZAPOJ); 
if (ZAPOJ=’P’) or (ZAPOJ=’p’) 
  then R:=R1*R2/(R1+R2) 
  else R:=R1+R2; 
writeln(’Vysledny odpor R = ’,R:4:2); 
readln 
end. 
 
Využitie funkcie upcase a upravenie vý stupu: 
program VYSLEDNY_ODPOR_2; 
uses crt; 
var R1, R2, R : real; 
        ZAPOJ : char; 
begin 
clrscr; 
write(’Zadaj hodnoty odporov R1 a R2 v ohmoch: ’); readln(R1,R2); 
write(’Sposob zapojenia (P - paralelne, iny znak - seriovo): ’); 
readln(ZAPOJ); 
ZAPOJ:=upcase(ZAPOJ); 
if ZAPOJ=’P’ 
  then R:=R1*R2/(R1+R2) 
  else R:=R1+R2; 
writeln; 
write(’Vysledny odpor pri ’); 
if ZAPOJ=’P’ 
  then write(’paralelnom’) 
  else write(’seriovom’); 
writeln(’ zapojeni R = ’,R:4:2,’ ohmov.’); 
readln 
end. 

Ď alšou možnosťou je ošetrenie zadania spôsobu zapojenia tak, aby program „zobral“ len S alebo P. 
program VYSLEDNY_ODPOR_3; 
uses crt; 
var R1, R2 : real; 
     ZAPOJ : char; 
begin 
clrscr; 
write(’Zadaj hodnoty odporov R1 a R2 v ohmoch: ’); readln(R1,R2); 
write(’Sposob zapojenia (S - seriovo, P - paralelne): ’); readln(ZAPOJ); 
if (ZAPOJ=’S’) or (ZAPOJ=’s’) 
  then writeln(’Vysledny odpor R = ’,R1+R2:3:1,’ ohmov’) 
  else if (ZAPOJ=’P’) or (ZAPOJ=’p’) 
         then writeln(’Vysledny odpor R = ’,R1*R2/(R1+R2):3:1,’ ohmov’) 
         else writeln(’Nebolo stlacene S alebo P!’); 
readln 
end. 
 

Ú loha: Program VYSLEDNY_ODPOR môžete upraviť aj tak, aby reagoval na vloženie celého slova seriovo 

alebo paralelne (zrejme premenná  ZAPOJ bude typu string[9]). 

 

Príklad VIF-4: Vytvorte program, ktorý  po zadaní pH ozná mi: prostredie kyslé, neutrá lne alebo zá sadité. 

Analý za: Trochu chémie nezaškodí. Tuším ná s učili: ak je pH < 7, tak je prostredie kyslé; ak je pH = 7, tak je 

neutrá lne a ak je pH > 7, tak je prostredie zá sadité. Použitie podmienený ch príkazov je zrejmé. 
 
program PH_PROSTREDIA; 
uses crt; 
var PH:real; 
begin 
clrscr; 
write(’Zadaj pH: ’); readln(PH); 
writeln; 
write(’Prostredie je ’); 
if PH<7 then writeln(’kysle.’); 
if PH=7 then writeln(’neutralne.’); 
if PH>7 then writeln(’zasadite.’); 
readln 
end. 
 

Príklad VIF-5: Vytvorte program analogický  programu S-5 (prepočíta čas v sekundá ch na hodiny : minú ty : 

sekundy s vý stupom HH : MM : SS (dvojčísla). 

Analý za: Rozdiel od príkladu S-5 je v tom, že ak je číselný  ú daj (hodiny, minú ty alebo sekundy) 

jednociferný , tak treba „predložiť“ nulu.  

 
program CAS_V_SEKUNDACH; 
uses crt; 
var  H,M,CAS:longint; 
begin 
clrscr; 
write(’Zadaj cas v sekundach: ’); readln(CAS); 
H:=CAS div 3600; CAS:=CAS mod 3600; 
M:=CAS div 60;   CAS:=CAS mod 60; 
if H<10 then write(’0’); 
write(H,’ : ’); 
if M<10 then write(’0’); 
write(M,’ : ’); 
if CAS<10 then write(’0’); 

writeln(CAS); 
readln 
end. 
 
Príklad VIF-6: Vytvorte program na určenie, či zadaný  rok je priestupný . 

Analý za: Platí, že rok je priestupný , ak je deliteľný  4 a nie je deliteľný  100, okrem rokov deliteľný ch 400. 

Napríklad rok 1600 je priestupný  ale rok 1700 nie. Ešte sa „s priestupný m rokom“ stretneme. 

 
program PRIESTUPNY_ROK; 
uses crt; 
var ROK:integer; 
begin 
clrscr; 
write(’Zadaj rok: ’); readln(ROK); 
if (ROK mod 4=0) and ((ROK mod 100<>0) or (ROK mod 400=0)) 
  then writeln(’Rok ’,ROK,’ - priestupny’) 
  else writeln(’Rok ’,ROK,’ - nepriestupny’); 
readln 
end. 
 

Ešte jedno riešenie (toto riešenie ešte použijeme neskôr): 
program PRIESTUPNY_ROK_2; 
uses crt; 
var ROK:integer; 
    PRIESTUPNY:boolean; 
begin 
clrscr; 
write(’Zadaj rok: ’); readln(ROK); 
PRIESTUPNY:=(ROK mod 100<>0) and (ROK mod 4=0) or (ROK mod 400=0); 
if PRIESTUPNY 
  then writeln(’Rok ’,ROK,’ - priestupny’) 
  else writeln(’Rok ’,ROK,’ - nepriestupny’); 
readln 
end. 
Upravte program tak, aby oznamoval aktuá lne: Rok ... bol / je / bude ... Použite príkaz GetDate z unitu Dos 

(pozri pomocou Ctrl+F1), ktorého vý sledkom je aktuá lny rok, mesiac, deň a poradové číslo dňa v tý ždni, 

pričom poradové číslo nedele je 0, pondelka 1 atď. Použité premenné sú  typu word (pozri aj príklad VCS-1: 

program POCET_DNI). 

 

Príklad VIF-7: Vytvorte program na vý počet doby splá cania - počtu rokov a mesiacov, bezú ročnej pôžičky 

po zadaní požičanej sumy a vý šky mesačnej splá tky. 

Analý za: Zovšeobecnite situá ciu, keď má te požičaný ch 12 000 Sk a mesačne splá cate po 100 Sk. Ako dlho 

budete platiť? Najprv nech ná s nezaujímajú  „zvyšky“ (rozumej mesiace), len celé roky, ktoré treba splá cať. 

Tie určí delenie bezo zvyšku, t.j. funkcia div. Ak ste našli odpoveď, môžeme ísť ďalej. Ak by ste mali 

požičaný ch 13 000 Sk, koľko mesiacov by ste museli ešte po desiatich rokoch platiť? Zovšeobecnite! 

Myslite aj na možnosť, že splá cať nebude treba ani jeden celý  rok, potom položka počet rokov splá cania by 

sa vôbec nemala zobraziť, podobne s mesiacmi, ak sa všetko splatí za celý  ná sobok roka. 

 
program SPLATKY; 
uses crt; 
var SUMA, MESACNE, ROKOV, MESIACOV, ZOSTAVASK : longint; 
begin 
clrscr; 

write(’Pozicana suma: ’); readln(SUMA); 
write(’Vyska mesacnej splatky: ’); readln(MESACNE); 
writeln; 
writeln(’Treba splacat: ’); 
ROKOV:=SUMA div (12*MESACNE); 
if ROKOV>0 
  then writeln(’- rokov: ’,ROKOV); 
ZOSTAVASK:=SUMA-ROKOV*12*MESACNE; 
MESIACOV:=ZOSTAVASK div MESACNE; 
if MESIACOV>0 
  then writeln(’- mesiacov: ’,MESIACOV); 
ZOSTAVASK:=SUMA-(ROKOV*12*MESACNE+MESIACOV*MESACNE); 
if ZOSTAVASK>0 
  then writeln(’- dalsi mesiac este splatit: ’,ZOSTAVASK,’ korun.’); 
readln 
end. 
 
Doplňujú ca ú loha: Vý stup s koncovkami v sprá vnom tvare: 

Treba splá cať: 

1 rok  alebo 2, 3, 4 roky alebo 5, 6, ... rokov 

1 mesiac alebo 2, 3, 4 mesiace alebo 5, 6, ... mesiacov 

... splatiť 1 korunu alebo 2, 3, 4 koruny alebo 5,6, ...korú n. 

 

Príklad VIF-8: Na pobavenie a doplnenie vôbec prvého programu v skriptá ch. Program, ktorý  po zadaní 

mena pozdraví, opý ta sa "Ako sa má š?" a primerane zareaguje. 
 
program POZDRAV; 
uses Crt; 
var  MENO, ODPOVED:string; 
begin 
clrscr; 
write(’Ako sa volas ? ’); 
readln(MENO); 
{ nasledujuce zmeni prve pismeno v mene na velke - pozri Help pomocou Ctrl+F1 } 
MENO:=upcase(MENO[1])+copy(MENO,2,length(MENO)); 
writeln; 
writeln(’Ahoj, ’,MENO,’!’); 
writeln; 
write(’Ako sa mas? ’); 
readln(ODPOVED); 
writeln; 
if (ODPOVED=’dobre’) or (ODPOVED=’vyborne’) or (ODPOVED=’OK’) 
  then writeln(’Som rad, ze sa mas ’,ODPOVED,’!’) 
  else if (ODPOVED=’zle’) or (ODPOVED=’nanic’) or (ODPOVED=’na...’) 
         then writeln(’Mrzi ma, ze sa mas ’,ODPOVED,’!’) 
         else writeln(’"’,ODPOVED,’" nepoznam!’); 
writeln; 
gotoxy(25,12); 
writeln(’A h o j , musim koncit!’); 
delay(3500);     { delay(CISLO) zastavi beh programu na CISLO milisekund } 
clrscr 
end. 
 

Príklad VIF-9: Už sme mali fyziku, chémiu, bankovníctvo, slušné sprá vanie, tak znova trocha matematiky. 

Vytvorte program, ktorý  zistí, či zadané číslo je deliteľné druhý m zadaný m číslom. 

Analý za: Napríklad číslo 21 je deliteľné 7, lebo zvyšok po delení čísla 21 siedmimi je nula! Ale zvyšok po 

celočíselnom delení dá va funkcia mod, preto ak... 

program DELI; 
uses crt; 
var DELENEC, DELITEL : integer; 
begin 
clrscr; 
write(’Zadaj delenca a delitela: ’); 
readln(DELENEC, DELITEL); 
if DELENEC mod DELITEL=0 
  then writeln(’Cislo ’,DELENEC,’ je delitelne ’,DELITEL) 
  else writeln(’Cislo ’,DELENEC,’ nie je delitelne ’,DELITEL); 
readln 
end. 

N-á rne vetvenie, podmienený  príkaz case 

Má  tvar:       v slovnom zá pise nemá  ekvivalent,  

        realizuje sa zá pisom: 

        ak podm1           tak p1 
        inak ak podm2   tak p2 
        inak ak podm3   tak p3 
        inak ... 
        inak ak podmn   tak pn  

 

h1, h2 až hn sú  hodnoty, ktoré môže   kde p1 až pn sú  príkazy 
 nadobudnú ť vý raz v 

V TP sa n-á rne vetvenie realizuje podmienený m príkazom case. 

Má  tvar:  case v of  kde v je tzv. vý berový  vý raz ordiná lneho typu, 

     h1 : p1;  h1, h2 až hn sú  hodnoty rovnakého typu ako vý berový  vý raz 

     h2 : p2;  a  p, p1, p2 až pn sú  príkazy 

         ... 

     hn : pn 

  [  else p  ]  do hranatý ch zá tvoriek sa umiestňuje nepovinná  časť  

  end   príkaz case končí vyhradený m slovom end 

Vykonanie: Vyhodnotí sa vý berový  vý raz a vykoná  príkaz, predznačený  hodnotou, ktorú  nadobudol 

vý berový  vý raz. Ak vý raz v nenadobudne ani jednu z hodnôt h1 až hn a príkaz case obsahuje časť else, 

vykoná  sa príkaz p; ak príkaz case neobsahuje časť else, príkaz case je bez ú činku. 

Príklad:        case MESIAC of 

  1,3,5,7,8,10,12 : DNI := 31; 

            2 :  if PRIESTUPNY then DNI := 29 else DNI := 28; 

                           4,6,9,11 : DNI := 30 

         end { do premennej DNI priradí počet dní v mesiaci } 

Dovolené sú  aj zá pisy s intervalmi (medzi minimá lnou a maximá lnou hodnotou sú  dve bodky): 

 case ZNAK of 
    ’A’..’V’ , ’a’..’v’ : write( succ(ZNAK) ); 
                  ’Z’ : write( ’A’ ) ; 
            ’z’ : write( ’a’ ) 
     else write(ZNAK) 
 end { príkaz zakóduje písmeno na nasledujú ce  (Z, z na A, a) a zobrazí ho } 
  { iné znaky nemení, zobrazí pôvodné } 

vý raz v 

p2 p1 pn 

h1 hn h2 



 Zá klady programovania  – Turbo Pascal 30  

Príklad VCS-1: Program na určenie aktuá lneho dňa v tý ždni pomocou príkazu GetDate (unit Dos), na 

určenie počtu dní do konca mesiaca a do konca roka. 

Analý za: Vý sledkom príkazu GetDate (pozri pomocou Ctrl+F1) je aktuá lny rok, mesiac, deň a poradové 

číslo dňa v tý ždni, pričom poradové číslo nedele je 0, pondelka 1 atď. Využili sme aj tvrdenie, že rok je 

priestupný  (februá r má  29 dní), ak je deliteľný  4 a pritom nie je deliteľný  100 alebo je deliteľný  400. Rok 

2000 by mal byť priestupný , lebo je deliteľný  4, nemal by byť priestupný , lebo je deliteľný  100, ale je 

priestupný , pretože je deliteľný  400. Zaujímavé. Ak rok nie je deliteľný  4, určite je nepriestupný . 

Dali sme si zá ležať aj na koncovke slova deň, dni, dní (slovenská  diakritika sa do TP dá  dostať). 

Riešenie vý počtu dní do konca roka nie je veľmi elegantné, pokú ste sa o krajšie riešenie . 
 
program POCET_DNI; 
uses crt,dos; 
var 
  d, m, r, por : word; 
  pdm, dkm: byte;    { pdm - pocet dni v mesiaci, dkm - do konca mesiaca } 
  dkr:integer;       { dkr - do konca roka } 
  priestupny:boolean;{ vysledkom je true, ak je rok priestupny, inak false } 
begin 
clrscr; 
GetDate(r,m,d,por); 
write(’Dnes je ’); 
case por of 
  0: write(’nedela’); 
  1: write(’pondelok’); 
  2: write(’utorok’); 
  3: write(’streda’); 
  4: write(’stvrtok’); 
  5: write(’piatok’); 
  6: write(’sobota’); 
end; 
writeln(’ ’, d, ’.’, m, ’.’, r, ’.’); 
writeln; 
priestupny:=(r mod 4=0) and ((r mod 100<>0) or (r mod 400=0)); 
case m of 
  1,3,5,7,8,10,12: pdm:=31; 
                2: if priestupny 
                     then pdm:=29 
                     else pdm:=28; 
         4,6,9,11: pdm:=30; 
end; 
dkm:=pdm-d; 
write(’Do konca mesiaca ’); 
case dkm of 
      1: writeln(’zostáva 1 deň.’); 
  2,3,4: writeln(’zostávajú ’, dkm,’ dni.’); 
  else writeln(’zostáva ’, dkm,’ dní.’) 
  end; 
writeln; 
case m of 
  1: if priestupny 
       then dkr:=366-d 
       else dkr:=365-d; 
  2: if priestupny 
       then dkr:=366-31-d 
       else dkr:=365-31-d; 
  3: dkr:=dkm+30+31+30+31+31+30+31+30+31; 
  4: dkr:=dkm+31+30+31+31+30+31+30+31; 
  5: dkr:=dkm+30+31+31+30+31+30+31; 
  6: dkr:=dkm+31+31+30+31+30+31; 
  7: dkr:=dkm+31+30+31+30+31; 

  8: dkr:=dkm+30+31+30+31; 
  9: dkr:=dkm+31+30+31; 
 10: dkr:=dkm+30+31; 
 11: dkr:=dkm+31; 
 12: dkr:=dkm; 
end; 
write(’Do konca roka ’); 
case dkr of 
      1: writeln(’zostáva 1 deň.’); 
  2,3,4: writeln(’zostávajú ’, dkr,’ dni.’); 
  else writeln(’zostáva ’, dkr,’ dní.’) 
end; 
readln 
end. 
 
Nerieš ené  úlohy na vetvenie: 

1. Vytvorte program na riešenie rovnice ax + b = 0,   a, b reá lne. 

2. Vytvorte program na riešenie kvadratickej rovnice  ax2 + bx + c = 0,  a, b, c reá lne, a ≠ 0. 

Ná vod: D = b
2 - 4.a.c ; pre D ≥ 0 platí x1,2 = ( -b ± √ D ) / (2.a).  

3. Vytvorte program na riešenie rovnice ax2 + bx + c = 0,  a, b, c reá lne. 

Ná vod: Pozri ú lohy č. 1 a 2. 

4. Vytvorte program na vý počet nezná mej strany pravouhlého trojuholníka po zadaní usporiadanej trojice 

[a, b, c] ( a, b odvesny, c prepona), kde nezná ma strana má  pri zadaní veľkosť 0. 

Ná vod: Použi Pythagorovu vetu, napríklad pre trojicu [3,0,5] sa počíta b =  (5
2 - 3

2
) = 4. 

5. Vytvorte program na  vý počet vý šky poštovného pri podaní peňažnej pouká žky (tarifná  tabuľka je na 

druhej strane pouká žky). 

6. Vytvorte program na vý počet povrchu a obvodu zvoleného rovinného ú tvaru. Ponuka napr.: štvorec, 

obdĺžnik, trojuholník, lichobežník, kruh a pod. (selekcia príkazom case!). 

7. Vytvorte program, ktorý  zistí, či zadané tri čísla môžu byť veľkosti strá n trojuholníka. 

Ná vod: Použi trojuholníkovú  nerovnosť: sú čet dvoch strá n v trojuholníku je vždy väčší ako tretia strana. 

8. Uká žte, že tri podmienky umožňujú  rozlíšiť 8 situá cií. 

9. Osoby majú  hmotnosť x , y a z. Nosnosť vý ťahu je w. Vytvorte  program na zistenie, koľko já zd je treba, 

aby sa všetci dostali vý ťahom z prízemia na 13. poschodie. 

10. Je daný  počet dní v mesiaci a informá cia, na ktorý  deň v tý ždni pripadá  prvý  deň v mesiaci (pondelok - 

1, utorok - 2, atď.).  Zistite, koľko je v danom mesiaci piatkov. 

11. Je daný  počet dní v mesiaci a informá cia, na ktorý  deň v tý ždni pripadá  prvý  deň v mesiaci (pondelok - 

1, utorok - 2, atď.).  Zistite, koľko je v danom mesiaci pracovný ch dní. 

12. Vytvorte program, ktorý  po zadaní postupnosti celý ch čísel ozná mi, či je postupnosť rastú ca, 

neklesajú ca, klesajú ca,  nerastú ca, konštantná  alebo nemá  žiadnu z tý chto vlastností. 

CYKLUS 
Použijeme, ak nejaký  príkaz alebo skupina príkazov sa má  opakovane vykoná vať, pokiaľ je splnená  alebo 

nesplnená  daná  podmienka. 

Cykly rozdeľujeme: 

1. podľa umiestnenia podmienky na: 

a) cyklus s podmienkou na začiatku 

b) cyklus s podmienkou na konci 

c) ú plný  cyklus (s podmienkou v strede) 

2. podľa toho, či je zná my alebo nezná my počet opakovaní v cykle na: 

a) cyklus s explicitne daný m počtom opakovaní 

b) cyklus s implicitne daný m počtom opakovaní. 

Cyklus s explicitne daným počtom opakovaní, príkaz for 

Cyklus s explicitne („zvonka“) daný m počtom opakovaní použijeme pri zná mom počte opakovaní príkazov 

v cykle. 

V algoritmizá cii nemá  tento cyklus jednoznačnú  formu zobrazenia, my sme sa rozhodli pre tvar: 

       v slovnom zá pise: 

        

       pre rpc := v1 až po (naspäť po) v2 opakuj 
            p 

        

       kde rpc je tzv. riadiaca premenná  cyklu (riadi počet 
       prechodov cyklom), v1 a v2 sú  vý razy a p je príkaz 

 

V TP cyklu so zná mym počtom opakovaní zodpovedá  príkaz for. 

Má  tvar: for rpc := v1 to (downto) v2 do 
      p 

kde rpc je tzv. riadiaca premenná  cyklu ordiná lneho typu, v1 a v2 sú  vý razy rovnakého typu 
ako rpc a p je príkaz; príkaz for má  dva varianty, buď s „to“ alebo s „downto“ 
 

Vykonanie príkazu for (v zá tvorká ch pre downto): 

1.  vyhodnotia sa vý razy v1 a v2; hodnota vý razu v1 sa priradí ako začiatočná  hodnota rpc, hodnota 

vý razu v2 ako koncová  hodnota rpc, 

2.  pokiaľ je hodnota rpc menšia (väčšia) alebo rovná  koncovej hodnote, opakovane sa vykoná va príkaz p 

a rpc nadobú da hodnoty succ(rpc) (pred(rpc)). 

Pozná mky:  

• Riadiaca premenná  cyklu je ordiná lneho typu, t.j. môže byť len typu integer, boolean alebo char. Z toho 

vyplý va, že nemôže nadobú dať hodnoty meniace sa o ľubovoľný  krok, napr. 0,5; 0,1 a pod. Programá tor 

s dobrý m programá torský m štý lom zá sadne nemení hodnotu rpc ani o krok napr. 2, 10 a pod., len na 

nasledovníka resp. predchodcu. Ak potrebujeme zrealizovať cyklus s „nevhodný m“ krokom pre príkaz 

for, použijeme cyklus s príkazom while alebo repeat. 

  rpc ≤ v2 

 p 

– 

+ 

  rpc := v1 

rpc := rpc+1 

• Začiatočná  a koncová  hodnota riadiacej premennej cyklu sa získa vyhodnotením vý razov v1 a v2 na 

začiatku vykoná vania príkazu for a počas cyklu ich nemožno meniť. 

• Z historický ch dôvodov (z čias programovacieho jazyka fortran) sa zaužívalo najčastejšie označiť 

riadiacu premennú  cyklu písmenom I, nie je to však pravidlo. 

Príklady: for I := 1 to 100 do write( I : 4 ) 

príkaz zobrazí čísla 1   2   3   4 ... 99 100 v piatich riadkoch po 20 čísel (80 znakov v riadku 

delené 4 znaky na číslo = 20 čísel v riadku) 

for I := 100 downto 1 do write ( I : 4 ) 

príkaz zobrazí čísla 100  99  98  ... 2   1, t.j. opačne ako v predchá dzajú com príkaze for 

  SUCET := 0; 

  for CISLO := 1 to POCET do  SUCET := SUCET + CISLO 

  príkaz sčíta čísla 0 + 1 + 2 + 3 + ... + POCET, 0 je počiatočná  hodnota premennej SUCET 

  MOCNINA := 1; 

  for I := 1 to N do MOCNINA := MOCNINA * X 

  príkaz vykoná  1.X.X. ... .X = XN, 1 je počiatočná  hodnota premennej MOCNINA 

  for Z := ’A’ to ’Z’ do writeln( Z : 5, ord( Z ) : 3 ) 

  príkaz zobrazí v riadkoch znaky A až Z a ich poradové čísla z ASCII tabuľky 

  for J := -10 to 10 do p 

  príkaz for vykoná  príkaz p 21-krá t (J bude postupne -10, -9, ... , 0, ... , 10) 

  for K := 0 to (downto) 0 do p 

  príkaz for vykoná  príkaz p jedenkrá t 

  for NIC := 100 to –100 do p 

príkaz for nevykoná  príkaz p ani raz (začiatočná  hodnota rpc nie je ani raz menšia alebo 

rovná  koncovej hodnote) 

  for NIC := 0 downto 10 do p 

príkaz for nevykoná  príkaz p ani raz (začiatočná  hodnota rpc nie je ani raz väčšia alebo 

rovná  koncovej hodnote) 

  for NIC := ’z’ to ’A’ do p 

príkaz for nevykoná  príkaz p ani raz, pretože ’z’ > ’A’ (znak z má  v ASCII tabuľke väčšie 

poradové číslo) 

for I := - trunc( sqrt(N) ) to trunc( sqrt(N) ) do p 

Vý razy v1 a v2 nemusia mať pri zá pise len konkrétne hodnoty, môžu to byť aj vý razy. 

 

Príklad CFR-1: Vytvorte program na zobrazenie hodnôt riadiacej premennej cyklu zo zadaného intervalu 

pre typ integer aj char. Uvedomte si, že ak variant „to“ zobrazí viac ako jednu hodnotu, variant „downto“ 

nemôže zobraziť ani jednu hodnotu a naopak. 
 
program HODNOTY_RPC; 
uses crt; 
var  ZAC, KON, I : integer; 
     ZACZ, KONZ, Z : char; 
begin 

clrscr; 
write(’Zadaj zaciatocnu a koncovu hodnotu typu integer: ’); 
readln(ZAC,KON); 
writeln(’Hodnoty rpc pre "to":’); 
for I:=ZAC to KON do 
  write(I:8); 
writeln; 
writeln(’Hodnoty rpc pre "downto":’); 
for I:=ZAC downto KON do 
  write(I:8); 
writeln; writeln; 
write(’Zadaj zaciatocnu hodnotu typu char: ’); 
readln(ZACZ); 
write(’Zadaj  koncovu  hodnotu  typu char: ’); 
readln(KONZ); 
writeln(’Hodnoty rpc pre "to":’); 
for Z:=ZACZ to KONZ do 
  write(Z:4); 
writeln; 
writeln(’Hodnoty rpc pre "downto":’); 
for Z:=ZACZ downto KONZ do 
  write(Z:4); 
writeln; 
readln 
end. 
Neuviedli sme analý zu, radšej upozorníme na „problém“ s načítaním viacerý ch hodnôt typu char v jednom 

príkaze readln. Keby sme, tak ako pri type integer, umiestnili premenné ZACZ a KONZ do jedného príkazu 

readln a hodnoty vložili oddelené medzerou, premennej KONZ by vždy priradilo medzeru (druhý  znak je 

predsa medzera)! Preto nečíselné hodnoty načítavame z klá vesnice väčšinou v samostatný ch príkazoch 

readln. Ak chceme načítať obe hodnoty v jednom príkaze readln, musíme ich vložiť za sebou bez oddelenia. 

 

Príklad CFR-2: Vytvorte program na vý počet sú čtu všetký ch prirodzený ch čísel od 1 po zadané N pomocou 

cyklu.  

Analý za:  

1. Akú  algoritmickú  konštrukciu treba použiť? 

Napríklad pre N = 100 treba sčítať 1 + 2 + 3 + ... + 99 + 100 = 5050. Vidíme, že sa opakuje sčítavanie.  

Keďže opakovane treba vykoná vať nejaký  príkaz, použijeme cyklus. Pozná me aj počet opakovaní, 

N-krá t (aj 1 treba pričítať k pôvodnej hodnote v pamäti), preto to je cyklus so zná mym počtom 

opakovaní a ten sa realizuje príkazom for:   for CISLO := 1 to N do 

2. Aký  príkaz sa má  v cykle opakovať?  

Vyzerá  to tak, že vždy k čiastočnému sú čtu, t.j. k sú čtu, ký m nie je všetko sčítané, treba pripočítať 

vhodné číslo. Vhodné číslo je pri prvom prechode cyklom 1, pri druhom 2, pri treťom 3 atď., až pri 

n-tom N. Ak sme pozorní, uvedomíme si, že prá ve tieto hodnoty nadobú da riadiaca premenná  cyklu for. 

Preto:          nový  čiastočný  sú čet = predchá dzajú ci čiastočný  sú čet + CISLO 

Ak dobre pozná me vykonanie príkazu priradenia, uvedomíme si, že ten presne pracuje s „nový m 

a predchá dzajú cim“, ak na jeho ľavej aj pravej strane je tá  istá  premenná  (to isté pamäťové miesto)! 

Preto v cykle treba opakovať príkaz:  SUCET := SUCET + CISLO 

V príkaze takéhoto typu musíte neustá le vidieť: nový  SUCET = starý  SUCET + ...  

alebo všeobecne: nová  HODNOTA = stará  HODNOTA +,*,-,/... 

3. Teraz je už všetko v poriadku? 

Ešte nie. Ak začneme simulovať vý počet v cykle, zistíme, že nepozná me hodnotu premennej SUCET na 

pravej strane pri prvom prechode cyklom. Nepozná me počiatočný  sú čet resp. nenastavili sme ešte 

počiatočnú  hodnotu premennej SUCET. Sprá vne vykonanie príkazu priradenia, v ktorom je rovnaká  

premenná  na ľavej aj pravej strane, si vyžaduje pred príkazom cyklu nastavenie počiatočnej hodnoty 

takejto premennej. Pre sčítanie to bý va (ale vo všeobecnosti nemusí byť) 0, pre ná sobenie 1. Preto pred 

príkaz for vložíme ešte príkaz:  SUCET := 0; 

Už počujeme hlasy, že vá š program počíta sprá vne aj bez tohto priradenia (spomínali sme, že TP 7.0 vloží 

do všetký ch číselný ch premenný ch počiatočnú  hodnotu 0). Čo však, ak program nespustíte v TP 7.0 alebo 

ho vložíte do cyklu, aby vá m viackrá t umožnil zopakovať vý počet? Potom určite nebude počítať sprá vne 

(počiatočnou hodnotou bude ná hodné číslo alebo hodnota predchá dzajú ceho sú čtu). 

Teraz je už všetko v poriadku a tu je program: 
program SucetPrvychN; 
uses crt; 
var  N, CISLO, SUCET : integer; { pripadne typ word alebo longint } 
begin 
clrscr; 
write(’Scitat prirodzene cisla po ’); 
readln(N); 
SUCET:=0; 
for CISLO:=1 to N do 
  SUCET:=SUCET+CISLO; 
writeln(’Sucet prvych ’,N,’ cisel je ’,SUCET); 
readln 
end. 

Pozná mka: Na vý počet sú čtu prvý ch n prirodzený ch čísel možno použiť aj vzorec SUCET = 
N
2  .(1+N), preto 

sme v zadaní ú lohy uviedli, že chceme vý počet pomocou cyklu. Zá roveň si uvedomte, že vý razne sa 

efektivitou vý počtu líšiacich algoritmov môže byť (a väčšinou aj bý va) viacej! 

Program upravte tak, aby: 

a) počítal sú čet prvý ch N prirodzený ch čísel od N po 1, 

b) počítal sú čet všetký ch prirodzený ch čísel od zadaného A po zadané B vrá tane, 

c) počítal sú čet všetký ch celý ch čísel od zadaného A po zadané B vrá tane. 

 

Príklad CFR-3: Vytvorte program na vý počet mocniny x na n-tú , x reá lne, n prirodzene číslo. 

Analý za: Začneme „sedliackym“ rozumom. Ak má me vypočítať napríklad 53, ná sobíme 5.5.5 = 125. Ď alej 

by sme mohli vedieť, že 00 nie je definované (asi preto, že 0 na „čokoľvek“ je 0 a „čokoľvek“ na 0-tú  je 1, 

preto 00 by malo byť 0 a zá roveň 1 a to nejde). 

1. Opakovane sa vykoná va ná sobenie, preto treba použiť cyklus; ná sobenie treba vykonať n-krá t, preto 

cyklus so zná mym počtom opakovaní, teda príkaz for. 

2. Vždy nový  sú čin sa rovná  predchá dzajú ci sú čin krá t x, preto sa v cykle bude opakovať príkaz SUCIN := 

SUCIN * X. 

3. Nesmieme zabudnú ť na počiatočnú  hodnotu premennej SUCIN, ktorá  môže byť 1. Keby sme zvolili 

nulu? 

4. Je tu však ešte jedno obmedzenie, ak zadá me N aj X nula, tak má  program ozná miť: „0 na 0-tú  

nedefinovane!“, inak má  „normá lne“ počítať – teda vetvenie. 

 
program MOCNINA; 
uses crt; 
var X, MOCNINA : real; 
     N, I, PDM : integer; 
begin 
clrscr; 
write(’Zadaj zaklad mocniny a exponent: ’); 
readln(X,N); 
write(’Zadaj pocet desatinnych miest vystupu: ’); 
readln(PDM); 
if (X=0) and (N=0) 
  then writeln(’0 ma 0-tu nedefinovane!’) 
  else begin 
       MOCNINA:=1; 
       for I:=1 to N do 
         MOCNINA:=MOCNINA*X; 
       writeln(X:PDM+2:PDM,’ na ’,N,’ = ’,MOCNINA:PDM+2:PDM) 
       end; 
readln 
end. 
 

Príklad CFR-4: Vytvorte program na vý počet aritmetického priemeru n reá lnych čísel. 

Analý za: Sná ď všetci vieme, že aritmetický  priemer sa vypočíta ako sú čet daný ch n čísel delený  počtom 

všetký ch čísel, t.j. n. Program si má  vypý tať n reá lnych čísel a sčítať ich, preto cyklus s „príkazmi“: zadaj 

číslo a k čiastočnému sú čtu pripočítaj nové číslo. Toto sa má  zopakovať pre n čísel, teda n-krá t. Takže cyklus 

so zná mym počtom opakovaní – príkaz for. 

V programe sme namiesto premennej n použili premennú  POCET a ošetrili sme aj prípad, keď bude zadaný  

počet čísel 0 - program by bez ošetrenia skončil chybový m hlá sením: Delenie nulou! 

 
program PRIEMER; 
uses crt; 
var  I, POCET : integer; 
     X, SUCET : real; 
begin 
clrscr; 
write(’Zadaj pocet cisel: ’); 
readln(POCET); 
SUCET:=0; 
for I:=1 to POCET do 
  begin 
  write(I:2,’. cislo: ’); 
  readln(X); 
  SUCET:=SUCET+X 
  end; 
if POCET>0 
  then writeln(’Priemer zaokruhleny na stotiny: ’,SUCET/POCET:4:2); 
readln 
end. 
 

Ď alšie ú lohy aj na príkaz for sú  uvedené v kapitole Ď alšie zaujímavé ú lohy. 

Cyklus s podmienkou na začiatku, príkaz while 

O typickom použití cyklu s podmienkou na začiatku si povieme v časti „Kedy ktorý  cyklus“. 

Má  tvar:      v slovnom zá pise: 

        

       pokiaľ podmienka opakuj 
            p 

 

       kde p je príkaz 

 

 

 

V TP má  tvar:  while b do 
     p    kde b je vý raz typu boolean a p príkaz 

Vykonanie: Pokiaľ vý raz b nadobú da hodnotu true, opakovane sa vykoná va príkaz p, ak vý raz b nadobudne 

hodnotu false, cyklus sa ukončí. 

Príklady:  I := 1;   
while I <= 10 do   
    I := I + 1;   príkaz v cykle sa vykoná  10-krá t 

SUCET := 0; 
while N > 0 do 
    begin 
    SUCET := SUCET + N; 
    N := N - 1 

     end    sčíta všetky prirodzené čísla od N po 1 

 DELITEL := 2; 
 while CISLO mod DELITEL <> 0  do 
      inc ( DELITEL )  ná jde deliteľa zadaného čísla ( CISLO > 1 ) 

 

Príklad CWH-1: Príkaz for je viac-menej špeciá lnym prípadom príkazu while. Demonštrujeme to 

v nasledujú com programe. 

 
program AKO_FOR; 
uses crt; 
var  N, I : integer; 
begin 
clrscr; 
write(’Zadaj pocet opakovani: ’); 
readln(N); 
writeln(’Ako prikaz for s "to" od 1 po ’,N); 
I:=1; 
while I<=N do 
  begin 
  writeln(I:6,’. opakovanie’); 
  I:=I+1 
  end; 
readln; 
 

podmienka 

  p 

– 

+ 

writeln(’Ako prikaz for s "downto" od ’,N,’ po 1’); 
while N>0 do 
  begin 
  writeln(N:6,’. opakovanie’); 
  N:=N-1 
  end; 
readln 
end. 
 

Príklad CWH-2: Vytvorte program na vý počet ciferného sú čtu zadaného prirodzeného čísla. 

Analý za: Ciferný  sú čet napríklad čísla 123 je 6, lebo 1 + 2 + 3 = 6. K prvej cifre (3) sa dostaneme 

predelením čísla 123 desiatimi: 123 : 10 = 12, zvyšok 3.  Pokú sme sa vystačiť s celý mi číslami. Ak spravíme 

123 mod 10, dostaneme rovno číslo 3. Ak spravíme 123 div 10 dostaneme 12, čo sa ná m tiež hodí, lebo ďalej 

potrebujeme to isté (mod a div) zopakovať s 12-kou (12 mod 10 = 2 a 12 div 10 = 1) a nakoniec s 1-kou 

(1 mod 10 = 1 a 1 div 10 = 0). Tý m sme prešli všetky cifry čísla. Opakovane spracú vam vý sledky po 

funkciá ch mod  a div s deliteľom 10, preto treba použiť cyklus. Zadané číslo môže byť jedno ale aj troj (to 

naše) prípadne desaťciferné a na začiatku cyklu nevieme, kedy spracujeme poslednú  cifru. Preto cyklus 

s nezná mym počtom opakovaní príkazov v cykle – príkaz while. Podmienkou ukončenia cyklu je, aby sme 

nemali už čo „spracovať“, t.j. aby aktuá lnym číslom bola nula (CISLO div 10 = 0). 

 
program CIFERNY_SUCET; 
uses crt; 
var  CISLO    : longint; 
     CIFSUCET : byte; 
begin 
clrscr; 
write(’Zadaj prirodzene cislo: ’); 
readln(CISLO); 
write(’Sucet cifier cisla ’,CISLO,’ je ’); 
CIFSUCET:=0; 
while CISLO>0 do 
  begin 
  CIFSUCET:=CIFSUCET+CISLO mod 10; 
  CISLO:=CISLO div 10 
  end; 
writeln(CIFSUCET); 
readln 
end. 
 
Prečo sme vý stup museli posunú ť hore? Ak sa vá m funkcie div a mod nepá čia, tu je riešenie bez nich: 
program CIFERNY_SUCET_2; 
uses crt; 
var  CISLO    : longint; 
     CIFSUCET : byte; 
begin 
clrscr; 
write(’Zadaj prirodzene cislo: ’);  
readln(CISLO); 
write(’Sucet cifier cisla ’,CISLO,’ je ’); 
CIFSUCET:=0; 
while CISLO>0 do 
  begin 
  CIFSUCET:=CIFSUCET+trunc(frac(CISLO/10)*10); 
  CISLO:=trunc(CISLO/10) 
  end; 
writeln(CIFSUCET); 
readln 
end. 

Príklad CWH-3: Vytvorte program na zobrazenie prevodnej tabuľky medzi slov. korunami a inou menou. 

Kurz je zadaný  v konštante. Tabuľka bude zobrazovať hodnoty od 0,- Sk po zadanú  sumu (premenná  

KoncovaSuma) so zadaný m prírastkom (premenná  Krok). Napríklad pre koncovú  sumu 100 s prírastkom 10 

má  vypočítať a zobraziť hodnoty pre 0, 10, 20, 30, 40, ... , 100 Sk. Vý počet hodnoty v cudzej mene je 

jednoduchý : suma v Sk krá t kurz. Keďže sa opakovane počíta hodnota v cudzej mene, treba použiť cyklus. 

Riadiaca premenná  cyklu sa však môže zväčšovať o teoreticky ľubovoľné kladné reá lne číslo, preto 

použijeme príkaz while. 

Program sme doplnili o jednoduchú  hlavičku tabuľky a hodnoty sú  zobrazované s presnosťou na desatiny, na 

miestach stotín sa zobrazujú  nuly. Doplňte tabuľku o zvislé „čiary“. 
program TABULKA; 
uses  crt; 
const KURZ=1.24; 
var   KoncovaSuma, Krok, RPC : real; 
begin 
clrscr; 
write(’Zadaj koncovu sumu a prirastok: ’); 
readln(KoncovaSuma, Krok); 
clrscr; 
writeln(’----------------------’:48); 
writeln(’      SK         MENA ’:48); 
writeln(’----------------------’:48); 
RPC:=0; 
while RPC<=KoncovaSuma do 
  begin 
  writeln(RPC:33:1,’0’,KURZ*RPC:12:1,’0’); 
  RPC:=RPC+KROK 
  end; 
writeln(’----------------------’:48); 
readln 
end. 
 

Cyklus s podmienkou na konci, príkaz repeat 

Má  tvar:      v slovnom zá pise: 

       opakuj 

          p1; 
          p2; 
           ... 
          pn 

       pokiaľ nebude podmienka (splnená ) 

        
       kde p1, p2 až pn sú  príkazy 

 

V TP má  tvar:  repeat 

     p1; 
     p2; 
      ... 
     pn 

until b  kde b je vý raz typu boolean a p1 až pn sú  príkazy 

Vykonanie: Vykonajú  sa príkazy p1, p2 až pn a opakovane sa budú  vykoná vať, pokiaľ vý raz b bude 

nadobú dať hodnotu false. Ak vý raz b nadobudne hodnotu true, cyklus sa ukončí. 

– podmienka 

p 

+ 

Príklady:  I := 0;    repeat     repeat 
repeat       write(’Zadaj číslo: ’);      p1; ... 

     I := I + 1      readln(X)       pn 
  until I = 10   until X = 0    until false 
  príkaz v cykle sa   príkazy v cykle sa budú    nekonečný  cyklus 

vykoná  10-krá t   opakovať, ký m nevložíme 0            ukončí sa  Ctrl+Break 
 

 

Príklad CRP-1: Vytvorte programovú  schému na zabezpečenie opakovania behu programu ľubovoľný  

počet krá t. 

Analý za: Už vá s nudí neustá le spú šťať ten istý  program, najmä keď ho testujete, či pracuje sprá vne? 

Riešením je príkaz repeat. Chceme dosiahnuť, aby sa po spustení program vykonal a na konci sa opý tal: 

„Opakovať?“. Ak odpovieme N = nie, program skončí, inak sa vykoná  znova. 

Má me viacej možností, najprv tá  najjednoduchšia: 

... 

begin 

repeat 

   { príkazy programu } 

   write(’Opakovať (nie = N) ?’); 

   readln(ODPOVED) 

until (ODPOVED = ’N’) or (ODPOVED = ’n’) 

end. 

Elegantnejším riešením je použiť funkciu unitu crt readkey. Vý sledkom funkcie readkey je stlačený  klá ves, 

ušetríme premennú  ODPOVED aj stlačenie klá vesu Enter. Readkey v kombiná cii s upcase: 

begin 

repeat 

   { príkazy programu } 

   writeln(’Opakovať (nie = N) ?’) 

until  upcase(readkey) = ’N’ 

end. 

Na ukončenie programov sa používa aj klá ves Esc. Pomocou programu na nasledujú cej strane zistíme, že 

jeho poradové číslo je 27. A tu je schéma:  const Esc = #27; 

begin      ... 

repeat      begin 

   { príkazy programu }    { príkazy programu }   

   writeln(’Koniec – stlač Esc!’:45)  writeln(’Koniec – stlač Esc!’:45) 

until  readkey = chr(27)    until  readkey = Esc  

end.      end. 

Zá pis chr(X) je rovnocenný  so zá pisom #X. 

Aj na zistenie poradového čísla stlačeného znaku môže poslú žiť malý  program s cyklom repeat: 
program KOD_STLACENEHO_KLAVESU; 
uses crt; 
var  stlaceny_klaves:char; 
begin 
clrscr; 
writeln(’Koniec – stlac C’:45); 
repeat 
  stlaceny_klaves:=readkey; 
  writeln(ord(stlaceny_klaves)); 
until upcase(stlaceny_klaves)=’C’; 
end. 
 

Príklad CRP-2: Vytvorte programovú  schému na ošetrenie sprá vnosti vstupný ch hodnôt. Ak vstupné 

hodnoty nespĺňajú  vstupné podmienky, majú  sa vyžiadať nové vstupné hodnoty. 

Analý za: Príkaz repeat je vhodný  aj na ošetrenie vstupný ch hodnôt, či spĺňajú  vstupné podmienky. Najprv 

predsa musíme zadať vstupné hodnoty a až potom môže podmienka rozhodnú ť, či treba vstup opakovať 

alebo môže program pokračovať. Ako uká žku ošetríme vý počet mocniny xn pre x = 0 a n = 0 (pozri príklad 

CFR-3):   
   repeat 

    write(’Zadaj zaklad mocniny a exponent: ’); 
   readln(X,N) 
 until (X<>0) or (N<>0) 

Sprá vnejšie je nie len „zacykliť“ vstup ale aj doplniť vý pisom, prečo sa príkazy vstupu opakujú . Teda: 
repeat 

   write(’Zadaj zaklad mocniny a exponent: ’); 
  readln(X,N); 
  if (X=0) and (N=0) then writeln(’0 na 0-tú nedefinované!’) 
until (X<>0) or (N<>0); 

 
Pod chybou vstupu môžeme rozumieť aj „preklep“, napríklad vložením písmena namiesto číslice 

(Error 106: Invalid numeric format.). K vyriešeniu tohto problému treba poznať funkciu IOResult, ktorá  po 

každej vstupno/vý stupnej operá cii nadobú da buď hodnotu 0, ak nedošlo ku chybe, alebo číslo chyby. Keďže 

sme zapli v Options – Compiler... prepínač I/O Checking – kontrolu všetký ch vstupno/vý stupný ch operá cií, 

program by hneď skončil s chybový m hlá sením. Tú to kontrolu však možno vypnú ť umiestnením direktívy 

{$I - } pred príkaz vstupu a ná sledne ju zapnú ť príkazom {$I +}. Schéma má  tvar: 

... 
var CHYBA : boolean; 
     X : integer;  { X aký koľvek číselný  typ } 
... 

 begin 

 repeat 

    write(’Vlož číslo: ’); 

    {$I-} readln(X); {$I+} 

    CHYBA := IOResult <> 0; 

    if CHYBA then writeln(’Chyba vstupu!’) 

 until not CHYBA; 

 napr. writeln(X) 

Ak dôjde ku chybe, IOResult bude rôzny od nuly a preto premenná  CHYBA nadobudne hodnotu TRUE. Na 

obrazovke sa zobrazí: Chyba vstupu! a znova Vlož číslo:, pretože CHYBA má  hodnotu true a negá cia (not) 

true je false – vtedy sa príkazy v repeat opakujú . Otestujte program pre dobré aj zlé vstupné hodnoty 

A teraz už „normá lny“ príklad: 

Príklad CRP-3: Vytvorte program – hru na uhá dnutie zadaného čísla, keď po vložení čísla program ozná mi 

„Veľa, uber!“, „Má lo, pridaj!“ alebo „Vý borne, uhá dol si!“. Nech treba uhá dnuť celé číslo z intervalu od 0 

po 100. 

Analý za: Pre získanie čísla, ktoré bude treba uhá dnuť sú  dve možnosti:  

• buď má me protihrá ča a ten zadá  číslo tak, aby to sú per nevidel (napríklad čiernym písmom – nastavuje 

sa príkazom textcolor na čiernom podklade – nastavuje sa príkazom textbackground), alebo 

• číslo pomocou funkcie random vyberie počítač (random(x) - vý sledkom je ná hodné celé číslo z intervalu 

<0,x); pred  príkazom random musí byť použitý  na začiatku programu príkaz randomize). 

Druhá  časť programu je vlastné há danie. „Sedliacky“ rozum hovorí: vložíme číslo, počítač ho porovná  

s číslom, ktoré treba uhá dnuť a ozná mi „Veľa, uber!“ alebo „Má lo, pridaj!“ a znova sa celé opakuje, až ký m 

neuhá dneme, vtedy há danie končí. Keďže nevieme, na koľký  pokus sa ná m podarí uhá dnuť hľadané číslo, je 

to cyklus s nezná mym počtom opakovaní. Najprv musíme vložiť číslo a porovnať a až potom vieme 

rozhodnú ť, či cyklus končí – preto cyklus s podmienkou na konci. A tu je hotový  program: 
program HADAJ; 
uses  crt; 
const MAX=100; 
var   CISLO,UHCISLO:integer; 
      ODP:char; 
begin 
randomize; 
repeat 
  clrscr; 
  gotoxy(1,12); writeln(’Cislo vyberie pocitac ....... P’:50); 
  writeln;      writeln(’Cislo zada protihrac ........ H’:50); 
  ODP:=upcase(readkey); 
  clrscr; 
  case ODP of 
    ’P’: UHCISLO:=random(MAX+1); 
    ’H’: begin 
         gotoxy(15,12); 
         write(’Zadaj cislo na uhadnutie < ’,MAX,’: ’); 
         textcolor(black); 
         readln(UHCISLO); 
         textcolor(white); 
         clrscr; 
         end; 
  end; 
  writeln(’H A D A J  C I S L O !’); 
  repeat 
    readln(CISLO); 
    if CISLO <> UHCISLO 
      then if CISLO<UHCISLO then writeln(’Malo, pridaj!’) 
                            else writeln(’Vela, uber!’); 
  until CISLO=UHCISLO; 
  clrscr; textcolor(yellow+blink); 
  gotoxy(30,12); write(’B  I  N  G  O  ! ! ! ’); 
  delay(2500); 
  normvideo; 
  gotoxy(30,24); write(’Este raz? (nie = N) ’); 
until upcase(readkey)=’N’; 
end. 

Príkaz normvideo by mal nastaviť štandardné atribú ty pre farby textu aj pozadia. Program môžete doplniť aj 

o zvukové efekty pomocou príkazov sound(frekvencia) a nosound – vypína zvuk, tento nesmiete zabudnú ť 

uviesť po príkaze sound. Dĺžka trvania tónu sa realizuje príkazom delay(milisekú nd). 

Kedy ktorý  cyklus 

 Keď už má me zá kladnú  predstavu o fungovaní jednotlivý ch cyklov a im zodpovedajú cich príkazov, 

môžeme si poznatky trochu zosystematizovať: 

• ak vieme počet opakovaní príkazov v cykle a premenná , ktorá  riadi počet prechodov cyklom, sa môže 

meniť na nasledovníka alebo predchodcu (krok +/- 1 alebo postupnosť za sebou idú cich znakov ASCII), 

použijeme príkaz for 

• ak nevieme počet opakovaní príkazov v cykle alebo premenná , ktorá  riadi počet prechodov cyklom, 

nenadobú da „pekné“ hodnoty (hodnoty nasledovníkov alebo predchodcov), použijeme príkazy while 

alebo repeat, pričom: 

• príkaz while, t.j. s podmienkou na začiatku použijeme, ak môže nastať situá cii, že príkaz v cykle sa 

nemá  vykonať ani raz 

• príkaz repeat, t.j. s podmienkou na konci použijeme, ak sa príkazy v cykle majú  vykonať aspoň raz. 

Uvedomte si, že v každom cykle musí byť premenná , ktorá  riadi počet prechodov daný m cyklom! 

Kontrolujte, či sa jej hodnoty zväčšujú  alebo zmenšujú  a či sú  ohraničené podmienkou ukončenia cyklu, iná č 

cyklus nikdy neskončí. 

Sem-tam sa u začiatočníkov stá va, v snahe nezabudnú ť bodkočiarku, že ju dajú  aj za slovo „do“ v príkaze for 

alebo while. Tý m vznikol tzv. prá zdny príkaz a cyklus zrejme pracuje celkom iná č, ako to bolo pôvodne 

plá nované. 

 

Organizácia programu 
 Poslednou teoretickou časťou je organizá cia programu v Turbo Pascale.  

Program sa skladá : 

program meno programu;    hlavička programu (môže chý bať) 

uses unity oddelené čiarkami;    deklará cia unitov 

const meno konš tanty = hodnota;   definícia konštá nt 

type meno typu = ...;     definícia typov       ú sek definícií 

var premenné oddelené čiarkami : typ premenný ch; deklará cia premenný ch       a deklará cií 

procedure ...      deklará cia 

function ...      procedú r a funkcií    blok 

begin              

príkaz1;      príkazová  časť 

príkaz2;      začína slovom begin 

...       a končí slovom end 

príkazn 

end.       bodka – koniec programu (nepatrí do bloku) 

Pozná mky:  

• Niektoré objekty z predchá dzajú cej schémy ešte nepozná me.  

• Príkazy sa v príkazovej časti oddeľujú  bodkočiarkou.  

• Všetky mená  objektov sú  identifiká tory. 

EXE verzie programov 

 Turbo Pascal umožňuje vytvá rať aj od prostredia TP nezá vislé verzie programov, priamo programy 

s príponou exe. Keď chceme získať exe verziu programu, musíme jeho kompilá ciu (preklad) presmerovať 

z pamäte na disk. Robí sa to prepnutím položky Destination v Compile z hodnoty Memory na hodnotu Disk. 

Program s rovnaký m ná zvom ako bol pôvodný  sú bor, len s príponou exe, sa uloží do aktuá lneho adresá ra.  

 

Ď alš ie rieš ené úlohy 
Príklad CWH-4: Vytvorte program na krá tenie zlomkov. 

Analý za: Napr. v zlomku 15/12 sa dá  čitateľ aj menovateľ deliť číslom 3 a zlomok sa dá  upraviť do 

zá kladného tvaru 5/4. Číslo 3, ktorý m sme krá tili zlomok, je najväčší spoločný  deliteľ (nsd) čitateľa a 

menovateľa zlomku. Nsd(15,12) sa vypočíta takto: nsd(15,12) = nsd(15-12,12) = nsd(3,12) = nsd(3,12-3) = 

nsd(3,9) = nsd(3,9-6) = nsd(3,3) = 3. Opakujeme odčítanie menšieho čísla od väčšieho dovtedy, ký m sa 

nerovnajú . Rovnaké číslo je najväčší spoločný  deliteľ pôvodný ch čísel. 

V slovnom zá pise:  pokiaľ je A<>B opakuj 

        ak A>B 

    tak A := A-B 

    inak B := B-A; 

   NSD := A  { alebo B, keďže sú  rovnaké } 

Vý stup programu nech je v tvare: 15 / 12 = 5 / 4, ak je podiel celé číslo, zlomková  čiara a menovateľ 1 nech 

sa nezobrazia, napr. 12 / 6 = 2. 

 
program ZLOMOK; 
uses crt; 
var CIT,MEN,A,B:integer; 
begin 
clrscr; 
repeat 
  write(’Zadaj citatela a menovatela zlomku: ’ ); 
  readln(CIT,MEN); 
  write(CIT,’ / ’,MEN,’ = ’); 
  A:=CIT; 
  B:=MEN; 
  while A<>B do 
    if A>B 
      then A:=A-B 
      else B:=B-A; 
  CIT:=CIT div A; 
  MEN:=MEN div B; 
  if MEN=1 
    then writeln(CIT) 
    else writeln(CIT,’ / ’,MEN); 
  writeln(’Koniec - stlac Esc!’) 
until readkey=chr(27) 
end. 

Príklad CRP-4: Vytvorte program na vý počet aritmetického priemeru vopred nezná meho počtu kladný ch 

reá lnych čísel. 

Analý za: Ú loha je podobná  ú lohe CFR-4, v nej však bol zná my počet čísel a preto sme mohli použiť príkaz 

for. Pri nezná mom počte čísel musíme použiť cyklus s nezná mym počtom opakovaní. Ukončenie cyklu je 

v takomto prípade na tzv. koncovú  hodnotu. V našom príklade sú  všetky zadá vané reá lne čísla kladné a preto 

môžeme zvoliť za koncovú  hodnotu nulu. Znamená  to, že príkazy v cykle sa budú  opakovať, pokiaľ 

nevložíme 0, vtedy sa cyklus ukončí. Najprv vložíme číslo a potom môžeme rozhodnú ť, či cyklus končí, teda 

podmienka je na konci, preto príkaz repeat. Ošetrite aj prípad, keď hneď prvý m vložený m číslom bude 0. 
 
program PRIEMER2; 
uses crt; 
var     POCET : integer; 
     X, SUCET : real; 
begin 
clrscr; 
POCET:=0; 
SUCET:=0; 
repeat 
  write(’Vloz cislo: ’); 
  readln(X); 
  if X>0 
    then begin 
         inc(POCET); 
         SUCET:=SUCET+X 
         end 
until X=0; 
if POCET>0 
  then writeln(’Priemer zaokruhleny na stotiny: ’,SUCET/POCET:4:2) 
  else writeln(’Bola zadana len koncova hodnota!’); 
readln 
end. 
 

Program CFR-5: Program na prevod malý ch písmen textu na veľké. 

Analý za: Prakticky si treba len osvojiť zá pis napr. TEXT[1] – prvý  znak reťazca v premennej TEXT, 

TEXT[5] – piaty znak reťazca, TEXT[I] – I-ty znak reťazca v premennej TEXT. Ak teda chceme zmeniť 

všetky písmená  v reťazci na veľké, musíme ísť od prvého znaku až po posledný . Počet znakov v reťazci ná m 

zistí funkcia length (vracia číslo – dĺžku reťazca). Funkcia upcase zmení malé písmeno na veľké, ostatné 

znaky nemení. 

 
program PREVOD; 
uses crt; 
var TEXT : string; 
    DLZKA, I : integer; 
begin 
clrscr; 
write(’Text: ’); 
readln(TEXT); 
DLZKA:=length(TEXT); 
for I:=1 to DLZKA do 
  TEXT[I]:=upcase(TEXT[I]); 
writeln(TEXT); 
readln 
end. 

Program CFR-6: Vytvorte program na  kódovanie textu posunutím o jeden znak vpravo (Zz -> Aa) a 

dekódovanie textu posunutím o jeden znak vľavo (Aa -> Zz). 

Analý za: Podstatou algoritmu je brať znak po znaku text a pokiaľ je znakom písmeno, zobraziť jeho 

nasledovníka – pri kódovaní, resp. predchodcu – pri dekódovaní. Situá ciu trochu komplikujú  okrajové 

písmená , pretože veľké Z a malé z sa majú  posunú ť na veľké A a malé a resp. Aa na Zz pri dekódovaní. 

Keďže je viacej možností, a na každú  sa má  iná č kódovať, vhodné je použiť príkaz case s časťou else (každý  

iný  znak ako písmeno sa nemá  zmeniť). Prejsť treba celý  text, od prvého po posledný  znak, čo ná m zaručí 

príkaz for (v2 = length(TEXT)). Program musí umožňovať vybrať si z dvoch možností: Kódovať alebo 

Dekódovať. 

 
program KODOVANIE_a_DEKODOVANIE_posunutim_o_znak; 
uses crt; 
var TEXT:string; 
    I:byte; 
    ODPOVED:char; 
begin 
clrscr; 
write(’Text: ’); 
readln(TEXT); 
writeln(’[K]odovat alebo [D]ekodovat?’); 
ODPOVED:=upcase(readkey); 
if ODPOVED=’K’ 
  then begin 
       for I:=1 to length(TEXT) do 
         case TEXT[i] of 
           ’A’..’V’,’a’..’v’: write(succ(TEXT[I])); 
                         ’Z’: write(’A’); 
                         ’z’: write(’a’); 
           else write(TEXT[I]) 
         end; 
       writeln 
       end 
   else if ODPOVED=’D’ 
          then begin 
               for I:=1 to length(TEXT) do 
                 case TEXT[i] of 
                   ’B’..’Z’,’b’..’z’: write(pred(TEXT[I])); 
                                 ’A’: write(’Z’); 
                                 ’a’: write(’z’); 
                   else write(TEXT[I]) 
                 end; 
               writeln 
               end 
          else writeln(’Nebolo stlacene K alebo D!’); 
readln 
end. 
 

Príklad CFR-7*: Príklad je určený  pre zdatnejších alebo hú ževnatejších programá torov. Je principiá lne 

odlišný  od predchá dzajú ceho. Program má  kódovať a dekódovať písmená  textu posunutím o POSUN písmen 

vpravo (vľavo). 

Analý za: Programy s hviezdičkou (nie sú  zaká zané) sú  určené na prá cu s hotový m programom. Ujasnite si 

jednotlivé časti programu a vyskú šajte, ako pracujú . Program si pamätá  predchá dzajú ci text (pri prvom 

spustení samozrejme nie), aby ste ho mohli po zakódovaní hneď dekódovať a opačne. Dekódovanie je 

v skutočnosti kódovanie o 26-POSUN znakov (neustá le vpravo). Číslo 26 je počet písmen (A až Z alebo 

malé a až z), pričom poradové číslo A je 65 a malého a 97. Pri kódovaní sa pohybujeme v uzavretom kruhu. 

program KODOVANIE_a_DEKODOVANIE_posunutim_o_POSUN_znakov; 
uses crt; 
var TEXT:string; 
    POSUN:integer; 
    DLZKA,i,ZacAbc,PozVAbc,RelPozKodu:byte; 
    ODPOVED:char; 
begin 
clrscr; 
repeat 
  writeln(’Novy text (ano - A)?’); 
  if upcase(readkey)=’A’ 
    then begin 
         writeln(’Text:’); 
         readln(TEXT); 
         DLZKA:=length(TEXT); 
         end; 
  repeat 
    write(’[K]odovat alebo [D]ekodovat? ’); 
    readln(ODPOVED); 
    ODPOVED:=upcase(ODPOVED); 
  until (ODPOVED=’K’) or (ODPOVED=’D’); 
  write(’Posun o znakov: ’); 
  readln(POSUN); 
  POSUN:=POSUN mod 26; 
  if POSUN<0 then POSUN:=POSUN+26; 
  if ODPOVED=’D’ then POSUN:=26-POSUN; 
  for i:=1 to DLZKA do 
    if TEXT[i] in [’A’..’Z’,’a’..’z’] 
      then begin 
           case TEXT[i] of 
             ’A’..’Z’:ZacAbc:=65; 
             ’a’..’z’:ZacAbc:=97; 
           end; 
           PozVAbc:=ord(TEXT[i])-ZacAbc; 
           RelPozKodu:=(PozVAbc+POSUN) mod 26; 
           TEXT[i]:=chr(ZacAbc+RelPozKodu) 
           end; 
  writeln(TEXT); 
  writeln(’KONIEC - stlac Esc’:45) 
until readkey=#27 
end. 
 

Ď alšou skupinou ú loh sú  „grafické“ ú lohy v textovom režime obrazovky. 

Príklad CFR-8: Program ROZSVECOVANIE_V_OKNACH ná hodný m generovaním polohy postupne 

vypĺňa plochu obrazovky „rozsvietený mi“ obdĺžnikmi (znak s poradový m číslo 219) . 

Analý za: Treba opakovane „rozsvecovať okná “, preto cyklus. My sme ho postavili na zná mom počte 

opakovaní a preto sa program opý ta, koľko okien má  rozsvietiť (to je počet opakovaní príkazov v cykle). No 

a potom už stačí len použiť funkciu random a nechať ná hodne vygenerovať číslo od 1 do 80 – stĺpec a číslo 

od 1 do 24 – riadok, v ktorom sa „rozsvieti“. Požadovaný  stĺpec a riadok sa nastaví príkazom gotoxy. 

 
program ROZSVECOVANIE_V_OKNACH; 
uses crt; 
var  I, POCET : word; 
begin 
randomize; 
clrscr; 
write(’Pocet rozsvietenych okien do 65 tis.: ’); 
readln(POCET); 
for I:=1 to POCET do 
  begin 

  gotoxy(random(80)+1,random(24)+1); 
  write(chr(219)); 
  end; 
readln 
end. 
 

Predchá dzajú ca ú loha je ešte zaujímavejšia, ak chceme, aby program generoval „rozsvecovanie“, až ký m 

nestlačíme ľubovoľný  klá ves. Funkcia keypressed unitu crt má  hodnotu true, ak bol stlačený  klá ves, inak má  

hodnotu false. Celý  problém je vo vytvorení tentoraz cyklu s nezná mym počtom opakovaní, ktorý  skončí na 

stlačenie klá vesu. Ak chceme „rozsvecovanie“ spomaliť, stačí vložiť do cyklu príkaz delay(milisekú nd). 

 
program ROZSVECOVANIE_V_OKNACH2; 
uses crt; 
begin 
randomize; 
clrscr; 
repeat 
  gotoxy(random(80)+1,random(24)+1); 
  write(chr(219)); 
until keypressed; 
readln 
end. 
 

V ďalšom príklade sa pokú sime usmerniť pohyb. 

Príklad CRP-5: Vytvorte program na pá d „vajíčka“ (písmena O) z hora nadol v strede riadkov. 

Analý za: Kto si poctivo ujasnil predchá dzajú ce dva programy, nemôže mať s nasledujú cimi dvoma  

problémy. Mali by sme použiť príkaz for RIA:=1 to 25 do ... ale chceme od vá s, aby ste použili príkaz 

repeat. Program sme doplnili aj o zvukový  efekt. Ď alšie zvukové efekty môžete doprogramovať pre let 

vajíčka. Ak sa chcete pohrať, môžete pridať aj „gravitá ciu“, aby vajíčko pri svojom pá de zrý chľovalo. 

 
program pad_vajca; 
uses crt; 
var  ria:byte; 
begin 
clrscr; 
ria:=1; 
repeat 
  gotoxy(40,ria); write(’.’); 
  gotoxy(40,ria+1); write(’O’); 
  delay(150); inc(ria) 
until ria=25; 
sound(60); 
gotoxy(36,12); write(’B U M !!!’); 
gotoxy(37,ria); write(’---^---’); 
delay(200); 
nosound; 
readln 
end. 
 
V ďalšej verzii programu PAD_VAJCA sa ho pokú sime zachrá niť pred rozbitím. Nech program vygeneruje 

ná hodne veľké písmeno (veľké písmená  začínajú  „na“ 65 a končia na 90) a zobrazí ho v ľavom hornom rohu 

obrazovky. Ak stihneme do dopadnutia vajíčka „do 25 riadku“ ná jsť na klá vesnici a stlačiť hľadaný  klá ves, 

pá d sa zastaví. 

program zachran_vajce; 
uses crt; 
var  RIA:byte; 
     PISMENO, STLACIL:char; 
begin 
randomize; 
clrscr; 
PISMENO:=chr(random(26)+65); 
gotoxy(1,5); write(’Najdi ’,PISMENO); 
RIA:=1; 
repeat 
  gotoxy(40,RIA); write(’.’); 
  gotoxy(40,RIA+1); write(’O’); 
  if keypressed then STLACIL:=upcase(readkey); 
  delay(100); inc(RIA) 
until (STLACIL=PISMENO) or (RIA=25); 
if RIA<25 
  then begin gotoxy(33,12); write(’V Y B O R N E !’) end 
  else begin 
       sound(60); 
       gotoxy(36,12); write(’B U M !!!’); 
       gotoxy(37,RIA); write(’---^---’); 
       delay(200); nosound 
       end; 
readln 
end. 
 
Ak vá s pohyb na obrazovke zaujal, môžeme skú šať ďalej. 

Príklad CRP-6: Vytvorte program na pohyb znaku po obrazovke. Na ovlá danie pohybu použite šípky na 

ovlá danie pohybu kurzora. Testujte aj „vybehnutie“ znaku z obrazovky. 

Analý za: Tento program uvedieme bez hlbšej analý zy. Nechá me ho na podrobné štú dium 

a experimentovanie. Ku kódom šípok sme sa dostali pomocou programu uvedeného nad príkladom CRP-2 (0 

si nevšímajte, sú  tam preto, že sa nejedná  o obyčajné klá vesy). Všimnite si cyklus repeat until keypressed, 

ktorý  zamedzuje neustá lemu prepisovaniu znaku na mieste a  jeho blikaniu (prakticky zacyklí program na 

danom mieste až do stlačenia ľubovoľného klá vesu). Program možno ná silne kedykoľvek ukončiť stlačením 

klá vesu Esc (príkaz halt – je to len uká žka, nenavykajte si naň!). 

Programy začínajú  byť zložitejšie a preto sme sa rozhodli začať nená padne používať podprogramy 

(procedú ry). Podprogram rieši nejaký  čiastkový  problém a je veľmi podobný  programu (namiesto slova 

program používa slovo procedure). V príkazovej časti hlavného programu stačí uviesť meno procedú ry 

a automaticky sa vykoná  všetko, čo v nej je naprogramované. 

 
program pohyb_znaku_pri_pridrzani_sipky_bez_blikania; 
uses  crt; 
 
const Esc=chr(27); 
      sipkaVPRAVO=chr(77);     {M} 
      sipkaVLAVO =chr(75);     {K} 
      sipkaHORE  =chr(72);     {H} 
      sipkaDOLE  =chr(80);     {P} 
 
var   stl,ria:byte; 
 
procedure inicializacia; 
  begin 
  clrscr; 
  stl:=40; ria:=12; gotoxy(stl,ria); write(’*’); 
  end; 

procedure pohyb; 
  var stlznak:char; 
  begin 
  repeat 
    repeat until keypressed; 
    stlznak:=upcase(readkey); 
    gotoxy(stl,ria); write(’ ’); 
    case stlznak of 
      sipkaVPRAVO : inc(stl); 
      sipkaVLAVO  : dec(stl); 
      sipkaHORE   : dec(ria); 
      sipkaDOLE   : inc(ria); 
      Esc         : halt; 
      end; 
    gotoxy(stl,ria); write(’*’); 
  until (stl=1)or(stl=80)or(ria=1)or(ria=25); 
  gotoxy(25,13); writeln(’B U M ! ! !  O H R A D A ! ! !’); 
  end; 
BEGIN 
inicializacia; 
pohyb; 
readln 
END. 
 

Pohyb je zaujímavejší, ak sa znak pohybuje určený m smerom, až ký m nezvolíme iný  smer. Čiže, ký m 

nezvolíme iný  smer (nestlačíme inú  šípku), znak sa má  uberať „starý m“ smerom. 

 
program pohyb_znaku_bez_podrzania_sipky; 
uses  crt; 
const Esc=chr(27); 
      sipkaVPRAVO=chr(77);     {M} 
      sipkaVLAVO =chr(75);     {K} 
      sipkaHORE  =chr(72);     {H} 
      sipkaDOLE  =chr(80);     {P} 
var   stl,ria:byte; 
 
procedure inicializacia; 
  begin 
  clrscr; 
  stl:=40; ria:=12; gotoxy(stl,ria); write(’*’); 
  end; 
 
procedure pohyb; 
  var stlznak:char; 
  begin 
  repeat 
    if keypressed then stlznak:=upcase(readkey); 
    gotoxy(stl,ria); write(’ ’); 
    case stlznak of 
      sipkaVPRAVO : inc(stl); 
      sipkaVLAVO  : dec(stl); 
      sipkaHORE   : dec(ria); 
      sipkaDOLE   : inc(ria); 
      Esc         : halt; 
      end; 
    gotoxy(stl,ria); write(’*’); delay(70); 
  until (stl=1)or(stl=80)or(ria=1)or(ria=25); 
  gotoxy(25,13); writeln(’B U M ! ! !  O H R A D A ! ! !’); 
  end; 
BEGIN 
inicializacia; 
pohyb; 
readln 
END. 

Spravme v programe ešte zopá r ú prav krá sy. Po prvé, vyrobme okolo obrazovky ohradu (v dolnej časti po 

riadok 24). Potrebné kódy znakov sme si našli po zobrazení ASCII tabuľky. Môžete postupovať aj tak, že 

v konštantá ch budete mať priamo zobrazené príslušné semigrafické znaky v apostrofoch. druhá  ú prava krá sy 

spočíva vo vypnutí zobrazovania kurzora, aby ná m „nekazil celkový  dojem“ (príkazy a potrebné premenné – 

registre pozná  unit dos). 

 
program pohyb_znaku_bez_podrzania_sipky2; 
uses  crt,dos; 
const ESC=chr(27); 
      sipkaVPRAVO=chr(77);     {M} 
      sipkaVLAVO =chr(75);     {K} 
      sipkaHORE  =chr(72);     {H} 
      sipkaDOLE  =chr(80);     {P} 
var   stl,ria:byte; 
 
procedure ohrada; 
  var i:integer; 
  begin 
  write(chr(201)); for i:=2 to 79 do write(chr(205)); write(chr(187)); 
  for i:=2 to 23 do begin write(chr(186)); write(chr(186):79) end; 
  write(chr(200)); for i:=2 to 79 do write(chr(205)); write(chr(188)); 
  end; 
 
procedure skry_kurzor; 
  var reg:registers; 
  begin                {ukaz} 
  reg.ah:=1;           {   1} 
  reg.ch:=32;          {  13} 
  reg.cl:=0;           {  40} 
  intr($10,reg) 
  end; 
 
procedure inicializacia; 
  begin 
  clrscr; 
  ohrada; 
  skry_kurzor; 
  stl:=40; ria:=12; gotoxy(stl,ria); write(’*’); 
  end; 
 
procedure pohyb; 
  var stlznak:char; 
  begin 
  repeat 
    if keypressed then stlznak:=readkey; 
    gotoxy(stl,ria); write(’ ’); 
    case stlznak of 
      sipkaVPRAVO : inc(stl); 
      sipkaVLAVO  : dec(stl); 
      sipkaHORE   : dec(ria); 
      sipkaDOLE   : inc(ria); 
      Esc         : halt; 
      end; 
    gotoxy(stl,ria); write(’*’); delay(70); 
  until (stl=1)or(stl=80)or(ria=1)or(ria=24); 
  gotoxy(25,13); write(’B U M ! ! !  O H R A D A ! ! !’); 
  end; 
BEGIN 
inicializacia; 
pohyb; 
readln 
END. 

Predchádzajúci pohyb doplníme do hry. Nech program vygeneruje na náhodnom, ale 
vhodnom(!) mieste napríklad O, ktoré treba pohybujúcim sa znakom trafiť. Princíp 
je jednoduchý, pribudne testovanie, či sa nenachádza pohybujúci sa znak na 
pozícii znaku O. 
 
program traf_znak; 
uses  crt,dos; 
 
const Esc=chr(27); 
      sipkaVPRAVO=chr(77);     {M} 
      sipkaVLAVO =chr(75);     {K} 
      sipkaHORE  =chr(72);     {H} 
      sipkaDOLE  =chr(80);     {P} 
 
var   stl,ria,x,y:byte; 
 
procedure skry_kurzor; 
  var reg:registers; 
  begin                
  reg.ah:=1;           
  reg.ch:=32;          
  reg.cl:=0;           
  intr($10,reg) 
  end; 
 
procedure ohrada; 
  var i:integer; 
  begin 
  textcolor(white); 
  write(chr(201)); for i:=2 to 79 do write(chr(205)); write(chr(187)); 
  for i:=2 to 23 do begin write(chr(186)); write(chr(186):79) end; 
  write(chr(200)); for i:=2 to 79 do write(chr(205)); write(chr(188)); 
  end; 
 
procedure inicializacia; 
  begin 
  randomize; 
  skry_kurzor; 
  clrscr; 
  ohrada; 
  x:=random(78)+2; y:=random(22)+2; gotoxy(x,y); write(’O’); 
  stl:=40; ria:=12; gotoxy(stl,ria); write(’*’); 
  end; 
 
procedure pohyb; 
  var stlznak:char; 
  begin 
  repeat 
    if keypressed then stlznak:=readkey; 
    gotoxy(stl,ria); write(’ ’); 
    case stlznak of 
      sipkaVPRAVO : inc(stl); 
      sipkaVLAVO  : dec(stl); 
      sipkaHORE   : dec(ria); 
      sipkaDOLE   : inc(ria); 
      Esc         : halt; 
      end; 
    gotoxy(stl,ria); write(’*’); delay(100); 
  until (stl=1)or(stl=80)or(ria=1)or(ria=24)or((stl=x)and(ria=y)); 
  if (stl=x)and(ria=y) 
    then begin gotoxy(32,12); write(’Z A S A H ! ! !’) end 
    else begin gotoxy(25,12); write(’B U M ! ! !  O H R A D A ! ! !’) end; 
  end; 

BEGIN 
inicializacia; 
pohyb; 
readln 
END. 
 
Program môžete ďalej vylepšovať, napríklad nech sa ná hodne pohybuje aj znak O. Ale to už nechá vame na 

vašu šikovnosť. 

Okrem dynamický ch (pohybujú cich sa) obrazcov je zaujímavé programovať aj statické obrazce. Väčšinou 

ide o požitie príkazov for, pričom 

• ak obrazec ne je plný , cykly bý vajú  susedné 

• ak je obrazec plný , cykly bý vajú  vnorené - vonkajší cyklus nastavuje riadok a vnú torný  cyklus 

zabezpečuje pohyb v danom riadku. 

Nasledujú ci program demonštruje uvedené tvrdenia. Susedné cykly sú  v procedú rach obdlznik1 a 2, stvorec 

a trojuholnik1 a 2; vnorené cykly sú  v procedú rach obdlznik3 a trojuholnik3. 
program OBRAZCE; 
uses crt; 
var n,i:integer; 
 
procedure obdlznik1; 
  begin 
  clrscr; 
  write(’Strana obdlznika: ’); readln(n); 
  for i:=1 to n do write(’*’); 
  writeln; 
  for i:=2 to n-1 do writeln(’*’,’*’:n-1); 
  for i:=1 to n do write(’*’); 
  writeln; 
  readkey 
  end; 
 
procedure obdlznik2; 
  begin 
  clrscr; 
  write(’Strana obdlznika: ’); readln(n); 
  for i:=1 to n do write(’*’); 
  writeln; 
  for i:=2 to n-1 do writeln(’*’,’*’:i-1,’*’:n-i); 
  for i:=1 to n do write(’*’); 
  writeln; 
  readkey 
  end; 
 
procedure stvorec; 
  begin 
  clrscr; 
  write(’Strana stvorca: ’); readln(n); 
  for i:=1 to n do write(’* ’); 
  writeln; 
  for i:=2 to n-1 do writeln(’*’,’*’:2*(n-1)); 
  for i:=1 to n do write(’* ’); 
  writeln; 
  readkey 
  end; 
 
procedure trojuholnik1; 
  begin 
  clrscr; 
  write(’Strana trojuholnika: ’); readln(n); 
  writeln(’*’); 

  for i:=2 to n-1 do writeln(’*’,’*’:i-1); 
  for i:=1 to n do write(’*’); 
  writeln; 
  readkey 
  end; 
 
procedure trojuholnik2; 
  begin 
  clrscr; 
  write(’Strana trojuholnika: ’); readln(n); 
  for i:=1 to n do write(’*’); 
  writeln; 
  for i:=2 to n-1 do writeln(’*’,’*’:n-i); 
  writeln(’*’); 
  readkey 
  end; 
 
procedure obdlznik3; 
  var j:integer; 
  begin 
  clrscr; 
  write(’Strana obdlznika: ’); readln(n); 
  for i:=1 to n do 
    begin 
    for j:=1 to n do  
      if i<>j then write(’*’) else write(’ ’); 
    writeln 
    end; 
  readkey 
  end; 
 
procedure trojuholnik3; 
  var j:integer; 
  begin 
  clrscr; 
  write(’Strana trojuholnika: ’); readln(n); 
  for i:=1 to n do 
    begin 
    for j:=1 to i do write(’*’); 
    writeln 
    end; 
  readkey 
  end; 
 
begin 
obdlznik1; 
obdlznik2; 
stvorec; 
trojuholnik1; 
trojuholnik2; 
obdlznik3; 
trojuholnik3; 
end. 
 

Pokú ste sa vytvoriť program ktorý  nakreslí rovnoramenný  trojuholník s podstavou šírky n znakov, obrá tenú  

pyramídu, kosoštvorec, štyri štvorce vedľa seba so stranou n znakov, štvorec s uhlopriečkami, plný  obdĺžnik 

bez znakov na opačnej uhlopriečke, ako to je v procedú re obdlznik3. 

Obrazce sú  nekonečnou témou. Môžete sa pokú siť naprogramovať aj pohyb napr. obdĺžnika (medzi 

jednotlivý mi polohami netreba mazať celú  obrazovku). 

Teraz trochu na oddý chnutie, aby sme však dlho neoddychovali, pobežia ná m hodiny. 

 



 Zá klady programovania  - Turbo Pascal   55 

Príklad CRP-7: Vytvorte program simulujú ci beh digitá lnych hodín v strede obrazovky v tvare HH : MM : 

SS. Využite príkaz GetTime z unitu Dos. 

Analý za: Program má  vlastne do nekonečna (ký m nestlačíme ľubovoľný  klá ves) čítať hodnoty systémového 

času pomocou príkazu gettime unitu dos a zobrazovať ich v strede obrazovky v predpísanom formá te. 

Problémom môže byť prechod z dvojciferného ú daja na jednociferný , keď cifra z desiatok nebude 

premazaná  – na to nesmieme zabudnú ť. 

 
program HODINY; 
uses crt,dos; 
var h,m,s,s100:word; 
           reg:registers; 
begin 
{skry kurzor}       
reg.ah:=1;          
reg.ch:=32;         
reg.cl:=0;          
intr($10,reg); 
clrscr; 
repeat 
  gettime(h,m,s,s100); 
  gotoxy(33,12); 
  if h<10 then write(’0’); 
  write(h,’ : ’); 
  if m<10 then write(’0’); 
  write(m,’ : ’); 
  if s<10 then write(’0’); 
  write(s); 
until keypressed; 
end. 
 

Doplňte program tak, aby hodiny pracovali ako budík („zazvonia“ v zadanom čase) prípadne časovač 

(„cinknú “ po uplynutí zadaného času). 

 

Na zá ver si ešte zopakujeme prá cu s textom. 

Príklad CWH-5: Vytvorte program, ktorý  zistí, či zadané slovo je symetrické, t.j. či sa rovnako číta zľava aj 

sprava. Symetrické sú  napríklad slová  RADAR a ABBA. 

Analý za: Zistiť, či zadané slovo je symetrické, znamená  porovnať jeho prvý  znak s posledný m, ak sa 

rovnajú , porovnať jeho druhý  znak s predposledný m, ak sa rovnajú , porovnať..., až ký m nenarazíme na dva 

rôzne znaky, alebo nie sme v strede slova (ľavá  polovica slova sa rovná  pravej polovici). Opakovane 

porovná vame vhodné znaky, preto treba použiť príkaz cyklu. Nevieme, kedy sa znaky už prestanú  rovnať, je 

to cyklus s nezná mym počtom opakovaní. Ú loha je riešiteľná  s príkazom while aj repeat, my sme sa rozhodli 

pre while. Keď sa po skončení cyklu s dvoma podmienkami ukončenia rozhodujete, pomocou ktorej 

testovať, prečo vlastne cyklus skončil, vyberte si vždy tú  dôležitejšiu. V našom prípade to znamená  testovať, 

či po skončení cyklu sa posledne testované písmená  rovnajú  alebo nie (if SLOVO[i]=SLOVO[DLZKA-

i+1]), či sme už v strede slova, nie je natoľko dôležité. 

 
program SYMETRICKE_SLOVO; 
uses crt; 
var SLOVO:string[80]; 
    i,DLZKA:byte; 

begin 
clrscr; 
repeat 
  write(’Vloz slovo: ’); readln(SLOVO); 
  DLZKA:=length(SLOVO); 
  i:=1; 
  while (SLOVO[i]=SLOVO[DLZKA-i+1]) and (i<DLZKA div 2) do 
    inc(i); 
  if SLOVO[i]=SLOVO[DLZKA-i+1] 
    then writeln(’Text je symetricky’) 
    else writeln(’Text nie je symetricky’); 
  writeln(’KONIEC - stlac Esc’) 
until readkey=#27; 
end. 
 

Uvedený  program možno zdokonaliť tý m, že nebude rozlišovať medzi veľký mi a malý mi písmenami 

a všetky ostatné znaky bude „ignorovať“. Potom aj text: „Jelenovi pivo nelej“ bude symetrický ! 

Analý za: Prvú  požiadavku, nerozlišovať medzi veľký mi a malý mi písmenami, možno vyriešiť pomocou 

funkcie upcase (zná ma vec). Druhú  požiadavku, aby všetky ostatné znaky boli ignorované, možno dosiahnuť 

vytvorením nového textu, ktorý  bude obsahovať len písmená . Využili sme množinový  operá tor in – je 

prvkom množiny (prvky množín sa uvá dzajú  v hranatý ch zá tvorká ch). 

 
program SYMETRICKY_TEXT_2; 
uses crt; 
var TEXT,novyTEXT:string[80]; 
    i,j,DLZKA:byte; 
begin 
clrscr; 
repeat 
  write(’Zadaj text: ’); readln(TEXT); 
  DLZKA:=length(TEXT); 
  j:=0; 
  for i:=1 to DLZKA do 
    if TEXT[i] in [’A’..’Z’,’a’..’z’] 
      then begin 
           inc(j); 
           novyTEXT[j]:=upcase(TEXT[i]) 
           end; 
  for i:=1 to j do 
    TEXT[i]:=novyTEXT[i]; 
  DLZKA:=j; 
  i:=1; 
  while (TEXT[i]=TEXT[DLZKA-i+1]) and (i<DLZKA div 2) do 
    inc(i); 
  if TEXT[i]=TEXT[DLZKA-i+1] 
    then writeln(’Text je symetricky’) 
    else writeln(’Text nie je symetricky’); 
  writeln(’KONIEC - stlac Esc’:45) 
until readkey=#27; 
end. 

Nerieš ené  úlohy na cykly: 

1. Vytvorte program na vý počet sú činu A.B dvoch prirodzený ch čísel pomocou sú čtu A + A + ...+ A 

(A ≥ B). 

2. Vytvorte program vý počet mocniny s krokom x2 (napr. x7 = x2. x2. x2.x). 

3. Vytvorte program na zistenie, či zadané číslo je prvočíslo. 

4. Vytvorte program na vypísanie všetký ch prvočísiel po zadané prirodzené číslo väčšie ako 1. 

5. Vytvorte program na zistenie, či prirodzené číslo N < 341 je  prvočíslo, ak viete, že platí veta: 

       „N je prvočíslom prá ve vtedy , ak je deliteľom čísla 2N-2 a N < 341“. 

6. Vytvorte program na ná jdenie všetký ch prirodzený ch čísel po  zadanú  hranicu, ktorý ch ciferný  sú čet sa 

rovná  zadanému číslu. 

7. Vytvorte program, ktorý  ako učebnú  pomôcku vygeneruje tabuľku ná sobilky zadaného ná sobiteľa. 

Obmedzte hodnotu ná sobiteľa do 100 a ná sobenec nech má  hodnoty z množiny {1,2,...,10 }. 

8. Vytvorte program na prevod čísla z desiatkovej do dvojkovej sú stavy. 

9. Vytvorte program na prevod čísla z  dvojkovej do desiatkovej sú stavy. 

10. Vytvorte program na prevod čísla z desiatkovej do šestná stkovej sú stavy. 

11. Vytvorte program na prevod čísla zo šestná stkovej do desiatkovej sú stavy. 

12. *Vytvorte program na zobrazenie podielu dvoch prirodzený ch čísel v tvare: 

{celá  časť},{predperióda}{(perióda)}. 

Napr.: 3 / 2 = 1,5(0);  7 / 11 = 0,(63);  11 / 7 = 1,(571428)  31 / 60 = 0,51(6) 

(Platí veta: Predperióda je tvorená  toľký mi ciframi, koľkokrá t je deliteľ deliteľný  číslami 10, 5 a 2; 

napríklad deliteľ 60 je deliteľný   10 a 2). 

13. Vytvorte program, ktorý  zobrazí prevodnú  tabuľku stupňov Celzia na Kelviny. Tabuľka začína 00C, 

končí hodnotou 100
0
C s krokom po 100C. Tabuľka má  mať hlavičku s nadpismi a pokú ste sa o rá mček 

zo semigrafický ch znakov ( write(chr(...)) ). 

14. Analogicky ako v predchá dzajú com príklade vytvorte prevodnú  tabuľku uhlové stupne - radiá ny. 

15. Vytvorte program , ktorý  vypočíta dĺžku lomenej čiary určenej bodmi so sú radnicami [ x,y ]. Na vý počet 

dĺžky ú sečky ( vzdialenosti dvoch bodov v rovine ) použite Pythagorovu vetu. 

16. Vytvorte program na ná jdenie najväčšieho spoločného deliteľa troch prirodzený ch čísel. 

17. Vytvorte program na vykreslenie grafu napr. funkcie sínus  s osou x zvisle. 

18. Vytvorte program na zistenie , či zadané prirodzené číslo je  dokonalé. Číslo N je dokonalé, ak sú čet 

všetký ch jeho deliteľov menších ako N sa rovná  N. Dokonalé je napríklad číslo 6. 

19. Vytvorte obmenu programu CRP-5, keď budú  postupne padať ná hodne generované veľké písmená  od 

prvého po 80. stĺpec. Na konci hry nech sa zobrazí ú spešnosť v percentá ch (počet „zachytený ch“ 

písmen). Ď alšie varianty hry: 

• padanie sa bude postupne zrý chľovať, napr. po „zachytení“ 5 písmen za sebou 

• pá d doplňte zvukový mi efektami 

• hru doplňte ponukou: začiatočník – pokročilý  – expert – koniec  

• hru nech možno kedykoľvek ukončiť stlačením klá vesu Esc 

• vytvorte „nekonečný “ cyklus, keď po 80. stlpci pokračuje hra automaticky v 1. stĺpci 

• písmená  nech padajú  v ná hodne generovaný ch stĺpcoch 

• padajú ce písmená  nech menia farbu 

• atď. 

Procedúry 
 Každý  problém (okrem najjednoduchších) možno rozložiť na čiastkové podproblémy. Ich 

vyriešením a vykonaním v sprá vnom poradí, dostá vame riešenie daného problému. Programovacie jazyky na 

riešenie „podú loh“ ponú kajú  podprogramy. Podprogram je relatívne samostatná  programová  jednotka 

(vý stavbou podobná  programu) riešiaca čiastkový  problém. Podprogramy používame najmä: 

• ak chceme sprehľadniť program - uvedením riešení jeho čiastkový ch problémov v podprogramoch 

(každý  poriadny program sa „rozpadá “ minimá lne na zadanie vstupný ch hodnôt, vý počet a vý stup 

získaný ch vý sledkov) alebo  

• ak potrebujeme vykonať rovnaký  „vý počet“1 viackrá t s rôznymi vstupný mi hodnotami (toto je praktický  

dôvod, aby sme neopisovali alebo nekopírovali tie isté príkazy viackrá t v programe všade tam, kde 

potrebujeme vykonať rovnakú  postupnosť príkazov). 

Programovací jazyk TP má  dva druhy podprogramov, procedú ry a funkcie. V tejto kapitole sa 

budeme venovať procedú ram. 

 Procedú ry môžu byť: 

• bez loká lnych objektov a bez parametrov, 

• s loká lnymi objektami a bez parametrov a 

• s parametrami. 

Procedú ry bez lokálnych objektov a bez parametrov 
Deklará cia procedúry bez loká lnych objektov a bez parametrov má  tvar: 

 procedure mp; { hlavička procedú ry } 

  begin 
 p1; 
 p2;  
 ...    
 pn 
 end; 

kde mp je meno procedú ry – identifiká tor a p1, p2 až pn sú  príkazy. 

Napríklad: procedure VSTUP; 

 begin 
 write(´Zadaj dve čísla: )́; 
 readln(A,B); 
 end; 

  procedure VYMENA; 

 begin 
 POM:=A; A:=B; B:=POM 
 end; 

  procedure VYSTUP; 

 begin 
 writeln(A,’ ’,B) 
 end; 

                                                 
1 V sú časnosti sa už počítače na klasické vý počty používajú  minimá lne, častejšie na vyhľadá vanie alebo triedenie dá t. 

príkazová  časť 

Procedú ra bez loká lnych objektov a bez parametrov používa len globá lne objekty (zavedené 

v nadradenej časti), hovoríme aj, že komunikuje s okolím len pomocou globá lnych premenný ch. Na mieste, 

kde chceme, aby došlo k vykonaniu príkazov uvedený ch v procedú re, stačí uviesť meno procedú ry - 

hovoríme o tzv. volaní procedú ry. Napríklad v príkazovej časti hlavného programu:  

BEGIN 
 clrscr; 
 VSTUP; 
 VYMENA; 
 VYSTUP; 
 readln 

END. 

Vidíme, že volanie procedú ry je na ú rovni príkazu (mená  procedú r sme použili ako nami definované 

príkazy). Všetky použité premenné musia byť deklarované v ú seku definícií a deklará cií hlavného programu. 

Ako v celom TP, aj pri podprogramoch platí, že každý  objekt musí byť najprv definovaný  alebo deklarovaný  

a až potom ho môžeme použiť. 

Procedú ry s lokálnymi objektami a bez parametrov 
 Keď sa pozrieme na procedú ru VYMENA, ľahko zistíme, že premennú  POM potrebujeme len počas 

vykoná vania tejto procedú ry. Taký chto objektov môže byť viacej a nie je dôvod zaťažiť pamäť počítača 

počas behu celého programu vyhradením pamäťový ch miest objektom, ktoré používame len loká lne. Preto 

takéto objekty stačí definovať a deklarovať len v danej procedú re, hovoríme, že sú  loká lne, čo znamená , že 

sú  použiteľné len v danej procedú re (alebo v podriadený ch procedú rach). 

Deklará cia procedúry s loká lnymi objektami má  tvar: 

 procedure mp; 

 ú sek definícií a deklará cií   

 príkazová  časť 

kde mp je meno procedú ry. 
 
Pamäťovo efektívnejší zá pis procedú ry VYMENA (s loká lnou premennou POM): 

  procedure VYMENA; 

 var POM : integer; 

 begin 
 POM:=A; A:=B; B:=POM 
 end; 

Volanie procedú ry s loká lnymi objektami je rovnaké ako procedú ry bez loká lnych objektov. Procedú ra 

naďalej komunikuje s okolím len cez globá lne premenné. Ak použijeme rovnaké pomenovanie pre loká lnu aj 

globá lnu premennú , dôjde k tzv. zatieneniu globá lnej premennej loká lnou premennou v danej procedú re, čo 

znamená , že daná  globá lna premenná  je nepoužiteľná  v danej procedú re.  

blok 

Procedú ry s parametrami 
 Ak by sme procedú ru POM chceli použiť viackrá t na vý menu hodnôt rôzne označený ch premenný ch 

(nie len A a B), najvý hodnejšie by bolo použiť parametre. Parametre umožňujú  efektívne komunikovať 

procedú re so svojim okolím, umožňujú  hodnoty do procedú ry dová žať prípadne aj vyvá žať. Pri písaní 

(deklarovaní) procedú ry nemusíme poznať hodnoty, ktoré budú  do procedú ry napr. dovezené, dokonca ani 

len označenie premenný ch, ktoré sa použije pri volaní procedú ry. Musíme však poznať počet parametrov 

a ich typ. Preto deklará ciu procedú ry píšeme s tzv. formá lnymi parametrami. 

Deklará cia procedúry s parametrami má  tvar: 

procedure mp (šfp1; šfp2; ... ; šfpn); 

 blok; 

kde mp je meno procedú ry a šfp1 až šfpn sú  špecifiká cie formá lnych parametrov. 

Napríklad:  procedure VYMENA ( var X , Y : integer ); 

  procedure NAJVACSIE ( A , B , C : integer; var MAX : integer ); 

  procedure ZASIFRUJ ( TEXT : string; POSUN : integer; var KOD : string ); 

Ak potrebujeme hodnoty do procedú ry len doviesť, špecifikujeme parametre nahradzované hodnotou. 

Š pecifiká cia parametrov nahradzovaných hodnotou má  tvar: 

  fp1, fp2, ... , fpn : tfp 

kde fp1 až fpn sú  identifiká tory formá lnych parametrov a tfp je identifiká tor ich typu. 

V príklade hlavičiek procedú r vyššie sú  parametre nahradzované hodnotou: A, B, C, TEXT a POSUN. 

Ak potrebujeme hodnoty z procedú ry aj vyviesť, špecifikujeme parametre nahradzované referenciou 

(odkazom).  

Š pecifiká cia parametrov nahradzovaných referenciou má  tvar: 

  var fp1, fp2, ... , fpn : tfp 

kde fp1 až fpn sú  identifiká tory formá lnych parametrov a tfp je identifiká tor ich typu. 

V príklade hlavičiek procedú r vyššie sú  parametre nahradzované odkazom: X, Y, MAX a KOD. 

 Pri volaní procedú ry s parametrami za formá lne parametre dosadzujeme skutočné parametre. Ich 

počet, poradie a typy musia sú hlasiť s formá lnymi parametrami. 

Napríklad:  VYMENA ( A , B );    VYMENA ( X , Y ); 

  NAJVACSIE ( 5 , -3 , 7 , MAXIMUM ); NAJVACSIE (  X , Y , Z , Q ); 

  ZASIFRUJ ( ´CEZAR ,́ 1 , POSLI );  ZASIFRUJ ( SPRAVA , N , KOD ); 

Pri volaní procedú ry môžu byť skutočné parametre nahradzované hodnotou konkrétne hodnoty (tieto 

hodnoty chceme do procedú ry len „doviesť“); pri ná hrade referenciou to musia byť premenné, lebo len tak sa 

môžu získané hodnoty „vyviesť“. Pri ná hrade hodnotou sa hodnoty skutočný ch parametrov pri volaní 

procedú ry len odovzdajú  formá lnym parametrom. Pri ná hrade referenciou sa skutočné parametre, počas 

vykoná vania príkazov v procedú re, stotožnia s formá lnymi parametrami – všetky zmeny vykonané 

s formá lnymi parametrami sa vykonajú  aj na skutočný ch parametroch – preto sa hodnoty z procedú ry aj 

„vyvezú “. Pre skutočné parametre môžeme, ale nemusíme, použiť rovnaké označenie ako pre formá lne 

parametre. 

 Praktické použitie procedú r si uká žeme po prebratí jednorozmerného poľa. 

Funkcie 
 Funkcia je špeciá lnym prípadom procedú ry. Procedú ru môžeme zapísať ako funkciu, keď jej 

vý sledkom je jedna jednoduchá  hodnota alebo typ string. 

Deklará cia funkcie má  tvar: 

  function mf ( šfp1; sfp2; ... ; šfpn ) : tvf; 

 blok; 

kde mf je meno funkcie – identifiká tor, šfp1 až šfpn sú  špecifiká cie formá lnych parametrov a tvf je typ 

vý sledku funkcie, ktorý  musí byť jednoduchý  typ alebo typ string. 

Napríklad: function MOCNINA ( X : real; N : integer ) : real; 

  function NACHADZA_SA : boolean; 

  function MAXIMUM ( A , B , C : integer ) : integer; 

Príklad: Vytvorte podprogram na vý počet mocniny xn, kde x je reá lne číslo a n prirodzené číslo. 

Analý za: Keďže vý sledkom vý počtu je jedna jednoduchá  hodnota, môžeme použiť aj funkciu. Na vý počet 

mocniny použijeme cyklus so zná mym počtom opakovaní: xn =  x.x.x. ... .x. 

  function MOCNINA (X : real; N : integer ) : real; 

 var MOC : real; 
   I : integer; 
 begin 
 MOC := 1; 
 for I:=1 to N do 
  MOC := MOC * I; 
 MOCNINA := MOC 
 end; 

Vý sledok sa z funkcie vyvá ža cez meno funkcie, preto identifiká tor mena funkcie sa musí aspoň raz 

vyskytnú ť na ľavej strane príkazu priradenia v tele danej funkcie. Aktivizá cia funkcie (vykonanie príkazov 

v tele funkcie) sa deje uvedením tzv. zá pisu funkcie (mena funkcie a skutočný ch parametrov) a nie je na 

ú rovni príkazu. Prakticky to znamená , že zá pis funkcie sa uvedie na mieste, kde chceme dosadiť vý sledok 

funkcie.  

Napríklad: writeln ( ´Vý sledok vý počtu mocniny: ,́ MOCNINA ( ZAKLAD , EXPONENT ) : 12 : 10 ); 

  VYSLEDOK := MOCNINA ( X , 3 ); 

 

Príklad FNK-1: Vytvorte program na vý počet počtu kombiná cií C(k,n) k-tej triedy z n prvkov 0≤k≤n. 

Analý za: Platí vzorec C(k,n) = 
n!

(n-k)!.k! . Vidíme, že vo vzorci sa trikrá t opakuje ten istý  vý počet – vý počet 

faktoriá lu (n! = 1.2.3. ... .n), len s rôznymi vstupný mi hodnotami (n, n-k a k). Preto je vý hodné použiť 

podprogram na vý počet faktoriá lu a keďže vý sledkom  je jedna jednoduchá  hodnota, konkrétne funkciu. 
program KOMBINACIE; 
uses  crt; 
const Esc=chr(27); 
var   N,K:integer; 
 
function KOMB(N,K:integer):integer; 
 
  function FAKT(N:integer):longint;  {lokalna funkcia} 
    var I:integer; 
        F:longint; 

    begin 
    F:=1; 
    for I:=1 to N do 
      F:=F*I;    
    FAKT:=F 
    end; 
 
  begin {príkazová časť funkcie KOMB } 
  KOMB:=FAKT(N) div FAKT(N-K) div FAKT(K) 
  end; 
 
BEGIN 
clrscr; 
writeln(’KOMBINACIE, KONIEC - Esc’:50); 
repeat 
  write(’Zadaj n a k ( n>=k>=0 ): ’);readln(N,K); 
  writeln(’C (’,N,’,’,K,’) = ’,KOMB(N,K)); 
until readkey=Esc; 
END. 
Program je skôr akademický , pretože hodnota n! už pre n = 13 prekročí maximá lnu hodnotu typu longint 

(13! = 6 227 020 800). 

 
Tak ako procedú ry, aj funkcie si precvičíme až v ďalších kapitolá ch. 

 
Rekurzia 
 Niektoré programovacie jazyky, vrá tane TP, umožňujú  používať veľmi elegantnú  programovaciu 

techniku, tzv. rekurziu. Ak procedú ra alebo funkcia vo svojom tele volá  samu seba, hovoríme o priamej 

rekurzii; ak napríklad procedú ra A volá  procedú ru B a naopak, hovoríme o nepriamej rekurzii. 

Príklad REK-1: Vytvorte rekurzívnu funkciu na vý počet n! 

Analý za: Vyjdeme z tzv. rekurentnej definície n-faktoriá lu: 

1. 0! = 1 
2. n! = n.(n-1)! pre n>0 

Ako vidieť z bodu 2, na vyčíslenie n! potrebujeme vykonať sú čin n krá t (n-1)!, pričom vý počet (n-1)! 

znamená  rovnaký  vý počet ako n!, len s hodnotou n zmenšenou o 1. 

 function FAKT ( N : integer) : longint; 
 begin 
 if N = 0 
 then FAKT := 1     { nerekurzívna vetva } 
 else FAKT := N * FAKT ( N-1 )  { rekurzívna vetva } 
 end; 

Keďže z príkladu FNC-1 vieme, že klasický  vý počet n! obsahuje cyklus, musí byť aj v príklade REK-1 

niekde „schovaný “. K cyklu ná s privedie poznanie, že počítač, ak nevie nejaký  vý raz, pre jeho zložitosť, 

vyčísliť, odkladá  požiadavky na operá cie do zá sobníka a vracia sa k nim, keď ich už vie finalizovať. Preto 

pri vý počte napríklad 5! počítač odloží najprv požiadavku 5 krá t „čosi“ (4! nepozná ), potom 4 krá t, 3 krá t, 2 

krá t, 1 krá t „čosi“, ale tu už „čosi“ je 0! a to je 1 (nerekurzívna vetva). Preto sa počítač vracia a vykoná va 

odložené operá cie, t.j. 1.1.2.3.4.5 čím získa vý sledok 120. Preto musíme vždy myslieť aj na ukončenie 

rekurzie, ukončenie volania danej procedú ry alebo funkcie, čo zabezpečí nerekurzívna vetva. 

Rekurzívny vý počet je časovo aj pamäťovo ná ročnejší ako iteračný  vý počet (pomocou cyklu), 

niektoré problémy však rieši veľmi elegantne a relatívne jednoducho. 

Príklad REK-2: Vytvorte rekurzívnu funkciu na vý počet mocniny xn, x reá lne číslo, n prirodzené číslo 

vrá tane nuly. 

Analý za: Potrebujeme znova poznať z matematiky rekurentnú  definíciu mocniny, ktorá  hovorí: 

Pre x ≠ 0: 
1. x0 = 1  

2. xn = x . xn-1  pre n>0 

Vidíme, že v bode 2 má me na ľavej aj pravej strane rovnaký  „problém“ – vý počet mocniny. 

  function MOCNINA ( X : real; N : word ): real; 
 begin 
 if N = 0 
 then MOCNINA := 1 
 else MOCNINA := X * MOCNINA ( X , N-1 ) 
 end; 

Parameter X nie je nevyhnutný , X môže byť aj globá lna premenná . Nezabudnite, že 00 nie je definované 

(ošetriť hneď na vstupe hodnôt X a N). 

Príklad REK-3*: Vytvorte program riešiaci problém Hanojské veže. 

Analý za: Ako uká žku sily a princípu prá ce rekurzie uvedieme zná my problém nazvaný  Hanojské veže. 

V starobylom klá štore ukrytom v ďalekom kú te Ázie vraj stoja tri zlaté kolíky. Na prvom z nich je 

nasunutý ch 64 diskov rôznych veľkostí, a to tak, že najväčší leží dole, nad ním leží o niečo menší, nad ním 

ešte menší atď. Ú lohou mníchov je premiestniť tú to vežu na druhý  kolík, pričom platí: 

• naraz sa môže premiestniť iba jeden disk 

• väčší disk sa nikdy nesmie položiť na menší 

• ktorý koľvek z troch kolíkov možno použiť ako „odkladací“ na dočasné umiestnenie disku 

Podľa legendy svet zanikne vo chvíli, keď mnísi svoju ú lohu splnia. 

Obrá zok sme nakreslili pre tri disky (A, B, C). Stojany (kolíky) sú  označené 1, 2 a 3 (skutočné parametre). 

Pre označenie kolíkov sme zvolili premenné ODKIAL, KAM a POMOCOU (formá lne parametre). 

 
Myšlienkový  postup pre N diskov je nasledovný :  

Aby sme najväčší disk (v obrá zku označený  A) mohli presunú ť z kolíka 1 na kolík 2, musíme: 

1. najprv presunú ť nad ním ležiace disky, t.j. vežu zloženú  z N-1 diskov, na kolík 3 pomocou kolíka 2, 

2. potom môžeme disk A premiestniť z kolíka 1 na kolík 2 a  

3. nakoniec presunú ť vežu zvyšný ch N-1 diskov z kolíka 3 na kolík 2 pomocou kolíka 1. 

Myšlienka použiť rekurziu by ná m mohla napadnú ť, ak si uvedomíme, že problém preniesť vežu s N diskami 

obsahuje v sebe problém preniesť vežu s N-1 diskami (ďalej už „len“ preložiť disk a naň vrá tiť vežu s N-1 

diskami).  

Tu je program: 
program HANOJ; 
uses crt; 
var  POCET:integer; 
 
procedure PRENES_VEZU(N,ODKIAL,KAM,POMOCOU:integer); 
 
  procedure PRENES_DISK(Z,NA:integer); 
    begin 
    writeln(Z,’ -> ’,NA) 
    end; 
 
begin 
if N>0 
  then begin 
       PRENES_VEZU(N-1,ODKIAL,POMOCOU,KAM); {1} 
       PRENES_DISK(ODKIAL,KAM); 
       PRENES_VEZU(N-1,POMOCOU,KAM,ODKIAL); {2} 
       end; 
end; 
 
BEGIN 
clrscr; 
write(’POCET DISKOV: ’); 
readln(POCET); 
PRENES_VEZU(POCET,1,2,3); 
readln; 
END. 
Jeho jednoduchosť asi prekvapí každého. Ú spešnosť rekurzie spočíva v tom, čo sme uviedli v príklade 

REK-1, totiž, že ak počítač „nevie niečo vypočítať“, „odloží to“ do zá sobníka. Tak počítač odkladá  

PRENES_VEZU pre čoraz menej diskov, aktualizuje ODKIAL, KAM, POMOCOU – uvedomte si, ktoré 

parametre sú  formá lne a ktoré skutočné(!), až niet čo preniesť a pokračuje ďalším príkazom PRENES_DISK 

atď. Zložitosť vý počtu by mala byť zrejmá  z vý pisu, ktorý  sme získali po doplnení vyššie uvedeného 

programu príkazmi na vý pis, pričom prvé volanie procedú ry PRENES_VEZU vo vetve then sme označili 

číslom 1 a druhé číslom 2. 
POCET DISKOV = 3 
Prenes vezu s 3 disk. pomocou volania 0 (prvé volanie z hlavného programu) 
Prenes vezu s 2 disk. pomocou volania 1 
Prenes vezu s 1 disk. pomocou volania 1 
Prenes vezu s 0 disk. pomocou volania 1 
1 -> 2 
Prenes vezu s 0 disk. pomocou volania 2 
1 -> 3 
Prenes vezu s 1 disk. pomocou volania 2 
Prenes vezu s 0 disk. pomocou volania 1 
2 -> 3 
Prenes vezu s 0 disk. pomocou volania 2 
1 -> 2 
Prenes vezu s 2 disk. pomocou volania 2 
Prenes vezu s 1 disk. pomocou volania 1 
Prenes vezu s 0 disk. pomocou volania 1 
3 -> 1 
Prenes vezu s 0 disk. pomocou volania 2 
3 -> 2 
Prenes vezu s 1 disk. pomocou volania 2 

Prenes vezu s 0 disk. pomocou volania 1 
1 -> 2 
Prenes vezu s 0 disk. pomocou volania 2 

Nerieš ené  úlohy na rekurziu: 

4. Vytvorte program na vý počet najväčšieho spoločného deliteľa (NSD) dvoch prirodzený ch čísel pomocou 

rekurzie, ak platí:   A   ak A=B 

NSD(A,B) = NSD(A-B,B) ak A>B 

    NSD(A,B-A) ak B>A 

5. Vytvorte program na vý počet najväčšieho spoločného deliteľa (NSD) dvoch prirodzený ch čísel pomocou 

rekurzie, ak platí:   A+B   ak A=0 alebo B=0 

NSD(A,B) =  NSD(A mod B,B) ak A>B 

   NSD(A,B mod A) ak B>A 

6. Objasnite prá cu rekurzívnej procedú ry OTOC: 
program ZRKADLO; 
uses crt; 
procedure OTOC; 
  var z:char; 
  begin 
  read(z); 
  if z<>’.’ 
    then OTOC 
    else writeln; 
  write(z) 
  end; 
BEGIN 
clrscr; 
writeln(’Veta konci bodkou !’:48);write(’ ’); 
OTOC; 
readkey 
END. 

7. Vytvorte program na vý počet ľubovoľného Fibonacciho čísla pomocou rekurzívnej a nerekurzívnej 

(iteračnej) funkcie a porovnajte časy vý počtu - urobte zá ver. 

 Fib. postupnosť: 0, 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144,... 

 vzorcami:  F0 = 0 , F1 = 1 

   Fi = Fi-1 + Fi-2 pre i ≥ 2 

8. Vytvorte podprogram na vý počet ľubovoľného člena Pascalovho trojuholníka, po zadaní nezá porný ch 

celý ch čísel n a k , ak platí: 

B(n, 0) = B(n, n) = 1 

B(n, k) = B(n-1, k-1) + B(n-1, k)  pre 1 ≤ k ≤ n 

9. Vytvorte funkciu CISLICA(n,k), ktorej vý sledkom je k-ta číslica sprava prirodzeného čísla n. 

10. *Vytvorte program, ktorý  zadané prirodzené číslo n rozloží na prvočinitele, t.j. na prvočísla c1, c2, ...,ck , 

pričom platí: n = c1 . c2 . ... . ck. 
11. *Vytvorte program, ktorý  zistí, na koľko častí sa rozpadne štvorčekový  papier po vystrihnutí niektorý ch 

štvorčekov. 

12. *Vytvorte program na zistenie všetký ch možný ch spôsobov vyplatenia zadanej sumy pri zná mych 

hodnotá ch použitý ch mincí (počet mincí rôznych hodnôt je neobmedzený ). 

Jednorozmerné pole 
 Predstavme si, že chceme vypočítať napríklad aritmetický  priemer z veľkého počtu čísel a vý stup má  

byť v tvare: Aritmetický  priemer z čísel: vypísané všetky čísla je .... Na uchovanie väčšieho počtu ú dajov 

počas vykoná vania programu s doteraz prebraný mi ú dajový mi typmi nevystačíme. Pomôže ná m nový  

ú dajový  typ pole. 

 Konečnú  skupinu rovnaký ch ú dajov (čísel, reťazcov a pod.) môžeme v programovaní vý hodne uložiť 

do jednej štruktú rovanej premennej, do poľa. Pole sa skladá  z pevného počtu zložiek rovnakého typu, pričom 

každú  zložku možno sprístupniť pomocou indexu. Rôzne polia môžu obsahovať rôzne počty prvkov a aj ich 

hodnoty môžu byť rôznych typov (raz sú  to čísla typu integer, druhý krá t reťazce typu string atď.). Preto 

rôzne polia predstavujú  rôzne typy poľa, ktoré sa líšia počtom prvkov, typom zložky alebo typom indexu. 

Aby počítač vedel, akú  veľkú  pamäť má  vyhradiť pre hodnoty premennej typu pole (skrá tene pre pole), 

musíme v programe najprv typ pole definovať.  

Definícia typu pole má  tvar: 

 type mt = array [ ti ] of tz 

kde mt je meno typu pole – identifiká tor, ti je typ indexu – musí byť ordiná lny typ a tz je typ zložky poľa. 

Ako typ indexu sa najčastejšie používa tzv. interval, definovaný : min .. max, kde min a max sú  hodnoty toho 

istého ordiná lneho typu (medzi min a max sú  prá ve dve bodky!). 

Napríklad:      Pole vľavo uvedeného typu bude obsahovať: 

type POLE1 = array [ 1..20 ] of integer;   20 čísel typu integer s indexami 1, 2, ..., 20 

type POLE2 = array [ -100..100 ] of  boolean;  201 hodnôt false alebo true s indexami -100, -99, ... 

type POCET = array [ ´A .́.´Z  ́] of byte;  26 celý ch čísel od 0 do 255 s indexmi ´A ,́ ´B ,́...,´Z  ́

type tVELA = array [ byte ] of string;   256 reťazcov s indexami 0, 1, 2 až 255 

type HLUPOST = array [ boolean ] of boolean; dva prvky s hodnotami false alebo true s indexami 

false a true 

Uvedomte si, že sú časťou definície typu pole je aj pomenovanie daného typu. Ide o ú dajový  typ s nami 

definovaný m rozsahom, obsahom, indexami aj menom. 

 V ú seku deklará cií premenný ch deklarujeme premenné typu pole rovnaký m spôsobom ako iné 

premenné. Napríklad zá pis: var A : POLE1 zavá dza pole s ná zvom A  typu POLE1. Sprístupnenie 

jednotlivý ch zložiek poľa A  sa realizuje zá pisom A [ index ], kde index je indexový  vý raz a musí byť 

rovnakého typu, ako je typ indexu z definície typu pole. Zá pisu A [index ] hovoríme indexovaná  premenná  

a jej typ je určený  typom zložky daného poľa. 

Napríklad:  

v poli A typu POLE1 znamená  zá pis:  A [1]   prvú  zložku poľa A s hodnotou napríklad -7 

     A [2]  druhú  zložku poľa A  

     A [I]  i-tu zložku poľa A 

     A [20] poslednú  zložku poľa A 

v poli B typu POLE2 znamená  zá pis: B [-100]  prvú  zložku poľa B s hodnotou napríklad true 

     B [0] 101. zložku poľa B  

v poli C typu POCET znamená  zá pis: C [´A ]́ prvú  zložku poľa C s hodnotou  napríklad 0 

     C [´Z ]́ poslednú  zložku poľa C 

 Ak sú  dve polia rovnakého typu, napríklad označené A a B, možno ich hodnoty odovzdá vať naraz, 

napríklad príkazom B := A. 

 Premenná  typu pole môže byť deklarovaná  aj bez definície príslušného typu. Pre pole s ná zvom A by 

deklará cia mala tvar: 

var A : array [ ti ] of tz 

kde ti a tz majú  rovnaký  vý znam ako v definícii typu pole. Najprv definovať typ poľa a až potom deklarovať 

premennú  daného typu má  niekoľko vý hod a preto odporú čame typ poľa najprv definovať. Napríklad 

v špecifiká cii formá lneho parametra v procedú re je dovolené na mieste typu parametra uviesť len meno poľa 

a nie zá pis array [ ti ] of tz. 

 Pre programy s typom pole odporú čame používať podobný  ú sek definícií a deklará cií ako uvá dzame 

nižšie: 

program ...; 

uses   crt; 

const MAXPP = 20;  { definujeme konštantu udá vajú cu maximá lny počet zložiek - prvkov poľa } 

type   POLE = array [ 1..MAXPP ] of integer; 

var     A : POLE; 
          N, I : integer;  { N bude aktuá lny počet prvkov poľa A } 

 

Načítanie prvkov poľa je čiastkový  problém, ktorý  je vhodné vyriešiť v procedú re. Ak v celom 

programe pracujeme len s jedný m poľom, môžeme použiť procedú ru bez parametrov, t.j. pracovať 

s globá lnymi premenný mi. Ak v programe pracujeme s viacerý mi poliami, treba použiť parametre (uká žeme 

si v príklade neskôr). Princípom vstupu prvkov poľa je zapísať prvú  zadanú  hodnotu ako 1. prvok poľa, 

druhú  ako 2. prvok poľa, atď. až po zá pis posledného N-tého prvku poľa. N-krá t sa opakuje rovnaká  činnosť 

– zá pis do poľa, preto použijeme cyklus so zná mym počtom opakovaní – príkaz for. 

Príklad procedú ry riešiacej načítanie prvkov globá lneho poľa A s počtom prvkov N: 

procedure VSTUP; { bez parametrov } 

begin 
write(´Počet prvkov poľa: )́; 
readln(N); 
for I := 1 to N do     { Premenná  I bude postupne nadobú dať hodnoty 1, 2, 3, ..., N } 

begin 
write(I:2, .́ prvok: )́;     { Zá pis I:2 spôsobí, že jednotky budú  pod jednotkami a desiatky pod desiatkami } 
readln(A[I])        
end 

end; 

 Aj zobrazenie prvkov poľa je čiastkový  problém riešiteľný  v procedú re. Jednotlivé hodnoty poľa 

môžu byť na samostatný ch riadkoch (nevý hodné), oddelené medzerami alebo čiarkami. 

Príklad procedú ry slú žiacej na zobrazenie prvkov globá lneho poľa A oddelený ch medzerami 

(predpokladá me, že vý stupné hodnoty majú  najviac 7 znakov): 

procedure VYSTUP; 

begin 
for I := 1 to N do 

write(A[I]:8); { keďže je v riadku 80 znakov, v každom bude zobrazený ch do desať hodnôt } 
writeln 
end; 

Príklad procedú ry slú žiacej na zobrazenie prvkov globá lneho poľa A oddelený ch čiarkami: 

procedure VYSTUP; 

begin 
for I := 1 to N-1 do  { čiarky majú  byť za prvou až predposlednou hodnotou } 

write(A[I], ,́ )́; 
writeln(A[N]) 
end; 

 

Príklad POLE-1: Vytvorte program, ktorý  načíta najviac 30 celý ch čísel z intervalu 0 až 999 a zobrazí ich. 

Analý za: Požadovaný  program stačí doslova poskladať z predchá dzajú cich procedú r, len s malý mi ú pravami: 

MAXPP treba zväčšiť na 30 a typ zložky poľa môže byť byte (potom v prvej procedú re VYSTUP vystačíme 

aj s formá tom A[I]:4). Nezná ma je už len príkazová  časť hlavného programu: 

BEGIN 
clrscr; 
VSTUP;   { podľa „chuti“ môžete doplniť príkaz clrscr; } 
writeln(´Prvky pola: )́; 
VYSTUP; 
readln 
END. 

Odporú čame uložiť si tento program do sú boru s ná zvom POLE a má me zá klad pre mnoho programov 

pracujú cich s poľom – každý  by predsa mal mať vstup a vý stup prvkov poľa. Po otvorení sú boru POLE.PAS 

ho treba cez položku File – Save as... premenovať na aktuá lny ná zov a doplniť o procedú ru riešiacu zadanú  

ú lohu. 

 Často prvky poľa môžu byť aj ná hodne počítačom vygenerované čísla, stačí, ak zadá me len ich 

počet. Procedú ra VSTUP_NAHODNE za ná s „vyberie“ prvky poľa: 

procedure VSTUP_NAHODNE; 

begin 
write(´Počet prvkov poľa: )́; 
readln(N); 
for I := 1 to N do 

A[I]:=random(10)  { pole A bude obsahovať celé čísla od 0 po 9 } 
end; 

V hlavnom programe za BEGIN treba vložiť príkaz randomize, aby sme zná hodnili vý ber prvkov poľa po 

každom spustení programu. 

 

 Prá cu s poľom si precvičíme najmä na činnosti nazvanej vyhľadá vanie. Vyhľadá vanie v poli je 

činnosť s cieľom zistiť, či zložka so zadaný mi vlastnosťami sa nachá dza v poli. Cieľom môže byť aj zistenie, 

koľkokrá t sa zložka so zadaný mi vlastnosťami nachá dza v poli, na ktorý ch miestach, prípadne požiadavka, 

ak sa zložka v poli nenachá dza, vložiť ju za posledný  prvok poľa a pod. 

Príklad POLE-2: Vytvorte podprogram na zistenie najväčšieho prvku daného poľa. 

Analý za: Maximum hľadajme v poli celý ch čísel. Na začiatku hľadania je najväčším prvkom prvý  prvok 

poľa (jeho hodnotu uložíme do premennej MAX). Potom musíme zistiť, či druhý  prvok poľa nie je väčší, ak 

á no, našli sme nové maximum (do premennej MAX musíme zapísať hodnotu druhého prvku poľa). Podobne 

porovná me tretí prvok poľa s MAX a ak treba aktualizujeme MAX (zmeníme hodnotu MAX na hodnotu 

tretieho prvku poľa), potom štvrtý  prvok, atď., až ký m neprejdeme celé pole. Prejsť celé pole od prvého resp. 

druhého po posledný  prvok znamená  použiť príkaz for. 
 
procedure MAXIMUM; 
  var MAX:integer;  { lokálna premenná MAX } 
  begin 
  MAX:=A[1];   { priradenie počiatočnej hodnoty premennej MAX } 
  for I:=2 to N do  { zabezpečí prejdenie celého poľa od druhého prvku} 
    if A[I]>MAX then MAX:=A[I]; 
  writeln(’Z cisel:’); 
  VYSTUP;     {zobrazí prvky poľa } 
  writeln(’je najvacsie ’,MAX); 
  end; 
BEGIN 
clrscr; 
VSTUP; 
MAXIMUM; 
readln 
END. 
Procedú ru MAXIMUM môžeme napísať aj ako funkciu, keďže vý sledkom danej procedú ry je jedna 

jednoduchá  hodnota. Funkciu je vý hodné použiť, ak chceme maximum vyviesť. 

Program od funkcie MAXIMUM: 
function MAXIMUM:integer; 
  var MAX:integer; 
  begin 
  MAX:=A[1]; 
  for I:=2 to N do 
    if A[I]>MAX then MAX:=A[I]; 
  MAXIMUM:=MAX;    {priradenie najväčšej hodnoty menu funkcie,} 
  end;     {cez ktoré sa vyvezie daná hodnota} 
BEGIN 
clrscr; 
VSTUP; 
VYSTUP; 
writeln(’je najvacsie ’,MAXIMUM); {zápis funkcie na mieste,kde chceme zobraziť} 
readln     {výsledok funkcie} 
END. 
 

Príklad POLE-3: Vytvorte podprogram na zistenie, koľkokrá t sa zadaný  prvok nachá dza v poli naviac 100 

celý ch čísel. 

Analý za: Zistiť, koľkokrá t sa zadaný  prvok nachá dza v poli znamená  prejsť celé pole a vždy keď sa 

prezeraný  prvok poľa rovná  hľadanému, zväčšiť počet vý skytov o jednotku. 
procedure ZISTI_POCET; 
  var HLPRVOK,POCET:integer; 
  begin 
  write(’Zadaj prvok, ktoreho pocet vyskytov mam zistit: ’); 
  readln(HLPRVOK); 
  POCET:=0; 

  for I:=1 to N do 
    if A[I]=HLPRVOK then POCET:=POCET+1;      { analogicky inc(POCET); } 
  if POCET=0 
    then writeln(’Hladany prvok sa v poli nevyskytuje!’) 
    else writeln(’Hladany prvok sa v poli vyskytuje ’, POCET,’-krat.’) 
  end; 
BEGIN 
clrscr; 
VSTUP; 
ZISTI_POCET; 
readln 
END. 
Procedú ra môže byť opäť napísaná  aj ako funkcia, jej použitie vzhľadom na požiadavku zadať hľadaný  

prvok je trochu komplikovanejšia: 

function ZISTI_POCET(HLPRVOK:integer):integer; {parameter nahradzovaný hodnotou} 
  var POCET:integer; 
  begin 
  POCET:=0; 
  for I:=1 to N do 
    if A[I]=HLPRVOK then inc(POCET); 
  ZISTI_POCET:=POCET 
  end; 
 
BEGIN 
clrscr; 
VSTUP; 
write(’Zadaj prvok, ktoreho pocet vyskytov mam zistit: ’); 
readln(HLADANY_PRVOK);   {HLADANY_PRVOK musí byť globálna premenná} 
writeln(’Hladany prvok sa v poli vyskytuje ’, ZISTI_POCET(HLADANY_PRVOK),’-
krat.’); 
readln 
END. 
 

Príklad POLE-4: Vytvorte podprogram na zistenie, či sa prvok s danou hodnotou vyskytuje v poli. 

Analý za: Ký m doteraz sme museli prejsť celé pole, v tejto ú lohe môžeme prehľadá vanie poľa ukončiť 

po ná jdení hľadanej hodnoty v poli. Z algoritmického hľadiska to znamená  použiť časovo efektívnejší cyklus 

s nezná mym počtom opakovaní. Vý sledkom prehľadá vania má  byť len zistenie, či sa prvok v poli nachá dza 

(vyvezie sa hodnota true) alebo nenachá dza (vyvezie sa hodnota false). 
function NACHADZA_SA:boolean; 
  begin 
  I:=0; 
  repeat 
    inc(I) 
  until (A[I]=HLPRVOK) or (I=N); 
  NACHADZA_SA:=A[I]=HLPRVOK   { Tento zápis nahrádza príkaz if } 
  end; 
 
BEGIN 
clrscr; 
VSTUP; 
write(’Zistit vyskyt prvku s hodnotou: ’); 
readln(HLPRVOK); 
writeln(’Prvok s hodnotou ’,HLPRVOK,’ sa v poli:’); 
VYSTUP; 
if NACHADZA_SA 
  then writeln(’nachadza.’) 
  else writeln(’nenachadza.’); 
readln 
END. 

Príklad POLE-4b: Ako zaujímavosť uvá dzame rovnaký  problém – zistiť, či sa prvok s danou vlastnosťou 

vyskytuje v poli, pričom však vieme, že pole je utriedené. Pod utriedený m poľom rozumieme vzostupné 

utriedenie, t.j. najmenší prvok je prvý  a najväčší prvok je posledný , preto platí: A[I]≤A[I+1] pre všetky 

prípustné hodnoty I. Použiť by sme mohli aj predchá dzajú cu funkciu NACHADZA_SA, ktorá  používa tzv. 

lineá rne vyhľadá vanie – pole prehľadá va od prvého prvku prvok za prvkom, my však chceme časovo 

efektívnejší algoritmus (ak sa v tisíc prvkovom poli nachá dza hľadané číslo napr. na 999 mieste, lineá rne 

vyhľadá vanie musí prezrieť 999 prvkov, nasledujú ca funkcia však najviac 10!). 

Analý za: Princípom biná rneho vyhľadá vanie, lebo tak je toto vyhľadá vanie pomenované, je určenie stredu 

prehľadá vanej oblasti a zistenie, či sme našli hľadaný  prvok a ak nie, či sa nachá dza v jeho dolnej (ľavej) 

časti alebo v hornej (pravej) časti. Tento postup sa opakuje dovtedy, ký m sa nená jde hľadaný  prvok, alebo 

keď už nie je čo prehľadá vať (dolná  hranica prehľadá vanej oblasti je nad jej hornou hranicou). 
 
function NACHADZA_SA:boolean; 
  var DH,HH,S:integer; 
  begin 
  DH:=1; HH:=N; 
  repeat 
    S:=(DH+HH) div 2; 
    if A[S]<>HLPRVOK 
      then if A[S]<HLPRVOK 
             then DH:=S+1 
             else HH:=S-1 
  until (A[S]=HLPRVOK) or (DH>HH); 
  NACHADZA_SA:=A[S]=HLPRVOK 
  end; 
 

Ď alším typom ú loh môžu byť ú lohy zistiť, na ktorý ch miestach v poli sa vyskytuje prvok so zadanou 

vlastnosťou. Ú loha môže byť formulovaná : zistiť prvý  vý skyt, všetky vý skyty, posledný  vý skyt a pod.  

 

Príklad POLE-5: Vytvorte podprogram na ná jdenie prvého vý skytu prvku so zadanou vlastnosťou.  

Analý za: Ná jsť miesto vý skytu znamená  vlastne určiť index prvku, ktorý  má  požadovanú  vlastnosť (ak sú  

indexy poľa 1, 2, ..., N). Ak sa hľadaná  hodnota v poli nevyskytuje, nech sa vyvezie 0. Keďže ide o jednu 

jednoduchú  hodnotu, môžeme použiť funkciu. My sme sa rozhodli pre procedú ru, aby sme uká zali použitie 

procedú ry namiesto funkcie a aj preto, že funkčný  zá pis by sme museli použiť dvakrá t (všade tam v hlavnom 

programe, kde sme použili premennú  MIESTO), čo nie je efektívne. Ak sa chceme vyhnú ť viacná sobnému 

použitiu funkčného zá pisu s nezmenený mi hodnotami, môžeme vý sledok funkcie uložiť v hlavnom 

programe do premennej a ďalej pracovať len s touto premennou. V nasledujú cej časti programu sú  premenné 

HLPRVOK a MIESTO globá lne, nezabudnite ich deklarovať. 
 
procedure PRVY_VYSKYT; 
  begin 
  I:=0; 
  repeat 
    inc(I) 
  until (A[I]=HLPRVOK) or (I=N); 
  if A[I]=HLPRVOK 
    then MIESTO:=I 
    else MIESTO:=0 
  end; 

BEGIN 
clrscr; 
VSTUP; 
write(’Zistit miesto prveho vyskytu prvku s hodnotou: ’); 
readln(HLPRVOK); 
clrscr; 
writeln(’Prvok s hodnotou ’,HLPRVOK,’ sa v poli:’); 
VYSTUP; 
PRVY_VYSKYT; 
if MIESTO>0 
  then writeln(’nachadza na ’,MIESTO,’. mieste.’) 
  else writeln(’nenachadza.’); 
readln 
END. 

Veľmi zaujímavé sú  aj kombiná cie predchá dzajú cich ú loh, ako napríklad zistiť, či sa prvok 

s požadovanou vlastnosťou nachá dza v poli a ak á no, koľkokrá t, prípadne na ktorý ch miestach, prípadne aj 

koľkokrá t a na ktorý ch miestach. Tieto kombinované ú lohy väčšinou neznamenajú  riešiť každú  ú lohu 

v samostatnom cykle (neefektívne), ale pri jednom prechode poľom možno „vyriešiť“ všetky ú lohy. Ú lohy 

podobného typu sme dali medzi neriešené ú lohy, teraz si uveďme aspoň jednu zložitejšiu. 

 

Príklad POLE-6: Vytvorte program na zistenie najmenšieho prvku v poli, koľkokrá t sa vyskytuje a na 

ktorý ch miestach v poli. 

Analý za: Program nechá vame na „samoštú dium“. Všimnite si priradenie počiatočný ch hodnôt v procedú re 

MINIMUM a ako sme sa „pohrali“ s vý stupom (VYSTUP2). 
program POLE_6; 
uses  crt; 
const MAXPP=20; 
type  POLE=array[1..MAXPP] of integer; 
var   A,VYSKYT:POLE; 
      N,I,MIN,POCET:integer; 
 
procedure VSTUP; 
  begin 
  write(’Pocet prvkov pola: ’); 
  readln(N); 
  for I:=1 to N do 
    begin 
    write(I:2,’. prvok: ’); 
    readln(A[I]); 
    end; 
  end; 
 
procedure VYSTUP1; 
  begin 
  for I:=1 to N-1 do 
    write(A[I],’, ’); 
  writeln(A[N]); 
  end; 
 
procedure MINIMUM; 
  begin 
  MIN:=A[1]; POCET:=1; VYSKYT[1]:=1; { priradenie počiatočných hodnôt } 
  for I:=2 to N do 
    if A[I]<=MIN 
      then if A[I]=MIN  { prvok s hodnotou rovnou momentálnemu minimu } 
             then begin 
                  inc(POCET); 
                  VYSKYT[POCET]:=I 
                  end 

             else begin   { nové minimum } 
                  MIN:=A[I]; 
                  POCET:=1; 
                  VYSKYT[1]:=I 
                  end 
  end; 
 
procedure VYSTUP2; 
  begin 
  writeln(’je najmensim prvok ’,MIN); 
  write(’a vyskytuje sa ’,POCET,’-krat ’); 
  if POCET=1 
    then write(’na mieste: ’,VYSKYT[1]) 
    else begin 
         write(’na miestach: ’); 
         for I:=1 to POCET do write(VYSKYT[I],’  ’) 
         end 
   end; 
 
BEGIN 
clrscr; 
VSTUP; 
clrscr; 
writeln(’V poli:’); 
VYSTUP1; 
MINIMUM; 
VYSTUP2; 
readln 
END. 
 

 Ď alšiu techniku, alebo ak chcete „fintu“, ktorú  by mal ovlá dať každý  programá tor, je použitie tzv. 

ná razníka. Jeho princíp a použitie si uká žeme na nasledujú com príklade: 

Príklad POLE-7: Vytvorte podprogram, ktorý  zistí, či sa hľadaný  prvok nachá dza v poli, a ak nie, vloží ho 

za posledný  prvok. 

Analý za: Princíp použitia ná razníka spočíva vo vložení hľadaného prvku do poľa, v tomto prípade za 

posledný  prvok. To zabezpečí zastavenie prehľadá vania (prvok sa určite v takto upravenom poli nachá dza) 

najneskôr na ná razníku. Ak sa prehľadá vanie poľa zastaví skôr, hľadaný  prvok sa nachá dza už 

v pôvodnom poli. Ak sa hľadaný  prvok v poli nenachá dza, už ho má me vložený  za posledný  prvok, len 

nesmieme zabudnú ť zväčšiť počet prvkov poľa (hodnotu N). 
procedure NARAZNIK; 
  begin 
  A[N+1]:=HLPRVOK;       { vlozili sme naraznik } 
  I:=0; 
  repeat 
    I:=I+1 
  until A[I]=HLPRVOK; 
  if I=N+1 then N:=N+1   { prvok sa v poli nenachadzal, pridany dalsi prvok }  
  end; 
 
BEGIN 
clrscr; 
VSTUP; 
write(’Zistit vyskyt prvku s hodnotou: ’); 
readln(HLPRVOK); 
writeln(’Povodne pole:’); VYSTUP; 
NARAZNIK; 
writeln(’Nove pole:’); VYSTUP; 
readln 
END. 

 Na nasledujú com príklade „z trochu iného sú dka“ si uká žeme, aké je dôležité dobre si zvoliť typ 

pole. 

Príklad POLE-8: Vytvorte program, ktorý  po zadaní textu prekonvertujte malé písmená  v texte na veľké a 

zobrazí tabuľku počtu vý skytov jednotlivý ch písmen abecedy v texte. 

Analý za: Pole zvolíme tak, aby indexami poľa boli postupne všetky veľké písmená  abecedy a hodnotami 

daného poľa budú  čísla udá vajú ce počet vý skytov jednotlivý ch písmen v texte. 
program POLE_8; 
uses  crt; 
type  POLE=array[’A’..’Z’] of byte; 
var   TEXT:string; 
      POCET:POLE; 
 
procedure VSTUP; 
  procedure KONVERZIA; 
    var I:byte; 
    begin 
    for I:=1 to length(TEXT) do   
      TEXT[I]:=upcase(TEXT[I])   { malé písmená v TEXTe zmení na veľké } 
    end;   
  begin 
  write(’Zadaj text: ’); readln(TEXT); 
  KONVERZIA 
end; 
 
procedure ZISTI_POCTY; 
  var I:byte; 
  procedure VYNULUJ; 
    var ZNAK:char; 
    begin 
    for ZNAK:=’A’ to ’Z’ do 
      POCET[ZNAK]:=0 
    end; 
  begin 
  VYNULUJ; 
  for I:=1 to length(TEXT) do 
    if TEXT[I] in [’A’..’Z’] 
      then inc(POCET[TEXT[I]]);   { TEXT[I] predstavuje konkrétne písmeno textu }  
  end; 
 
procedure VYSTUP; 
  var ZNAK:char; 
  begin 
  writeln(’Znak          Pocetnost’:45); 
  for ZNAK:=’A’ to ’Z’ do 
    begin 
    writeln(ZNAK:24,POCET[ZNAK]:17); 
    if ZNAK=’W’ then begin readkey; clrscr end 
    end 
  end; 
 
BEGIN 
clrscr; 
VSTUP; 
ZISTI_POCTY; 
VYSTUP; 
readln 
END. 

Typové konš tanty 
 Typové konštanty môžeme prirovnať k premenný m, ktorý ch počiatočné hodnoty sú  uvedené už 

v ú seku definícií a deklará cií. Môžeme ich aj používať rovnako ako premenné takého istého typu. 

 Deklará cia typovej konštanty má  tvar: 

const identifiká tor_premennej : identifiká tor_typu = typová _konštanta  

Napríklad: 

const  SYM : boolean = TRUE; 

 stringANO : string[3] = ´ANO  ́ ; 

A : array[1..10] of integer = (15,2,-4,6,9,12,0,-7,6,11); 

CISLICE : array[0..9] of char = ( 0́ ,́ 1́ ,́ 2́ ,́ 3́ ,́ 4́ ,́ 5́ ,́´6 ,́ 7́ ,́ 8́ ,́ 9́ )́; 

CISLICE : array[0..9] of char = 0́123456789 ;́         { tá to deklará cia je totožná  s predchá dzajú cou } 

 

Príklad POLE-9: V príklade VCS-1 sme v programe použili na vý počet počtu dní do konca roka (dkr) 

„otrasný “ zá pis v príkaze case. Pri použití typovej konštanty poľa sa zá pis podstatne zjednoduší. Do uká žky 

sme pridali aj vý počet počtu dní od začiatku roka (ozr). 
const dvm:array[1..12] of 1..31=(31,28,31,30,31,30,31,31,30,31,30,31); 
      { dvm – počet dní v mesiaci } 
...  
writeln;  
dkr:=dkm; 
for i:=12 downto m+1 do 
  dkr:=dkr+dvm[i]; 
if (m<2) and priestupny then inc(dkr); 
write(’Do konca roka ’); 
case dkr of 
      1: writeln(’zostáva 1 deň.’); 
  2,3,4: writeln(’zostávajú ’, dkr,’ dni.’); 
  else writeln(’zostáva ’, dkr,’ dní.’) 
end; 
{ výpočet počtu dní od začiatku roka } 
writeln; 
ozr:=d; 
for i:=1 to m-1 do 
  ozr:=ozr+dvm[i]; 
if (m>2) and priestupny then inc(ozr); 
write(’Od zaciatku roka ’); 
case ozr of 
      1: writeln(’ubehol 1 deň.’); 
  2,3,4: writeln(’ubehli ’, ozr,’ dni.’); 
  else writeln(’ubehlo ’, ozr,’ dní.’) 
end; 
readln 
end. 
Samozrejme, ak pozná me počet dní do konca roka, počet dní od začiatku roka sa dá  vypočítať aj 

jednoduchšie. 

 

Príklad POLE-10: Vytvorte program na prevod sprá vy – textu do morseovej abecedy. 

Analý za: Príklad je typický  pre použitie typovej konštanty pole. Doplnili sme ho aj o zvukový  vý stup. 

Program ako „nepovinný “ ponechá vame na samoštú dium. 

program POLE_10; 
uses  crt; 
const TAB:array[’A’..’Z’] of string[4] = (’.-’,’-...’,’-.-.’,’-..’,’.’,’..-.’, 
’--.’,’....’,’..’,’.---’,’-.-’,’.-..’,’--’,’-.’,’---’,’.--.’,’--.-’,’.-.’,’...’, 
’ ’,’..-’, ’...-’,’.--’,’-..-’,’-.--’,’--..’); 
      takt=100; 
      bodka=takt; 
      ciarka=3*takt; 
      medzi_znakmi=takt; 
      medzi_bc=takt; 
      Hz=1000; 
var   s:string; 
      i:byte; 
 
procedure zvuk(MSznak:string); 
  var i:byte; 
  begin 
  write(MSznak); 
  for i:=1 to length(MSznak) do 
    begin 
    if MSznak[i]=’|’ 
      then Delay(medzi_znakmi) 
      else begin 
           Sound(Hz); 
           if MSznak[i]=’.’ 
             then Delay(bodka) 
             else Delay(ciarka); 
           NoSound 
           end; 
    Delay(medzi_bc) 
    end 
  end; 
 
BEGIN 
clrscr; 
write(’Zadaj retazec: ’); readln(s); 
for i:=1 to length(s) do 
  case s[i] of 
    ’A’..’Z’,’a’..’z’: zvuk( TAB[ UpCase(s[i]) ] + ’|’ ); 
    ’ ’              : writeln 
  end; 
readln 
END. 
 

 Okrem vyhľadá vania sa často ú daje aj triedia. Utriediť prvky poľa znamená  vykonať činnosť, po 

končení ktorej pre prvky poľa napr. A platí: A[I] ≤ A[I+1] pre všetky dovolené hodnoty I. Triediacich 

algoritmov je veľa. Každý  by mal ovlá dať aspoň ten najjednoduchší, ktorý m môže byť napríklad bubblesort. 

Príklad POLE-11: Vytvorte podprogram na utriedenie číselného poľa bubblesortom. 

Analý za: Princípom prá ce bubblesortu je prechá dzať poľom a porovná vať dva vedľa seba stojace prvky; ak 

je ľavý  prvok väčší ako pravý , vymeniť ich. Ak si predstavíte tú to činnosť, malo by byť zrejmé, že po prvom 

prechode poľom je najväčší prvok na svojom miesta, t.j. na konci poľa. Po druhom prechode poľom je na 

svojom mieste druhý  najväčší prvok, atď. Pri n-prvkovom poli musí byť po n-1 prechodoch pole utriedené. 

Podstatou algoritmu sú  dva vnorené cykly: vonkajší cyklus zabezpečuje pri n-prvkovom poli n-1 prechodov 

daný m poľom (pp – počet prechodov) a vnú torný  cyklus zabezpečuje pohyb v poli od prvého po n-tý  prvok 

(premenná  i). Vnú torný  cyklus možno ešte zefektívniť, ak zohľadníme aj to, že po prvom prechode poľom je 

najväčší prvok na svojom mieste a preto pri ďalších prechodoch poľom už nemusíme ísť až na koniec poľa; 

analogicky po druhom prechode poľom je už aj druhý  najväčší prvok poľa na svojom mieste a preto 

v nasledujú cich prechodoch poľom nemusíme porovná vať už ani predposledný  prvok, atď. 

1. prechod treba porovnať n prvkov t.j. ísť po n-1. prvok 

2. prechod treba porovnať n-1 prvkov t.j. ísť po n-2. prvok 

3. prechod treba porovnať n-2 prvkov t.j. ísť po n-3. prvok 
... 
pp. prechod     treba ísť po n-pp. prvok 
procedure bubblesort; 
  var pp:integer; 
  procedure vymena; 
    var pom:real; 
    begin 
    pom:=a[i]; a[i]:=a[i+1]; a[i+1]:=pom; 
    end; 
  begin 
  for pp:=1 to n-1 do  { počet prechodov } 
    for i:=1 to n-pp do  {pohyb v poli } 
      if a[i]>a[i+1] then vymena; 
  end; 
 
BEGIN 
clrscr; 
vstup; 
writeln(’Povodne pole:’); vystup; 
bubblesort; 
writeln(’Utriedene pole:’); vystup; 
readln 
END. 
 
 Aspoň jedný m príkladom si pripomenieme, že aj ú dajový  typ string (reťazec) je špeciá lnym 

prípadom jednorozmerného poľa. Teoreticky ide o pole typu array [0..255] of char, ktorého prvky môžeme 

načítať „naraz“ príkazom read a zobraziť tiež „naraz“ bez cyklu for príkazom write (znak s indexom 0 má  

osobitnú  funkciu – je v ňom uložená  aktuá lna dĺžka reťazca). Môžeme však sprístupniť i-ty znak reťazca 

indexovanou premennou ako u poľa. 

 

Príklad POLE-12: Vytvorte program, ktorý  zistí, či zadané dva reťazce sa skladajú  z tý ch istý ch písmen 

a v rovnakom počte. Takéto reťazce nazý vame anagramy. 

Analý za: Najľahšie sa zistí, či majú  zadané reťazce rovnakú  dĺžku. Ak ich dĺžky sú  rôzne, nemôžu byť 

anagramy. Ak sa ich dĺžky rovnajú , musíme zistiť, či sa skladajú  z rovnaký ch písmen a v rovnakom počte. 

Znamená  to zobrať prvý  znak v prvom reťazci a hľadať jeho vý skyt v druhom reťazci, ak je všetko 

„v poriadku“, zobrať druhý  znak z prvého reťazca atď. Teda cyklus v cykle, kde vonkajší cyklus nastavuje 

prvý , druhý ,... znak prvého reťazca a vnorený  cyklus „prechá dza“ druhý m reťazcom a hľadá  v ňom 

zodpovedajú ci znak. Najefektívnejšie sú  cykly s nezná mym počtom opakovaní. Ď alším problémom je 

zabezpečiť, aby sme zistili, či sa písmená  zhodujú  aj v počtoch. Najjednoduchšie je asi „vyrá bať“ pri 

prechode druhý m reťazcom postupne reťazec totožný  s prvý m. Pre ozrejmenie programu si spravte pomocný  

vý pis upravovaného druhého reťazca príkazom write(y), ako to je v pozná mke programu. 

program POLE_12; 
uses crt; 
var  slovo1,slovo2:string[80]; 
 
function anagram(x,y:string):boolean; 
  var i,j,dlzkax,dlzkay:byte; 
      anag:boolean; 
      pom:char; 
  begin 
  dlzkax:=length(x); dlzkay:=length(y); 
  if dlzkax=dlzkay 
    then begin 
         anag:=true; 
         i:=1; 
         while anag and (i<=dlzkax) do   { vonkajší cyklus } 
           begin 
           j:=i; 
           while (x[i]<>y[j]) and (j<dlzkay) do  { vnútorný cyklus } 
             j:=j+1; 
           anag:=x[i]=y[j]; pom:=y[i]; y[i]:=y[j]; y[j]:=pom; 
           i:=i+1;      { doplňte príkaz writeln(y) } 
           end; 
         anagram:=anag 
         end 
    else anagram:=false 
  end; 
 
BEGIN 
clrscr; 
repeat 
  write(’Prve  slovo: ’); readln(slovo1); 
  write(’Druhe slovo: ’); readln(slovo2); 
  if anagram(slovo1,slovo2) 
    then writeln(’JE TO ANAGRAM’:44) 
    else writeln(’NIE JE TO ANAGRAM’:46); 
  writeln(’KONIEC - stlac Esc’); 
  writeln 
until readkey=chr(27); 
END. 
 

 Ak sa niekomu zapá čil pohyb znaku po obrazovke, môže pokračovať v príklade CRP-6, a pokú siť sa 

naprogramovať jednu z prvý ch hier na počítačoch – červíka Wurmi.  

 

Príklad POLE-13:  Vytvorte program – hru, v ktorej budú  na obrazovke ná hodne vygenerované krížiky (x) 

- jed a listy - potrava (♠ chr(6)). Šípkami na riadenie pohybu kurzora nech sa ovlá da pohyb červíka Wurmi 

(znak chr(1)), ktorého ú lohou je „požierať“ listy a vyhý bať sa jedu. Ak narazí na jed, hra končí. Hra končí aj 

„narazením do okrajov obrazovky“. 

Analý za: Pri programovaní tejto hry použite príklad CRP-6, v ktorom má me vyriešené ovlá danie pohybu 

znaku na obrazovke vrá tane ošetrenia „trafenia“ znaku na obrazovke a „narazenie do okraja obrazovky“. 

Ď alšie zlepšenie hry si vyžaduje použitie polí, aby bolo možné zapamätať si pozície (x-ové a y-nové 

sú radnice) jedu a potravy. Po každom premiestnení Wurmiho treba testovať, či „nestojí“ na jede alebo 

potrave. Ak stojí na potrave, treba započítať, že „zjedol“ ďalší list. Hra končí, keď: 

• „zožerie“ všetky vygenerované listy 

• „zožerie“ jed 

• „narazí“ do okrajov obrazovky. 

program POLE_13; 
uses  crt,dos; 
 
const pocet=5;   { počet vygenerovaných jedov a listov } 
      ESC=chr(27); 
      sipkaVPRAVO=chr(77); 
      sipkaVLAVO =chr(75); 
      sipkaHORE  =chr(72); 
      sipkaDOLE  =chr(80); 
      list       =chr(6); 
 
var   stl,ria,body:integer; 
      px,py,jx,jy:array[1..pocet] of integer; 
 
procedure skry_kurzor; 
  var reg:registers; 
  begin 
  reg.ah:=1; 
  reg.ch:=32; 
  reg.cl:=0; 
  intr($10,reg) 
  end; 
 
procedure ohrada; 
  var i:integer; 
  begin 
  textcolor(white); 
  write(chr(201)); for i:=2 to 79 do write(chr(205)); write(chr(187)); 
  for i:=2 to 23 do begin write(chr(186)); write(chr(186):79) end; 
  write(chr(200)); for i:=2 to 79 do write(chr(205)); write(chr(188)); 
  end; 
 
procedure generuj_potravu_jed; 
  var i:integer; 
  begin 
  for i:=1 to pocet do 
    begin 
    px[i]:=random(78)+2; py[i]:=random(22)+2; { generuje potravu } 
    jx[i]:=random(78)+2; jy[i]:=random(22)+2; { generuje jed } 
    end 
  end; 
 
procedure vykresli_potravu_jed; 
  var i:integer; 
  begin 
  for i:=1 to pocet do 
    begin 
    gotoxy(px[i],py[i]); write(list); 
    gotoxy(jx[i],jy[i]); write(’x’); 
    end; 
  end; 
 
procedure inicializacia; 
  var limit:integer; 
  begin 
  randomize; 
  generuj_potravu_jed; 
  skry_kurzor; 
  clrscr; 
  ohrada; 
  vykresli_potravu_jed; 
  stl:=40; ria:=12; gotoxy(stl,ria); write(chr(1)); 
  body:=0 
  end; 

function jed:boolean; { testuje, či Wurmi nie je na poličku jed } 
  var i:integer; 
  begin 
  i:=0; 
  repeat 
    inc(i) 
  until ((jx[i]=stl) and (jy[i]=ria)) or (i=pocet); 
  jed:=(jx[i]=stl) and (jy[i]=ria) 
  end; 
 
procedure potrava; { testuje, či Wurmi nie je na políčku list } 
  var i:integer; 
  begin 
  i:=0; 
  repeat 
    inc(i) 
  until ((px[i]=stl) and (py[i]=ria)) or (i=pocet); 
  if (px[i]=stl) and (py[i]=ria) 
    then begin inc(body); px[i]:=0; py[i]:=0 end; 
  gotoxy(3,2); write(body); 
  end; 
 
procedure pohyb; 
  var stlznak:char; 
  begin 
  repeat 
    if keypressed then stlznak:=readkey; 
    gotoxy(stl,ria); write(’ ’); 
    case stlznak of 
      ESC:halt; 
      sipkaVPRAVO:inc(stl); 
      sipkaVLAVO :dec(stl); 
      sipkaHORE  :dec(ria); 
      sipkaDOLE  :inc(ria); 
      end; 
    gotoxy(stl,ria); write(chr(1)); potrava; delay(150); 
  until (stl=1) or (stl=80) or (ria=1) or (ria=24) or jed or (body=pocet); 
  if jed 
    then begin gotoxy(33,12); write(’ J E D ! ! ! ’) end 
    else if body=pocet 
           then begin gotoxy(20,12); write(’V Y B O R N E !!! VYHRAL SI!!!’) end 
           else begin gotoxy(25,12); write(’B U M !!! OHRADA!!!’) end 
  end; 
 
BEGIN 
inicializacia; 
pohyb; 
readln 
END. 
 

 

Nerieš ené  úlohy na jednorozmerné  pole a typ string: 

14. Vytvorte program na zistenie najmenšieho a najväčšieho prvku v poli najviac 100 celý ch čísel a vý menu 

najmenšieho prvku s prvý m a najväčšieho prvku s posledný m prvkom v poli. 

15. Vytvorte program na ná jdenie prvý ch dvoch najväčších (najmenších) čísel v poli najviac 20 rôznych 

celý ch čísel. 

16. Vytvorte program na ná jdenie k (k≤n) maxím (1. môžu byť aj rovnaké, 2. k rôznych maxím) v poli s n 

prvkami. 

Ná vod: využite upravený  bubblesort. 

17. Vytvorte program na zistenie počtu nú l (kladný ch čísel, pá rnych čísel, čísel deliteľný ch zadaný m číslom, 

čísel zo zadaného intervalu a pod.) v zadanom číselnom poli. 

18. Vytvorte program na zistenie posledného vý skytu (všetký ch vý skytov) zadaného znaku v texte. 

19. Vytvorte program na zistenie polohy maxima a minima v danom poli (ošetrite aj viacná sobný  vý skyt). 

20. Vytvorte program, ktorý  prvky poľa napr. 2, 4, 5, 3, 8, 1, 3, 5, 9, 4, 7 zobrazí v tvare: 

2, 4, 5, 

3, 8,  (3 < 5) 

1, 3, 5, 9, (1 < 8) 

4, 7  (4 < 9) 

21. Vytvorte program, ktorý  z pôvodného poľa odstrá ni všetky prvky so zadanou vlastnosťou (nepá rne čísla, 

zá porné čísla, rovné zadanej hodnote, znaky zo zadaného intervalu a pod.). 

22. Vytvorte program, ktorý  zlú či dve polia s prvkami rovnakého typu (prvky druhého poľa dá  za prvky 

prvého poľa). 

23. Vytvorte program, ktorý  zlú či dve usporiadané polia s prvkami rovnakého typu tak, že vý sledné pole 

bude tiež usporiadané. 

24. Vytvorte program na vý počet a zobrazenie Pascalovho trojuholníka v tvare: 

 1 
 1        1 

 1        2        1 
 1        3        3        1 
 1        4        6        4        1 
 1    1+4    4+6    6+4    4+1    1 
 ... 

Ná vod: Využite vzorce v zadaní REK-8, nepoužite však rekurziu ale pole. 

25. Vytvorte program na uloženie prvkov pôvodného poľa opačne, vymeniť prvý  prvok s posledný m, druhý  

s predposledný m atď. 

26. Vytvorte program na posunutie prvkov poľa o k miest doprava (doľava). Posledný ch k prvkov poľa bude 

posunutý ch na začiatok poľa. 

27. Vytvorte program na usporiadanie prvkov (čísel, znakov, reťazcov) daného poľa zostupne (od 

najväčšieho po najmenší). 

28. Vytvorte program na vý počet ciferného sú čtu prirodzeného čísla využitím typu string. 

29. *Vytvorte program na zobrazenie všetký ch prvočísiel od 2 po 60 000 metódou tzv. Eratosténovho sita. 

Ná vod: Napíšeme si čísla od 2 do 60 000. Dá me vypísať 2 a škrtneme všetky jej ná sobky po 60 000. 

Zoberieme a vypíšeme ďalšie číslo: 3 (nie je škrtnuté, preto je prvočíslo) a škrtneme všetky jeho 

ná sobky. Zoberieme ďalšie číslo: 4, je škrtnuté a preto nie je prvočíslo. Zoberieme ďalšie číslo: 5, nie je 

škrtnuté, preto ho vypíšeme ako prvočíslo a škrtneme všetky jeho ná sobky. Číslo 6 je škrtnuté. Číslo 7 

vypíšeme (nie je škrtnuté) a škrtneme všetky jeho ná sobky (14, 21, 28,...). Takto pokračujeme až po 

60 000. Môžeme použiť napríklad typ pole array [2..60000] of boolean, kde true znamená , že dané číslo 

nie je škrtnuté (je prvočíslo) a false znamená , že je škrtnuté. 

30. Zefektívnite bubblesort tý m, že ukončíte triedenie (prechody poľom), ak pri poslednom prechode už 

nedošlo ani k jednej vý mene susedný ch prvkov (pole musí byť už utriedené). 

31. *Vytvorte program, ktorý  zistí, koľko anagramov sa nachá dza v zadanom zozname slov. 

Dvojrozmerné pole 
 V definícii typu pole: 

 type mt = array [ ti ] of tz 

sme doteraz dosadzovali za typ zložky len jednoduchý  typ alebo typ reťazec. V princípe typ zložky môže byť 

aký koľvek, okrem typu sú bor. V prípade, že typ zložky bude opäť typ pole, dostá vame tzv. typ dvojrozmerné 

pole.  

Definícia typu dvojrozmerné pole má  tvar: 

type mt = array [ ti1 ] of array [ ti2 ] of tz 

kde mt je meno typu – identifiká tor, ti1 a ti2 sú  ordiná lne typy indexov typu pole a tz je typ zložky. 

Dovolený  je aj skrá tený  zá pis definície typu pole: 

type mt = array [ ti1 , ti2 ] of tz 

Napríklad: 

type  MATICA1 = array [1..10 ] of array [1..10] of integer; 

 MATICA2 = array [1..10,1..10] of integer;  { definícia rovnocenná  s MATICA1 } 

 SACHOVNICA=array[ á .́. h́ ,́1..8] of boolean; 

 NEZMYSEL=array[boolean,boolean] of char; 

 

 Dvojrozmerné pole nazý vame aj pole polí. Ná zov vznikol zrejme z poznatku, že jednorozmerné 

pole, resp. každý  jeho prvok je znova jednorozmerný m poľom. Z hľadiska štruktú ry dvojrozmernému poľu 

zodpovedá  matematický  pojem matica. Pre ľahšie pochopenie príkladov budeme ďalej používať nasledujú cu 

symboliku (častejšie ako dvojrozmerné pole budeme používať pojem matica, pretože tá  predstavuje model na 

riešenie ná šho problému; až keď je zná my model na riešenie, môžeme ho realizovať v počítači 

dvojrozmerný m poľom): 

• matica typu N x M („en krá t em“) obsahuje N riadkov a M stĺpcov 

• pre zmenu riadku, t.j. pohyb v stĺpci budeme používať premennú  I 

• pre zmenu stĺpca, t.j. pohyb v riadku budeme používať premennú  J 

• schematicky (matica typu N x M): 

J sa mení od 1 po M (pohyb v riadku, zmena stĺpca) 
 

a 1,1 a 1,2  a 1,3 ... a 1,j ... a 1,m 

a 2,1 a 2,2 a 2,3 ... a 2,j ... a 2,m 

a 3,1 a 3,2 a 3,3 ... a 3,j ... a 3,m 

... ... ... ... ... ... ... 

  a i,1  a i,2 a i,3 ...  a i,j ... a i,m 

   ... ... ... ... ... ... ... 

  a n,1 a n,2 a n,3 ... a n,j ... a n,m 

 

I sa mení 

od 1 po N 

(pohyb v stĺpci, 

zmena riadku) 

 Z hľadiska algoritmický ch konštrukcií pre prá cu s maticou väčšinou potrebujeme cyklus v cykle. 

Vonkajší cyklus zabezpečuje najčastejšie nastavenie príslušného riadku (for I:=1 to N do) a vnú torný  cyklus 

pohyb v nastavenom riadku t.j. zmenu stĺpca od 1 po M (for J:=1 to M do). 

 Aj pri dvojrozmerný ch poliach pracujeme s indexovaný mi premenný mi a používame zá pis, 

napríklad pre pole A: A [I][J] resp. skrá tený  zá pis A[I,J] - prvok poľa A v I-tom riadku a J-tom stĺpci. 

 

Príklad MAT-1: Vytvorte program zabezpečujú ci načítanie typu (rozmerov) a prvkov matice a jej 

zobrazenie na obrazovke monitora. 

Analý za: Program obsahuje dva vstupy. V procedú re VSTUP treba zadať okrem rozmerov aj všetky prvky 

matice z klá vesnice, v procedú re VSTUP_NAHODNE po zadaní rozmerov matice program ná hodne 

vygeneruje prvky matice – celé čísla od 0 po 9. Keďže z procedú r vstupu sa majú  novozískané hodnoty 

vyviesť, parametre musia byť nahradzované referenciou. Pri vý stupe, hodnoty stačí len doviesť do 

procedú ry, ide o ná hradu formá lnych parametrov hodnotou. 
program MATICA_1; 
uses  crt; 
const MAXPP=10; 
type  MATICA=array[1..MAXPP,1..MAXPP] of integer; 
var   A:MATICA; 
      N,M,I,J:integer; 
 
procedure VSTUP(var N,M:integer;var A:MATICA); 
  begin 
  write(’Zadaj rozmery matice <= ’,MAXPP,’: ’); 
  readln(N,M); 
  for I:=1 to N do   { nastaví riadok } 
    for J:=1 to M do   { nastaví stĺpec } 
      begin 
      write(’Prvok ’,I,’,’,J,’: ’); 
      readln(A[I,J]); 
      end; 
  end; 
 
procedure VSTUP_NAHODNE(var N,M:integer;var A:MATICA); 
  begin 
  write(’Zadaj rozmery matice <= ’,MAXPP,’: ’); 
  readln(N,M); 
  randomize; 
  for I:=1 to N do 
    for J:=1 to M do 
      A[I,J]:=random(10); 
  end; 
 
procedure VYSTUP(N,M:integer;A:MATICA); 
  begin 
  for I:=1 to N do  { nastavuje riadok } 
    begin 
    for J:=1 to M do  { nastavuje stĺpec – pohyb v riadku } 
      write(A[I,J]:4);  { hodnoty zobrazené v riadku } 
    writeln;   { presunutie „kurzora“ na nový riadok } 
    end; 
  end; 
 
BEGIN 
clrscr; 
writeln(’Prvky matice nahodne (nie - N)?’); 
if upcase(readkey)=’N’ 
  then VSTUP(N,M,A) 

  else VSTUP_NAHODNE(N,M,A); 
VYSTUP(N,M,A); 
readln; 
END. 
 Najprv si uvedieme niekoľko matematický ch príkladov na prá cu s maticami.  

Príklad MAT-2: Vytvorte program na vyná sobenie matice celý ch čísel celý m číslom k. 

Analý za: Vyná sobiť číselnú  maticu číslom znamená  vyná sobiť každý  jej prvok daný m číslom, t.j. pre každé 

dovolené i, j: B[i,j] novej matice sa rovná  k∗A[i,j].  
procedure VYNASOB(N,M:integer;A:MATICA;var B:MATICA); 
  var K:integer; 
  begin 
  write(’Maticu vynasobit celym cislom: ’); 
  readln(K); 
  for I:=1 to N do 
    for J:=1 to M do 
      B[I,J]:=K*A[I,J] 
  end; 
 
BEGIN 
clrscr; 
writeln(’Prvky matice nahodne (nie - N)?’); 
if upcase(readkey)=’N’ 
  then VSTUP(N,M,A) 
  else VSTUP_NAHODNE(N,M,A); 
VYSTUP(N,M,A); 
VYNASOB(N,M,A,B); 
VYSTUP(N,M,B); 
readln; 
END. 
 
Príklad MAT-3: Vytvorte program na vý počet sú čtu dvoch matíc rovnaký ch typov. 

Analý za: Sú čtom dvoch matíc A a B typu NxM je matica C typu NxM, kde C[i,j] = A[i,j] + B[i,j]. 

V procedú rach vstupu a vý stupu je nevyhnutné použiť parametre, keďže procedú ry chceme použiť pre 

matice A, B aj C! 
program MATICA_3; 
uses  crt; 
const MAXPP=10; 
type  MATICA=array[1..MAXPP,1..MAXPP] of integer; 
var   A,B,C:MATICA; 
      N,M,I,J:integer; 
 
procedure ROZMERY(var N,M:integer); 
  begin 
  write(’Zadaj rozmery matice <= ’,MAXPP,’: ’); 
  readln(N,M); 
  end; 
 
procedure VSTUP(N,M:integer;var A:MATICA); 
... 
 
procedure VSTUP_NAHODNE(N,M:integer;var A:MATICA); 
... 
 
procedure VYSTUP(N,M:integer;A:MATICA); 
... 
 
procedure SCITAT(N,M:integer;A,B:MATICA;var C:MATICA); 
  begin 
  for I:=1 to N do 
    for J:=1 to M do 

      C[I,J]:=A[I,J]+B[I,J] 
  end; 
 
 
BEGIN 
randomize; 
clrscr; 
ROZMERY(N,M); 
VSTUP_NAHODNE(N,M,A); 
VSTUP_NAHODNE(N,M,B); 
writeln(’Prva matica: ’); 
VYSTUP(N,M,A); 
writeln(’Druha matica: ’); 
VYSTUP(N,M,B); 
SCITAT(N,M,A,B,C); 
writeln(’Sucet prvej a druhej: ’); 
VYSTUP(N,M,C); 
readln; 
END. 
 
Príklad MAT-4: Pre zasvätený ch, ktorí vedia, ako sa ná sobia matice, uvá dzame bez komentá ra procedú ru 

na sú čin dvoch matíc. Matica A musí byť typu NxL, matice B typu LxM a vý sledná  matica C bude typu 

NxM. 
procedure SUCIN(N,L,M:integer;A,B:MATICA;var C:MATICA); 
  begin 
  for I:=1 to N do 
    for J:=1 to M do 
  begin 
  C[I,J]:=0; 
  for K:=1 to L do 
       C[I,J]:=C[I,J] + A[I,K] * B[K,J] 
  end; 
  end; 
 

Príklad MAT-5: Vytvorte program, ktorý  zistí, či zadaná  štvorcová  matica je symetrická . 

Analý za: Štvorcová  matica (počet riadkov sa rovná  počtu stĺpcov) je symetrická , ak je symetrická  podľa 

hlavnej diagoná ly (hlavná  diagoná la je tvorená  prvkami A[1,1], A[2,2], A[3,3], ..., A[N,N]). Preto pre každý  

prvok matice musí platiť: A[I,J] = A[J,I] pre všetky dovolené hodnoty I a J.  

Nie najefektívnejší, ale jednoduchší program znamená  použitie cyklov for, ktoré “zistia“, či sa prvky pod 

hlavnou diagoná lou rovnajú  príslušný m prvkom nad hlavnou diagoná lou (ak sa „dolný  trojuholník rovná  

hornému, musí sa aj horný  rovnať dolnému“). Keďže vý sledkom podprogramu je hodnota true – ak je matica 

symetrická  alebo hodnota false – ak nie je symetrická , môžeme použiť aj funkciu. 
program MATICA_5; 
uses  crt; 
const MAXPP=10; 
type  MATICA=array[1..MAXPP,1..MAXPP] of integer; 
var   A:MATICA; 
      N,I,J:integer; 
procedure VSTUP(var N:integer;var A:MATICA); 
.... 
procedure VYSTUP(N:integer;A:MATICA); 
... 
function SYMETRICKA(N:integer;A:MATICA):boolean; 
  var SYM:boolean; 
  begin 
  SYM:=true; 
  for I:=2 to N do 

    for J:=1 to I-1 do 
      if A[I,J]<>A[J,I] then SYM:=false; { SYM:=SYM and (A[I,J]=A[J,I]); } 
  SYMETRICKA:=SYM 
  end; 
BEGIN 
clrscr; 
VSTUP(N,A); 
writeln(’Matica: ’); 
VYSTUP(N,A); 
if SYMETRICKA(N,A) 
  then writeln(’je symetricka.’) 
  else writeln(’nie je symetricka.’); 
readln; 
END. 
 

Príklad MAT-6: Vytvorte program na ná jdenie najmenšej a najväčšej hodnoty v každom riadku matice. 

Analý za: Každý  riadok matice je jednorozmerné pole „doplnené“ o index riadku, v ktorom sa nachá dza. Ak 

vieme hľadať minimum a maximum v jednorozmernom poli,  stačí toto hľadanie vložiť do cyklu, ktorý  bude 

meniť riadky od prvého po posledný . 
program MATICA_6; 
uses  crt; 
const MAXPP=10; 
type  MATICA=array[1..MAXPP,1..MAXPP] of integer; 
      POLE=array[1..MAXPP] of integer; 
var   A:MATICA; 
      MIN,MAX:POLE; 
      N,M,I,J:integer; 
 
procedure VSTUP(var N,M:integer;var A:MATICA); 
... 
procedure VSTUP_NAHODNE(var N,M:integer;var A:MATICA); 
... 
procedure VYSTUP(N,M:integer;A:MATICA); 
... 
 
procedure MIN_MAX(N,M:integer;A:MATICA;var MIN,MAX:POLE); 
  begin 
  for I:=1 to N do 
    begin 
    MIN[I]:=A[I,1]; 
    MAX[I]:=A[I,1]; 
    for J:=2 to M do 
      begin 
      if A[I,J]<MIN[I] then MIN[I]:=A[I,J]; 
      if A[I,J]>MAX[I] then MAX[I]:=A[I,J] 
      end; 
    end; 
  end; 
 
procedure VYSTUPPOLA(N:integer;A:POLE); 
  begin 
  for I:=1 to N-1 do 
    write(A[I],’, ’); 
  writeln(A[N]) 
  end; 
 
BEGIN 
clrscr; 
randomize; 
writeln(’Prvky matice nahodne (nie - N)?’); 
if upcase(readkey)=’N’ 
  then VSTUP(N,M,A) 

  else VSTUP_NAHODNE(N,M,A); 
MIN_MAX(N,M,A,MIN,MAX); 
writeln(’V matici: ’); 
VYSTUP(N,M,A); 
writeln(’minima v riadkoch: ’); 
VYSTUPPOLA(N,MIN); 
writeln(’maxima v riadkoch: ’); 
VYSTUPPOLA(N,MAX); 
readln; 
END. 
 

Príklad MAT-7: Vytvorte program na utriedenie prvkov v každom riadku matice. 

Analý za: Aby sme zvlá dli tento príklad, musíme vedieť aspoň jeden triediaci algoritmus. Pri 

jednorozmernom poli sme sa zaoberali bubblesortom, ktorý  použijeme aj teraz. Ako v predchá dzajú com 

príklade, aj tu treba len bubblesort „aplikovať“ na každý  riadok matice, čo znamená  vložiť bubblesort do 

cyklu na nastavenie riadku. 
procedure UTRIED_V_RIADKU(N,M:integer;var A:MATICA); 
  var PP:integer; 
  procedure VYMENA(var X,Y:integer); 
    var POM:integer; 
    begin 
    POM:=X; X:=Y; Y:=POM; 
    end; 
  begin 
  for I:=1 to N do   { nastaví riadok } 
    for PP:=1 to M-1 do   { počet prechodov riadkom } 
      for J:=1 to M-PP do  { pohyb v riadku } 
        if A[I,J]>a[I,J+1] then VYMENA(A[I,J],A[I,J+1]); 
  end; 
 
BEGIN 
clrscr; 
writeln(’Prvky matice nahodne (nie - N)?’); 
if upcase(readkey)=’N’ 
  then VSTUP(N,M,A) 
  else VSTUP_NAHODNE(N,M,A); 
VYSTUP(N,M,A); 
UTRIED_V_RIADKU(N,M,A); 
writeln(’Matica utriedenia v riadkoch: ’); 
VYSTUP(N,M,A); 
readln; 
END. 
Ako treba upraviť program, aby sa zachovala aj pôvodná  matica A? Podprogram, ktorý  utriedi prvky matice 

v stĺpcoch, sa len má lo líši od podprogramu UTRIED_V_RIADKU. Odlaďte ho. 

 

Príklad MAT-8: Vytvorte program, ktorý  utriedi riadky matice podľa hodnoty prvého prvku v riadku, t.j. 

podľa prvého stĺpca. 

Analý za:  

Napríklad v matici:   2 4 1 5 

    3 5 2 7 

    1 2 1 4 

bude posledný  riadok presunutý  do prvého, prvý  do stredného a stredný  riadok do tretieho. 

Utriedená  matica:  1 2 1 4 

    2 4 1 5 

    3 5 2 7 

Keďže je potrebné vymieňať celé riadky, už v ú seku definícií typ dvojrozmerné pole zadefinujeme ako pole 

riadkov. Tiež si treba uvedomiť, že všetky prvky v prvom stĺpci sú  prístupné cez zá pis A[I,1], kde I určuje 

riadok. No a dokonale treba ovlá dať nejaký  triediaci algoritmus, napríklad bubblesort. 
program MATICA_8; 
uses  crt; 
const MAXPP=10; 
type  RIADOK=array[1..MAXPP] of integer; 
      MATICA=array[1..MAXPP] of RIADOK; 
var   A:MATICA; 
      N,M,I,J:integer; 
 
procedure VSTUP_NAHODNE(var N,M:integer;var A:MATICA); 
... 
procedure VYSTUP(N,M:integer;A:MATICA); 
... 
procedure UTRIED_PODLA_PRVEHO(N,M:integer;var A:MATICA); 
  var R:RIADOK; 
      PP:integer; 
  procedure VYMENA(var X,Y:RIADOK);  { typ RIADOK! } 
    var POM:RIADOK; 
    begin 
    POM:=X; X:=Y; Y:=POM 
    end; 
  begin 
  for PP:=1 to N-1 do 
    for I:=1 to N-PP do 
      if A[I,1]>A[I+1,1] then VYMENA(A[I],A[I+1]) 
  end; 
 
BEGIN 
clrscr; 
VSTUP_NAHODNE(N,M,A); 
writeln(’Povodna matica: ’); 
VYSTUP(N,M,A); 
UTRIED_PODLA_PRVEHO(N,M,A); 
writeln(’Riadky utriedene podla prveho prvku v riadku: ’); 
VYSTUP(N,M,A); 
readln; 
END. 
 

Na zá ver „matematický ch“ príkladov s maticami uvedieme „hviezdičkový “ príklad. 

Príklad MAT-9*: Vytvorte program, ktorý  „rozvinie“ riadky a stĺpce matice do jednorozmerného poľa 

v smere chodu hodinový ch ručičiek. 

Analý za: 

Napríklad maticu:   2 4 1 5 

    3 5 2 7 

    1 2 1 4 

rozvinie program do poľa: 2    4    1    5    7    4    1    2    1    3    5    2 
program MAT-9; 
uses  crt; 
const MAXPP=10; 
type  MATICA=array[0..MAXPP,0..MAXPP] of integer; 
      POLE=array[0..MAXPP*MAXPP] of integer; 
var   A:MATICA; 
      B:POLE; 
      i,j,k,n,m,l,p,h,d:integer; 

procedure VSTUP; 
... 
procedure VYSTUP; 
... 
procedure VYTVOR_POLE; 
 procedure vpravo; 
   begin 
   for j:=l to p do begin k:=k+1; B[k]:=A[h,j]; end; 
   end; 
 procedure dole; 
   begin 
   for i:=h to d do begin k:=k+1; B[k]:=A[i,p]; end; 
   end; 
 procedure vlavo; 
   begin 
   for j:=p downto l do begin k:=k+1; B[k]:=A[d,j]; end; 
   end; 
 procedure hore; 
   begin 
   for i:=d downto h do begin k:=k+1; B[k]:=A[i,l]; end; 
   end; 
 begin 
 k:=0; l:=1; p:=m; h:=1; d:=n; 
 repeat 
   if k<n*m then vpravo; h:=h+1; 
   if k<n*m then dole;   p:=p-1; 
   if k<n*m then vlavo;  d:=d-1; 
   if k<n*m then hore;   l:=l+1; 
 until k=n*m; 
 end; 
procedure VYSTUP_POLA; 
 begin 
 writeln(’"Rozvinuta" matica:’); 
 for k:=1 to n*m do write(B[k]:4); 
 writeln; 
 end; 
BEGIN 
clrscr; 
VSTUP; 
VYSTUP; 
VYTVOR_POLE; 
VYSTUP_POLA; 
readln 
END. 
 

 Mnohí z ná s poznajú  hru ŽIVOT, v ktorej sa simuluje život jednobunkový ch organizmov v rovine. 

Po vložení počiatočnej generá cie program „počíta“ ďalšie generá cie. Organizmus prežije do ďalšej 

generá cie, ak ma 2 alebo 3 susedov, iná č zomrie. Živý  organizmus vznikne, ak má  prá ve 3 susedov. 

Príklad MAT-10: Vytvorte program simulujú ci život jednobunkový ch organizmov za vyššie uvedený ch 

podmienok. 

Analý za: Potrebné sú  dve matice. V jednej „prebieha život“, druhá  je pomocná  pri vý počte novej generá cie. 

Pre získanie nový ch poznatkov sme namiesto čísel 0 – prá zdna pozícia a 1 – živý  organizmus použili 

programá torom definovaný  typ s hodnotami nezivý  (ordiná lne číslo 0) a zivy (ordiná lne číslo 1). Program 

ponechá vame na samoštú dium.  
program MAT-10; 
uses  crt; 
const okraj=24; 
      plot =23; 

type  torganizmu=(nezivy,zivy); { programátorom vymenovaný typ } 
      tsuradnic=0..okraj; 
      tpola=array[tsuradnic,tsuradnic] of torganizmu; 
var   a,pom:tpola; 
      q:char; 
 
procedure vycisti_pole; 
  var i,j:tsuradnic; 
  begin 
  clrscr; 
  for i:=0 to okraj do 
    for j:=0 to okraj do 
      begin 
      a[i,j]:=nezivy; 
      pom[i,j]:=nezivy; 
      end; 
  end; 
procedure vystup; 
  var i,j:tsuradnic; 
  begin 
  clrscr; 
  for i:=1 to plot do 
    begin 
    for j:=1 to plot do 
      if a[i,j]=zivy 
        then write(’*’) 
        else write(’ ’); 
    writeln 
    end; 
  end; 
 
procedure vstup; 
  var i,j:integer; 
  begin 
  repeat 
    write(’Zadaj suradnice x,y ziveho organizmu ( KONIEC -> 0 ) ’); 
    read(i); 
    if i>0 
      then begin 
           readln(j); 
           if (i<okraj)and(j>0)and(j<okraj) 
             then a[i,j]:=zivy 
             else begin writeln(’Chybne suradnice’);delay(2000) end; 
           vystup 
           end 
  until i=0; 
  end; 
 
procedure nova_generacia; 
  var i,j:tsuradnic; pocet:0..8; 
  begin 
  for i:=1 to plot do 
    for j:=1 to plot do 
      begin 
      pocet:=ord(a[i-1,j-1])+  { počíta počet susedov okolo a[i,j] } 
             ord(a[i-1,j])+ 
             ord(a[i-1,j+1])+ 
             ord(a[i,j-1])+ 
             ord(a[i,j+1])+ 
             ord(a[i+1,j-1])+ 
             ord(a[i+1,j])+ 
             ord(a[i+1,j+1]); 
      if (pocet=3)or((pocet=2)and(a[i,j]=zivy)) 
        then pom[i,j]:=zivy 
        else pom[i,j]:=nezivy; 
      end; 

  a:=pom; 
  vystup; 
  end; 
BEGIN 
vycisti_pole; 
repeat 
  vstup; 
  repeat 
    nova_generacia; 
    write(’Chces dalsiu generaciu (N) ? ’);q:=upcase(readkey); 
  until q=’N’; 
  gotoxy(1,24); 
  write(’Chces doplnit zadanie (A) ? ’);q:=upcase(readkey); 
until q<>’A’; 
END. 
 

 

Nerieš ené  úlohy na dvojrozmerné  pole: 

11. Vytvorte program na vý počet sú čtu všetký ch prvkov číselnej matice. 

12. Vytvorte program, ktorý  sčíta všetky prvky na hlavnej diagoná le (ide z ľavého horného rohu matice do 

pravého dolného rohu) štvorcovej matice. 

13. Vytvorte program, ktorý  sčíta všetky prvky na vedľajšej diagoná le (ide z pravého horného rohu matice 

do ľavého dolného rohu) štvorcovej  matice. 

14. Vytvorte program na zistenie počtu nú l (zadaného čísla, kladný ch čísel, deliteľný ch zadaný m číslom, 

zo zadaného intervalu a pod.) v zadanej číselnej matici. 

15. Vytvorte program na ná jdenie najmenšieho (najväčšieho) prvku matice a jeho vý menu s prvý m 

(posledný m) prvkom matice. 

16. Vytvorte program na ná jdenie najmenšieho a najväčšieho prvku každého riadku matice a jeho vý menu 

s prvý m a posledný m prvkom v danom riadku matice. 

17. Vytvorte program na ná jdenie najmenšieho a najväčšieho prvku v každom stĺpci matice a jeho vý menu 

s prvý m a posledný m prvkom v danom stĺpci. 

18. Vytvorte program na ná jdenie zadanej hodnoty v matici a jej nahradenie novou hodnotou. 

19. Vytvorte program na zistenie, či sa zadaný  znak nachá dza v matici znakov a koľkokrá t sa nachá dza. 

20. Vytvorte program na zistenie miesta vý skytu (všetký ch, prvého, posledného) prvku so zadanou 

vlastnosťou v matici. 

21. Vytvorte program, ktorý  vo štvorcovej matici vymení prvky podľa hlavnej diagoná ly. 

22. Vytvorte program, ktorý  v matici vymení prvý  stĺpec (riadok) s posledný m, druhý  s predposledný m, atď. 

23. Vytvorte program, ktorý  zistí, či zadaná  štvorcová  matica je diagoná lna (okrem hlavnej diagoná ly 

a[i,j] = 0). 

24. Vytvorte program na otočenie štvorcovej matice o 900. 

25. Vytvorte program, ktorý  od miesta „čľupnutia kameňa do matice“ zvý ši pôvodne nulové hodnoty matice 

s narastajú cou vzdialenosťou od miesta „čľupnutia“ o 1. 

26. Vytvorte program, ktorý  pri pohybe v matici bude meniť hodnoty prechá dzaný ch prvkov. 

27. Vytvorte program, ktorý  pri pohybe v matici bude meniť hodnoty v okolí prechá dzaný ch prvkov. 

Ú dajový  typ záznam 
 Najmä pri hromadnom spracovaní ú dajov, pri informačný ch systémoch, často treba spojiť do 

jedného celku rôzne ú dajové typy (integer, string, boolean a pod.). Umožňuje to ú dajový  typ zá znam. 

Zá znam je nehomogénny štruktú rovaný  ú dajový  typ, ktorý  sa skladá  z pevného počtu položiek, 

vo všeobecnosti rôznych typov. 

Definícia typu zá znam má  tvar: 

type mt = record 

 p1 : tp1;  kde mt je meno typu 
 p2 : tp2;  p1 až pn sú  identifiká tory položiek zá znamu 
 ...   a tp1 až tpn sú  typy jednotlivý ch položiek 
 pn : tpn 

end 

Napríklad: type KARTA = record 
 OsCislo : integer; 
 Meno, 
 Priezv : string[25]; 
 DatNar : record 
 Den : 1..31; 
 Mes : 1..12; 
 Rok : 1900..2100 
 end; 
 Adresa : record 
 Obec, 
 Ulica : string[25]; 
 PSC : string[5] 
 end; 
 Priemer : array [1..5] of real; 
 Moze : boolean 
 end; 

  type BOD = record 
 x, 
 y : real; 
 farba : 0..14; 
 blik : boolean 
 end; 

  type ZLOZKA = record 
 HODNOTA : string; 
 POCET        : byte 
 end; 

 var KC : record Re, Im: real end; { deklarovaná  premenná  typu record } 

 
 Ak p je premenná  typu zá znam, k jej položke s ná zvom pi sa dostaneme zá pisom: p.pi. Napríklad, ak 

K je premenná  typu KARTA, možno použiť zá pisy: K.OsCislo, K.Meno, K.DatNar.Den, K.DatNar.Rok, 

K.Adresa.Obec, K.Priemer[1] a pod.  

 
Príklad ZAZ-1: Vytvorte program na určenie vzá jomnej polohy bodu a kružnice v rovine s využitím 

ú dajového typu zá znam. 

Analý za: Bod napr. B je určený  v rovine dvoma sú radnicami Bx a By. Kružnica je určená  stredom S 

so sú radnicami Sx a Sy a polomerom r. Vzdialenosť medzi bodom B a stredom kružnice k, bodom S, možno 

vypočítať pomocou Pythagorovej vety. Bod B leží vo vnú tornej oblasti kružnice k (S;r), ak jeho vzdialenosť 

od stredu kružnice k je menšia ako polomer r. Ak vzdialenosť /BS/ = r, bod B leží na kružnici a ak je tá to 

vzdialenosť väčšia ako polomer r, bod leží mimo oblasť kružnice k. 
program ZAZ_1; 
uses crt; 
type BOD=record 
            x,y:real; 
         end; 
     KRUZNICA=record 
                 s:BOD; 
                 r:real; 
              end; 
var A:BOD; 
    k:KRUZNICA; 
    vzd :real; 
 
procedure VSTUP; 
  begin 
  writeln(’Zadaj suradnice bodu ’); 
  readln(A.x,A.y); 
  writeln(’Zadaj stred kruznice a polomer’); 
  readln(k.s.x,k.s.y,k.r); 
  end; 
 
function VZDIALENOST:real; 
  begin 
  vzdialenost:=sqrt(sqr(A.x-k.s.x)+sqr(A.y-k.s.y)); 
  end; 
 
BEGIN 
clrscr; 
VSTUP; 
vzd:=VZDIALENOST; 
if vzd>k.r  
  then writeln(’Bod lezi vo vonkajsej oblasti kruznice.’) 
  else if vzd=k.r  
         then writeln(’Bod lezi na kruznici.’) 
         else writeln(’Bod lezi vo vnutornej oblasti kruznice.’); 
readln; 
END. 
 

Príkaz with 
 Ak v časti programu používame opakovane tú  istú  položku alebo používame viacej položiek tej istej 

premennej typu zá znam, príkaz with umožňuje zjednodušiť resp. skrá tiť zá pis k prístupu k tý mto položká m. 

Príkaz with má  tvar:  with z do p  kde z je premenná  typu zá znam a p je príkaz 

V príkaze p je dovolené označovať položky premennej z len identifiká tormi, t.j. zá pis z.položka skrá tiť na 

zá pis položka. 

Napríklad:  with k do begin writeln(’Zadaj stred kruznice a polomer’); readln(s.x , s.y , r) end; 

 with Z , DatNar do begin OsCislo:=99; Den:=1; Mes:=1; Rok:=2000 end; 

Pri zá pise with z1,z2 do p možno v príkaze p skrá tene označovať jednak položky zá znamu z2, ale aj tie 

položky zá znamu z1, ktoré sa nezhodujú  v označení so žiadnou položkou zá znamu z2. 

Príklad ZAZ-2: Vytvorte jednoduchú  kartotéku na evidenciu osôb (evidenčné číslo, meno a adresa: miesto, 

ulica). 

Analý za: Na precvičenie ú dajového typu record vytvoríme jednoduchú  kartotéku. Jednotlivé karty budú  

uložené v poli, čo v skutočnosti nikdy nebý va, lebo ukončením programu by sme prišli o všetky ú daje 

v kartá ch. Na zapamätanie ú dajov aj po vypnutí počítača slú žia vonkajšie pamäte (najmä pevný  disk). 

Ukladať ú daje na vonkajšie pamäte umožňuje ú dajový  typ sú bor, o ktorom si povieme už v nasledujú cej 

kapitole. 
program ZAZ_2; 
uses  crt; 
const pk=10;        { pocet kariet } 
      dmena=15; 
      dadr=25; 
type  karta=record 
           ec:byte; 
           meno:string[dmena]; 
           adr:record 
               miesto, 
               ulica:string[dadr]; 
               end; 
            end; 
      kartoteka=array[1..pk] of karta; 
var   k:kartoteka; 
      apk,i:byte; 
      q:char; 
 
procedure vstup_karta; 
  begin 
  clrscr; 
  apk:=apk+1; 
  with k[apk],adr do 
    begin 
    ec:=apk; 
    gotoxy(1,8); 
    writeln(’Evidencne cislo : ’:40,ec); 
    writeln; 
    write(’Meno a priezvisko : ’:40);readln(meno); 
    writeln; 
    write(’  Adresa - miesto : ’:40);readln(miesto); 
    writeln; 
    write(’         -  ulica : ’:40);readln(ulica); 
    end; 
  end; 
 
procedure vystup_kartoteka; 
  procedure vystup_karta(i:byte); 
    begin 
    clrscr; 
    writeln(’Vystup’); 
    gotoxy(1,8); 
    with k[i],adr do 
      begin 
      writeln(’  Evidencne cislo : ’:40,ec); 
      writeln; 
      writeln(’Meno a priezvisko : ’:40,meno); 
      writeln; 
      writeln(’  Adresa - miesto : ’:40,miesto); 
      writeln; 
      writeln(’          - ulica : ’:40,ulica); 
      end; 
    readln; 
    end; 

  begin   {procedura vystup_kartoteka } 
  for i:=1 to apk do vystup_karta(i); 
  end; 
 
procedure menu; 
  begin 
  repeat 
    clrscr; 
    gotoxy(1,10); 
    writeln(’VLOZIT KARTU........................ +’:55); 
    writeln; 
    writeln(’VYPISAT KARTOTEKU............... Enter’:55); 
    writeln; 
    writeln(’KONIEC.............................. K’:55); 
    q:=upcase(readkey); 
    case q of 
      ’+’     : vstup_karta; 
      chr(13) : vystup_kartoteka; 
    end; 
  until q=’K’; 
  end; 
 
BEGIN 
apk:=0; 
vstup_karta; 
menu; 
END. 

Variantný  záznam 
 Pri používaní ú dajového typu zá znam môže vyvstať požiadavka na ná vrh takej štruktú ry zá znamu, 

v ktorej sa na daný  prvok nemusí vzťahovať celý  zoznam položiek uvedený ch v zá zname, ale iba jeho určitá  

časť. Takto môže vzniknú ť niekoľko alternatívnych typov – variant - jedného typu zá znam. Takú to situá ciu 

ná m umožňuje riešiť tzv. variantný  zá znam, ktorý  sa skladá  z pevnej časti a z variantnej časti. 

Variantný  zá znam má  tvar: record 

 pevná  časť; 

 case rozlišovacia_položka : rozlišovací_typ_variantnej_časti of 

 rozlišovacia_konštanta1 : ( položka1 : typ1;  

 položka2 : typ2; 

 ...      ); 

 rozlišovacia_konštanta2 : (položka1 : typ1;  

 položka2 : typ2; 

 ...      ); 

 ... 

 end 

kde rozlišovací typ variantnej časti môže byť len ordiná lny typ. 

Napríklad: 

type  TypPohlavia =(MUZ,ZENA); 

 TypMiery=(PRSIA,PAS,BOKY); 

OSOBA = record 

 Priezvisko, 

 Meno   : string[15]; 

 DatNar : string[10]; 

 case Pohlavie : TypPohlavia of 

 MUZ : (Hmotnost : real; 

 Brada   : boolean); 

 ZENA : (Miery : array[TypMiery] of  byte) 

 end; 

Príklad ZAZ-3: Program ZAZ-2 doplňte o variantný  zá znam evidujú ci u mužov fú zy a u žien počet detí. 

Analý za: Aj keď je program veľmi podobný  predchá dzajú cemu, uvá dzame ho podrobne až po celkom 

spoločnú  časť. 
program ZAZ_3; 
uses  crt; 
const pk=10;            { pocet kariet } 
      dmena=15; 
      dadr=25; 
type  karta=record 
           ec:byte; 
           meno:string[dmena]; 
           adr:record 
               miesto, 
               ulica:string[dadr]; 
               end; 
           case pohl:(muz,zena) of 
             muz:(fuzy:boolean); 
             zena:(deti:integer); 
           end; 
     kartoteka=array[1..pk] of karta; 
var  k:kartoteka; 
     apk,i:byte; 
     q:char; 
 
procedure vstup_karta; 
  var odp:char; 
  begin 
  clrscr; 
  apk:=apk+1; 
  with k[apk],adr do 
    begin 
    ec:=apk; 
    gotoxy(1,8); 
    writeln(’Evidencne cislo : ’:40,ec); 
    writeln; 
    write(’Meno a priezvisko : ’:40);readln(meno); 
    writeln; 
    write(’  Adresa - miesto : ’:40);readln(miesto); 
    writeln; 
    write(’         -  ulica : ’:40);readln(ulica); 
    writeln; 
    write(’  Muz (ano - A) ? : ’:40); 
    if upcase(readkey)=’A’ 
      then begin 
           writeln(’ano’); 
           pohl:=muz; 
           writeln; 
           write(’Ma fuzy (ano - A) ? : ’:40); 
           fuzy:=upcase(readkey)=’A’; 

           if fuzy then writeln(’ano’) else writeln(’nie’) 
           end 
      else begin 
           writeln(’nie’); 
           pohl:=zena; 
           writeln; 
           write(’Pocet deti : ’:40);readln(deti); 
           end; 
    delay(500) 
    end; 
  end; 
 
procedure vystup_kartoteka; 
  procedure vystup_karta(i:byte); 
    begin 
    clrscr; 
    writeln(’Vystup’); 
    gotoxy(1,8); 
    with k[i],adr do 
      begin 
      writeln(’  Evidencne cislo : ’:40,ec); 
      writeln; 
      writeln(’Meno a priezvisko : ’:40,meno); 
      writeln; 
      writeln(’  Adresa - miesto : ’:40,miesto); 
      writeln; 
      writeln(’          - ulica : ’:40,ulica); 
      writeln; 
      if pohl=muz 
        then if fuzy 
               then writeln(’Fuzy : ano’:43) 
               else writeln(’Fuzy : nie’:43) 
        else writeln(’Pocet deti : ’:40,deti); 
      end; 
    readln; 
    end; 
  begin 
  for i:=1 to apk do vystup_karta(i); 
  end; 
... 
 

Na zá ver ú dajového typu zá znam uvá dzame „červíka Wurmiho“ (Príklad POLE-13) doplneného 

o jeho rast, pridanie člá nku tela, pri každom zožratí potravy (listu). 

 

Príklad ZAZ-4: Program POLE-13 doplňte o „telo“ červíka Wurmiho, ktoré narastie o jeden člá nok pri 

každom zožratí potravy. 

Analý za: Program POLE-13 treba doplniť o proces zapamätania si polohy člá nkov tela Wurmiho.  

Z viacerý ch možností sme vybrali na zapamätanie polohy člá nkov tela jednorozmerné pole TELO, v ktorom 

sú  ako položky recordu zapamätané x-ová  a y-nová  sú radnica 1., 2. atď. člá nku tela. 

Program ponechá vame na samoštú dium, podprogramy totožné s programom POLE-13 neuvá dzame. 
program ZAZ_4; 
uses  crt,dos; 
const pocet=5; 
... 
var   body,dlzka:integer; 
      px,py,jx,jy:array[1..pocet] of integer; 
      telo:array[0..pocet] of record x,y:integer end; 
 
procedure skry_kurzor; 
... 

procedure ohrada; 
... 
procedure generuj_potravu_jed; 
... 
procedure vykresli_potravu_jed; 
... 
procedure inicializacia; 
  var limit:integer; 
  begin 
  randomize; 
  generuj_potravu_jed; 
  skry_kurzor; 
  clrscr; 
  ohrada; 
  vykresli_potravu_jed; 
  dlzka:=0; telo[0].x:=40; telo[0].y:=12; 
  gotoxy(telo[0].x,telo[0].y); write(tvaricka); 
  body:=0; 
  end; 
 
function jed:boolean; 
  var i:integer; 
  begin 
  i:=0; 
  repeat 
    inc(i) 
  until ((jx[i]=telo[0].x) and (jy[i]=telo[0].y)) or (i=pocet); 
  jed:=(jx[i]=telo[0].x) and (jy[i]=telo[0].y) 
  end; 
 
function potrava:boolean; 
  var i:integer; 
  begin 
  i:=0; 
  repeat 
    inc(i) 
  until ((px[i]=telo[0].x) and (py[i]=telo[0].y)) or (i=pocet); 
  if (px[i]=telo[0].x) and (py[i]=telo[0].y) 
    then begin potrava:=true; inc(body); px[i]:=0; py[i]:=0 end 
    else potrava:=false; 
  gotoxy(3,2); write(body); 
  end; 
 
procedure pohyb; 
  var stlznak:char; 
      i:integer; 
  begin 
  repeat 
    if keypressed then stlznak:=readkey; 
    if potrava then inc(dlzka); 
    gotoxy(telo[dlzka].x,telo[dlzka].y); write(’ ’); 
    for i:=dlzka downto 1 do 
      begin 
      telo[i]:=telo[i-1]; 
      gotoxy(telo[i].x,telo[i].y); write(’0’); 
      end; 
    case stlznak of 
      ESC:halt; 
      sipkaVPRAVO:inc(telo[0].x); 
      sipkaVLAVO :dec(telo[0].x); 
      sipkaHORE  :dec(telo[0].y); 
      sipkaDOLE  :inc(telo[0].y); 
      end; 
    gotoxy(telo[0].x,telo[0].y); write(tvaricka); 
    delay(100); 
  until (telo[0].x=1)or(telo[0].x=80)or(telo[0].y=1)or(telo[0].y=24)or jed or 



 Zá klady programovania  - Turbo Pascal   101 

(body=pocet); 
  if jed 
    then begin gotoxy(33,12); write(’ J E D ! ! ! ’) end 
    else if body=pocet 
           then begin gotoxy(20,12); write(’V Y B O R N E !!! VYHRAL SI!!!’) end 
           else begin gotoxy(25,12); write(’B U M ! ! !  O H R A D A ! ! !’) end  
  end; 
 
BEGIN 
inicializacia; 
pohyb; 
readln 
END. 
 

 

Nerieš ené  úlohy na zá znam: 

5. Navrhnite ú sek definícií a deklará cií pre evidenčnú  kartu knihy a čitateľa v knižnici. 

6. Vytvorte program na evidenciu kníh (čitateľov) knižnice. 

7. Program z ú lohy 6 doplňte o vyradenie čitateľa (knihy) z evidencie. 

8. Program z ú lohy 6 doplňte o vyhľadanie knihy (čitateľa) po zadaní mena. 

9. Vytvorte program na evidenciu motorový ch vozidiel. 

10. Program z ú lohy 9 doplňte o variantný  zá znam, ktorý  podľa typu vozidla bude evidovať napr. pri 

osobnom aute počet miest a nosnosť, pri autobuse počet miest, klimatizá ciu a pod., pri ná kladnom aute 

nosnosť, ná kladnú  plošinu a pod. 

11. Vytvorte program pracujú ci ako elektronický  dvojjazyčný  (viacjazyčný ) slovník. 

12. Vytvorte program, ktorý  po zadaní textu zistí počet vý skytov jednotlivý ch slov v texte. 

13. Pre typ RETAZEC = record  

 POCETZN : 0..255;  
 ZNAKY : array[1..255] of char 

 end; 

       napíšte procedú ru, ktorá  spojí dva reťazce do tretieho. 

14. K ú dajovému typu z príkladu 13 napíšte funkciu, ktorá  zistí, či jeden reťazec je obsiahnutý  v druhom (je 

podreťazcom). 

Ú dajový  typ súbor 
 Ak potrebujeme uchovať ú daje aj po vypnutí počítača resp. po ukončení programu, musíme ich 

uložiť na vonkajšie pamäťové médium – disk, disketu a pod. Zapisovať a čítať ú daje z vonkajších 

pamäťový ch médií ná m v TP umožňuje ú dajový  typ sú bor. 

Súbor je štruktú rovaný  ú dajový  typ, ktorý  sa skladá  z teoreticky neobmedzeného počtu zložiek, všetky 

rovnakého typu. Prakticky je počet zložiek sú boru obmedzený  kapacitou vonkajšej pamäte. 

Definícia typu sú bor má  tvar:   type mt = file of tz 

kde mt je meno typu a tz je typ zložky, ktorý  nesmie byť typ sú bor ani typ obsahujú ci ako zložku typ sú bor. 

Napríklad: type CELE = file of integer;   typ sú bor obsahujú ci celé čísla 

  type KARTOTEKA = file of KARTA;  typ umožňujú ci uloženie kartotéky do sú boru 

  type BODY = file of record x,y,z:real end; typ sú bor obsahujú ci trojice x,y,z 

  var F : file of string;    premenná  typu sú bor obsahujú ci reťazce 

 

 Sú bory môžu byť vnú torné (pracovné) a vonkajšie. Pracovné sú bory si môže vytvoriť program pri 

riešení pamäťovo ná ročnejších ú loh, TP im dá va príponu $$$. Po skončení programu sa môžu z vonkajšej 

pamäte odstrá niť. Ď alej sa budeme zaoberať len vonkajšími sú bormi, ktoré existujú  aj po skončení programu 

a umožňujú  trvalé uloženie dá t. 

 Zá kladné príkazy pre prá cu so sú bormi umožňujú  len sekvenčné spracovanie sú boru, t.j. vytvá ranie 

aj čítanie sú boru len zložku po zložke (od prvej k druhej atď.). V každom okamihu spracovania sú boru je 

prístupná  len jedna zložka sú boru, na ktorú  „ukazuje“ prístupová  premenná  sú boru. Ak F je premenná  typu 

sú bor, k nej prislú chajú ca prístupová  premenná  má  označenie F^ a vznikne automaticky pri deklará cii 

premennej typu sú bor. 

 Deklará ciou premennej typu sú bor vznikne len abstraktný  vonkajší sú bor, sú  určené len jeho logické 

vlastnosti. K spojeniu logického sú boru s fyzický m – skutočne existujú cim na konkrétnom vonkajšom 

pamäťovom médiu, dôjde príkazom  assign ( logické_meno , fyzické_meno )  kde logické meno je 

premenná  typu sú bor a fyzické meno je reťazec obsahujú ci ná zov sú boru na disku, prípadne doplnený  aj 

o cestu. Napríklad: assign(F,´CITATEL.DAT )́, assign(SUB,´C:\TP\CELE1.DBF )́. 

Zá pis ú dajov do sú boru (napríklad F):  

 Zá pis ú dajov do sú boru (po stotožnení logického a fyzického mena sú boru!) prebieha v dvoch 

fá zach: 

1. sú bor sa pripraví na zá pis príkazom rewrite(F). Vytvorí sa prá zdny sú bor (prá zdna hodnota sú boru), 

prístupová  premenná  F^ sa nastaví na „koniec“ sú boru. Ak sú bor už obsahuje nejaké zložky, budú  

odstrá nené! 

2. vlastný  zá pis do sú boru sa realizuje príkazom write(F,V), kde V je vý raz rovnakého typu, ako je typ 

zložky sú boru. Po vyhodnotení vý razu V, získaní konkrétnej hodnoty, dôjde k jej zá pisu do sú boru F. 

Bod 2 možno ľubovoľný  počet krá t opakovať. 

Čítanie ú dajov zo sú boru (napríklad F): 

 Aj sekvenčné čítanie ú dajov zo sú boru prebieha v dvoch fá zach: 

1. sú bor sa pripraví na čítanie príkazom reset(F). Príkaz spôsobí nastavenie prístupovej premennej F^ na 

začiatok sú boru. 

2. vlastné čítanie zo sú boru sa realizuje príkazom read(F,X), kde X je premenná  rovnakého typu ako je typ 

zložky sú boru. Po vykonaní príkazu je v premennej X hodnota načítaná  zo sú boru. Čítanie zo sú boru 

možno opakovať, až ký m prístupová  premenná  nie je za poslednou položkou sú boru.  

Na zistenie pozície prístupovej premennej v sú bore slú ži funkcia eof (end of file), ktorá  má  hodnotu true, ak 

prístupová  premenná  je na konci sú boru (hodnota F^ nie je definovaná ), inak má  hodnotu false. To možno 

vý hodne využiť na načítanie všetký ch zložiek sú boru schémou: 

  reset(F); 

  while not eof(F) do { pokiaľ nie je koniec sú boru opakuj } 

 begin 
 read(F,X);  { načíta hodnotu zo sú boru do premennej X } 
 write(X)  { zobrazí načítanú  hodnotu na obrazovke } 
 end; 

Ukončenie prá ce so sú borom (napríklad F): 

 Po ukončení prá ce so sú borom sa sú bor musí zavrieť príkazom close(F). 

 

Príklad SUB-1: Vytvorte program na vytvorenie sú boru a zobrazenie zložiek sú boru obsahujú ceho čísla 

typu integer. 

Analý za: Program obsahuje dve samostatné procedú ry. Procedú ra VSTUP umožňuje vytvoriť, po zadaní 

mena fyzického sú boru, sú bor obsahujú ci celé čísla a procedú ra VYSTUP zobrazuje zložky – celé čísla, 

zvoleného sú boru. Program je pre pohodlie doplnený  o menu. 
program SUB_1; 
uses  crt; 
var   F:file of integer; 
      ODP:char; 
 
procedure VSTUP; 
  var FM:string[12]; 
      X:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  rewrite(F); 
  repeat 
    write(’Vlozit cislo: ’); 
    readln(X); 
    write(F,X); 
    writeln(’Vlozit dalsie cislo (nie - N) ?’:52); 
  until upcase(readkey)=’N’; 
  close(F); 
  end; 
 
procedure VYSTUP; 
  var FM:string[12]; 
      X:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  reset(f); 
  while not eof(F) do 

    begin 
    read(F,X); 
    write(X:4); 
    end; 
  close(F); 
  readln 
  end; 
 
BEGIN 
repeat 
  clrscr; 
  gotoxy(1,10); 
  writeln(’Vytvorit subor ............... V’:50); 
  writeln; 
  writeln(’Zobrazit subor ............... Z’:50); 
  writeln; 
  writeln(’Koniec ....................... K’:50); 
  repeat ODP:=upcase(readkey); until (ODP=’V’)or(ODP=’Z’)or(ODP=’K’); 
  clrscr; 
  case ODP of 
    ’V’ : VSTUP; 
    ’Z’ : VYSTUP; 
  end 
until ODP=’K’ 
END. 
 

TP má  pre prá cu so sú bormi ďalšie procedú ry a funkcie. Funkcia FileSize(F) vracia číslo udá vajú ce 

počet zložiek sú boru F. Procedú ra Seek(F,n) nastaví prístupovú  premennú  za n-tú  zložku sú boru F. Preto 

príkaz Seek(F,FileSize(F)) nastaví prístupovú  premennú  F^ za poslednú  zložku sú boru a umožní ná m 

napríklad zá pis na koniec sú boru F. Príkaz Rename(F,nové_fyz._meno) premenuje externý  sú bor (po príkaze 

assign). Príkaz Erase(F) zmaže externý  sú bor (po príkaze assign). 

Uvedené príkazy si precvičíme v nasledujú com príklade. 

 

Príklad SUB-2: Program SUB-1 (predchá dzajú ci príklad) doplňte o podprogramy na vytvorenie prá zdneho 

sú boru, nazistenie počtu zložiek sú boru, na pridanie zložky na koniec sú boru a na odstrá nenie zvolenej 

zložky zo sú boru. 

Analý za: Program SUB-2 je určený  na ozrejmenie zá kladný ch techník prá ce s ú dajový m typom sú bor. 

Ošetrili sme otvorenie sú boru funkciou IOResult, ak sú bor so zadaný m menom na disku neexistuje, vypíše sa 

sprá va: Sú bor neexistuje! Najkomplikovanejšie je odstrá nenie zvolenej zložky zo sú boru, kde sa musí použiť 

aj pomocný  sú bor (zo zdrojového sú boru sa číta a do pomocného zapisuje), ktorý  po zapísaní zvolený ch 

zložiek je premenovaný  na pôvodný  zdrojový  sú bor (pôvodný  sú bor musí byť najprv vymazaný ). 
program SUB_2; 
uses  crt; 
var   F:file of integer; 
      FM:string[12]; 
      ODP:char; 
      CHYBA:boolean; 
 
procedure PRAZDNY; 
  var FM:string[12]; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  rewrite(F); 
  close(F) 

  end; 
 
procedure VSTUP; 
  var FM:string[12]; 
      X:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  rewrite(F); 
  repeat 
    write(’Vlozit cislo: ’); 
    readln(X); 
    write(F,X); 
    writeln(’Vlozit dalsie cislo (nie - N) ?’:52); 
  until upcase(readkey)=’N’; 
  close(F); 
  end; 
 
procedure VYSTUP; 
  var FM:string[12]; 
      X:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  {$I-} reset(F); {$I+} 
  CHYBA:=IOResult<>0; 
  if CHYBA 
    then writeln(’Subor neexistuje!’) 
    else begin 
         if filesize(F)=0 then writeln(’Subor je prazdny!’); 
         while not eof(F) do 
           begin 
           read(F,X); 
           write(X:4) 
           end; 
         close(F) 
         end; 
  readln 
  end; 
 
procedure POCET; 
  var FM:string[12]; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  {$I-} reset(F); {$I+} 
  CHYBA:=IOResult<>0; 
  if CHYBA 
    then writeln(’Subor neexistuje!’) 
    else begin 
         writeln(’Pocet zloziek: ’,filesize(F)); 
         close(F); 
         end; 
  readln 
  end; 
 
procedure PRIDAT_NA_KONIEC; 
  var FM:string[12]; 
      X:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 

  {$I-} reset(F); {$I+} 
  CHYBA:=IOResult<>0; 
  if CHYBA 
    then writeln(’Subor neexistuje!’) 
    else begin 
         write(’Pridat hodnotu: ’); 
         readln(X); 
         seek(F,filesize(F)); 
         write(F,X); 
         close(F); 
         end 
  end; 
 
procedure ODSTRANIT; 
  var FPom:file of integer; 
      FM:string[12]; 
      X,HlHodnota:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  {$I-} reset(F); {$I+} 
  CHYBA:=IOResult<>0; 
  if CHYBA 
    then writeln(’Subor neexistuje!’) 
    else begin 
         assign(FPom,’POMOCNY.$$$’); 
         rewrite(FPom); 
         write(’Odstranit hodnotu: ’); 
         readln(HlHodnota); 
         while not eof(F) do 
           begin 
           read(F,X); 
           if X<>HlHodnota then write(FPom,X) 
           end; 
         close(F); close(FPom); 
         assign(F,FM); erase(F); 
         assign(FPom,’POMOCNY.$$$’); rename(FPom,FM) 
         end 
  end; 
 
BEGIN 
repeat 
  clrscr; 
  gotoxy(1,6); 
  writeln(’Vytvorit prazdny subor ....... S’:50); 
  writeln; 
  writeln(’Vytvorit neprazdny subor ..... V’:50); 
  writeln; 
  writeln(’Zobrazit subor ............... Z’:50); 
  writeln; 
  writeln(’Pocet zloziek suboru ......... P’:50); 
  writeln; 
  writeln(’Pridat zlozku na koniec ...... C’:50); 
  writeln; 
  writeln(’Odstranit zlozku zo suboru ... O’:50); 
  writeln; 
  writeln(’Koniec ....................... K’:50); 
  repeat ODP:=upcase(readkey); until ODP in [’S’,’V’,’Z’,’P’,’C’,’O’,’K’]; 
  clrscr; 
  case ODP of 
    ’S’ : PRAZDNY; 
    ’V’ : VSTUP; 
    ’Z’ : VYSTUP; 
    ’P’ : POCET; 
    ’C’ : PRIDAT_NA_KONIEC; 

    ’O’ : ODSTRANIT; 
  end 
until ODP=’K’ 
END. 
 

Príklad SUB-3: Kartotéku v programe ZAZ_2 pracujú cu s poľom prerobte na „skutočnú “ kartotéku 

pracujú cu so sú borom. 

Analý zy: Program využíva len už vyššie použité procedú ry a funkcie a preto ho nebudeme podrobnejšie 

opisovať. Odporú čame premenovať si sú bor ZAZ-2 na SUB-3 a potom urobiť príslušné zmeny. 
program SUB_3; 
uses  crt; 
const dmena=15; 
      dadr=25; 
type  karta=record 
           ec:integer; 
           meno:string[dmena]; 
           adr:record 
               miesto, 
               ulica:string[dadr]; 
               end; 
            end; 
      kartoteka=file of karta; 
var   f:kartoteka; 
      k:karta; 
      q:char; 
 
procedure vstup_karta; 
  var pocet:integer; 
  begin 
  reset(f); 
  pocet:=filesize(f); 
  seek(f,pocet); 
  with k,adr do 
    begin 
    inc(pocet); 
    ec:=pocet; 
    gotoxy(1,8); 
    writeln(’Evidencne cislo : ’:40,ec); 
    writeln; 
    write(’Meno a priezvisko : ’:40);readln(meno); 
    writeln; 
    write(’  Adresa - miesto : ’:40);readln(miesto); 
    writeln; 
    write(’         -  ulica : ’:40);readln(ulica); 
    end; 
  write(f,k); 
end; 
 
procedure vystup_karta; 
   begin 
   clrscr; 
   gotoxy(1,8); 
   with k,adr do 
     begin 
     writeln(’  Evidencne cislo : ’:40,ec); 
     writeln; 
     writeln(’Meno a priezvisko : ’:40,meno); 
     writeln; 
     writeln(’  Adresa - miesto : ’:40,miesto); 
     writeln; 
     writeln(’          - ulica : ’:40,ulica); 
     end; 
   readln; 

   end; 
 
procedure vystup_kartoteka; 
  begin 
  reset(f); 
  while not eof(f) do 
    begin 
    read(f,k); 
    vystup_karta; 
    end; 
  end; 
 
procedure najdi; 
  var hlmeno:string[dmena]; 
  begin 
  gotoxy(20,12); 
  write(’Zadaj meno hladaneho: ’); 
  readln(hlmeno); 
  reset(f); 
  while not eof(f) do 
    begin 
    read(f,k); 
    if k.meno=hlmeno then vystup_karta 
    end 
  end; 
 
procedure odstran; 
  var hlmeno:string[dmena]; 
      fn:kartoteka; 
  begin 
  gotoxy(20,12); 
  write(’Zadaj meno na odstranenie: ’); 
  readln(hlmeno); 
  assign(fn,’pracov.pom’); 
  reset(f);rewrite(fn); 
  while not eof(f) do 
    begin 
    read(f,k); 
    if k.meno<>hlmeno then write(fn,k); 
    end; 
  close(f); close(fn); 
  assign(f,’pracov.dat’);  erase(f); 
  assign(fn,’pracov.pom’); rename(fn,’pracov.dat’); 
  assign(f,’pracov.dat’); 
end; 
 
procedure pocet; 
  begin 
  reset(f); 
  gotoxy(30,12); 
  writeln(’Pocet pracovnikov: ’,filesize(f)); 
  readln; 
  end; 
 
procedure menu; 
  begin 
  repeat 
    clrscr; 
    gotoxy(1,6); 
    writeln(’            ZALOZIT KARTU....................... 1’); 
    writeln; 
    writeln(’            NAJST PODLA MENA.................... 2’); 
    writeln; 
    writeln(’            ODSTRANIT S MENOM................... 3’); 
    writeln; 
    writeln(’            POCET PRACOVNIKOV................... 4’); 

    writeln; 
    writeln(’            ZOZNAM PRACOVNIKOV.................. 5’); 
    writeln; 
    writeln(’            KONIEC.............................. 0’); 
    q:=readkey; 
    clrscr; 
    case q of 
      ’1’: vstup_karta; 
      ’2’: najdi; 
      ’3’: odstran; 
      ’4’: pocet; 
      ’5’: vystup_kartoteka; 
    end; 
  until q=’0’; 
  close(f); 
  end; 
 
BEGIN 
clrscr; 
assign(f,’pracov.dat’); 
gotoxy(10,12); 
write(’Vytvorit novy subor pracovnikov (ano - A) ? ’); 
{$I-} reset(f); {$I+} 
if (IOResult<>0) or (upcase(readkey)=’A’) then rewrite(f); 
menu; 
END. 
 
Príklad SUB-4: Vytvorte program, ktorý  spojí dva sú bory do jedného. Nech sú bory obsahujú  najviac 

dvadsaťznakové slová . 

Analý za: Podprogram zabezpečujú ci spojenie dvoch sú borov je pomerne jednoduchý . Môžeme nastaviť 

prístupovú  premennú  na koniec v prvom sú bore a pokračovať v zá pise prvkov z druhého sú boru. Ak má  mať 

spojený  sú bor nový  ná zov, stačí prvý  sú bor (v ňom sú  všetky prvky) premenovať. Program sme doplnili aj 

o utriedenie prvkov sú boru bubblesortom. Na triedenie sú borov síce existujú  špeciá lne triediace algoritmy, 

pre zaujímavosť sme sa však „pohrali“ s bubblesortom. Všimnite si, že dve rôzne logické premenné sú boru 

majú  priradené to isté fyzické meno. Je to možné len za určitý ch podmienok (v bubblesorte sa nemení počet 

zložiek sú boru, len ich poradie) a preto sa taký mto „perličká m“ radšej vyhý bajte. 
program SUB_4; 
uses  crt; 
const DLSLOVA=20; 
type  SLOVO=string[DLSLOVA]; 
      SUBOR=file of SLOVO; 
      NAZOV=string[12]; 
var   FM1,FM2,FMS:NAZOV; 
 
procedure VSTUP(FM:NAZOV); 
  var F:SUBOR; 
      S:SLOVO; 
  begin 
  assign(F,FM); 
  rewrite(F); 
  repeat 
    write(’Vlozit slovo: ’); 
    readln(S); 
    write(F,S); 
    writeln(’Koniec - stlac 0’:45) 
  until upcase(readkey)=’0’; 
  close(F) 
  end; 
 
procedure VYSTUP(FM:NAZOV); 

  var F:SUBOR; 
      S:SLOVO; 
  begin 
  assign(F,FM); 
  reset(F); 
  while not eof(F) do 
    begin 
    read(F,S); 
    write(S,’ ’) 
    end; 
  writeln; 
  close(F) 
  end; 
 
procedure SPOJ(FM1,FM2,FMS:NAZOV); 
  var F1,F2:SUBOR; 
      S:SLOVO; 
  begin 
  assign(F1,FM1); reset(F1); 
  assign(F2,FM2); reset(F2); 
  seek(F1,filesize(F1)); 
  while not eof(F2) do 
    begin 
    read(F2,S); 
    write(F1,S) 
    end; 
  close(F1); close(F2); 
  assign(F1,FMS); erase(F1); 
  assign(F1,FM1); rename(F1,FMS); 
  end; 
 
procedure BUBBLESORT(FM:NAZOV); 
  var F,FPom:SUBOR; 
      A,B:SLOVO; 
      DL,PP,I:integer; 
  begin 
  assign(F,FM); assign(FPom,FM); 
  reset(F); 
  DL:=filesize(F); 
  for PP:=1 to DL-1 do 
    begin 
    reset(F); reset(FPom); 
    read(F,A); 
    for I:=1 to DL-PP do 
      begin 
      read(F,B); 
      if A<B 
        then begin write(FPom,A); A:=B end 
        else write(FPom,B) 
      end; 
      write(FPom,A) 
    end; 
  close(F);close(FPom) 
  end; 
 
BEGIN 
clrscr; 
write(’Meno prveho suboru na disku: ’); readln(FM1); 
VSTUP(FM1); 
write(’Meno druheho suboru na disku: ’); readln(FM2); 
VSTUP(FM2); 
write(’Meno spojeneho suboru na disku: ’); readln(FMS); 
clrscr; 
writeln; writeln(’Prvy subor:’); 
VYSTUP(FM1); 
writeln; writeln(’Druhy subor:’); 

VYSTUP(FM2); 
writeln; writeln(’Spojeny subor:’); 
SPOJ(FM1,FM2,FMS); 
VYSTUP(FMS); 
writeln; writeln(’Utriedeny subor:’); 
BUBBLESORT(FMS); 
VYSTUP(FMS); 
readln 
END. 
 

Textové sú bory 
 Existujú  sú bory dá t, ktoré sú  členené na riadky. Takéto sú bory nazý vame textové a ich typ sa 

označuje štandardný m identifiká torom typu text. V textovom sú bore F okrem funkcie eof(F) je deklarovaná  

aj funkcia eoln(F) (end of line), slú žiaca na rozpoznanie konca riadku. Príkazy read(F,P) a write(F,V) sú  

rozšírené o príkazy readln(F,P) – z textového sú boru F sa prečíta lexiká lna jednotka (hodnota) 

zodpovedajú ca typu premennej P a prejde sa na nový  riadok, a writeln(F,V) – do textového sú boru F sa 

zapíše hodnota vý razu V a oddeľovač riadkov eoln. 

Najzná mejšie sú bory typu text sú  štandardný  vstupný  sú bor input, ktorému je priradené vstupné 

zariadenia klá vesnica, a štandardný  vý stupný  sú bor output, ktorému je priradené vý stupné zariadenie 

monitor. Definovanie ich typu ako sú borov typu text je implicitné. K otvoreniu oboch sú borov dochá dza 

automaticky po spustení programu. Príkazy read(input,P), readln(input,P), write(output,V) 

a writeln(output,V) sa používajú  bez identifiká torov input a output. 

 

Príklad SUB-5: Vytvorte program, ktorý  bude čítať znaky z klá vesnice (vstupný  sú bor input) a pre každý  

riadok zobrazí, koľkokrá t sa v ňom vyskytuje jeho prvý  znak. 

Analý za: Program musí obsahovať dva vnorené cykly. Vonkajší cyklus zabezpečuje čítanie znakov, až ký m 

nie je načítaný  koniec sú boru. Vnú torný  cyklus číta znaky, až ký m nie je koniec riadku, potom musí 

odriadkovať. Vložiť znak koniec riadku nie je problém, stačí stlačiť Enter (chr(13)). Ak si spomenieme, že 

„v starom dobrom DOSe“ sa koniec sú boru vkladal kombiná ciou klá ves Ctrl+Z, sme „zachrá není“. 
program SUB_5; 
var pocet:integer; 
    z,s:char; 
begin 
writeln(’Vkladaj text, ukoncenie - Ctrl+Z !!!’); 
while not eof do 
  begin 
  read(z); 
  pocet:=1; 
  while not eoln do 
    begin 
    read(s); 
    if s=z then inc(pocet); 
    end; 
  writeln(’Pocet vyskytov znaku ’,z,’ je ’,pocet,’-krat’); 
  readln; 
  end; 
end. 
 

Príklad SUB-6: Pokú ste sa vysvetliť, prečo oba ďalej uvedené programy SUB-6a aj SUB-6b pracujú  

rovnako. Ich ú lohou je uložiť do sú boru na disk text z klá vesnice. 

Program SUB_6a; 
var z:char; 
    T:text; 
    FM:string[12]; 
 
BEGIN 
write(’Meno suboru na disku: ’); 
readln(FM); 
assign(T,FM); 
rewrite(T); 
writeln(’Vkladaj text, koniec - vloz Ctrl+Z !!!’); 
while not eof do 
  begin 
  while not eoln do 
    begin 
    read(z); 
    write(T,z) 
    end; 
  readln; 
  writeln(T) 
  end; 
writeln; 
writeln(’Ulozeny subor:’); 
reset(T); 
while not eof(T) do 
  begin 
  while not eoln(T) do 
    begin 
    read(T,z); 
    write(z); 
    end; 
  readln(T); 
  writeln 
  end; 
close(T); 
readln 
END. 
 
program SUB_6b; 
uses crt; 
var  z:char; 
     T:text; 
     FM:string[12]; 
 
BEGIN 
clrscr; 
write(’Meno suboru na disku: ’); 
readln(FM); 
assign(T,FM); 
rewrite(T); 
writeln(’Vkladaj text, koniec - stlac ‘’); 
repeat 
  read(z); 
  if z<>’‘’ then write(T,z) 
until z=’‘’; 
writeln; 
writeln(’Ulozeny subor:’); 
reset(T); 
while not eof(T) do 
  begin 
  read(T,z); 
  write(z); 
  end; 
close(T); 
readln;readln 
END. 

Nerieš ené  úlohy na súbor: 

7. Kartotéku z príkladu SUB-3 doplňte o položku rodné číslo a podprogram, ktorý  zistí počet osôb 

mužského a ženského pohlavia v kartotéke (u mužov je v rodnom čísle na 3. mieste 0 alebo1, u žien 5 

alebo 6). 

8. Kartotéku z príkladu SUB-3 doplňte o položku plat a podprogram, ktorý  vypíše osoby, ktorý ch plat spĺňa 

zadané kritériá  (menší, väčší, rovný ). 

9. Vytvorte program, ktorý  zo sú boru reá lnych čísel prepíše do nového sú boru len celé čísla. 

10. Ú loha z jednorozmerného poľa o vyhľadá vaní, počte vý skytov, miesta vý skytu a pod. zrealizujte pre 

sú bor celý ch čísel (znakov, reťazcov). 

11. Vytvorte program, ktorý  z daného textového sú boru odstrá ni všetky viacná sobné medzery a prá zdne 

riadky. 

12. Vytvorte program slú žiaci ako elektronický  dvojjazyčný  slovník. 

13. Vytvorte program, ktorý  bude čítať znaky z klá vesnice a pre každý  riadok určí sú čet ordiná lnych čísel 

znakov vyskytujú cich sa v riadku. 

Dynamické premenné 
 Ú dajové typy, s ktorý mi sme sa zaoberali doteraz, patria medzi tzv. statické ú dajové typy. Premenné 

prislú chajú ce statický m typom sú  zavedené v ú seku definícií a deklará cií a je im trvalo vyhradené miesto 

v pamäti počas behu programu. Počet a rozsah premenný ch sa počas prá ce v bloku, t.j. aj v programe,  

nemôže meniť. 

 Jazyk pascal umožňuje vytvá rať premenné nie len v ú seku deklará cií premenný ch, ale aj 

v príkazovej časti. Dokonca možno tieto premenné v príkazovej časti aj rušiť. Premenné s tý mito 

vlastnosťami nazý vame dynamické  premenné  a príslušné ú dajové typy nazý vame dynamické ú dajové 

štruktú ry. 

 Z toho, čo sme uviedli doteraz, vyplý va, že dynamické premenné nemožno zaviesť v ú seku 

deklará cií a teda zabezpečiť prístup k ich hodnotá m pomocou ich identifiká torov. Dynamické premenné 

vznikajú  a zanikajú  počas realizá cie programu a sprístupnenie ich hodnôt sa robí pomocou nového 

ú dajového typu ukazovateľ. 

Ak t je identifiká tor nejakého typu, tak typ ukazovateľ definujeme zá pisom: 

type u = ^ t 

kde u je meno (identifiká tor) typu ukazovateľ. Hovoríme aj, že typ t je zviazaný  s typom u a množina hodnôt 

premennej typu ukazovateľ „ukazuje“ na prvky typu t. 

Napríklad: 

 type UKAZ1 = ^ integer; 

UKAZ2 = ^ PRVOK; 

Ak p je premenná  typu ukazovateľ, tak premenná  typu zviazaného s typom ukazovateľ (premenná  typu t) sa 

nazý va dynamická  premenná  a označuje sa p^. 

Grafická  interpretá cia: 

            p : ^ t „ukazuje“ p ^ : t  

 

 premenná  typu dynamická  
 ukazovateľ premenná  

 

Pomocou hodnoty premennej typu ukazovateľ sprístupňujeme dynamickú  premennú . V ú seku 

definícií a deklará cií je zavedená  len premenná  typu ukazovateľ! Do množiny hodnôt premennej typu 

ukazovateľ patrí vždy aj hodnota nil, jediná  konštanta typu ukazovateľ, ktorá  neukazuje na nijaký  prvok (na 

žiadnu dynamickú  premennú ). 

Graficky:     p : ^ t 

 

      p := nil 

Premenná  p má  síce priradenú  hodnotu, ale „neukazuje“ na žiadnu dynamickú  premennú . 

Príkazom new(p), kde p je premenná  typu ukazovateľ, sa vytvorí dynamická  premenná  typu t 

s nedefinovanou hodnotou a ukazovateľ na tú to premennú  sa uloží do premennej p. Príkazom new(p) 

priraďujeme premennej p hodnotu, ktorou je zodpovedajú ca dynamická  premenná  a ktorú  označujeme p^. 

Graficky:        p : ^ t    p^ : t 

                                

 new(p) 

 

Príkazom dispose(p), kde p je premenná  typu ukazovateľ, sa zruší dynamická  premenná  p^, na ktorú  

ukazovala premenná  p. Hodnota premennej p nie je po ukončení príkazu dispose(p) definovaná . Pamäťový  

priestor, ktorý  zaberala dynamická  premenná , sa dal k dispozícii na ďalšie použitie procesorom. 

Zhrnutie:                

var  p : ^t ;                  p 

 (v pamäti počítača sa vyhradí pamäťové miesto pre statickú  premennú  typu ukazovateľ, t.j. 32 bitov pre 

„akú si“ adresu) 

p := nil ;        p 

(do pamäťového miesta vyhradeného pre premennú  typu ukazovateľ sa uloží číslo neexistujú cej adresy, 

napríklad 0) 

new (p) ;       p 

                                        p^ : t 

(do pamäťového miesta vyhradeného pre premennú  typu ukazovateľ sa uloží adresa vytvorenej dynamickej 

premennej p^) 

 

Lineárny jednosmernezreť azený  zoznam 
 Z dynamický ch štruktú r sa najčastejšie využíva lineá rny jednosmernezreťazený  zoznam. Graficky ho 

možno zná zorniť: 

 

 

kde Z je premenná  typu ukazovateľ (pointer), ktorá  ukazuje na prvú  dynamickú  premennú  v zozname, 

ktorého prvky sú  typu zá znam (record) s dvoma položkami: s hodnotou (h1 až h5) a s ukazovateľom na ďalší 

prvok zoznamu. Ukazovateľ posledného prvku zoznamu už nemá  kde ukazovať a preto má  hodnotu nil. 

Postup v zozname je sekvenčný , od ukazovateľa na prvý  prvok cez ukazovatele na nasledujú ci prvok až na 

koniec zoznamu. V ú seku definícií a deklará cií je zavedený  len ukazovateľ na zoznam (premenná  Z typu ^t) 

a typ dynamickej premennej (t). Vlastné prvky zoznamu, dynamické premenné, vznikajú  podľa potreby 

počas prá ce programu príkazom new. Na pohyb v zozname používame pomocnú  premennú  typu ukazovateľ 

a premennú  Z nechá vame väčšinou ukazovať vždy na začiatok zoznamu (inak sa nemá me možnosť dostať na 

začiatok zoznamu). 

? 

? 

 h1            h2           h3         h4       h5  Z 

Program pracujú ci so zoznamom by začínal: 

type UKAZ=^PRVOK; 

 PRVOK=record 

 hodnota: typ_hodnoty; 

 dalsi: UKAZ 

 end; 

var Z : UKAZ; 

Prá cu s lineá rnym jednosmernezreťazený m zoznamom demonštruje nasledujú ci príklad. 

 

Program DYN-1: Vytvorte program umožňujú ci vo forme menu vytvoriť prá zdny zoznam, naplniť ho 

hodnotami, vypísať hodnoty zo zoznamu, odstrá niť zvolenú  hodnotu, uložiť hodnoty zoznamu do sú boru 

a naopak, načítať ich zo sú boru do zoznamu. 

Analý za: Program je ponechaný  na samoštú dium, odporú čame však kresliť si jednotlivé situá cie pomocou 

zavedenej symboliky. Všimnite si, v akom poradí prvky ukladané do zoznamu! 
program DYN_1; 
uses crt; 
type ukazovatel=^prvok; 
     prvok=record 
             h:integer; 
             u:ukazovatel 
           end; 
     subor=file of integer; 
var  z:ukazovatel; 
     sub:subor; 
     q:char; 
 
procedure vytvor(var z:ukazovatel); 
  begin 
  z:=nil 
  end; 
 
function prazdny(z:ukazovatel):boolean; 
  begin 
  prazdny:=z=nil 
  end; 
 
procedure nahodne(var z:ukazovatel); 
  var p:ukazovatel; 
      i:integer; 
  begin 
  z:=nil; randomize; 
  for i:=1 to random(10)+2 do 
    begin 
    new(p); 
    with p^ do 
      begin 
      h:=random(10); 
      u:=z;z:=p; 
      end; 
    end; 
  end; 
 
procedure napln(var z:ukazovatel); 
  var p:ukazovatel; 
      q:char; 
  begin 
  z:=nil; 

  repeat 
    writeln(’Vlozit prvok ( ano -> A ) ?’);q:=upcase(readkey); 
    if q=’A’ 
      then begin 
           new(p); 
           with p^ do 
             begin 
             write(’Hodnota: ’);readln(h); 
             u:=z;z:=p; 
             end; 
           end; 
  until q<>’A’; 
  end; 
 
procedure dopln(var z:ukazovatel); 
  var p:ukazovatel; 
      q:char; 
  begin 
  repeat 
    writeln(’Vlozit prvok ( ano -> A ) ?’);q:=upcase(readkey); 
    if q=’A’ 
      then begin 
           new(p); 
           with p^ do 
             begin 
             write(’Hodnota: ’);readln(h); 
             u:=z;z:=p; 
             end; 
           end; 
  until q<>’A’; 
  end; 
 
procedure vymaz(var z:ukazovatel); 
  var p,p1:ukazovatel; 
  begin 
  if not prazdny(z) 
    then begin 
         repeat 
           p:=z; 
           writeln(p^.h:5,’ - vymazat ( ano -> A ) ’);q:=upcase(readkey); 
           if q=’A’ 
             then begin 
                  z:=p^.u; 
                  dispose(p); 
                  p:=z 
                  end 
             else begin 
                  p1:=p; 
                  p:=p^.u 
                  end; 
         until q<>’A’; 
         while p<>nil do 
           begin 
           writeln(p^.h:5,’ - vymazat ( ano -> A ) ’);q:=upcase(readkey); 
             if q=’A’ 
               then begin 
                    p1^.u:=p^.u; 
                    dispose(p); 
                    p:=p1^.u 
                    end 
               else begin 
                    p1:=p; 
                    p:=p^.u 
                    end; 
           end 
         end 



 Zá klady programovania  – Turbo Pascal 118  

  end; 
 
procedure vypis(z:ukazovatel); 
  var p:ukazovatel; 
  begin 
  if prazdny(z) 
    then writeln(’Zoznam je prazdny’) 
    else begin 
         p:=z; 
         write(p^.h:5); 
         while p^.u<>nil do 
           begin 
           p:=p^.u; 
           write(p^.h:5) 
           end 
         end 
  end; 
 
procedure utried(var z:ukazovatel); 
  var p,q,k:ukazovatel; pom:integer; 
      bola_vymena:boolean; 
  begin 
  if (z<>nil)and(z^.u<>nil) 
    then repeat 
         p:=z^.u; q:=z; bola_vymena:=false; 
         while (p<>nil)and(p<>k) do 
           begin 
           if p^.h<q^.h 
             then begin 
                  pom:=p^.h; p^.h:=q^.h; q^.h:=pom; 
                  bola_vymena:=true 
                  end; 
           q:=p; p:=p^.u 
           end; 
         k:=q 
         until not bola_vymena; 
  end; 
 
procedure zapis_sub(z:ukazovatel); 
  var p:ukazovatel; 
  begin 
  rewrite(sub); 
  p:=z; 
  repeat 
    write(sub,p^.h); 
    p:=p^.u 
  until p^.u=nil; 
  write(sub,p^.h); 
end; 
 
procedure citaj_sub(var z:ukazovatel); 
  var p:ukazovatel; 
  begin 
  reset(sub); 
  z:=nil; 
  while not eof(sub) do 
    begin 
    new(p); 
    with p^ do 
      begin 
      read(sub,h); 
      u:=z;z:=p; 
      end; 
    end; 
end; 

BEGIN 
assign(sub,’dynamic.sub’); 
rewrite(sub); 
vytvor(z); 
repeat 
  clrscr;gotoxy(1,5); 
  WRITELN(’VYTVOR PRAZDNY ................ 1’:53); 
  WRITELN(’VYTVOR NAHODNE ................ 2’:53); 
  WRITELN(’NAPLN ......................... 3’:53); 
  WRITELN(’DOPLN ......................... 4’:53); 
  WRITELN(’VYMAZ ......................... 5’:53); 
  WRITELN(’UTRIED ........................ 6’:53); 
  WRITELN; 
  WRITELN(’VYPIS ......................... 7’:53); 
  WRITELN; 
  WRITELN(’NACITAJ ZO SUBORU ............. 8’:53); 
  WRITELN(’ULOZ DO SUBORU ................ 9’:53); 
  WRITELN; 
  WRITELN(’KONIEC ........................ 0’:53); 
  q:=readkey;clrscr; 
  case q of 
    ’1’:vytvor(z); 
    ’2’:nahodne(z); 
    ’3’:napln(z); 
    ’4’:dopln(z); 
    ’5’:vymaz(z); 
    ’6’:utried(z); 
    ’7’:vypis(z); 
    ’8’:citaj_sub(z); 
    ’9’:zapis_sub(z); 
  end; 
if (q>’2’) and (q<’5’) or (q=’7’) then readln 
until q=’0’; 
close(sub) 
END. 
 

Na zá ver uvedieme príklad, ktorý  využíva lineá rny obojsmernezreťazený  zoznam, ktorý  možno 

graficky zná zorniť (3-prvkový ): 

 

 

 

Program DYN-2: Vytvorte program na vý počet cifier n! s využitím dynamický ch dá tový ch štruktú r.  

Analý za: Vý počet n! zlyhá va u štandardný ch celočíselný ch typov už pri n=13. Nasledujú ci program počíta 

cifry n-faktoriá lu pre „neobmedzené“ n. Podstatou je „manuá lne“ ná sobenie, cifru za cifrou. 
program DYN_3; 
uses  crt; 
const Esc=chr(27); 
type  ukaz=^prvok; 
      prvok=record 
              cifra:0..9; 
              pred,za:ukaz 
            end; 
var   cislica,prva:ukaz; 
      posl:pointer; 
      i,n,pocet,prenos,stl,ria:integer; 
                         sucin:integer; { longint } 
procedure text_pocitam; 
  begin 
    stl:=wherex;ria:=wherey; 
    gotoxy(34,1);textcolor(15+128);writeln(’POCITAM  ’); 

h2 h3 h1   Z 

    gotoxy(stl,ria);normvideo; 
  end; 
 
procedure text_faktorial; 
  begin 
    stl:=wherex;ria:=wherey;if ria=1 then ria:=2; 
    gotoxy(34,1);writeln(’FAKTORIAL’); 
    gotoxy(stl,ria) 
  end; 
 
BEGIN 
repeat 
  clrscr; 
  text_faktorial; 
  write(’N = ’);readln(n);write(n,’ ! = ’); 
  text_pocitam; 
  new(cislica); 
  with cislica^ do begin cifra:=1;pred:=nil;za:=nil end; 
  posl:=cislica;prva:=cislica;prenos:=0;sucin:=1; 
  for i:=1 to n do 
    begin 
    cislica:=posl; 
    repeat 
      with cislica^ do 
        begin 
        sucin:=cifra*i+prenos; 
        cifra:=sucin mod 10; 
        prenos:=sucin div 10; 
        cislica:=pred 
        end 
    until cislica=nil; 
    while prenos<>0 do 
      begin 
      new(cislica); 
      with cislica^ do 
        begin 
        pred:=nil;za:=prva;prva^.pred:=cislica;prva:=cislica; 
        cifra:=prenos mod 10;prenos:=prenos div 10; 
        end; 
      end; 
    end; 
  text_faktorial; 
  cislica:=prva;pocet:=0; 
  repeat 
    write(cislica^.cifra);pocet:=pocet+1;cislica:=cislica^.za; 
    dispose(prva); prva:=cislica; 
  until cislica=nil; 
  writeln;writeln(’Pocet cifier v cisle: ’,pocet, ’        Koniec - Esc’) 
until readkey=Esc 
END. 

Nerieš ené  úlohy na dynamické  dá tové  š truktúry: 

3. Zá sobník je dynamická  dá tová  štruktú ra, na ktorej sú  definované operá cie: 

a) pridať prvok na vrch zá sobníka 

b) odobrať prvok z vrchu zá sobníka  

c) zistiť, či je zá sobník prá zdny 

d) vytvoriť prá zdny zá sobník 

Zrealizujte zoznam pracujú ci ako zá sobník (zoznam typu LIFO, last-in first-out). 

4. Vytvorte program kontrolujú ci sprá vnosť ozá tvorkovania vý razu s využitím v príklade 3 opísaného 

zá sobníka. 

5. Rad je dynamická  dá tová  štruktú ra, na ktorej sú  definované operá cie: 

a) pridať prvok na koniec radu 

b) odobrať prvok zo začiatku radu 

c) zistiť, či je rad prá zdny 

d) vytvoriť prá zdny rad 

Zrealizujte zoznam pracujú ci ako rad (zoznam typu FIFO, first-in first-out). 

6. Vytvorte program na vý počet Fibonacciho čísel pomocou trojprvkového radu. 

Fibonacciho postupnosť: 0, 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144,... 

vzorcami: F0 = 0 F1 = 1 Fi = Fi-1 + Fi-2  pre i ≥ 2 

7. Vytvorte program na vý počet hodnôt Pascalovho trojuholníka pomocou radu. 

8. Vytvorte podprogramy na vyhľadá vanie, zistenie počtu vý skytov, miesta vý skytu a pod. prvkov 

so zvolený mi vlastnosťami v lineá rnom jednosmernezreťazenom zozname. 

9. Vytvorte program, ktorý  spojí dva lineá rne jednosmernezreťazené zoznamy do jedného. 

 Ú vod 
 

 Skriptá  Zá klady programovania – Turbo Pascal sú  sumarizá ciou poznatkov z klasického kurzu 

programovania. Nekladú  si za cieľ podrobne vyložiť všetky pojmy. Tiež vý klad a poradie tém má  klasické 

usporiadanie, od najjednoduchšieho k zložitejšiemu, od zá kladný ch pojmov k zložitejším štruktú ram. Vý klad 

a príklady sa nevyhý bajú  ú lohá m s matematický m zá kladom, ako sa o to pokú šajú  modernejšie učebnice 

programovania. Domnievam sa, že ak má te odmietavý  vzťah k matematike, nemôže vá s programovanie 

a algoritmizá cia zaujať. Musíte byť trochu matematik, aby ste presne a jednoznačne vedeli formulovať 

a zapísať svoje myšlienky a trochu umelec, aby ste doká zali abstraktne myslieť a vytvoriť si svoje originá lne 

riešenie daného problému. Zá kladné „ťahy štetcom“ sa možno naučíte prá ve z tý chto skrípt. 

 Obsahom vý kladu sú  zá kladné algoritmické konštrukcie sekvencia, vetvenie a cyklus, a príkazy 

Turbo Pascalu, umožňujú ce zrealizovať tieto konštrukcie v programe. Preto sa v príkladoch používajú  len 

štandardné ú dajové typy integer, real, boolean a char a jediný  štruktú rovaný  typ string. Neklá dol som si za 

cieľ o každom zavedenom pojme uviesť všetko, skôr uká zať jeho najtypickejšie použitie, preto sa miestami 

vedome dopú šťam nepresností alebo skôr neú plností, mali by však byť skutočne nepodstatné; možno sa 

dopú šťam aj chý b, ale tý ch určite nevedome. Tiež si nekladiem ná roky na autorstvo použitý ch príkladov. 

Vybral som ich z veľkého množstva príkladov, s ktorý mi som sa stretol za viac ako desať rokov praxe. 

O originá lnosti riešenia ťažko hovoriť, keďže ide o „triviá lne“ príklady. Tý mto spĺňam aj požiadavku 

študentky Maji S., ktorá  ma raz poprosila o vydanie Triviá lností pá na procesora Pirošku (premenovanie je 

tiež dielom študenta, tentoraz Tomá ša Z.). 

 Skriptá , okrem vý kladu zá kladný ch štruktú r a príkazov, obsahujú  aj veľké množstvo riešený ch a 

neriešený ch ú loh. Najprv by ste sa sami mali potrá piť nad ú lohou, a až potom siahnuť po riešení v skriptá ch. 

Riešenia si treba ozrejmiť, tý m sa naučíte minimá lne „čítať“ programy a možno aj nové riešenie daného 

problému. Ideá lne by bolo, keby ste si pri štú diu programu klá dli a najmä odpovedali na otá zky charakteru: 

Počíta program sprá vne aj pre takéto hodnoty?, Musí to byť zapísané prá ve takto?, Počiatočná  hodnota 

premennej musí byť prá ve taká to?, Čo sa stane, keď to napíšem takto?, Čo keď zamením poradie tý chto 

príkazov? A ak tento príkaz nahradím tý mto? atď. Svoje tvrdenia si vždy overte priamo v TP odladením 

zmeneného programu.  

 Na zá ver vá m prajem veľa originá lnych riešení, najprv sa však treba naučiť „v pote tvá re zopá r 

zá kladný ch ťahov štetcom“. 

Algoritmus, vlastnosti algoritmu, formy zápisu algoritmu, algoritmické 
konš trukcie 

Ak sa začnete zaujímať o počítače a najmä programovanie, určite narazíte na slovo algoritmus. 

Algoritmus je ná vod, postup, ako vyriešiť určitý  problém. Tý m problémom môže byť vý mena pneumatiky 

pri defekte, zostavenie mlynčeka na mletie mäsa tak, aby to, čo vznikne zomletím, bolo ešte ďalej 

použiteľné, bezpečné prejdenie na druhú  stranu cesty, „vzorné“ umytie zubov atď. Z tý chto príkladov by 

malo byť zrejmé, že algoritmy ná s učia od narodenia a na každom kroku („toto sprav takto a toto zas 

takto...“). Sná ď v každej domá cnosti sa nachá dza zbierka algoritmov na prípravu jedá l – kuchá rska kniha. 

Každý  ná vod na obsluhu, ktorý  by ste mali dostať pri kú pe nového vý robku, je vlastne algoritmom. 

Vo všetký ch tý chto príkladoch je vykoná vateľom algoritmu človek. Pri zá pise algoritmu určeného pre 

človeka si môžeme dovoliť použiť spojenia „krá tko povariť“, „pritiahnuť až na doraz“, „vyladiť stanicu“ 

a pod. Ak je však vykoná vateľom algoritmu stroj, napr. počítač, sotva by porozumel uvedený m príkazom. 

Preto aj my prejdeme na odbornejšiu terminológiu (chceme sa predsa venovať počítačom a nie vareniu). 

Algoritmus je konečná  postupnosť elementá rnych príkazov, vykonanie ktorý ch, pre prípustné vstupné ú daje, 

po konečnom počte krokov (akcií) vedie mechanicky k získaniu korektný ch vý stupný ch ú dajov. Strašná  veta, 

skú sme ju „rozpitvať“. Konečná  postupnosť elementá rnych príkazov znamená , že ide o príkazy, ktoré sú  

zapísané v presne stanovenom poradí, je ich konečný  počet a pre vykoná vateľa algoritmu sú  elementá rne, t.j. 

každý  z príkazov pozná  a vie ho vykonať. Vykoná vanie algoritmu je proces, do ktorého vo všeobecnosti 

vstupujú  na začiatku prípustné vstupné ú daje a ich spracovaním získavame nové – vý stupné ú daje (na čo by 

sme iná č „trá pili“ počítač). Celý  proces samozrejme nesmie trvať nekonečne dlho (to by sme sa 

nedopracovali k žiadnym vý sledkom) a, keďže počítač je síce usilovný  ale hlupá k1, celý  proces musí 

prebiehať mechanicky - bez rozmý šľania. Pod korektný mi vý stupný mi ú dajmi rozumieme zmysluplné 

vý sledky adekvá tne vstupný m ú dajom; ak napríklad opakovane použijeme rovnaké vstupné ú daje, musíme 

dostať tie isté vý sledky.  

Z tejto analý zy vyplý vajú  aj vlastnosti algoritmu, z ktorý ch by sme uviedli len niekoľko: 

1. hromadnosť – algoritmus slú ži na riešenie celej skupiny ú loh určitého typy2; aby sme dodržali tú to 

vlastnosť, musí byť algoritmus napísaný  všeobecne (univerzá lne). Z matematiky vieme, že zovšeobecniť 

zá pis ná m umožňujú  premenné. Už na zá kladnej škole ná s učili, že napríklad obsah obdĺžnika sa 

vypočíta S = a.b, kde a a b sú  strany obdĺžnika a nie S = 5.7 pre nejaký  konkrétny obdĺžnik. Preto prvý  

zá pis môže byť príkazom algoritmu, druhý  nie. Zaujímavé je, že algoritmus musí byť napísaný  

všeobecne ale jeho realizá cia vždy prebieha s konkrétnymi ú dajmi; premenné nadobú dajú  konkrétne 

hodnoty v zá vislosti od zadaný ch vstupný ch hodnôt. Ak budú  vstupné ú daje - strany obdĺžnika a = 5 

a b = 7, tak konkrétny vý počet bude S = 5.7 a po ňom premenná  S nadobudne hodnotu 35. Hromadnosť 

vlastne znamená , že algoritmus slú ži na riešenie určitej skupiny konkrétnych ú loh, ktoré sa líšia len 

vstupný mi hodnotami. 

                                                 
1 Jeden mú dry pá n raz povedal: „Počítač je usilovný  hlupá k na triedenie jednotiek a nú l“. Dú fam, že s ním sú hlasíte. 
2 My predsa nevieme vymeniť len tú  konkrétnu pneumatiku na aute, na ktorom ná s to učili, ale u daného typu auta 

kedykoľvek ktorú koľvek pneumatiku; alebo prejsť cez cestu by sme mali vedieť bezpečne cez ktorú koľvek cestu, 

podobnú  tej, na ktorej ná s to učili; zuby si tiež vieme umyť, či ich má me 32 alebo menej, atď.  

2. deterministickosť (jednoznačnosť) – jednotlivé kroky (akcie) a ich poradie je presne a jednoznačne 

určené. Deterministickosť zaručuje, že pri opätovnom použití algoritmu pre tie isté vstupné ú daje 

dostaneme rovnaký  vý sledok. Umožňuje tiež realizá ciu algoritmu počítačom (počítač sa nemusí sá m 

rozhodovať). 

Ď alšie vlastnosti sú  konečnosť, rezultatívnosť, elementá rnosť atď. 

Formy zá pisu algoritmu: 

• zá pis v jazyku vý vojový ch diagramov (JVD) používa štá tnou normou stanovené značky, prehľadne 

(graficky) vyjadruje tok riadenia a dá t; ide o staršiu formu zá pisu algoritmu. 

Príklad: Algoritmus na ná jdenie najväčšieho z troch čísel vo forme vý vojového diagramu. 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

Algoritmus sa vykoná va zhora nadol. Vetvou „+“ sa pôjde, keď bude podmienka splnená  a vetvou „ -“, 

keď podmienka nebude splnená . Symbol „:=“ čítame „priraď“. 

• slovný  zá pis používa tzv. kľú čové (vyhradené) slová , napr. ak – tak – inak, čítaj, začiatok a pod.  

Príklad: Slovný  zá pis algoritmu na ná jdenie najväčšieho z troch čísel. 

 začiatok 

 čítaj ( A , B , C ); 

 ak A > B 

        tak   MAX := A 

      inak MAX := B; 

 ak C > MAX 

      tak   MAX :=C; 

 píš ( MAX ); 

 koniec. 

začiatok 

vstup: 
A, B, C 

A > B 

MAX := A MAX := B 

C > MAX 

MAX := C 

vý stup: 
MAX 

koniec 

• zá pis štruktú rogramom1 – najnovšia grafická  forma zá pisu algoritmu. 

Príklad: Algoritmus na ná jdenie najväčšieho z troch čísel zapísaný  N-S diagramom. 

 

 

 

 

 

 

 

 

 

 

Položme si otá zku: Je každá  ú loha algoritmicky riešiteľná , t.j. existuje na každý  problém algoritmus, 

pomocou ktorého ho vieme vyriešiť? Intuitívne cítime, že asi nie. Na algoritmicky neriešiteľné ú lohy síce 

nenarazíme „na každom kroku“, ale predsa existujú . Podarilo sa napríklad doká zať tvrdenie: Neexistuje 

algoritmus, ktorý  by pre aký koľvek daný  program rozhodol (na zá klade zdrojového tvaru programu 

a vstupný ch ú dajov), či tento program skončí.2 Zatiaľ sme neobjasnili slovo program. Veľmi jednoducho: 

program je algoritmus ú lohy zapísaný  v jazyku, ktorému rozumie počítač, v tzv. programovacom jazyku. 

Onedlho sa jedný m z programovacích jazykov začneme podrobnejšie zaoberať. 

Ak je ú loha algoritmicky riešiteľná , jej algoritmus možno vytvoriť kombiná ciou len troch 

algoritmický ch konštrukcií (Na prvý  pohľad aké jednoduché!). Algoritmické  konš trukcie sú : sekvencia, 

vetvenie a cyklus.  

Teraz prerušíme vý klad algoritmický ch konštrukcií a povieme si niečo o programovaní 

a programovacom jazyku Pascal.  

 

Programovanie, programovací jazyk Turbo Pascal 
 Tvorbu algoritmov nazý vame algoritmizá cia a tvorbu programov programovanie. Keďže program 

je „len“ algoritmus zapísaný  v programovacom jazyku, prioritnou činnosťou je algoritmizá cia. Má lokto však 

vydrží najprv „vyrobiť“ nedozerné množstvo algoritmov (naučiť sa algoritmizá ciu) a až potom začať 

programovať. Je to pochopiteľné, pretože až počítač pomocou programu mu dokonale „zhmotní“ to, čo 

vymyslel na algoritmizá cii. Preto aj my, aby sme si hneď overili, čo sme vymysleli, spojíme algoritmizá ciu 

a programovanie. Pri preberaní algoritmický ch konštrukcií si hneď uvedieme aj príkazy programovacieho 

jazyka, ktoré ná m umožňujú  danú  konštrukciu zrealizovať v programe. Budeme sa zaoberať zá kladmi 

programovacieho jazyka Turbo Pascal, implementá ciou jazyka Pascal, ktorý  navrhol v roku 1971 profesor 

                                                 
1 Tento ná zov sme zvolili pre podobné formy zá pisu algoritmu, ako sú  napríklad kopenogramy (autori Kofrá nek, 

Pecinovský  a Nová k) alebo N-S diagramy (autori Nassi a Schneiderman). 
2 Toto tvrdenie napríklad vedie k všeobecnejšiemu tvrdeniu: Neexistuje algoritmus, ktorý  rozhodne, či v programe je 

alebo nie je chyba. Už vieme, prečo je „programá torský  chlebík“ taký  ťažký . 

vstup: A , B , C 

A > B 
  
  

MAX := A    MAX := B 
 

C > MAX 

 
 MAX := C 
 

vý stup: MAX 

á no nie 

á no nie 

informatiky Niklaus Wirth1. Pascal bol vytvorený  na vyučovanie programovania a má  vlastnosti a kontrolné 

mechanizmy nú tiace, alebo aspoň umožňujú ce, tvoriť zrozumiteľné a prehľadné programy s čo najmenším 

počtom začiatočníckych chý b. 

 Program nie je len mechanický m prepisom algoritmu do programovacieho jazyka. Ký m v algoritme 

sa snažíme vystihnú ť podstatu riešenia problému a nezaoberá me sa podrobnosťami komuniká cie 

s užívateľom, program sa snažíme spraviť užívateľsky „prítulný “ (budeme hovoriť o užívateľskom 

komforte). Na to slú žia ďalšie, rozširujú ce príkazy. Tieto a ďalšie vlastnosti programovacieho jazyka Turbo 

Pascal si budeme postupne ozrejmovať pri konkrétnych programoch. 

 Pri prá ci v Turbo Pascale odporú čame: 

• v Options – Environment – v položke Preferences... v skupine Auto save zapnú ť voľbu Editor files 

a zapamätať ju položkou Save \...\ TURBO.TP; tá to voľba spôsobí, že pred prekladom programu sa tento 

automaticky najprv uloží na disk (ak sme sú bor s programom ešte nepomenovali, vyzve ná s vložiť ná zov 

sú boru), čo pri prípadnom použití tlačidla RESET neznamená  stratu programu (najprv však skú ste 

Ctrl+Break alebo v prostredí Windows 9x Ctrl+Alt+Del) 

• v Options – Compiler... zapnú ť v skupine Runtime errors všetky kontroly na chyby počas behu programu 

a zapamätať ich položkou Save \...\ TURBO.TP (jednotlivé kontroly preberieme podľa aktuá lnosti) 

• aj písanie nový ch programov začínať cez F3 Open a hneď vložiť ná zov sú boru, do ktorého má  byť nový  

program ukladaný  

• čo najviac používať kontextový  help: napísať kľú čové slovo, presunú ť pod neho kurzor a stlačiť Ctrl+F1 

• používať po označení textu pomocou myši alebo Shift a „kurzorový ch šípok“ na kopírovanie kombiná ciu 

Ctrl+C a K, premiestňovanie Ctrl+C a V a mazanie Ctrl+C a Y (alebo položky z ponuky Edit) 

• programy spú šťať pomocou Ctrl+F9. Ak sme program prá ve napísali alebo v ňom spravili zmenu, dôjde 

najprv k prekladu programu, pred ktorý m sa vykoná  aj kontrola sprá vnosti zá pisu (odhalia sa skomolené 

vyhradené slová , chý bajú ce bodkočiarky a pod.) a ú plnosti programu (odhalia sa nedeklarované 

premenné, nedovolené priradenia a pod.). Tá to kontrola neodhalí tzv. logické chyby (chyby v algoritme) 

alebo chyby, ktoré môžu nastať počas behu programu (zle zadaná  vstupná  hodnota, delenie nulou2 

a pod.) 

Pri písaní programu si programá tor volí ná zov programu, mená  použitý ch premenný ch, konštá nt, 

typov a iný ch objektov. Keďže norma Pascalu dovoľuje používať len písmená  anglickej abecedy, 

vo zvolený ch ná zvoch nesmú  byť použité diakritické znamienka (dĺžne, mäkčene a pod.). Tieto ná zvy sú  tzv. 

identifiká tory. Identifiká tor je postupnosť písmen (anglickej abecedy) a číslic začínajú ca písmenom; medzi 

písmená  je dodefinovaný  aj podčiarnik „_“. Identifiká tory sú  napríklad: CISLO, SUCET, obvod_obdlznika, 

ObsahStvorca, Premenna1, MENO, S, X, y, A, a, N5, z_13, ZNAK, text, OBSAHUJE, ... Identifiká tory nie 

sú  napríklad: 1A (nezačína písmenom), SUCET CISEL (použitá  medzera), premenná 1 (nedovolené 

písmeno á ), suma$ (nedovolený  znak $). Nerozlišujú  sa malé a veľké písmená , zá pisy sucet, Sucet, SuCeT, 

                                                 
1 Narodil sa v roku 1934, predná šal na Vysokej škole technickej a univerzite v Zürichu. 
2 Ak spú šťate Turbo Pascal na najnovších počítačoch Pentium, pri použití unitu Crt môže byť po spustení programu 

zahlá sená  chyba Error 200: Division by zero. Odstr á niť ju možno použitím unitu FDelay pred Crt (FDelay sa 

nenachá dza v štandardnej knižnici UNITS). 

SUCET pomenú vajú  v Pascale tú  istú  premennú . Ná zov objektu je dobré voliť si tak, aby ná m napovedal 

o jeho ú lohe, veľmi dlhé ná zvy však nie sú  praktické. 

Iná  situá cia je pri znakový ch reťazcoch alebo skrá tene len reťazcoch. Reťazec je text uzavretý  

v apostrofoch. Napríklad: ’Sú čet čísel = ’, ’Zadaj tri celé čísla: ’, ’Najväčšie číslo je ’, ’Vlož svoje meno’. 

V reťazcoch sa rozlišujú  veľké a malé písmená , dovolené sú  diakritické znamienka, medzery, interpunkčné 

znamienka. Príkazom write možno reťazce zobraziť na obrazovke v tvare, ako sú  zapísané, uľahčujú  najmä 

komuniká ciu medzi užívateľom a programom. 

S identifiká tormi a reťazcami sa budeme v programoch „stretá vať na každom kroku“. Identifiká tory 

budeme, aspoň zo začiatku, písať veľký mi písmenami, aby sa čitateľ ľahšie zorientoval. A teraz jednoduchý  

program na zozná menie sa so štruktú rou, spú šťaním a ú pravou programov. 

Príklad S-1: Vytvorte algoritmus a program, ktorý  pozdraví vypísaním slova Ahoj! na obrazovke. 
algoritmus POZDRAV_1;  program POZDRAV_1; 
začiatok    begin 
píš(’Ahoj’);   write(’Ahoj!’); 
koniec.    end.       
 
V ľavom stĺpci je slovný  zá pis algoritmu a v pravom program – prakticky preklad do angličtiny. Po spustení 

Turbo Pascalu (TP) pomocou sú boru turbo.exe v adresá ri BIN odporú čame stlačiť F3 Open (alebo F10 – File 

– Open...) a hneď zadať meno sú boru, do ktorého chceme program uložiť. Je dobré si uvedomiť rozdiel 

medzi menom programu a menom sú boru, v ktorom je uložený  daný  program. Meno programu je 

identifiká tor, ktorý  môže mať aj viac ako osem znakov. Meno sú boru môže mať najviac osem dovolený ch 

znakov (sú bor s programom POZDRAV_1 môžete nazvať napríklad s-1 alebo pozdrav1; príponu .pas pridá  

Turbo Pascal sá m). Po  zadaní mena sú boru a kliknutí na tlačidlo Open alebo stlačení klá vesu Enter Turbo 

Pascal automaticky prejde do editora („modré“ okno), slú žiaceho na písanie a upravovanie programov (ak sa 

v editore zobrazil text, znamená  to, že so zadaný m ná zvom už sú bor v adresá ri existuje a prá ve sa otvoril – 

pre nový  program musíme zvoliť iné meno sú boru). Po napísaní programu sa môžeme pokú siť spustiť 

program stlačením klá ves Ctrl+F9 alebo kliknutím na Run – Run (horné menu TP možno zaktívniť stlačením 

klá vesu F10). Po odstrá není prípadný ch chý b sa vykonajú  príkazy programu a automaticky sa TP prepne do 

editora. Vý stupy na obrazovku, v našom prípade pozdrav Ahoj!, je na užívateľskej obrazovke, do ktorej 

prepneme stlačením Alt+F5 alebo F10 - Debug - User screen. Po prezretí vý sledku prá ce programu stlačením 

ľubovoľného klá vesu sa vrá time do prostredia TP, presnejšie editora. 

 Ak chceme ukončiť prá cu v prostredí TP, môžeme použiť položky horného menu F10 – File – Exit 

alebo klá vesovú  skratku Alt+X. Pri neuložený ch programoch sa ná s TP pred ukončením opý ta na uloženie. 

Uzavrieť sú bor (program) bez ukončenia TP možno stlačením Alt+F3 alebo kliknutím na zeleno-žltý  

štvorček v ľavom hornom rohu okna (v editore). 

Ď alšia verzia programu POZDRAV demonštruje použitie príkazu ClrScr (Clear Screen), slú žiaceho na 

zmazanie obrazovky. Príkaz clrscr nepatrí do štandardu TP, nachá dza sa v jednotke (v unite) Crt, ktorá  

obsahuje rozširujú ce príkazy pre prá cu s obrazovkou, klá vesnicou, kurzorom, zvukom atď. Jednotlivé 

príkazy a funkcie unitu crt si môžeme prezrieť po vyvolaní kontextového helpu (po stlačení Ctrl+F1 

s kurzorom v slove crt) a vý bere položky Crt – Procedures. Dôležité je vedieť, že pri použití príkazov z unitu 

musíme hneď za menom programu uviesť uses (použi) a meno unitu, v našom prípade crt. Do programu sme 

vložili aj pozná mky, ktoré sa v TP vkladajú  medzi množinové zá tvorky – prekladač všetko medzi { a } 

ignoruje1. Rovnakú  funkciu plnia aj okrú hle zá tvorky s hviezdičkami, t.j. zá pisy (* a *). 
program POZDRAV_2; 
uses crt; { unit-jednotka obsahujuca prikazy rozsirujuce standard Pascalu } 
begin 
clrscr;   { prikaz na zmazanie obrazovky, nachadza sa v jednotke crt } 
write(’Ahoj!’); 
end. 
 
Tretia verzia programu odstraňuje nepohodlné prepínanie medzi užívateľskou obrazovkou a editorom. Príkaz 

vstupu readln (podrobnejšie sa ním budeme zaoberať neskôr) zastaví beh programu, môžeme si pozrieť 

vý stup na užívateľskej obrazovke, a až po stlačení klá vesu Enter program pokračuje resp. skončí, keďže 

nasleduje end s bodkou (bodka signalizuje prekladaču: koniec programu). 

 
program POZDRAV_3; 
uses crt; 
begin 
clrscr; 
write(’Ahoj!’); 
readln    { pocitac caka na stlacenie klavesu Enter, 
            pri verzii TP7 mozete pouzit aj prikaz readkey } 
end. 
 
V ďalšej verzii programu POZDRAV chceme, aby sa ná s počítač opý tal na meno a slušne ná s pozdravil: 

„Ahoj,  naše meno !“. Nemôžeme zvoliť konkrétne meno, lebo sa všetci nevolá me rovnako. Preto musíme 

použiť premennú , ktorá  bude nadobú dať hodnoty raz Jano, potom Miša, Peter, Alexandra,... Počítač musí 

mať v pamäti vyhradené miesto pre hodnotu každej premennej (ako by si ju iná č pamätal?). Aby vedel, že 

má  rezervovať pamäťové miesto a aké veľké, musíme mu to povedať. Robí sa to tzv. deklará ciou 

premenný ch, ktorá  začína slovkom var (variable – premenná ). Cez meno premennej sa odvolá vame na 

priradené pamäťové miesto a jeho veľkosť určujeme typom premennej. Ak je hodnotou premennej celé číslo, 

hovoríme o type integer, ak reá lne číslo, hovoríme o type real. O ú dajový ch typoch si povieme neskôr. 

Hodnotou našej premennej MENO je reťazec znakov, anglicky string. Pri použití typu string sa v pamäti 

vyhradí miesto až pre 255 znakov (má  niekto dlhšie krstné meno, asi nie). 
program POZDRAV_4; 
uses crt; 
var  MENO:string; 
begin 
clrscr; 
writeln(’Ako sa volas ?’); 
readln(MENO); 
writeln(’Ahoj, ’,MENO,’!’); 
readln 
end. 
 
Posledná  verzia programu POZDRAV je určená  pre tý ch, ktorí si chcú  vyskú šať príkaz GotoXY z unitu Crt. 

Potrebujete vedieť, že obrazovka má  štandardne 80 znakov v riadku (stĺpcov v smere x) a 25 riadkov na 

strane (v smere -y).  
 
program POZDRAV_5; 
uses crt; 
var  MENO:string; 

                                                 
1 Dopustili sme sa malej nepresnosti, lebo keby prvý m znakom za zá tvorkou bol $, prekladač by očaká val tzv. direktívu. 

begin 
clrscr; 
writeln(’Ako sa volas ?’); 
readln(MENO); 
clrscr; 
gotoxy(35,12); 
writeln(’Ahoj, ’,MENO,’!’); 
readln 
end. 
 

To bola odbočka do prostredia Turbo Pascalu a k prvému programu a teraz späť k algoritmický m 

konštrukciá m. 

 

SEKVENCIA 
Sekvencia je najjednoduchšou algoritmickou konštrukciou. Použijeme ju, ak sa majú  príkazy vykonať za 

sebou, v poradí, ako sú  zapísané. 

Má  tvar:   

 v slovnom zá pise a v TP: 

  p1; 

  p2; 

  ... 

 ...  pn; 

 

  kde p1, p2, ... , pn sú  príkazy 

 Príkazy od seba oddeľujeme bodkočiarkou! 

 

Vykonanie sekvencie: Príkazy p1, p2 až pn sa vykonajú  za sebou v poradí, ako sú  zapísané. 

Každá  z verzií programu POZDRAV (príklad S-1) obsahuje príkazy, ktoré sa po spustení programu 

vykonajú  postupne za sebou, ako sú  zapísané. Príklad S-1 je príkladom na sekvenciu. 

 

Príkaz priradenia 

Slú ži na priradenie hodnoty premennej. 

Má  tvar: p := v   kde p je premenná  a v vý raz; symbol „:=“ sa číta „priraď“ 

Vykonanie: vyhodnotí sa vý raz na pravej strane a jeho hodnota sa priradí premennej na ľavej strane príkazu 

priradenia. 

Príklady: A := 7; MENO := ’Peter’; I := 5; I := I + 1; OBSAH := A*A (hviezdička je symbol pre ná sobenie) 

Je dobré predstaviť si, čo sa v skutočnosti v počítači udeje. Napríklad pre príkaz I := I + 1. 

Premenná  I zaberá  v pamäti určité pamäťové miesto, zná zornime si ho takto:   I                      Tu je 

zapamätaná  hodnota premennej I. Po spustení programu tam môže byť ná hodné číslo, v Turbo Pascale 

vo verzii 7 tam je automaticky vložená  0. V našom prípade nech tam je hodnota 5, t.j. I                              

Príkaz I := I + 1 zoberie hodnotu premennej I, t.j. 5, zväčší ju o 1 na 6 a tú to hodnotu zapíše do premennej na 

ľavej strane príkazu priradenia, teda do I      . Predchá dzajú ca hodnota 5 je nená vratne stratená ! 

p1 

p2 

pn 

5 

6 

Aj keď to je asi najjednoduchší a najčastejší príkaz, dobre si ho premyslite. Zapamätajte si, že sa vykoná va 

sprava doľava a predchá dzajú ca hodnota premennej na ľavej strane sa prepíše novou hodnotou. 

Príkaz priradenia má  rovnaký  tvar v algoritmizá cii aj v Pascale, tu sa však vyžaduje, aby premenná  p 

a vý raz v boli rovnakého typu, čo je pochopiteľné, lebo ťažko by sa napríklad vý raz typu string (reťazec 

znakov) priraďoval premennej typu integer (celé čísla); dovolená  je len jedna vý nimka: celé číslo (integer) 

môže byť priradené premennej pre reá lne čísla (typu real). 

Príklad: Napíšte postupnosť (sekvenciu) príkazov priradenia, ktoré vymenia hodnoty dvoch premenný ch. 

Napríklad premenná  A nech má  hodnotu 5 a premenná  B hodnotu 7. Po vykonaní sekvencie nech je 

v A hodnota 7 a v B hodnota 5. 

Analý za:  A     B 

      ? 
   A    B 

 

Aj keď nechceme čitateľa podceňovať, sekvencia 

  A := B;  

B := A; 

nie je sprá vna (odsimulujte si, čo sa udeje v počítači). Ak si uvedomíte prečo, sprá vne riešenie ľahko zistíte. 

Vykonaním príkazu A := B sa prepíše pôvodná  hodnota premennej A (5), ktorú  však ešte potrebujeme. 

Riešením je odložiť si ju skôr, než dôjde k jej prepísaniu, niekde „bokom“, do pomocnej premennej napr. 

POM. Sprá vna sekvencia: POM := A; A :=B; B := POM;  

Pozná mky: 

1. Ú loha má  ešte jedno analogické riešenie (POM := B;...).  

2. Ak hodnoty premenný ch A a B sú  čísla, nemusíme použiť ani tretiu premennú  (napr.: A := A-B; 

B := B+A; A := B-A;). 

3. Ú loha vý meny hodnôt dvoch premenný ch je analogická  s ú lohou vý meny tekutín (napr. voda a mlieko) 

v dvoch pohá roch navzá jom. 

 

Príklad S-2: Vytvorte algoritmus a program na vý počet obsahu a obvodu obdĺžnika. 

Analý za: Algoritmus často rieši určitý  problém z nejakej vednej disciplíny, v tomto prípade z matematiky. 

Preto treba mať znalosti z daného vedného odboru, v našom prípade poznať vzorce na vý počet obsahu 

a obvodu obdĺžnika. Ď alej je dobré si uvedomiť, čo musíme poznať pre daný  vý počet (rozmery obdĺžnika - 

vstupné ú daje) a čo očaká vame, že vykonaním algoritmu získame (obsah a obvod - vý stupné ú daje).  Keďže 

zá kladnou vlastnosťou algoritmu je hromadnosť, vstupné aj vý stupné ú daje zadá vame cez premenné. 

Pre rozmery obdĺžnika sú  to bežne písmená  A a B, pre obsah a obvod si zvolíme ná zvy premenný ch OBSAH 

a OBVOD. Pri zostavovaní algoritmu uvažujte, ako by ste daný  problém riešili bez počítača („manuá lne“) 

a tento postup zapíšte vo forme algoritmu. 

 

 

 

5 7 

7 5 

manuá lne algoritmus program 
Začínam 
potrebujem vedieť rozmery obdĺžnika 
použijem vzorec na vý počet obsahu S = a.b 
použijem vzorec na vý počet obvodu 
O = 2.(a+b) 
ozná mim vý sledok 
skončil som. 

začiatok 
čítaj (A, B); 
OBSAH := A*B; 
 
OBVOD := 2*(A+B); 
píš (OBSAH, OBVOD); 
koniec. 

begin 
readln(A, B); 
OBSAH := A*B; 
 
OBVOD := 2*(A+B); 
write(OBSAH, OBVOD); 
end. 

 

Program treba doplniť o deklará ciu premenný ch. V prvej alternatíve budeme počítať len s celý mi číslami 

(typ integer), v druhej s reá lnymi (typ real). Programy sú  doplnené o užívateľsky  komfort, pod ktorý m 

rozumieme predovšetký m pred každý  príkaz vstupu read dať príkaz vý stupu write, ktorý  vypíše na 

obrazovke, čo treba zadať a tiež jednoznačný  vý stup vý sledný ch hodnôt.  

Program očaká va vloženie dvoch celý ch čísel, ktoré môžeme vložiť naraz, oddelené medzerou (nie čiarkou!) 

alebo napíšeme jedno, stlačíme Enter a potom druhé a Enter. 
program OBDLZNIK_CELE; 
uses crt; 
var A, B, OBSAH, OBVOD: integer; 
begin 
clrscr; 
write(’Zadaj strany obdlznika (cele cisla): ’); 
readln(A,B); 
OBSAH:=A*B;  
OBVOD:=2*(A+B); 
writeln(’Obsah obdlznika = ’,OBSAH); 
writeln(’Obvod obdlznika = ’,OBVOD); 
readln 
end. 
 
Typickou chybou začiatočníkov je, že po odladení programu (odstrá není chý b, ktoré ná jde prekladač TP) už 

ďalej „slepo veria počítaču“ a program ďalej netestujú . Musíte sa prinú tiť po napísaní programu ho 

otestovať, t.j. zadať čo najviac taký ch vstupný ch hodnôt, pre ktoré pozná te vý sledky a tiež vložiť tzv. 

hraničné hodnoty, ktoré sú  na hranici dovolený ch vstupný ch hodnôt alebo pri ktorý ch sa mení charakter 

vý počtu. Program OBDLZNIK by sme mohli otestovať pre A a B rovné 0, 1 a napríklad 2 a 4 (2.4=8 a 

2.(2+4)=12) alebo iné čísla, pre ktoré si ľahko spamäti zistíme sprá vne vý sledky. Skú ste aj  A = 1000 

a B = 1000. Ak má te zapnuté kontroly chý b počas behu programu (ako sme to odporučili na strane 6), 

počítač ukončí vý počet chybový m hlá sením Error 215: Arithmetic overflow – preplnenie, vý sledkom 

vý počtu je číslo väčšie ako preň vyhradené pamäťové miesto. Prečo je tomu tak, si povieme pri ú dajový ch 

typoch na nasledujú cej strane. Ak kontrolu nemá te zapnutú , možno si zlý  vý sledok 16 960 (pre obsah) ani 

nevšimnete! Takže dôveruj ale preveruj! 
program OBDLZNIK_REALNE; 
uses crt; 
var A, B, OBSAH, OBVOD: real; 
begin 
clrscr; 
write(’Zadaj strany obdlznika: ’); 
readln(A,B); 
OBSAH:=A*B;  
OBVOD:=2*(A+B); 
writeln(’Obsah obdlznika s presnostou na stotiny = ’,OBSAH:4:2); 
writeln(’Obvod obdlznika s presnostou na stotiny = ’,OBVOD:4:2); 
readln 
end. 

Reá lne čísla vkladá me podobne ako celé, len ich zá pis môže obsahovať aj desatinnú  bodku. Znova všetkému 

nerozumiete, ale naraz sa všetko nedá . Tie tajomné čísla :4:2 objasníme pri príkaze write. 

 
Ú dajové typy 

Jednoduché  štandardné  údajové  typy: 

Meno typu: Integer (ordiná lny typ) 

Množina hodnôt: celé čísla z intervalu –32 768 po 32 767 (konštanta MaxInt) 

Dovolené operá cie: + (sčítanie), - (odčítanie), * (ná sobenie), div (celočíselné delenie), mod (zvyšok po 

celočíselnom delení) 

Funkcie: abs(x), sqr(x), odd(x), trunc(x), round(x) 

  succ(x), pred(x), ord(x) 

Relačné operá tory: <, <=, =, <>, >=, > 

Vysvetlivky: 

V TP sú  pre hodnoty typu integer vyhradené v pamäti 2 B, t.j. 16 bitov. V jednom bite sa zapamätá va 

znamienko (0+/1-) a do zvyšný ch pätná stich bitov sa ukladá  hodnota bez znamienka pomocou 0 a 1. 

Všetký ch možností je 215 = 32 768. Preto do daného pamäťového miesta možno uložiť všetky nezá porné celé 

čísla od 0 po 32 767 alebo zá porné celé čísla od –1 po – 32 768. Ak je vý sledkom operá cie s typom integer 

číslo mimo tento rozsah, nastane v príklade S-2 opísaná  chyba, tzv. preplnenie (v Options – Compiler... 

prepínač Overflow Checking – kontrola preplnenia). Ak ná m uvedený  rozsah nevyhovuje, môžeme použiť 

ďalšie celočíselné typy: Shortint, Longint, Byte a Word, ktorý ch rozsahy si môžete pozrieť pomocou 

kontextového helpu (Ctrl+F1 s kurzorom pod napr. integer). Najväčší rozsah má  typ longint, kde konštanta 

MaxLongInt má  hodnotu 2 147 483 647 (= 231 –1). 

U celočíselný ch typov nie je dovolené klasické delenie, lebo jeho vý sledkom nemusí byť celé číslo. 

Dovolené je tzv. celočíselné delenie (celočíselný  podiel) div a zvyšok po celočíselnom delení mod. Tieto 

funkcie si treba ozrejmiť, lebo sa v numerický ch algoritmoch a programoch často využívajú . Napríklad 

15 div 6 = 2 a 15 mod 6 = 3 lebo 15:6 = 2 zvyšok 3, alebo 7 div 10 = 0 a 7 mod 10 = 7 lebo 7:10 = 0 

zvyšok 7, alebo 21 div 3 = 7 a 21 mod 3 = 0 lebo 21:3 = 7 zvyšok 0. Všeobecne pre kladné celé čísla A a B 

platí: Nech A div B = C a A mod B = D potom B.C + D = A a 0<=D<B. Pre zá porné celé čísla funkcie div 

a mod v TP pracujú  iná č ako v matematike! 

Vý sledkom: abs(x) je absolú tna hodnota čísla x, napr. abs(8) = 8, abs(-5) = 5, abs(0) = 0 

sqr(x) je druhá  mocnina čísla x, napr. sqt(8) = 64, sqr(-5) = 25, sqr(0) = 0 

odd(x) je true (pravda), ak x je nepá rne číslo, inak false (nepravda), napr. odd(0) = false 

trunc(x) je celá  časť reá lneho čísla x, napr. trunc(8.9) = 8, trunc(-5.6) = -5, trunc(0.5) = 0 

round(x) je zaokrú hlené reá lne číslo x, napr. round(8.9) = 9, round(-5.6) = -6, round(0.5) = 1 

succ(x) je nasledovník x, napr. succ(5) = 6, succ(-1) = 0, succ(-5) = -4 

pred(x) je predchodca x, napr. pred(5) = 4, pred(-1) = -2, pred(1) = 0 

ord(x) je poradové číslo x, napr. ord(5) = 5, ord (0) = 0, ord (-12) = -12 

Meno typu: Real (nie je ordiná lny typ!) 

Množina hodnôt: niektoré reá lne čísla z intervalu približne –1,7.1038 po 1,7.1038 

Dovolené operá cie: +, -, *, / (delenie) 

Funkcie: abs(x), sqr(x), sqrt(x), trunc(x), round(x), int(x), frac(x), sin(x), ln(x), exp(x), arctan(x) 

Relačné operá tory: <, <=, =, <>, >=, > 

Vysvetlivky: 

Keďže reá lnych čísel v matematike je nekonečne veľa (aj medzi dvoma ľubovoľný mi reá lnymi číslami) 

a pamäť počítača je konečná , počítač si doká že zapamätať len niektoré reá lne čísla. 

Ď alšie ú dajové typy pre reá lne čísla sú  single, double, extended a comp, ich použitie je však podmienené 

prítomnosťou alebo emulá ciou matematického koprocesora (funkcie v F10 – Options - Compiler... – 

Numeric processing).  

Niektoré funkcie sú  popísané vyššie, u typu integer, a má lo používané funkcie nepopíšeme. 

Vý sledkom: sqrt(x) je druhá  odmocnina z nezá porného čísla x, napr. sqrt(9) = 3, sqrt(2) = 1,4142135624 

  int(x) je celá  časť (cifry pred desatinnou bodkou) z reá lneho čísla x, napr. int(123.456) = 123 

  frac(x) je desatinná  časť z reá lneho čísla x, napr. frac(123.456) = 0.456 

exp(x) je ex 

  ln(x) je prirodzený  logaritmus kladného čísla x1 

 

Meno typu: Boolean (ordiná lny typ) 

Množina hodnôt: {False, True} 

Dovolené operá cie: not (negá cia), and (logický  sú čin), or (logický  sú čet), xor (logický  xor) 

Funkcie: succ(x), pred(x), ord(x) 

Relačné operá tory: <, <=, =, <>, >=, >, in (je prvkom) 

Vysvetlivky: 

Ide o logické hodnoty False – nepravda, 0 a True – pravda, nie 0; false < true.  

pred(true) = false, succ(false) = true; ord(false) = 0, ord(true) = 1 

 

Meno typu: Char (ordiná lny typ) 

Množina hodnôt: znaky tabuľky ASCII 

Funkcie: upcase(z) 

               succ(z), pred(z), ord(z), chr(x) 

Relačné operá tory: <, <=, =, <>, >=, >, in 

Vysvetlivky: 
Znaky tabuľky ASCII aj s poradový mi číslami možno zobraziť pomocou programu: 
program ZNAKY_ASCII; 
uses crt; 
var I:byte; 
begin 
clrscr; 
for I:=32 to 255 do 
  write(I:6,chr(I):2);             { vypis v tabulke } 
                                                 
1 Posledné dve funkcie možno použiť na vý počet xy pomocou vý razu exp(y*ln(x)), x kladné reá lne a y reá lne číslo. 

readln; 
clrscr; 
for I:=1 to 255 do 
  begin 
  writeln(I:6,chr(I):4);  { vypis pod sebou } 
  if I mod 23 = 0 then readln  { zastavi vypis vzdy po 23 riadkoch } 
  end 
end. 
 
Prvý ch 32 znakov (0-31) je riadiacich. Znak s poradový m číslom 32 je medzera atď.  

Vý sledkom funkcie chr(x) je znak zodpovedajú ci poradovému číslu x. 

Napríklad succ(’A’) = B, pred(’A’) = @, ord(’A’) = 65 a chr(65) = A, chr(ord(z)) = z, ord(chr(x)) = x. 

Všimnite si, že ak sa jedná  o konkrétny znak, t.j. konštantu, musí byť v apostrofoch, podobne ako reťazcová  

konštanta. 

Funkcia UpCase(z) zmení malé písmeno na veľké, na iné znaky nemá  vplyv, napr. upcase (’b’) = B. 

 

Ú dajové typy, pre ktoré sú  definované funkcie succ (successor - nasledovník), pred (predecessor - 

predchodca) a ord (ordinal - poradové číslo) nazý vame ordiná lne (každá  hodnota má  presne určené miesto). 

Sú  to typy integer, boolean a char. V TP sú  pre ne zavedené ešte dva praktické príkazy inc (increase - 

zväčšiť) a dec (decrease - zmenšiť). Príkaz inc(p) zväčší hodnotu premennej p o jednu pozíciu vpravo, napr. 

inc(5) = 6, inc(’A’) = B, inc(false) = true a opačne dec(p) zmenší hodnotu premennej p o jednu pozíciu vľavo, 

napr. dec(5) = 4, dec(’B’) = A, dec(true) = false. Tieto príkazy pracujú  podobne ako funkcie succ a pred 

avšak môžu mať aj tvar inc(p,n) resp. dec(p,n), kde n je celé číslo (môže byť aj zá porné) udá vajú ce, o koľko 

treba posunú ť hodnotu premennej p. Napr. inc(5,10) = 15, inc(’a’,10) = k, dec(5,10) = -5, dec(’a’,5) =  W. 

Š truktúrovaný  údajový  typ: 

Meno typu: string alebo string[n] kde n je celé číslo od 1 po 255 – udá va maximá lnu dĺžku reťazca 

Množina hodnôt: reťazce zo znakov ASCII 

Dovolené operá cie:  + (spá ja reťazce) 

Funkcie: length, copy, delete, concat, insert, pos 

Relačné operá tory: <, <=, =, <>, >=, > 

Vysvetlivky: 

String vyhradí v pamäti miesto pre 255 znakový  reťazec (pre najviac 255 znakov), napríklad string[5] 

vyhradí v pamäti miesto pre päť znakový  reťazec (pre najviac 5 znakov, ostatné nebudú  zapamätané). 

Vý sledkom funkcie length(r) je celé číslo - dĺžka reťazca r. Prá zdny reťazec ’’ má  dĺžku 0. 

Pri relačný ch operá ciá ch sa najprv porovnajú  prvé znaky v porovná vaný ch reťazcoch, ak sú  zhodné, druhé 

atď. 

Príkazy vstupu, príkaz read 
Príkazy vstupu slú žia na vstup ú dajov z klá vesnice. 

Majú  tvar: 

čítaj (p1, p2, ... , pn) 

       kde p1 až pn sú  premenné 

 

V TP sú  to modifiká cie príkazu read: 

  read (p1, p2, ... , pn ) 
  readln (p1, p2, ... , pn )   kde p1 až pn sú  premenné 
  readln 

Vykonanie: zastaví sa beh programu a počítač čaká  na vloženie príslušného počtu hodnôt v stanovenom 

poradí a deklarovaný ch typov. 

Napríklad: readln(A,B,C); readln(TEXT); readln(X); readln(MENO); readln(ODPOVED); readln; 

Pozná mky: 

• príkazy ukončené ln (line) možno interpretovať: načítaj hodnoty premenný ch p1 až pn a nastav čítanie na 

nový  riadok 

• jednotlivé číselné hodnoty môžu byť oddelené medzerami alebo vkladané v samostatný ch riadkoch 

zadaním hodnoty a stlačením klá vesu Enter 

• „prá zdny“ príkaz readln neočaká va hodnotu, len stlačenie klá vesu Enter 

• príkaz read(p1, ... , pn) nebudeme predbežne používať 

• po zadaní väčšieho čísla, ako je dovolený  rozsah daného typu, sa zobrazí chybové hlá senie Error 201: 

Range check error - prekročenie rozsahu (v Options – Compiler... prepínač Range Checking – kontrola 

rozsahu hodnôt). Napríklad u typu integer po zadaní čísla väčšieho ako je maxint (32 767). Ak nedošlo 

k zobrazeniu chybového hlá senia, pozri str. 6 

• ak pri zadá vaní hodnôt zadaná  hodnota nezodpovedá  očaká vanému typu, napr. u číselného typu vložíme 

miesto číslice písmeno, zobrazí sa chybové hlá senie Error 106: Invalid numeric format – zlý  číselný  

formá t (v Options – Compiler... prepínač I/O Checking – kontrola vstupno/vý stupný ch operá cií) 

 
Príkazy vý stupu, príkaz write 
Príkazy vý stupu slú žia na zobrazenie vý stupný ch hodnôt na obrazovke monitora. 

Majú  tvary: 

       píš  (h1, h2, ... , hn) 

       kde h1 až hn sú  vý stupné hodnoty 

 

V TP sú  to modifiká cie príkazu write: 

  write (f1, f2, ... , fn ) 
  writeln (f1, f2, ... , fn )    kde f1 až fn sú  formá tovacie vý razy 
  writeln 

Vykonanie: na obrazovke monitora sa zobrazia vý stupné hodnoty v predpísaný ch formá toch (buď hodnoty 

premenný ch alebo text pôvodne uvedený  v apostrofoch). 

vstup: 
p1,p2,...,pn 

vý stup: 
h1,h2,...,hn 

Napríklad: writeln(’Dĺžka čiary je ’,DLZKA); write(’Zadaj kladné číslo: ’); writeln(A,’ ’,B); writeln; 

Pozná mky: 

• príkazy ukončené ln (line) možno interpretovať: zobraz vý stupné hodnoty a nastav vý stup (kurzor) 

na začiatok nového riadku 

• „prá zdny“ príkaz writeln nastaví vý stup na nový  riadok 

 

Formá tovacie výrazy: 

1) h   h je vý raz reprezentujú ci vý stupnú  hodnotu 

2) h : pzv  h ako vyššie, pzv je vý raz reprezentujú ci počet znakov vý stupu 

3) h : pzv : pdm h a pzv ako vyššie, pdm je vý raz reprezentujú ci počet desatinný ch miest vý stupu 

u typu real 

 

Napríklad: 

a) nech A je premenná  typu integer 

hodnota A  príkaz  na obrazovke  pozná mka 

12345  write(A) 12345   znamienko plus sa nezobrazuje 
-12345  write(A) -12345   znamienko mínus zaberá  jeden znak 
12345  write(A:10)      12345  zľava sa doplní päť medzier 
-12345  write(A:7)   -12345  zľava sa doplní jedna medzera 
12345  write(A:3) 12345   ignoruje počet znakov vý stupu 

b) nech X je premenná  typu real (matematický  koprocesor vypnutý ) 

hodnota X  príkaz  na obrazovke  pozná mka 

123,456789 write(X)  1.2345678900E+02 tzv. semilogaritmický  tvar, E – exponent sa 
číta „krá t desať na“, pre „+“ vynechaný  znak 

123,456789 write(X:10)  1.235E+02  neprehliadnite zaokrú hlenie mantisy 
123,456789 write(X:2)  1.2E+02  minimá lny vý pis s exponentom (8 znakov) 
123,456789 write(X:6:2) 123.46   6 znakov vý stupu, 2 za des. bodkou 
123,456789 write(X:4:2) 123.46   ignoruje počet znakov vý stupu 
123,456789 write(X:10:3)     123.457  zľava sa doplnia tri medzery 
123,456789 write(X:12:10) 123.4567890000 maximá lna presnosť zobrazenia 
123,456789 write(X:5:0)    123   zľava sa doplnia dve medzery  
0,0000987654321 write(X)   9.8765432100E-05 semilogaritmický  tvar 
0   write(X)   0.0000000000E+00  semilogaritmický  tvar 

c) nech S je premenná  typu string 

hodnota S  príkaz  na obrazovke  pozná mka 

Janosik  write(S) Janosik 
Janosik  write(S:3) Janosik   ignoruje počet znakov vý stupu 
Janosik  write(S:10)    Janosik  zľava sa doplnia tri medzery 

d) nech Q je premenná  typu boolean 

hodnota Q  príkaz  na obrazovke  pozná mka 

true  write(Q) TRUE 
false  write(Q:10)            FALSE  zľava sa doplní päť medzier 
true  write(Q:1) TRUE   ignoruje počet znakov vý stupu 

Pozná mky: 

• neprehliadnite zaokrú hľovanie reá lnych čísel na zadaný  počet desatinný ch miest 

• bez použitia formá tovacích vý razov napríklad v príkaze write(A,B), kde A a B sú  typu integer, neodlíšite 

od seba cifry kladný ch čísel! Okrem formá tovania je riešením aj vloženie medzery, t.j. write(A,’ ’,B). 

• formá tovacie vý razy sú  vý razy, t.j. môžu mať aj tvar napríklad writeln(A*B), write(X : PRESNOST+2 : 

PRESNOST), kde premenná  PRESNOST typu integer môže nadobú dať hodnoty 0 (jednotky), 

1 (desatiny), 2 (stotiny), ... , 10 

• ujasnite si vykonanie príkazu writeln(’Obvod = ’:40,2*(A+B):4:2) 

 

Vrá ťme sa však k sekvencii.  

Príklad S-3: Vytvorte program simulujú ci palubný  počítač auta. Zná my je objem palivovej ná drže a 

priemerná  spotreba paliva počas jazdy. Vždy pri natankovaní plnej ná drže sa vynuluje tzv. denné počítadlo 

kilometrov. Po zadaní prejdený ch kilometrov (z denného počítadla kilometrov) počítač vypočíta aktuá lny 

dosah auta. Po zadaní vzdialenosti do cieľa vypočíta priemernú  dobu jazdy. 

Analý za: Dosah auta pri plnej ná drži sa vypočíta ako podiel: objem palivovej ná drže / priemerná  spotreba 

paliva na 100 km, krá t 100 km. Keďže tento podiel nemusí byť celé číslo, zaokrú hlime ho na celé km, 

samozrejme nadol. Aktuá lny dosah sa zrejme rovná : dosah pri plnej ná drži – prejdené km. Priemerná  doba 

jazdy sa vypočíta ako podiel: vzdialenosť do cieľa / priemerná  rý chlosť  za hodinu (konštanta). Vý sledok je 

v hodiná ch a zobrazíme ho s presnosťou na desatiny hodiny. 

 
program POCITAC; 
uses  crt; 
const ObjemPalNadrze = 46;    { litrov } 
      PriemSpotPaliva = 6.5;  { litra na 100 km } 
      PriemRychlost = 90;     { km/hod. } 
var   PrejdeneKM, AktDosah, Dosah, VzdDoCiela : integer; 
                               PriemDobaJazdy : real; 
begin 
clrscr; 
write(’Prejdene kilometre (z denneho pocitadla km): ’); 
readln(PrejdeneKM); 
Dosah:=trunc(ObjemPalNadrze/PriemSpotPaliva*100); 
AktDosah:=Dosah-PrejdeneKM; 
writeln(’Aktualny dosah: ’,AktDosah,’km’); 
writeln; 
write(’Vzdialenost do ciela: ’); 
readln(VzdDoCiela); 
PriemDobaJazdy:=VzdDoCiela/PriemRychlost; 
writeln(’Priemerna doba jazdy: ’,PriemDobaJazdy:3:1,’ hod.’); 
readln 
end. 
 

Príklad S-4: Vytvorte program na „uhá dnutie“ celého čísla z intervalu 0 až 100, ak viete, že po zadaní 

zvyškov po delení 3 (zvysok3), 5 (zvysok5) a 7 (zvysok7) mysleného čísla sa myslene číslo rovná  zvyšku po 

celočíselnom delení vý razu (70*zvysok3+21*zvysok5+15*zvysok7) 105-timi. Strašné, ani len zadaniu sa 

nedá  rozumieť! 

Analý za: Treba zadať zvyšky po delení 3, 5 a 7 mysleného čísla – to sú  vstupné ú daje a my pre ne zvolíme 

vstupné premenné označené zvysok3, zvysok5 a zvysok7. Myslené číslo sa rovná  zvyšku po delení vý razu..., 

preto si tento vý raz vypočítajme. VYRAZ = 70*zvysok3+21*zvysok5+15*zvysok7. Myslené číslo sa rovná  

zvyšku po celočíselnom delení VYRAZu 105-timi. Zvyšok po celočíselnom delení ná m dá va funkcia mod. 

A tu je program: 
program UHADNEM_CISLO; 
uses crt; 
var zv3,zv5,zv7,myslene_cislo:byte; 
begin 
clrscr; 
writeln(’U H A D N E M   M Y S L E N E   C I S L O  D O  100 !’:60); 
write(’Zvysok po deleni 3: ’); readln(zv3); 
write(’Zvysok po deleni 5: ’); readln(zv5); 
write(’Zvysok po deleni 7: ’); readln(zv7); 
myslene_cislo:=(70*zv3+21*zv5+15*zv7) mod 105; 
writeln; 
writeln(’Myslel si si cislo ’,myslene_cislo); 
readln 
end. 
 

Príklad S-5: Vytvorte program, ktorý  prepočíta čas v sekundá ch na hodiny : minú ty : sekundy. 

Analý za: Vieme, že hodina má  3 600 sekú nd. Koľkokrá t sa 3 600 nachá dza v sekundá ch (bezo zvyšku), je 

hľadaný  počet hodín – to vie vypočítať funkcia div. Funkcia mod určí zvyšok sekú nd, čo zostal na minú ty 

a sekundy. Počet minú t zistíme opäť celočíselný m delením zvyšný ch sekú nd číslom 60. No a to čo zostane, 

sú  už len sekundy (< 60). Vý počet funkciou mod možno nahradiť aj príkazom CAS := CAS – 3600*H resp. 

CAS := CAS – 60*M. 
program CAS_V_SEKUNDACH; 
uses crt; 
var  H, M, CAS : longint; 
begin 
clrscr; 
write(’Zadaj cas v sekundach: ’); 
readln(CAS); 
H:=CAS div 3600; CAS:=CAS mod 3600; 
M:=CAS div 60;   CAS:=CAS mod 60; 
writeln(H,’ : ’,M,’ : ’,CAS); 
readln 
end. 
 
Ú loha: Doká žete upraviť predchá dzajú ci program tak, aby počítal iba s jednou premennou CAS? 
program CAS_V_SEKUNDACH_2; 
uses crt; 
var  CAS:longint; 
begin 
clrscr; 
write(’Zadaj cas v sekundach: ’); readln(CAS); 
gotoxy(34,12); 
writeln(CAS div 3600,’ : ’,CAS mod 3600 div 60,’ : ’,CAS mod 3600 mod 60); 
readln 
end. 

Nerieš ené  úlohy na sekvenciu: 

1. Vytvorte program na vý počet obsahu a obvodu štvorca s presnosťou na desatiny. 

Ná vod: Vzorec pre vý počet obsahu S = a2 a obvodu O = 4a. 

2. Vytvorte program na vý počet obsahu a obvodu kruhu s presnosťou na tisíciny. 

Ná vod: Vzorec pre vý počet obsahu S = π.r2 a obvodu O = 2π.r. TP pozná  konštantu PI = 3,1415... 

3. Vytvorte program na vý počet objemu, povrchu a telesovej uhlopriečky kvá dra s presnosťou na stotiny. 

Ná vod: Vzorec pre vý počet objemu V = a.b.c, povrchu S = 2(a.b+a.c+b.c), uhlopriečky u = a2+b2+c2 . 

4. Vytvorte program na vý počet objemu a povrchu gule s najväčšou možnou presnosťou. 

Ná vod: Vzorec pre vý počet objemu V = 4/3π.r3, povrchu S = 4π.r2. 

5. Vytvorte program na vý počet počtu 10, 5, 2 a 1 Sk mincí na vyplatenie zadanej sumy (použite funkcie 

div a mod ). 

Pozná mka: Pozri program CAS_V_SEKUNDACH. 

6. Vytvorte program na ná kup valú t za určitú  sumu korú n. Zmená reň si ú čtuje manipulačný  poplatok vo 

vý ške 1% z vloženej sumy.  Skutočne platená  čiastka sa zaokrú hľuje na celé koruny. Ná zov a kurz meny 

si zvoľte sami. 

7. Priemerná  denná  teplota sa počíta ako aritmetický  priemer teplôt nameraný ch o 6-tej rá no, o 12-tej 

a o 18-tej hodine,  pričom ú daj nameraný  o 18-tej hodine sa započítava dvakrá t. Vytvorte program, ktorý  

zo zadaný ch ú dajov vypočíta priemernú  dennú  teplotu. 

8. Sú  zadané odpory dvoch rezistorov v ohmoch. Určte hodnotu  vý sledného odporu pri ich sériovom 

a paralelnom zapojení. 

Ná vod: Pri sériovom zapojení R = R1+R2, pri paralelnom R = R1.R2 / (R1+R2). 

9. Vytvorte program na riešenie lineá rnej rovnice ax + b = 0    a, b reá lne, a ≠ 0. 

Ná vod: Koreň x = -b/a. 

10. Vytvorte program na prevod veľkosti uhla v radiá noch na stupne a minú ty. 

Ná vod: π radiá nov = 1800, 10=60’. TP pozná  konštantu PI. 

VETVENIE, ROZHODOVANIE 
Vetvenie použijeme, ak vykonanie príkazu (alebo skupiny príkazov) je podmienené splnením alebo 

nesplnením určitej podmienky. 

S rozhodovaním sa stretá vame v bežnom živote na každom kroku. Ak bude teplo, pôjdem na kú palisko, inak 

budem sedieť za počítačom. Vykonanie akcie „pôjdem na kú palisko“ je podmienené splnením podmienky 

„bude teplo“. Vykonanie akcie „budem sedieť za počítačom“ je podmienené nesplnením podmienky! 

Vetvenie pozná me biná rne (dve možnosti) a n-á rne (n možností, n≥ 2). 

Ú plné  biná rne vetvenie, podmienený  príkaz if 

Má  tvar: 

 

  ak podmienka 
 tak p1 
 inak p2 

 kde p1 a p2 sú  príkazy 

Vykonanie: Ak je podmienka splnená , vykoná  sa 

príkaz p1, ak nie je splnená , vykoná  sa príkaz p2. 

 

V TP ú plnému biná rnemu vetveniu zodpovedá  ú plný  príkaz if, ktorý  má  tvar: 

 if b 
 then p1 
 else p2 kde b je vý raz typu boolean, p1 a p2 sú  príkazy 

Vykonanie ú plného príkazu if: Ak vý raz b nadobudne hodnotu true, vykoná  sa príkaz p1, ak nadobudne 

hodnotu false, vykoná  sa príkaz p2. 

Napríklad: if X < 0 then writeln(’Číslo je zá porné’) else writeln(’Číslo je nezá porné’) 

  if PRIESTUPNY_ROK then DNI := 29 else DNI := 28 

  if (ZNAK >= ’A’) and (ZNAK <= ’Z’) then writeln(’veľké písmeno’) else writeln(’neviem’) 

  if odd(CISLO) then writeln(’nepá rne’) else writeln(’pá rne’) 

  if true then writeln(’vždy toto’) else writeln(’toto nikdy’) 

  if A mod B = 0 then writeln(A,’ je deliteľné ’,B) else writeln(A,’ nie je deliteľné ’,B) 

Neúplné  biná rne vetvenie, neúplný  príkaz if 

Má  tvar: 

 

  ak podmienka 
 tak p 

        kde p je príkaz 

Vykonanie: Ak je podmienka splnená , vykoná  sa 

príkaz p, inak je vetvenie bez ú činku. 

 

 

podmienka 

p1 p2 

+ – 

podmienka 

p 

+ 

– 

V TP neú plnému biná rnemu vetveniu zodpovedá  neú plný  podmienený  príkaz if, ktorý  má  tvar: 

 if b 
 then p   kde b je vý raz typu boolean a p je príkaz  

Vykonanie neú plného príkazu if: Ak vý raz b nadobudne hodnotu true, vykoná  sa príkaz p, ak nadobudne 

hodnotu false, príkaz if je bez ú činku. 

Napríklad: if MENO = ’Peter’ then writeln(’Ď alší Peter’) 

  if CISLO > NAJVACSIE then NAJVACSIE := CISLO 

  if (CISLO > MaxInt) or (CISLO <= -MaxInt) then writeln(’mimo rozsah integer’) 

  if ord(ZNAK) < 32 then writeln(’riadiaci znak’) 

  if X = HLADANE_CISLO then POCET := POCET + 1 

 

Príklad VIF-1: Vytvorte program na ná jdenie najväčšieho z troch celý ch čísel. 

Analý za: Znova odporú čame algoritmus tohto problému zostaviť obyčajný m „sedliackym“ rozumom. Čo 

spravíme prvé, keď sa dozvieme čísla, z ktorý ch treba ná jsť najväčšie? Asi porovná me prvé dve čísla 

a z nich vyberieme väčšie (ak sú  rovnaké, je jedno, ktoré vyberieme) a to ná sledne porovná me s tretím 

číslom. A má me najväčšie – maximum. Algoritmus je na strane 4.  
program MAXIMUM_1; 
uses crt; 
var A, B, C, MAX : integer; 
begin 
clrscr; 
write(’Zadaj tri cele cisla: ’); 
readln(A,B,C); 
if A>B 
  then MAX:=A 
  else MAX:=B; 
if C>MAX 
  then MAX:=C; 
writeln(’Z cisel ’,A,’, ’,B,’, ’,C,’ je najvacsie ’,MAX,’.’); 
readln 
end. 
 

Aby sme demonštrovali variabilnosť riešenia už tak jednoduchého problému, uvá dzame ďalšie programy 

riešiace rovnakú  ú lohu. 

Iná  možnosť (jej vý hodou je jednoduchosť a použiteľnosť pre ľubovoľný  počet čísel, treba len opakovať 

príkaz if ? > MAX then MAX := ?): 
program MAXIMUM_2; 
uses crt; 
var A, B, C, MAX : integer; 
begin 
clrscr; 
write(’Zadaj tri cele cisla: ’); 
readln(A,B,C); 
MAX:=A;  { priradenie pociatocnej hodnoty premennej MAX } 
if B>MAX 
  then MAX:=B; 
if C>MAX 
  then MAX:=C; 
writeln(’Z cisel ’,A,’, ’,B,’, ’,C,’ je najvacsie ’,MAX,’.’); 
readln 
end. 

V nasledujú com programe je logická  chyba, viete ju ná jsť? 
program MAXIMUM_3; 
uses crt; 
var A, B, C : integer; 
begin 
clrscr; 
write(’Zadaj tri cele cisla: ’); 
readln(A,B,C);   
write(’Z cisel ’,A,’, ’,B,’, ’,C,’ je najvacsie ’); 
if (A>B) and (A>C)     { preco sme presunuli vystup (v nom nie je chyba!)? } 
  then writeln(A); 
if (B>A) and (B>C) 
  then writeln(B); 
if (C>A) and (C>B) 
  then writeln(C); 
readln 
end. 
 

Takto to je už dobre? 
program MAXIMUM_4; 
uses crt; 
var A, B, C : integer; 
begin 
clrscr; 
write(’Zadaj tri cele cisla: ’); 
readln(A,B,C); 
write(’Z cisel ’,A,’, ’,B,’, ’,C,’ je najvacsie ’); 
if (A>B) and (A>C) 
  then writeln(A) 
  else if (B>A) and (B>C) 
         then writeln(B) 
         else writeln(C); 
readln 
end. 

Programá torský  štý l 

 Ak sa pozriete na zá pisy príkazov if aj celý ch programov, vidíte v nich určitú  štruktú ru, ktorou sa 

snažíme vyjadriť podriadenosť niektorý ch častí nad iný mi. Rovnocenné príkazy sú  na jednej ú rovni, 

podriadené časti sa snažíme posunú ť napríklad o tri medzery vpravo. Zá roveň taký to zá pis sprehľadňuje celý  

algoritmus alebo program. Veľmi vá m odporú čame používať určité logické schémy pri zá pisoch programov 

a algoritmov. Budú  pre všetký ch prehľadnejšie, zrozumiteľnejšie, pri tvorbe spravíte menej chý b a ľahšie sa 

v nich budú  hľadať chyby. Odporú čame dve schémy: 

1. zá pis  if   podmienka 

     then príkaz1 

     else príkaz2 

dobre graficky zná zorňuje, že ide o ú plné biná rne vetvenie s oboma vetvami na samostatný ch riadkoch; 

riadky s then a else sú  podriadené riadku if, kde príkaz začína, ďalší príkaz by bol už na ú rovni if; 

2. zá pis if    podmienka then príkaz1 

else príkaz2 

zdôrazňuje prítomnosť vetvy else, t.j. ú plného biná rneho vetvenia, vetva then je „schovaná “, keďže sa 

musí vyskytovať v každom príkaze if.  

Pri písaní programov v editore TP si všimnite, že podporuje štruktú rovaný  zá pis. 

 Prakticky prekladaču TP je jedno, ako je daný  program napísaný . Pre neho je rozhodujú ca len 

prítomnosť bodkočiarok medzi jednotlivý mi príkazmi. Celý  program by mohol byť napísaný  v jednom 

riadku (ak by sa tam zmestil). Sú časťou dobrého programá tora je aj dobrý  programá torský  štý l, aj čo sa tý ka 

zá pisu programov, ktorý  sa určite neprezentuje programom v jednom stĺpci alebo riadku! 

 A propos bodkočiarky. Platí: príkazy sa od seba oddeľujú  bodkočiarkami. Preto za begin, pred end 

a iný mi vyhradený mi slovami bodkočiarky nemusia byť, neoddeľovali by príkazy. V niektorý ch prípadoch 

sú  dokonca chybou (uvidíme pri cykloch). My v programoch na miestach, kde nemusia byť, nebudeme dá vať 

bodkočiarky, aj keď väčšinou by tam mohli byť. 

Zložený  príkaz 

 Ak si dobre všimnete príkaz if, za then aj else je dovolený  len jeden (!) príkaz. Ak potrebujeme na 

tý chto miestach použiť viacej príkazov, musíme vytvoriť tzv. zložený  príkaz.  

Zložený  príkaz použijeme na mieste, kde je dovolený  len jeden príkaz a my potrebujeme, aby došlo 

k vykonaniu viacerý ch príkazov. Zložený  príkaz vytvá ra zo skupiny príkazov jeden príkaz. 

Má  tvar:  začiatok  v TP:  begin 
  p1;     p1; 
  p2;     p2; 
  ...     ... 
  pn;     pn; 
  koniec     end  kde p1 až pn sú  príkazy 

Vykonanie: Z príkazov p1, p2 až pn je vytvorený  jeden (zložený ) príkaz. 

Napríklad: if A > B   if  X<=0 
 then begin      then writeln(’Pre X<=0 neviem vypočítať!’) 
 POM := A;     else  begin 
 A := B;   XnaN := exp( N*ln(X) ); 
 B := POM;   writeln( X, ’ na ’ , N , ’ = ’ , XnaN ) 
 end                 end 
 { po skončení A ≤ B } { vý počet mocniny xn } 

Poradie vyhodnocovania operá cií (priorita operá torov) 

 Všimnite si, že v programoch MAXIMUM 3 a 4 sme v podmienkach príkazov if použili zá tvorky. 

Otá zka znie: Kedy musíme použiť zá tvorky v zá pise vý razov a kedy nie? O tom rozhoduje priorita 

jednotlivý ch operá torov. Podobne ako v matematike má  aj v TP ná sobenie a delenie „prednosť“ pred 

sčítaním a odčítaním a pod. Všeobecne platí, že najprv sa vyhodnotia funkcie (abs, sqr, sqrt, odd,...).  

Priorita operá torov je nasledovná : 

1. uná rne mínus (zmena znamienka), not 

2. *, / , div, mod, and 

3. +, - , or, xor 

4. <, <=, =, <>, >=, >, in 

pri rovnosti operá torov sa postupuje zľava doprava; prvé sa vyhodnocujú  vý razy v zá tvorká ch, 

od vnú torný ch k vonkajším. 

Príklady: ( I = N ) or ( ODP = ’K’ ) Bez zá tvoriek by došlo ku chybe, lebo or má  vyššiu prioritu ako = 
     a preto by sa najprv vykonalo N or ODP pričom N je číselný  typ a 
 boolean boolean  ODP je typu char, čo nie sú  typy boolean očaká vané „okolo“ or. 

Vý raz  
a + b

c   musí byť zapísaný  pomocou zá tvoriek (A+B)/C, inak by došlo najprv k deleniu B/C (delenie 

má  vyššiu prioritu ako sčítanie) a až potom k sčítaniu, čo nezodpovedá  pôvodnému vý razu. 

Vý raz  
a

b.c  musí byť zapísaný  A/(B*C), iná č by nezodpovedal pôvodnému vý razu, pretože zá pis A/B*C  

(pravidlo „zľava doprava“) zodpovedá  vý razu  
a
b .c (čo nie je pôvodný  vý raz). Pôvodnému vý razu však 

zodpovedá  aj zá pis A/B/C! 

Vý raz  
a.b
c   môže byť zapísaný  bez zá tvoriek A*B/C (pravidlo „zľava doprava“; najprv sa vykoná  ná sobenie 

a potom delenie). 

 

Znova sme trochu mú drejší v teórii, vrá ťme sa však k príkladom využívajú cim vetvenie a príkaz if. 

Príklad VIF-2: Vytvorte program na usporiadanie troch celý ch čísel (zostupne). 

Analý za: Podľa potreby budeme vymieňať hodnoty premenný ch A, B, C, ký m nebude A<=B<=C. 
program USPORIADAJ; 
uses crt; 
var A, B, C, POM : integer; 
begin 
clrscr; 
write(’Zadaj tri cele cisla: ’); 
readln(A,B,C); 
if A>B 
  then begin POM:=A; A:=B; B:=POM end; 
if B>C 
  then begin POM:=B; B:=C; C:=POM end; 
if A>B 
  then begin POM:=A; A:=B; B:=POM end; 
writeln(A,’ <= ’,B,’ <= ’,C); 
readln 
end. 
 

Venujme sa ešte jednej „teoretickej“ otá zke. V príkaze if nič nebrá ni tomu, aby na mieste príkazu p1 

alebo p2 mohol byť znova, podľa potreby, použitý  príkaz if, ako to vidieť v programe MAXIMUM_4.  

Pri štruktú rach  if b1     if b1 
   then if b2       then  if b2 
              then p1       then p1 
else p2      else p2 
     else p3 

vzniká  otá zka, ku ktorému if patria vetvy else? 

Platí: Vetva else patrí vždy k najbližšiemu predchá dzajú cemu then.   

Vyjadrené štruktú rovane:  
  if b1     if b1 

   then if b2       then  if b2 
              then p1       then p1 

 else  p2       else  p2 
        else p3 

Ak chceme, aby else p2 v prvej štruktú re patrilo prvému then musíme to prekladaču povedať zá pisom: 

  if b1  
   then begin  
           if b2 
              then p1 
           end 
   else  p2   

 

Príklad VIF-3: Vytvorte program na vý počet vý sledného odporu dvoch rezistorov zapojený ch sériovo 

alebo paralelne s presnosťou na stotiny. 

Analý za: Z fyziky treba vedieť, že pri sériovom zapojení rezistorov vý sledný  odpor R=R1+R2, pri paralelnom  

 R1 . R2 
zapojení R =  ––––––––––– .  
 R1 + R2 

Doriešiť treba ešte zadanie spôsobu zapojenia. Rozhodli sme sa pre vloženie prvého písmena (typ char), 

presnejšie, pri vložení znaku p alebo P je zapojenie paralelné, inak sériové. 

 
program VYSLEDNY_ODPOR; 
uses crt; 
var R1, R2, R : real; 
        ZAPOJ : char; 
begin 
clrscr; 
write(’Zadaj hodnoty odporov R1 a R2: ’); readln(R1,R2); 
write(’Sposob zapojenia (P - paralelne, iny znak - seriovo): ’); 
readln(ZAPOJ); 
if (ZAPOJ=’P’) or (ZAPOJ=’p’) 
  then R:=R1*R2/(R1+R2) 
  else R:=R1+R2; 
writeln(’Vysledny odpor R = ’,R:4:2); 
readln 
end. 
 
Využitie funkcie upcase a upravenie vý stupu: 
program VYSLEDNY_ODPOR_2; 
uses crt; 
var R1, R2, R : real; 
        ZAPOJ : char; 
begin 
clrscr; 
write(’Zadaj hodnoty odporov R1 a R2 v ohmoch: ’); readln(R1,R2); 
write(’Sposob zapojenia (P - paralelne, iny znak - seriovo): ’); 
readln(ZAPOJ); 
ZAPOJ:=upcase(ZAPOJ); 
if ZAPOJ=’P’ 
  then R:=R1*R2/(R1+R2) 
  else R:=R1+R2; 
writeln; 
write(’Vysledny odpor pri ’); 
if ZAPOJ=’P’ 
  then write(’paralelnom’) 
  else write(’seriovom’); 
writeln(’ zapojeni R = ’,R:4:2,’ ohmov.’); 
readln 
end. 

Ď alšou možnosťou je ošetrenie zadania spôsobu zapojenia tak, aby program „zobral“ len S alebo P. 
program VYSLEDNY_ODPOR_3; 
uses crt; 
var R1, R2 : real; 
     ZAPOJ : char; 
begin 
clrscr; 
write(’Zadaj hodnoty odporov R1 a R2 v ohmoch: ’); readln(R1,R2); 
write(’Sposob zapojenia (S - seriovo, P - paralelne): ’); readln(ZAPOJ); 
if (ZAPOJ=’S’) or (ZAPOJ=’s’) 
  then writeln(’Vysledny odpor R = ’,R1+R2:3:1,’ ohmov’) 
  else if (ZAPOJ=’P’) or (ZAPOJ=’p’) 
         then writeln(’Vysledny odpor R = ’,R1*R2/(R1+R2):3:1,’ ohmov’) 
         else writeln(’Nebolo stlacene S alebo P!’); 
readln 
end. 
 

Ú loha: Program VYSLEDNY_ODPOR môžete upraviť aj tak, aby reagoval na vloženie celého slova seriovo 

alebo paralelne (zrejme premenná  ZAPOJ bude typu string[9]). 

 

Príklad VIF-4: Vytvorte program, ktorý  po zadaní pH ozná mi: prostredie kyslé, neutrá lne alebo zá sadité. 

Analý za: Trochu chémie nezaškodí. Tuším ná s učili: ak je pH < 7, tak je prostredie kyslé; ak je pH = 7, tak je 

neutrá lne a ak je pH > 7, tak je prostredie zá sadité. Použitie podmienený ch príkazov je zrejmé. 
 
program PH_PROSTREDIA; 
uses crt; 
var PH:real; 
begin 
clrscr; 
write(’Zadaj pH: ’); readln(PH); 
writeln; 
write(’Prostredie je ’); 
if PH<7 then writeln(’kysle.’); 
if PH=7 then writeln(’neutralne.’); 
if PH>7 then writeln(’zasadite.’); 
readln 
end. 
 

Príklad VIF-5: Vytvorte program analogický  programu S-5 (prepočíta čas v sekundá ch na hodiny : minú ty : 

sekundy s vý stupom HH : MM : SS (dvojčísla). 

Analý za: Rozdiel od príkladu S-5 je v tom, že ak je číselný  ú daj (hodiny, minú ty alebo sekundy) 

jednociferný , tak treba „predložiť“ nulu.  

 
program CAS_V_SEKUNDACH; 
uses crt; 
var  H,M,CAS:longint; 
begin 
clrscr; 
write(’Zadaj cas v sekundach: ’); readln(CAS); 
H:=CAS div 3600; CAS:=CAS mod 3600; 
M:=CAS div 60;   CAS:=CAS mod 60; 
if H<10 then write(’0’); 
write(H,’ : ’); 
if M<10 then write(’0’); 
write(M,’ : ’); 
if CAS<10 then write(’0’); 

writeln(CAS); 
readln 
end. 
 
Príklad VIF-6: Vytvorte program na určenie, či zadaný  rok je priestupný . 

Analý za: Platí, že rok je priestupný , ak je deliteľný  4 a nie je deliteľný  100, okrem rokov deliteľný ch 400. 

Napríklad rok 1600 je priestupný  ale rok 1700 nie. Ešte sa „s priestupný m rokom“ stretneme. 

 
program PRIESTUPNY_ROK; 
uses crt; 
var ROK:integer; 
begin 
clrscr; 
write(’Zadaj rok: ’); readln(ROK); 
if (ROK mod 4=0) and ((ROK mod 100<>0) or (ROK mod 400=0)) 
  then writeln(’Rok ’,ROK,’ - priestupny’) 
  else writeln(’Rok ’,ROK,’ - nepriestupny’); 
readln 
end. 
 

Ešte jedno riešenie (toto riešenie ešte použijeme neskôr): 
program PRIESTUPNY_ROK_2; 
uses crt; 
var ROK:integer; 
    PRIESTUPNY:boolean; 
begin 
clrscr; 
write(’Zadaj rok: ’); readln(ROK); 
PRIESTUPNY:=(ROK mod 100<>0) and (ROK mod 4=0) or (ROK mod 400=0); 
if PRIESTUPNY 
  then writeln(’Rok ’,ROK,’ - priestupny’) 
  else writeln(’Rok ’,ROK,’ - nepriestupny’); 
readln 
end. 
Upravte program tak, aby oznamoval aktuá lne: Rok ... bol / je / bude ... Použite príkaz GetDate z unitu Dos 

(pozri pomocou Ctrl+F1), ktorého vý sledkom je aktuá lny rok, mesiac, deň a poradové číslo dňa v tý ždni, 

pričom poradové číslo nedele je 0, pondelka 1 atď. Použité premenné sú  typu word (pozri aj príklad VCS-1: 

program POCET_DNI). 

 

Príklad VIF-7: Vytvorte program na vý počet doby splá cania - počtu rokov a mesiacov, bezú ročnej pôžičky 

po zadaní požičanej sumy a vý šky mesačnej splá tky. 

Analý za: Zovšeobecnite situá ciu, keď má te požičaný ch 12 000 Sk a mesačne splá cate po 100 Sk. Ako dlho 

budete platiť? Najprv nech ná s nezaujímajú  „zvyšky“ (rozumej mesiace), len celé roky, ktoré treba splá cať. 

Tie určí delenie bezo zvyšku, t.j. funkcia div. Ak ste našli odpoveď, môžeme ísť ďalej. Ak by ste mali 

požičaný ch 13 000 Sk, koľko mesiacov by ste museli ešte po desiatich rokoch platiť? Zovšeobecnite! 

Myslite aj na možnosť, že splá cať nebude treba ani jeden celý  rok, potom položka počet rokov splá cania by 

sa vôbec nemala zobraziť, podobne s mesiacmi, ak sa všetko splatí za celý  ná sobok roka. 

 
program SPLATKY; 
uses crt; 
var SUMA, MESACNE, ROKOV, MESIACOV, ZOSTAVASK : longint; 
begin 
clrscr; 

write(’Pozicana suma: ’); readln(SUMA); 
write(’Vyska mesacnej splatky: ’); readln(MESACNE); 
writeln; 
writeln(’Treba splacat: ’); 
ROKOV:=SUMA div (12*MESACNE); 
if ROKOV>0 
  then writeln(’- rokov: ’,ROKOV); 
ZOSTAVASK:=SUMA-ROKOV*12*MESACNE; 
MESIACOV:=ZOSTAVASK div MESACNE; 
if MESIACOV>0 
  then writeln(’- mesiacov: ’,MESIACOV); 
ZOSTAVASK:=SUMA-(ROKOV*12*MESACNE+MESIACOV*MESACNE); 
if ZOSTAVASK>0 
  then writeln(’- dalsi mesiac este splatit: ’,ZOSTAVASK,’ korun.’); 
readln 
end. 
 
Doplňujú ca ú loha: Vý stup s koncovkami v sprá vnom tvare: 

Treba splá cať: 

1 rok  alebo 2, 3, 4 roky alebo 5, 6, ... rokov 

1 mesiac alebo 2, 3, 4 mesiace alebo 5, 6, ... mesiacov 

... splatiť 1 korunu alebo 2, 3, 4 koruny alebo 5,6, ...korú n. 

 

Príklad VIF-8: Na pobavenie a doplnenie vôbec prvého programu v skriptá ch. Program, ktorý  po zadaní 

mena pozdraví, opý ta sa "Ako sa má š?" a primerane zareaguje. 
 
program POZDRAV; 
uses Crt; 
var  MENO, ODPOVED:string; 
begin 
clrscr; 
write(’Ako sa volas ? ’); 
readln(MENO); 
{ nasledujuce zmeni prve pismeno v mene na velke - pozri Help pomocou Ctrl+F1 } 
MENO:=upcase(MENO[1])+copy(MENO,2,length(MENO)); 
writeln; 
writeln(’Ahoj, ’,MENO,’!’); 
writeln; 
write(’Ako sa mas? ’); 
readln(ODPOVED); 
writeln; 
if (ODPOVED=’dobre’) or (ODPOVED=’vyborne’) or (ODPOVED=’OK’) 
  then writeln(’Som rad, ze sa mas ’,ODPOVED,’!’) 
  else if (ODPOVED=’zle’) or (ODPOVED=’nanic’) or (ODPOVED=’na...’) 
         then writeln(’Mrzi ma, ze sa mas ’,ODPOVED,’!’) 
         else writeln(’"’,ODPOVED,’" nepoznam!’); 
writeln; 
gotoxy(25,12); 
writeln(’A h o j , musim koncit!’); 
delay(3500);     { delay(CISLO) zastavi beh programu na CISLO milisekund } 
clrscr 
end. 
 

Príklad VIF-9: Už sme mali fyziku, chémiu, bankovníctvo, slušné sprá vanie, tak znova trocha matematiky. 

Vytvorte program, ktorý  zistí, či zadané číslo je deliteľné druhý m zadaný m číslom. 

Analý za: Napríklad číslo 21 je deliteľné 7, lebo zvyšok po delení čísla 21 siedmimi je nula! Ale zvyšok po 

celočíselnom delení dá va funkcia mod, preto ak... 

program DELI; 
uses crt; 
var DELENEC, DELITEL : integer; 
begin 
clrscr; 
write(’Zadaj delenca a delitela: ’); 
readln(DELENEC, DELITEL); 
if DELENEC mod DELITEL=0 
  then writeln(’Cislo ’,DELENEC,’ je delitelne ’,DELITEL) 
  else writeln(’Cislo ’,DELENEC,’ nie je delitelne ’,DELITEL); 
readln 
end. 

N-á rne vetvenie, podmienený  príkaz case 

Má  tvar:       v slovnom zá pise nemá  ekvivalent,  

        realizuje sa zá pisom: 

        ak podm1           tak p1 
        inak ak podm2   tak p2 
        inak ak podm3   tak p3 
        inak ... 
        inak ak podmn   tak pn  

 

h1, h2 až hn sú  hodnoty, ktoré môže   kde p1 až pn sú  príkazy 
 nadobudnú ť vý raz v 

V TP sa n-á rne vetvenie realizuje podmienený m príkazom case. 

Má  tvar:  case v of  kde v je tzv. vý berový  vý raz ordiná lneho typu, 

     h1 : p1;  h1, h2 až hn sú  hodnoty rovnakého typu ako vý berový  vý raz 

     h2 : p2;  a  p, p1, p2 až pn sú  príkazy 

         ... 

     hn : pn 

  [  else p  ]  do hranatý ch zá tvoriek sa umiestňuje nepovinná  časť  

  end   príkaz case končí vyhradený m slovom end 

Vykonanie: Vyhodnotí sa vý berový  vý raz a vykoná  príkaz, predznačený  hodnotou, ktorú  nadobudol 

vý berový  vý raz. Ak vý raz v nenadobudne ani jednu z hodnôt h1 až hn a príkaz case obsahuje časť else, 

vykoná  sa príkaz p; ak príkaz case neobsahuje časť else, príkaz case je bez ú činku. 

Príklad:        case MESIAC of 

  1,3,5,7,8,10,12 : DNI := 31; 

            2 :  if PRIESTUPNY then DNI := 29 else DNI := 28; 

                           4,6,9,11 : DNI := 30 

         end { do premennej DNI priradí počet dní v mesiaci } 

Dovolené sú  aj zá pisy s intervalmi (medzi minimá lnou a maximá lnou hodnotou sú  dve bodky): 

 case ZNAK of 
    ’A’..’V’ , ’a’..’v’ : write( succ(ZNAK) ); 
                  ’Z’ : write( ’A’ ) ; 
            ’z’ : write( ’a’ ) 
     else write(ZNAK) 
 end { príkaz zakóduje písmeno na nasledujú ce  (Z, z na A, a) a zobrazí ho } 
  { iné znaky nemení, zobrazí pôvodné } 

vý raz v 

p2 p1 pn 

h1 hn h2 



 Zá klady programovania  – Turbo Pascal 30  

Príklad VCS-1: Program na určenie aktuá lneho dňa v tý ždni pomocou príkazu GetDate (unit Dos), na 

určenie počtu dní do konca mesiaca a do konca roka. 

Analý za: Vý sledkom príkazu GetDate (pozri pomocou Ctrl+F1) je aktuá lny rok, mesiac, deň a poradové 

číslo dňa v tý ždni, pričom poradové číslo nedele je 0, pondelka 1 atď. Využili sme aj tvrdenie, že rok je 

priestupný  (februá r má  29 dní), ak je deliteľný  4 a pritom nie je deliteľný  100 alebo je deliteľný  400. Rok 

2000 by mal byť priestupný , lebo je deliteľný  4, nemal by byť priestupný , lebo je deliteľný  100, ale je 

priestupný , pretože je deliteľný  400. Zaujímavé. Ak rok nie je deliteľný  4, určite je nepriestupný . 

Dali sme si zá ležať aj na koncovke slova deň, dni, dní (slovenská  diakritika sa do TP dá  dostať). 

Riešenie vý počtu dní do konca roka nie je veľmi elegantné, pokú ste sa o krajšie riešenie . 
 
program POCET_DNI; 
uses crt,dos; 
var 
  d, m, r, por : word; 
  pdm, dkm: byte;    { pdm - pocet dni v mesiaci, dkm - do konca mesiaca } 
  dkr:integer;       { dkr - do konca roka } 
  priestupny:boolean;{ vysledkom je true, ak je rok priestupny, inak false } 
begin 
clrscr; 
GetDate(r,m,d,por); 
write(’Dnes je ’); 
case por of 
  0: write(’nedela’); 
  1: write(’pondelok’); 
  2: write(’utorok’); 
  3: write(’streda’); 
  4: write(’stvrtok’); 
  5: write(’piatok’); 
  6: write(’sobota’); 
end; 
writeln(’ ’, d, ’.’, m, ’.’, r, ’.’); 
writeln; 
priestupny:=(r mod 4=0) and ((r mod 100<>0) or (r mod 400=0)); 
case m of 
  1,3,5,7,8,10,12: pdm:=31; 
                2: if priestupny 
                     then pdm:=29 
                     else pdm:=28; 
         4,6,9,11: pdm:=30; 
end; 
dkm:=pdm-d; 
write(’Do konca mesiaca ’); 
case dkm of 
      1: writeln(’zostáva 1 deň.’); 
  2,3,4: writeln(’zostávajú ’, dkm,’ dni.’); 
  else writeln(’zostáva ’, dkm,’ dní.’) 
  end; 
writeln; 
case m of 
  1: if priestupny 
       then dkr:=366-d 
       else dkr:=365-d; 
  2: if priestupny 
       then dkr:=366-31-d 
       else dkr:=365-31-d; 
  3: dkr:=dkm+30+31+30+31+31+30+31+30+31; 
  4: dkr:=dkm+31+30+31+31+30+31+30+31; 
  5: dkr:=dkm+30+31+31+30+31+30+31; 
  6: dkr:=dkm+31+31+30+31+30+31; 
  7: dkr:=dkm+31+30+31+30+31; 

  8: dkr:=dkm+30+31+30+31; 
  9: dkr:=dkm+31+30+31; 
 10: dkr:=dkm+30+31; 
 11: dkr:=dkm+31; 
 12: dkr:=dkm; 
end; 
write(’Do konca roka ’); 
case dkr of 
      1: writeln(’zostáva 1 deň.’); 
  2,3,4: writeln(’zostávajú ’, dkr,’ dni.’); 
  else writeln(’zostáva ’, dkr,’ dní.’) 
end; 
readln 
end. 
 
Nerieš ené  úlohy na vetvenie: 

1. Vytvorte program na riešenie rovnice ax + b = 0,   a, b reá lne. 

2. Vytvorte program na riešenie kvadratickej rovnice  ax2 + bx + c = 0,  a, b, c reá lne, a ≠ 0. 

Ná vod: D = b
2 - 4.a.c ; pre D ≥ 0 platí x1,2 = ( -b ± √ D ) / (2.a).  

3. Vytvorte program na riešenie rovnice ax2 + bx + c = 0,  a, b, c reá lne. 

Ná vod: Pozri ú lohy č. 1 a 2. 

4. Vytvorte program na vý počet nezná mej strany pravouhlého trojuholníka po zadaní usporiadanej trojice 

[a, b, c] ( a, b odvesny, c prepona), kde nezná ma strana má  pri zadaní veľkosť 0. 

Ná vod: Použi Pythagorovu vetu, napríklad pre trojicu [3,0,5] sa počíta b =  (5
2 - 3

2
) = 4. 

5. Vytvorte program na  vý počet vý šky poštovného pri podaní peňažnej pouká žky (tarifná  tabuľka je na 

druhej strane pouká žky). 

6. Vytvorte program na vý počet povrchu a obvodu zvoleného rovinného ú tvaru. Ponuka napr.: štvorec, 

obdĺžnik, trojuholník, lichobežník, kruh a pod. (selekcia príkazom case!). 

7. Vytvorte program, ktorý  zistí, či zadané tri čísla môžu byť veľkosti strá n trojuholníka. 

Ná vod: Použi trojuholníkovú  nerovnosť: sú čet dvoch strá n v trojuholníku je vždy väčší ako tretia strana. 

8. Uká žte, že tri podmienky umožňujú  rozlíšiť 8 situá cií. 

9. Osoby majú  hmotnosť x , y a z. Nosnosť vý ťahu je w. Vytvorte  program na zistenie, koľko já zd je treba, 

aby sa všetci dostali vý ťahom z prízemia na 13. poschodie. 

10. Je daný  počet dní v mesiaci a informá cia, na ktorý  deň v tý ždni pripadá  prvý  deň v mesiaci (pondelok - 

1, utorok - 2, atď.).  Zistite, koľko je v danom mesiaci piatkov. 

11. Je daný  počet dní v mesiaci a informá cia, na ktorý  deň v tý ždni pripadá  prvý  deň v mesiaci (pondelok - 

1, utorok - 2, atď.).  Zistite, koľko je v danom mesiaci pracovný ch dní. 

12. Vytvorte program, ktorý  po zadaní postupnosti celý ch čísel ozná mi, či je postupnosť rastú ca, 

neklesajú ca, klesajú ca,  nerastú ca, konštantná  alebo nemá  žiadnu z tý chto vlastností. 

CYKLUS 
Použijeme, ak nejaký  príkaz alebo skupina príkazov sa má  opakovane vykoná vať, pokiaľ je splnená  alebo 

nesplnená  daná  podmienka. 

Cykly rozdeľujeme: 

1. podľa umiestnenia podmienky na: 

a) cyklus s podmienkou na začiatku 

b) cyklus s podmienkou na konci 

c) ú plný  cyklus (s podmienkou v strede) 

2. podľa toho, či je zná my alebo nezná my počet opakovaní v cykle na: 

a) cyklus s explicitne daný m počtom opakovaní 

b) cyklus s implicitne daný m počtom opakovaní. 

Cyklus s explicitne daným počtom opakovaní, príkaz for 

Cyklus s explicitne („zvonka“) daný m počtom opakovaní použijeme pri zná mom počte opakovaní príkazov 

v cykle. 

V algoritmizá cii nemá  tento cyklus jednoznačnú  formu zobrazenia, my sme sa rozhodli pre tvar: 

       v slovnom zá pise: 

        

       pre rpc := v1 až po (naspäť po) v2 opakuj 
            p 

        

       kde rpc je tzv. riadiaca premenná  cyklu (riadi počet 
       prechodov cyklom), v1 a v2 sú  vý razy a p je príkaz 

 

V TP cyklu so zná mym počtom opakovaní zodpovedá  príkaz for. 

Má  tvar: for rpc := v1 to (downto) v2 do 
      p 

kde rpc je tzv. riadiaca premenná  cyklu ordiná lneho typu, v1 a v2 sú  vý razy rovnakého typu 
ako rpc a p je príkaz; príkaz for má  dva varianty, buď s „to“ alebo s „downto“ 
 

Vykonanie príkazu for (v zá tvorká ch pre downto): 

1.  vyhodnotia sa vý razy v1 a v2; hodnota vý razu v1 sa priradí ako začiatočná  hodnota rpc, hodnota 

vý razu v2 ako koncová  hodnota rpc, 

2.  pokiaľ je hodnota rpc menšia (väčšia) alebo rovná  koncovej hodnote, opakovane sa vykoná va príkaz p 

a rpc nadobú da hodnoty succ(rpc) (pred(rpc)). 

Pozná mky:  

• Riadiaca premenná  cyklu je ordiná lneho typu, t.j. môže byť len typu integer, boolean alebo char. Z toho 

vyplý va, že nemôže nadobú dať hodnoty meniace sa o ľubovoľný  krok, napr. 0,5; 0,1 a pod. Programá tor 

s dobrý m programá torský m štý lom zá sadne nemení hodnotu rpc ani o krok napr. 2, 10 a pod., len na 

nasledovníka resp. predchodcu. Ak potrebujeme zrealizovať cyklus s „nevhodný m“ krokom pre príkaz 

for, použijeme cyklus s príkazom while alebo repeat. 

  rpc ≤ v2 

 p 

– 

+ 

  rpc := v1 

rpc := rpc+1 

• Začiatočná  a koncová  hodnota riadiacej premennej cyklu sa získa vyhodnotením vý razov v1 a v2 na 

začiatku vykoná vania príkazu for a počas cyklu ich nemožno meniť. 

• Z historický ch dôvodov (z čias programovacieho jazyka fortran) sa zaužívalo najčastejšie označiť 

riadiacu premennú  cyklu písmenom I, nie je to však pravidlo. 

Príklady: for I := 1 to 100 do write( I : 4 ) 

príkaz zobrazí čísla 1   2   3   4 ... 99 100 v piatich riadkoch po 20 čísel (80 znakov v riadku 

delené 4 znaky na číslo = 20 čísel v riadku) 

for I := 100 downto 1 do write ( I : 4 ) 

príkaz zobrazí čísla 100  99  98  ... 2   1, t.j. opačne ako v predchá dzajú com príkaze for 

  SUCET := 0; 

  for CISLO := 1 to POCET do  SUCET := SUCET + CISLO 

  príkaz sčíta čísla 0 + 1 + 2 + 3 + ... + POCET, 0 je počiatočná  hodnota premennej SUCET 

  MOCNINA := 1; 

  for I := 1 to N do MOCNINA := MOCNINA * X 

  príkaz vykoná  1.X.X. ... .X = XN, 1 je počiatočná  hodnota premennej MOCNINA 

  for Z := ’A’ to ’Z’ do writeln( Z : 5, ord( Z ) : 3 ) 

  príkaz zobrazí v riadkoch znaky A až Z a ich poradové čísla z ASCII tabuľky 

  for J := -10 to 10 do p 

  príkaz for vykoná  príkaz p 21-krá t (J bude postupne -10, -9, ... , 0, ... , 10) 

  for K := 0 to (downto) 0 do p 

  príkaz for vykoná  príkaz p jedenkrá t 

  for NIC := 100 to –100 do p 

príkaz for nevykoná  príkaz p ani raz (začiatočná  hodnota rpc nie je ani raz menšia alebo 

rovná  koncovej hodnote) 

  for NIC := 0 downto 10 do p 

príkaz for nevykoná  príkaz p ani raz (začiatočná  hodnota rpc nie je ani raz väčšia alebo 

rovná  koncovej hodnote) 

  for NIC := ’z’ to ’A’ do p 

príkaz for nevykoná  príkaz p ani raz, pretože ’z’ > ’A’ (znak z má  v ASCII tabuľke väčšie 

poradové číslo) 

for I := - trunc( sqrt(N) ) to trunc( sqrt(N) ) do p 

Vý razy v1 a v2 nemusia mať pri zá pise len konkrétne hodnoty, môžu to byť aj vý razy. 

 

Príklad CFR-1: Vytvorte program na zobrazenie hodnôt riadiacej premennej cyklu zo zadaného intervalu 

pre typ integer aj char. Uvedomte si, že ak variant „to“ zobrazí viac ako jednu hodnotu, variant „downto“ 

nemôže zobraziť ani jednu hodnotu a naopak. 
 
program HODNOTY_RPC; 
uses crt; 
var  ZAC, KON, I : integer; 
     ZACZ, KONZ, Z : char; 
begin 

clrscr; 
write(’Zadaj zaciatocnu a koncovu hodnotu typu integer: ’); 
readln(ZAC,KON); 
writeln(’Hodnoty rpc pre "to":’); 
for I:=ZAC to KON do 
  write(I:8); 
writeln; 
writeln(’Hodnoty rpc pre "downto":’); 
for I:=ZAC downto KON do 
  write(I:8); 
writeln; writeln; 
write(’Zadaj zaciatocnu hodnotu typu char: ’); 
readln(ZACZ); 
write(’Zadaj  koncovu  hodnotu  typu char: ’); 
readln(KONZ); 
writeln(’Hodnoty rpc pre "to":’); 
for Z:=ZACZ to KONZ do 
  write(Z:4); 
writeln; 
writeln(’Hodnoty rpc pre "downto":’); 
for Z:=ZACZ downto KONZ do 
  write(Z:4); 
writeln; 
readln 
end. 
Neuviedli sme analý zu, radšej upozorníme na „problém“ s načítaním viacerý ch hodnôt typu char v jednom 

príkaze readln. Keby sme, tak ako pri type integer, umiestnili premenné ZACZ a KONZ do jedného príkazu 

readln a hodnoty vložili oddelené medzerou, premennej KONZ by vždy priradilo medzeru (druhý  znak je 

predsa medzera)! Preto nečíselné hodnoty načítavame z klá vesnice väčšinou v samostatný ch príkazoch 

readln. Ak chceme načítať obe hodnoty v jednom príkaze readln, musíme ich vložiť za sebou bez oddelenia. 

 

Príklad CFR-2: Vytvorte program na vý počet sú čtu všetký ch prirodzený ch čísel od 1 po zadané N pomocou 

cyklu.  

Analý za:  

1. Akú  algoritmickú  konštrukciu treba použiť? 

Napríklad pre N = 100 treba sčítať 1 + 2 + 3 + ... + 99 + 100 = 5050. Vidíme, že sa opakuje sčítavanie.  

Keďže opakovane treba vykoná vať nejaký  príkaz, použijeme cyklus. Pozná me aj počet opakovaní, 

N-krá t (aj 1 treba pričítať k pôvodnej hodnote v pamäti), preto to je cyklus so zná mym počtom 

opakovaní a ten sa realizuje príkazom for:   for CISLO := 1 to N do 

2. Aký  príkaz sa má  v cykle opakovať?  

Vyzerá  to tak, že vždy k čiastočnému sú čtu, t.j. k sú čtu, ký m nie je všetko sčítané, treba pripočítať 

vhodné číslo. Vhodné číslo je pri prvom prechode cyklom 1, pri druhom 2, pri treťom 3 atď., až pri 

n-tom N. Ak sme pozorní, uvedomíme si, že prá ve tieto hodnoty nadobú da riadiaca premenná  cyklu for. 

Preto:          nový  čiastočný  sú čet = predchá dzajú ci čiastočný  sú čet + CISLO 

Ak dobre pozná me vykonanie príkazu priradenia, uvedomíme si, že ten presne pracuje s „nový m 

a predchá dzajú cim“, ak na jeho ľavej aj pravej strane je tá  istá  premenná  (to isté pamäťové miesto)! 

Preto v cykle treba opakovať príkaz:  SUCET := SUCET + CISLO 

V príkaze takéhoto typu musíte neustá le vidieť: nový  SUCET = starý  SUCET + ...  

alebo všeobecne: nová  HODNOTA = stará  HODNOTA +,*,-,/... 

3. Teraz je už všetko v poriadku? 

Ešte nie. Ak začneme simulovať vý počet v cykle, zistíme, že nepozná me hodnotu premennej SUCET na 

pravej strane pri prvom prechode cyklom. Nepozná me počiatočný  sú čet resp. nenastavili sme ešte 

počiatočnú  hodnotu premennej SUCET. Sprá vne vykonanie príkazu priradenia, v ktorom je rovnaká  

premenná  na ľavej aj pravej strane, si vyžaduje pred príkazom cyklu nastavenie počiatočnej hodnoty 

takejto premennej. Pre sčítanie to bý va (ale vo všeobecnosti nemusí byť) 0, pre ná sobenie 1. Preto pred 

príkaz for vložíme ešte príkaz:  SUCET := 0; 

Už počujeme hlasy, že vá š program počíta sprá vne aj bez tohto priradenia (spomínali sme, že TP 7.0 vloží 

do všetký ch číselný ch premenný ch počiatočnú  hodnotu 0). Čo však, ak program nespustíte v TP 7.0 alebo 

ho vložíte do cyklu, aby vá m viackrá t umožnil zopakovať vý počet? Potom určite nebude počítať sprá vne 

(počiatočnou hodnotou bude ná hodné číslo alebo hodnota predchá dzajú ceho sú čtu). 

Teraz je už všetko v poriadku a tu je program: 
program SucetPrvychN; 
uses crt; 
var  N, CISLO, SUCET : integer; { pripadne typ word alebo longint } 
begin 
clrscr; 
write(’Scitat prirodzene cisla po ’); 
readln(N); 
SUCET:=0; 
for CISLO:=1 to N do 
  SUCET:=SUCET+CISLO; 
writeln(’Sucet prvych ’,N,’ cisel je ’,SUCET); 
readln 
end. 

Pozná mka: Na vý počet sú čtu prvý ch n prirodzený ch čísel možno použiť aj vzorec SUCET = 
N
2  .(1+N), preto 

sme v zadaní ú lohy uviedli, že chceme vý počet pomocou cyklu. Zá roveň si uvedomte, že vý razne sa 

efektivitou vý počtu líšiacich algoritmov môže byť (a väčšinou aj bý va) viacej! 

Program upravte tak, aby: 

a) počítal sú čet prvý ch N prirodzený ch čísel od N po 1, 

b) počítal sú čet všetký ch prirodzený ch čísel od zadaného A po zadané B vrá tane, 

c) počítal sú čet všetký ch celý ch čísel od zadaného A po zadané B vrá tane. 

 

Príklad CFR-3: Vytvorte program na vý počet mocniny x na n-tú , x reá lne, n prirodzene číslo. 

Analý za: Začneme „sedliackym“ rozumom. Ak má me vypočítať napríklad 53, ná sobíme 5.5.5 = 125. Ď alej 

by sme mohli vedieť, že 00 nie je definované (asi preto, že 0 na „čokoľvek“ je 0 a „čokoľvek“ na 0-tú  je 1, 

preto 00 by malo byť 0 a zá roveň 1 a to nejde). 

1. Opakovane sa vykoná va ná sobenie, preto treba použiť cyklus; ná sobenie treba vykonať n-krá t, preto 

cyklus so zná mym počtom opakovaní, teda príkaz for. 

2. Vždy nový  sú čin sa rovná  predchá dzajú ci sú čin krá t x, preto sa v cykle bude opakovať príkaz SUCIN := 

SUCIN * X. 

3. Nesmieme zabudnú ť na počiatočnú  hodnotu premennej SUCIN, ktorá  môže byť 1. Keby sme zvolili 

nulu? 

4. Je tu však ešte jedno obmedzenie, ak zadá me N aj X nula, tak má  program ozná miť: „0 na 0-tú  

nedefinovane!“, inak má  „normá lne“ počítať – teda vetvenie. 

 
program MOCNINA; 
uses crt; 
var X, MOCNINA : real; 
     N, I, PDM : integer; 
begin 
clrscr; 
write(’Zadaj zaklad mocniny a exponent: ’); 
readln(X,N); 
write(’Zadaj pocet desatinnych miest vystupu: ’); 
readln(PDM); 
if (X=0) and (N=0) 
  then writeln(’0 ma 0-tu nedefinovane!’) 
  else begin 
       MOCNINA:=1; 
       for I:=1 to N do 
         MOCNINA:=MOCNINA*X; 
       writeln(X:PDM+2:PDM,’ na ’,N,’ = ’,MOCNINA:PDM+2:PDM) 
       end; 
readln 
end. 
 

Príklad CFR-4: Vytvorte program na vý počet aritmetického priemeru n reá lnych čísel. 

Analý za: Sná ď všetci vieme, že aritmetický  priemer sa vypočíta ako sú čet daný ch n čísel delený  počtom 

všetký ch čísel, t.j. n. Program si má  vypý tať n reá lnych čísel a sčítať ich, preto cyklus s „príkazmi“: zadaj 

číslo a k čiastočnému sú čtu pripočítaj nové číslo. Toto sa má  zopakovať pre n čísel, teda n-krá t. Takže cyklus 

so zná mym počtom opakovaní – príkaz for. 

V programe sme namiesto premennej n použili premennú  POCET a ošetrili sme aj prípad, keď bude zadaný  

počet čísel 0 - program by bez ošetrenia skončil chybový m hlá sením: Delenie nulou! 

 
program PRIEMER; 
uses crt; 
var  I, POCET : integer; 
     X, SUCET : real; 
begin 
clrscr; 
write(’Zadaj pocet cisel: ’); 
readln(POCET); 
SUCET:=0; 
for I:=1 to POCET do 
  begin 
  write(I:2,’. cislo: ’); 
  readln(X); 
  SUCET:=SUCET+X 
  end; 
if POCET>0 
  then writeln(’Priemer zaokruhleny na stotiny: ’,SUCET/POCET:4:2); 
readln 
end. 
 

Ď alšie ú lohy aj na príkaz for sú  uvedené v kapitole Ď alšie zaujímavé ú lohy. 

Cyklus s podmienkou na začiatku, príkaz while 

O typickom použití cyklu s podmienkou na začiatku si povieme v časti „Kedy ktorý  cyklus“. 

Má  tvar:      v slovnom zá pise: 

        

       pokiaľ podmienka opakuj 
            p 

 

       kde p je príkaz 

 

 

 

V TP má  tvar:  while b do 
     p    kde b je vý raz typu boolean a p príkaz 

Vykonanie: Pokiaľ vý raz b nadobú da hodnotu true, opakovane sa vykoná va príkaz p, ak vý raz b nadobudne 

hodnotu false, cyklus sa ukončí. 

Príklady:  I := 1;   
while I <= 10 do   
    I := I + 1;   príkaz v cykle sa vykoná  10-krá t 

SUCET := 0; 
while N > 0 do 
    begin 
    SUCET := SUCET + N; 
    N := N - 1 

     end    sčíta všetky prirodzené čísla od N po 1 

 DELITEL := 2; 
 while CISLO mod DELITEL <> 0  do 
      inc ( DELITEL )  ná jde deliteľa zadaného čísla ( CISLO > 1 ) 

 

Príklad CWH-1: Príkaz for je viac-menej špeciá lnym prípadom príkazu while. Demonštrujeme to 

v nasledujú com programe. 

 
program AKO_FOR; 
uses crt; 
var  N, I : integer; 
begin 
clrscr; 
write(’Zadaj pocet opakovani: ’); 
readln(N); 
writeln(’Ako prikaz for s "to" od 1 po ’,N); 
I:=1; 
while I<=N do 
  begin 
  writeln(I:6,’. opakovanie’); 
  I:=I+1 
  end; 
readln; 
 

podmienka 

  p 

– 

+ 

writeln(’Ako prikaz for s "downto" od ’,N,’ po 1’); 
while N>0 do 
  begin 
  writeln(N:6,’. opakovanie’); 
  N:=N-1 
  end; 
readln 
end. 
 

Príklad CWH-2: Vytvorte program na vý počet ciferného sú čtu zadaného prirodzeného čísla. 

Analý za: Ciferný  sú čet napríklad čísla 123 je 6, lebo 1 + 2 + 3 = 6. K prvej cifre (3) sa dostaneme 

predelením čísla 123 desiatimi: 123 : 10 = 12, zvyšok 3.  Pokú sme sa vystačiť s celý mi číslami. Ak spravíme 

123 mod 10, dostaneme rovno číslo 3. Ak spravíme 123 div 10 dostaneme 12, čo sa ná m tiež hodí, lebo ďalej 

potrebujeme to isté (mod a div) zopakovať s 12-kou (12 mod 10 = 2 a 12 div 10 = 1) a nakoniec s 1-kou 

(1 mod 10 = 1 a 1 div 10 = 0). Tý m sme prešli všetky cifry čísla. Opakovane spracú vam vý sledky po 

funkciá ch mod  a div s deliteľom 10, preto treba použiť cyklus. Zadané číslo môže byť jedno ale aj troj (to 

naše) prípadne desaťciferné a na začiatku cyklu nevieme, kedy spracujeme poslednú  cifru. Preto cyklus 

s nezná mym počtom opakovaní príkazov v cykle – príkaz while. Podmienkou ukončenia cyklu je, aby sme 

nemali už čo „spracovať“, t.j. aby aktuá lnym číslom bola nula (CISLO div 10 = 0). 

 
program CIFERNY_SUCET; 
uses crt; 
var  CISLO    : longint; 
     CIFSUCET : byte; 
begin 
clrscr; 
write(’Zadaj prirodzene cislo: ’); 
readln(CISLO); 
write(’Sucet cifier cisla ’,CISLO,’ je ’); 
CIFSUCET:=0; 
while CISLO>0 do 
  begin 
  CIFSUCET:=CIFSUCET+CISLO mod 10; 
  CISLO:=CISLO div 10 
  end; 
writeln(CIFSUCET); 
readln 
end. 
 
Prečo sme vý stup museli posunú ť hore? Ak sa vá m funkcie div a mod nepá čia, tu je riešenie bez nich: 
program CIFERNY_SUCET_2; 
uses crt; 
var  CISLO    : longint; 
     CIFSUCET : byte; 
begin 
clrscr; 
write(’Zadaj prirodzene cislo: ’);  
readln(CISLO); 
write(’Sucet cifier cisla ’,CISLO,’ je ’); 
CIFSUCET:=0; 
while CISLO>0 do 
  begin 
  CIFSUCET:=CIFSUCET+trunc(frac(CISLO/10)*10); 
  CISLO:=trunc(CISLO/10) 
  end; 
writeln(CIFSUCET); 
readln 
end. 

Príklad CWH-3: Vytvorte program na zobrazenie prevodnej tabuľky medzi slov. korunami a inou menou. 

Kurz je zadaný  v konštante. Tabuľka bude zobrazovať hodnoty od 0,- Sk po zadanú  sumu (premenná  

KoncovaSuma) so zadaný m prírastkom (premenná  Krok). Napríklad pre koncovú  sumu 100 s prírastkom 10 

má  vypočítať a zobraziť hodnoty pre 0, 10, 20, 30, 40, ... , 100 Sk. Vý počet hodnoty v cudzej mene je 

jednoduchý : suma v Sk krá t kurz. Keďže sa opakovane počíta hodnota v cudzej mene, treba použiť cyklus. 

Riadiaca premenná  cyklu sa však môže zväčšovať o teoreticky ľubovoľné kladné reá lne číslo, preto 

použijeme príkaz while. 

Program sme doplnili o jednoduchú  hlavičku tabuľky a hodnoty sú  zobrazované s presnosťou na desatiny, na 

miestach stotín sa zobrazujú  nuly. Doplňte tabuľku o zvislé „čiary“. 
program TABULKA; 
uses  crt; 
const KURZ=1.24; 
var   KoncovaSuma, Krok, RPC : real; 
begin 
clrscr; 
write(’Zadaj koncovu sumu a prirastok: ’); 
readln(KoncovaSuma, Krok); 
clrscr; 
writeln(’----------------------’:48); 
writeln(’      SK         MENA ’:48); 
writeln(’----------------------’:48); 
RPC:=0; 
while RPC<=KoncovaSuma do 
  begin 
  writeln(RPC:33:1,’0’,KURZ*RPC:12:1,’0’); 
  RPC:=RPC+KROK 
  end; 
writeln(’----------------------’:48); 
readln 
end. 
 

Cyklus s podmienkou na konci, príkaz repeat 

Má  tvar:      v slovnom zá pise: 

       opakuj 

          p1; 
          p2; 
           ... 
          pn 

       pokiaľ nebude podmienka (splnená ) 

        
       kde p1, p2 až pn sú  príkazy 

 

V TP má  tvar:  repeat 

     p1; 
     p2; 
      ... 
     pn 

until b  kde b je vý raz typu boolean a p1 až pn sú  príkazy 

Vykonanie: Vykonajú  sa príkazy p1, p2 až pn a opakovane sa budú  vykoná vať, pokiaľ vý raz b bude 

nadobú dať hodnotu false. Ak vý raz b nadobudne hodnotu true, cyklus sa ukončí. 

– podmienka 

p 

+ 

Príklady:  I := 0;    repeat     repeat 
repeat       write(’Zadaj číslo: ’);      p1; ... 

     I := I + 1      readln(X)       pn 
  until I = 10   until X = 0    until false 
  príkaz v cykle sa   príkazy v cykle sa budú    nekonečný  cyklus 

vykoná  10-krá t   opakovať, ký m nevložíme 0            ukončí sa  Ctrl+Break 
 

 

Príklad CRP-1: Vytvorte programovú  schému na zabezpečenie opakovania behu programu ľubovoľný  

počet krá t. 

Analý za: Už vá s nudí neustá le spú šťať ten istý  program, najmä keď ho testujete, či pracuje sprá vne? 

Riešením je príkaz repeat. Chceme dosiahnuť, aby sa po spustení program vykonal a na konci sa opý tal: 

„Opakovať?“. Ak odpovieme N = nie, program skončí, inak sa vykoná  znova. 

Má me viacej možností, najprv tá  najjednoduchšia: 

... 

begin 

repeat 

   { príkazy programu } 

   write(’Opakovať (nie = N) ?’); 

   readln(ODPOVED) 

until (ODPOVED = ’N’) or (ODPOVED = ’n’) 

end. 

Elegantnejším riešením je použiť funkciu unitu crt readkey. Vý sledkom funkcie readkey je stlačený  klá ves, 

ušetríme premennú  ODPOVED aj stlačenie klá vesu Enter. Readkey v kombiná cii s upcase: 

begin 

repeat 

   { príkazy programu } 

   writeln(’Opakovať (nie = N) ?’) 

until  upcase(readkey) = ’N’ 

end. 

Na ukončenie programov sa používa aj klá ves Esc. Pomocou programu na nasledujú cej strane zistíme, že 

jeho poradové číslo je 27. A tu je schéma:  const Esc = #27; 

begin      ... 

repeat      begin 

   { príkazy programu }    { príkazy programu }   

   writeln(’Koniec – stlač Esc!’:45)  writeln(’Koniec – stlač Esc!’:45) 

until  readkey = chr(27)    until  readkey = Esc  

end.      end. 

Zá pis chr(X) je rovnocenný  so zá pisom #X. 

Aj na zistenie poradového čísla stlačeného znaku môže poslú žiť malý  program s cyklom repeat: 
program KOD_STLACENEHO_KLAVESU; 
uses crt; 
var  stlaceny_klaves:char; 
begin 
clrscr; 
writeln(’Koniec – stlac C’:45); 
repeat 
  stlaceny_klaves:=readkey; 
  writeln(ord(stlaceny_klaves)); 
until upcase(stlaceny_klaves)=’C’; 
end. 
 

Príklad CRP-2: Vytvorte programovú  schému na ošetrenie sprá vnosti vstupný ch hodnôt. Ak vstupné 

hodnoty nespĺňajú  vstupné podmienky, majú  sa vyžiadať nové vstupné hodnoty. 

Analý za: Príkaz repeat je vhodný  aj na ošetrenie vstupný ch hodnôt, či spĺňajú  vstupné podmienky. Najprv 

predsa musíme zadať vstupné hodnoty a až potom môže podmienka rozhodnú ť, či treba vstup opakovať 

alebo môže program pokračovať. Ako uká žku ošetríme vý počet mocniny xn pre x = 0 a n = 0 (pozri príklad 

CFR-3):   
   repeat 

    write(’Zadaj zaklad mocniny a exponent: ’); 
   readln(X,N) 
 until (X<>0) or (N<>0) 

Sprá vnejšie je nie len „zacykliť“ vstup ale aj doplniť vý pisom, prečo sa príkazy vstupu opakujú . Teda: 
repeat 

   write(’Zadaj zaklad mocniny a exponent: ’); 
  readln(X,N); 
  if (X=0) and (N=0) then writeln(’0 na 0-tú nedefinované!’) 
until (X<>0) or (N<>0); 

 
Pod chybou vstupu môžeme rozumieť aj „preklep“, napríklad vložením písmena namiesto číslice 

(Error 106: Invalid numeric format.). K vyriešeniu tohto problému treba poznať funkciu IOResult, ktorá  po 

každej vstupno/vý stupnej operá cii nadobú da buď hodnotu 0, ak nedošlo ku chybe, alebo číslo chyby. Keďže 

sme zapli v Options – Compiler... prepínač I/O Checking – kontrolu všetký ch vstupno/vý stupný ch operá cií, 

program by hneď skončil s chybový m hlá sením. Tú to kontrolu však možno vypnú ť umiestnením direktívy 

{$I - } pred príkaz vstupu a ná sledne ju zapnú ť príkazom {$I +}. Schéma má  tvar: 

... 
var CHYBA : boolean; 
     X : integer;  { X aký koľvek číselný  typ } 
... 

 begin 

 repeat 

    write(’Vlož číslo: ’); 

    {$I-} readln(X); {$I+} 

    CHYBA := IOResult <> 0; 

    if CHYBA then writeln(’Chyba vstupu!’) 

 until not CHYBA; 

 napr. writeln(X) 



 Zá klady programovania  – Turbo Pascal 42  

Ak dôjde ku chybe, IOResult bude rôzny od nuly a preto premenná  CHYBA nadobudne hodnotu TRUE. Na 

obrazovke sa zobrazí: Chyba vstupu! a znova Vlož číslo:, pretože CHYBA má  hodnotu true a negá cia (not) 

true je false – vtedy sa príkazy v repeat opakujú . Otestujte program pre dobré aj zlé vstupné hodnoty 

A teraz už „normá lny“ príklad: 

Príklad CRP-3: Vytvorte program – hru na uhá dnutie zadaného čísla, keď po vložení čísla program ozná mi 

„Veľa, uber!“, „Má lo, pridaj!“ alebo „Vý borne, uhá dol si!“. Nech treba uhá dnuť celé číslo z intervalu od 0 

po 100. 

Analý za: Pre získanie čísla, ktoré bude treba uhá dnuť sú  dve možnosti:  

• buď má me protihrá ča a ten zadá  číslo tak, aby to sú per nevidel (napríklad čiernym písmom – nastavuje 

sa príkazom textcolor na čiernom podklade – nastavuje sa príkazom textbackground), alebo 

• číslo pomocou funkcie random vyberie počítač (random(x) - vý sledkom je ná hodné celé číslo z intervalu 

<0,x); pred  príkazom random musí byť použitý  na začiatku programu príkaz randomize). 

Druhá  časť programu je vlastné há danie. „Sedliacky“ rozum hovorí: vložíme číslo, počítač ho porovná  

s číslom, ktoré treba uhá dnuť a ozná mi „Veľa, uber!“ alebo „Má lo, pridaj!“ a znova sa celé opakuje, až ký m 

neuhá dneme, vtedy há danie končí. Keďže nevieme, na koľký  pokus sa ná m podarí uhá dnuť hľadané číslo, je 

to cyklus s nezná mym počtom opakovaní. Najprv musíme vložiť číslo a porovnať a až potom vieme 

rozhodnú ť, či cyklus končí – preto cyklus s podmienkou na konci. A tu je hotový  program: 
program HADAJ; 
uses  crt; 
const MAX=100; 
var   CISLO,UHCISLO:integer; 
      ODP:char; 
begin 
randomize; 
repeat 
  clrscr; 
  gotoxy(1,12); writeln(’Cislo vyberie pocitac ....... P’:50); 
  writeln;      writeln(’Cislo zada protihrac ........ H’:50); 
  ODP:=upcase(readkey); 
  clrscr; 
  case ODP of 
    ’P’: UHCISLO:=random(MAX+1); 
    ’H’: begin 
         gotoxy(15,12); 
         write(’Zadaj cislo na uhadnutie < ’,MAX,’: ’); 
         textcolor(black); 
         readln(UHCISLO); 
         textcolor(white); 
         clrscr; 
         end; 
  end; 
  writeln(’H A D A J  C I S L O !’); 
  repeat 
    readln(CISLO); 
    if CISLO <> UHCISLO 
      then if CISLO<UHCISLO then writeln(’Malo, pridaj!’) 
                            else writeln(’Vela, uber!’); 
  until CISLO=UHCISLO; 
  clrscr; textcolor(yellow+blink); 
  gotoxy(30,12); write(’B  I  N  G  O  ! ! ! ’); 
  delay(2500); 
  normvideo; 
  gotoxy(30,24); write(’Este raz? (nie = N) ’); 
until upcase(readkey)=’N’; 
end. 

Príkaz normvideo by mal nastaviť štandardné atribú ty pre farby textu aj pozadia. Program môžete doplniť aj 

o zvukové efekty pomocou príkazov sound(frekvencia) a nosound – vypína zvuk, tento nesmiete zabudnú ť 

uviesť po príkaze sound. Dĺžka trvania tónu sa realizuje príkazom delay(milisekú nd). 

Kedy ktorý  cyklus 

 Keď už má me zá kladnú  predstavu o fungovaní jednotlivý ch cyklov a im zodpovedajú cich príkazov, 

môžeme si poznatky trochu zosystematizovať: 

• ak vieme počet opakovaní príkazov v cykle a premenná , ktorá  riadi počet prechodov cyklom, sa môže 

meniť na nasledovníka alebo predchodcu (krok +/- 1 alebo postupnosť za sebou idú cich znakov ASCII), 

použijeme príkaz for 

• ak nevieme počet opakovaní príkazov v cykle alebo premenná , ktorá  riadi počet prechodov cyklom, 

nenadobú da „pekné“ hodnoty (hodnoty nasledovníkov alebo predchodcov), použijeme príkazy while 

alebo repeat, pričom: 

• príkaz while, t.j. s podmienkou na začiatku použijeme, ak môže nastať situá cii, že príkaz v cykle sa 

nemá  vykonať ani raz 

• príkaz repeat, t.j. s podmienkou na konci použijeme, ak sa príkazy v cykle majú  vykonať aspoň raz. 

Uvedomte si, že v každom cykle musí byť premenná , ktorá  riadi počet prechodov daný m cyklom! 

Kontrolujte, či sa jej hodnoty zväčšujú  alebo zmenšujú  a či sú  ohraničené podmienkou ukončenia cyklu, iná č 

cyklus nikdy neskončí. 

Sem-tam sa u začiatočníkov stá va, v snahe nezabudnú ť bodkočiarku, že ju dajú  aj za slovo „do“ v príkaze for 

alebo while. Tý m vznikol tzv. prá zdny príkaz a cyklus zrejme pracuje celkom iná č, ako to bolo pôvodne 

plá nované. 

 

Organizácia programu 
 Poslednou teoretickou časťou je organizá cia programu v Turbo Pascale.  

Program sa skladá : 

program meno programu;    hlavička programu (môže chý bať) 

uses unity oddelené čiarkami;    deklará cia unitov 

const meno konš tanty = hodnota;   definícia konštá nt 

type meno typu = ...;     definícia typov       ú sek definícií 

var premenné oddelené čiarkami : typ premenný ch; deklará cia premenný ch       a deklará cií 

procedure ...      deklará cia 

function ...      procedú r a funkcií    blok 

begin              

príkaz1;      príkazová  časť 

príkaz2;      začína slovom begin 

...       a končí slovom end 

príkazn 

end.       bodka – koniec programu (nepatrí do bloku) 

Pozná mky:  

• Niektoré objekty z predchá dzajú cej schémy ešte nepozná me.  

• Príkazy sa v príkazovej časti oddeľujú  bodkočiarkou.  

• Všetky mená  objektov sú  identifiká tory. 

EXE verzie programov 

 Turbo Pascal umožňuje vytvá rať aj od prostredia TP nezá vislé verzie programov, priamo programy 

s príponou exe. Keď chceme získať exe verziu programu, musíme jeho kompilá ciu (preklad) presmerovať 

z pamäte na disk. Robí sa to prepnutím položky Destination v Compile z hodnoty Memory na hodnotu Disk. 

Program s rovnaký m ná zvom ako bol pôvodný  sú bor, len s príponou exe, sa uloží do aktuá lneho adresá ra.  

 

Ď alš ie rieš ené úlohy 
Príklad CWH-4: Vytvorte program na krá tenie zlomkov. 

Analý za: Napr. v zlomku 15/12 sa dá  čitateľ aj menovateľ deliť číslom 3 a zlomok sa dá  upraviť do 

zá kladného tvaru 5/4. Číslo 3, ktorý m sme krá tili zlomok, je najväčší spoločný  deliteľ (nsd) čitateľa a 

menovateľa zlomku. Nsd(15,12) sa vypočíta takto: nsd(15,12) = nsd(15-12,12) = nsd(3,12) = nsd(3,12-3) = 

nsd(3,9) = nsd(3,9-6) = nsd(3,3) = 3. Opakujeme odčítanie menšieho čísla od väčšieho dovtedy, ký m sa 

nerovnajú . Rovnaké číslo je najväčší spoločný  deliteľ pôvodný ch čísel. 

V slovnom zá pise:  pokiaľ je A<>B opakuj 

        ak A>B 

    tak A := A-B 

    inak B := B-A; 

   NSD := A  { alebo B, keďže sú  rovnaké } 

Vý stup programu nech je v tvare: 15 / 12 = 5 / 4, ak je podiel celé číslo, zlomková  čiara a menovateľ 1 nech 

sa nezobrazia, napr. 12 / 6 = 2. 

 
program ZLOMOK; 
uses crt; 
var CIT,MEN,A,B:integer; 
begin 
clrscr; 
repeat 
  write(’Zadaj citatela a menovatela zlomku: ’ ); 
  readln(CIT,MEN); 
  write(CIT,’ / ’,MEN,’ = ’); 
  A:=CIT; 
  B:=MEN; 
  while A<>B do 
    if A>B 
      then A:=A-B 
      else B:=B-A; 
  CIT:=CIT div A; 
  MEN:=MEN div B; 
  if MEN=1 
    then writeln(CIT) 
    else writeln(CIT,’ / ’,MEN); 
  writeln(’Koniec - stlac Esc!’) 
until readkey=chr(27) 
end. 

Príklad CRP-4: Vytvorte program na vý počet aritmetického priemeru vopred nezná meho počtu kladný ch 

reá lnych čísel. 

Analý za: Ú loha je podobná  ú lohe CFR-4, v nej však bol zná my počet čísel a preto sme mohli použiť príkaz 

for. Pri nezná mom počte čísel musíme použiť cyklus s nezná mym počtom opakovaní. Ukončenie cyklu je 

v takomto prípade na tzv. koncovú  hodnotu. V našom príklade sú  všetky zadá vané reá lne čísla kladné a preto 

môžeme zvoliť za koncovú  hodnotu nulu. Znamená  to, že príkazy v cykle sa budú  opakovať, pokiaľ 

nevložíme 0, vtedy sa cyklus ukončí. Najprv vložíme číslo a potom môžeme rozhodnú ť, či cyklus končí, teda 

podmienka je na konci, preto príkaz repeat. Ošetrite aj prípad, keď hneď prvý m vložený m číslom bude 0. 
 
program PRIEMER2; 
uses crt; 
var     POCET : integer; 
     X, SUCET : real; 
begin 
clrscr; 
POCET:=0; 
SUCET:=0; 
repeat 
  write(’Vloz cislo: ’); 
  readln(X); 
  if X>0 
    then begin 
         inc(POCET); 
         SUCET:=SUCET+X 
         end 
until X=0; 
if POCET>0 
  then writeln(’Priemer zaokruhleny na stotiny: ’,SUCET/POCET:4:2) 
  else writeln(’Bola zadana len koncova hodnota!’); 
readln 
end. 
 

Program CFR-5: Program na prevod malý ch písmen textu na veľké. 

Analý za: Prakticky si treba len osvojiť zá pis napr. TEXT[1] – prvý  znak reťazca v premennej TEXT, 

TEXT[5] – piaty znak reťazca, TEXT[I] – I-ty znak reťazca v premennej TEXT. Ak teda chceme zmeniť 

všetky písmená  v reťazci na veľké, musíme ísť od prvého znaku až po posledný . Počet znakov v reťazci ná m 

zistí funkcia length (vracia číslo – dĺžku reťazca). Funkcia upcase zmení malé písmeno na veľké, ostatné 

znaky nemení. 

 
program PREVOD; 
uses crt; 
var TEXT : string; 
    DLZKA, I : integer; 
begin 
clrscr; 
write(’Text: ’); 
readln(TEXT); 
DLZKA:=length(TEXT); 
for I:=1 to DLZKA do 
  TEXT[I]:=upcase(TEXT[I]); 
writeln(TEXT); 
readln 
end. 

Program CFR-6: Vytvorte program na  kódovanie textu posunutím o jeden znak vpravo (Zz -> Aa) a 

dekódovanie textu posunutím o jeden znak vľavo (Aa -> Zz). 

Analý za: Podstatou algoritmu je brať znak po znaku text a pokiaľ je znakom písmeno, zobraziť jeho 

nasledovníka – pri kódovaní, resp. predchodcu – pri dekódovaní. Situá ciu trochu komplikujú  okrajové 

písmená , pretože veľké Z a malé z sa majú  posunú ť na veľké A a malé a resp. Aa na Zz pri dekódovaní. 

Keďže je viacej možností, a na každú  sa má  iná č kódovať, vhodné je použiť príkaz case s časťou else (každý  

iný  znak ako písmeno sa nemá  zmeniť). Prejsť treba celý  text, od prvého po posledný  znak, čo ná m zaručí 

príkaz for (v2 = length(TEXT)). Program musí umožňovať vybrať si z dvoch možností: Kódovať alebo 

Dekódovať. 

 
program KODOVANIE_a_DEKODOVANIE_posunutim_o_znak; 
uses crt; 
var TEXT:string; 
    I:byte; 
    ODPOVED:char; 
begin 
clrscr; 
write(’Text: ’); 
readln(TEXT); 
writeln(’[K]odovat alebo [D]ekodovat?’); 
ODPOVED:=upcase(readkey); 
if ODPOVED=’K’ 
  then begin 
       for I:=1 to length(TEXT) do 
         case TEXT[i] of 
           ’A’..’V’,’a’..’v’: write(succ(TEXT[I])); 
                         ’Z’: write(’A’); 
                         ’z’: write(’a’); 
           else write(TEXT[I]) 
         end; 
       writeln 
       end 
   else if ODPOVED=’D’ 
          then begin 
               for I:=1 to length(TEXT) do 
                 case TEXT[i] of 
                   ’B’..’Z’,’b’..’z’: write(pred(TEXT[I])); 
                                 ’A’: write(’Z’); 
                                 ’a’: write(’z’); 
                   else write(TEXT[I]) 
                 end; 
               writeln 
               end 
          else writeln(’Nebolo stlacene K alebo D!’); 
readln 
end. 
 

Príklad CFR-7*: Príklad je určený  pre zdatnejších alebo hú ževnatejších programá torov. Je principiá lne 

odlišný  od predchá dzajú ceho. Program má  kódovať a dekódovať písmená  textu posunutím o POSUN písmen 

vpravo (vľavo). 

Analý za: Programy s hviezdičkou (nie sú  zaká zané) sú  určené na prá cu s hotový m programom. Ujasnite si 

jednotlivé časti programu a vyskú šajte, ako pracujú . Program si pamätá  predchá dzajú ci text (pri prvom 

spustení samozrejme nie), aby ste ho mohli po zakódovaní hneď dekódovať a opačne. Dekódovanie je 

v skutočnosti kódovanie o 26-POSUN znakov (neustá le vpravo). Číslo 26 je počet písmen (A až Z alebo 

malé a až z), pričom poradové číslo A je 65 a malého a 97. Pri kódovaní sa pohybujeme v uzavretom kruhu. 

program KODOVANIE_a_DEKODOVANIE_posunutim_o_POSUN_znakov; 
uses crt; 
var TEXT:string; 
    POSUN:integer; 
    DLZKA,i,ZacAbc,PozVAbc,RelPozKodu:byte; 
    ODPOVED:char; 
begin 
clrscr; 
repeat 
  writeln(’Novy text (ano - A)?’); 
  if upcase(readkey)=’A’ 
    then begin 
         writeln(’Text:’); 
         readln(TEXT); 
         DLZKA:=length(TEXT); 
         end; 
  repeat 
    write(’[K]odovat alebo [D]ekodovat? ’); 
    readln(ODPOVED); 
    ODPOVED:=upcase(ODPOVED); 
  until (ODPOVED=’K’) or (ODPOVED=’D’); 
  write(’Posun o znakov: ’); 
  readln(POSUN); 
  POSUN:=POSUN mod 26; 
  if POSUN<0 then POSUN:=POSUN+26; 
  if ODPOVED=’D’ then POSUN:=26-POSUN; 
  for i:=1 to DLZKA do 
    if TEXT[i] in [’A’..’Z’,’a’..’z’] 
      then begin 
           case TEXT[i] of 
             ’A’..’Z’:ZacAbc:=65; 
             ’a’..’z’:ZacAbc:=97; 
           end; 
           PozVAbc:=ord(TEXT[i])-ZacAbc; 
           RelPozKodu:=(PozVAbc+POSUN) mod 26; 
           TEXT[i]:=chr(ZacAbc+RelPozKodu) 
           end; 
  writeln(TEXT); 
  writeln(’KONIEC - stlac Esc’:45) 
until readkey=#27 
end. 
 

Ď alšou skupinou ú loh sú  „grafické“ ú lohy v textovom režime obrazovky. 

Príklad CFR-8: Program ROZSVECOVANIE_V_OKNACH ná hodný m generovaním polohy postupne 

vypĺňa plochu obrazovky „rozsvietený mi“ obdĺžnikmi (znak s poradový m číslo 219) . 

Analý za: Treba opakovane „rozsvecovať okná “, preto cyklus. My sme ho postavili na zná mom počte 

opakovaní a preto sa program opý ta, koľko okien má  rozsvietiť (to je počet opakovaní príkazov v cykle). No 

a potom už stačí len použiť funkciu random a nechať ná hodne vygenerovať číslo od 1 do 80 – stĺpec a číslo 

od 1 do 24 – riadok, v ktorom sa „rozsvieti“. Požadovaný  stĺpec a riadok sa nastaví príkazom gotoxy. 

 
program ROZSVECOVANIE_V_OKNACH; 
uses crt; 
var  I, POCET : word; 
begin 
randomize; 
clrscr; 
write(’Pocet rozsvietenych okien do 65 tis.: ’); 
readln(POCET); 
for I:=1 to POCET do 
  begin 

  gotoxy(random(80)+1,random(24)+1); 
  write(chr(219)); 
  end; 
readln 
end. 
 

Predchá dzajú ca ú loha je ešte zaujímavejšia, ak chceme, aby program generoval „rozsvecovanie“, až ký m 

nestlačíme ľubovoľný  klá ves. Funkcia keypressed unitu crt má  hodnotu true, ak bol stlačený  klá ves, inak má  

hodnotu false. Celý  problém je vo vytvorení tentoraz cyklu s nezná mym počtom opakovaní, ktorý  skončí na 

stlačenie klá vesu. Ak chceme „rozsvecovanie“ spomaliť, stačí vložiť do cyklu príkaz delay(milisekú nd). 

 
program ROZSVECOVANIE_V_OKNACH2; 
uses crt; 
begin 
randomize; 
clrscr; 
repeat 
  gotoxy(random(80)+1,random(24)+1); 
  write(chr(219)); 
until keypressed; 
readln 
end. 
 

V ďalšom príklade sa pokú sime usmerniť pohyb. 

Príklad CRP-5: Vytvorte program na pá d „vajíčka“ (písmena O) z hora nadol v strede riadkov. 

Analý za: Kto si poctivo ujasnil predchá dzajú ce dva programy, nemôže mať s nasledujú cimi dvoma  

problémy. Mali by sme použiť príkaz for RIA:=1 to 25 do ... ale chceme od vá s, aby ste použili príkaz 

repeat. Program sme doplnili aj o zvukový  efekt. Ď alšie zvukové efekty môžete doprogramovať pre let 

vajíčka. Ak sa chcete pohrať, môžete pridať aj „gravitá ciu“, aby vajíčko pri svojom pá de zrý chľovalo. 

 
program pad_vajca; 
uses crt; 
var  ria:byte; 
begin 
clrscr; 
ria:=1; 
repeat 
  gotoxy(40,ria); write(’.’); 
  gotoxy(40,ria+1); write(’O’); 
  delay(150); inc(ria) 
until ria=25; 
sound(60); 
gotoxy(36,12); write(’B U M !!!’); 
gotoxy(37,ria); write(’---^---’); 
delay(200); 
nosound; 
readln 
end. 
 
V ďalšej verzii programu PAD_VAJCA sa ho pokú sime zachrá niť pred rozbitím. Nech program vygeneruje 

ná hodne veľké písmeno (veľké písmená  začínajú  „na“ 65 a končia na 90) a zobrazí ho v ľavom hornom rohu 

obrazovky. Ak stihneme do dopadnutia vajíčka „do 25 riadku“ ná jsť na klá vesnici a stlačiť hľadaný  klá ves, 

pá d sa zastaví. 

program zachran_vajce; 
uses crt; 
var  RIA:byte; 
     PISMENO, STLACIL:char; 
begin 
randomize; 
clrscr; 
PISMENO:=chr(random(26)+65); 
gotoxy(1,5); write(’Najdi ’,PISMENO); 
RIA:=1; 
repeat 
  gotoxy(40,RIA); write(’.’); 
  gotoxy(40,RIA+1); write(’O’); 
  if keypressed then STLACIL:=upcase(readkey); 
  delay(100); inc(RIA) 
until (STLACIL=PISMENO) or (RIA=25); 
if RIA<25 
  then begin gotoxy(33,12); write(’V Y B O R N E !’) end 
  else begin 
       sound(60); 
       gotoxy(36,12); write(’B U M !!!’); 
       gotoxy(37,RIA); write(’---^---’); 
       delay(200); nosound 
       end; 
readln 
end. 
 
Ak vá s pohyb na obrazovke zaujal, môžeme skú šať ďalej. 

Príklad CRP-6: Vytvorte program na pohyb znaku po obrazovke. Na ovlá danie pohybu použite šípky na 

ovlá danie pohybu kurzora. Testujte aj „vybehnutie“ znaku z obrazovky. 

Analý za: Tento program uvedieme bez hlbšej analý zy. Nechá me ho na podrobné štú dium 

a experimentovanie. Ku kódom šípok sme sa dostali pomocou programu uvedeného nad príkladom CRP-2 (0 

si nevšímajte, sú  tam preto, že sa nejedná  o obyčajné klá vesy). Všimnite si cyklus repeat until keypressed, 

ktorý  zamedzuje neustá lemu prepisovaniu znaku na mieste a  jeho blikaniu (prakticky zacyklí program na 

danom mieste až do stlačenia ľubovoľného klá vesu). Program možno ná silne kedykoľvek ukončiť stlačením 

klá vesu Esc (príkaz halt – je to len uká žka, nenavykajte si naň!). 

Programy začínajú  byť zložitejšie a preto sme sa rozhodli začať nená padne používať podprogramy 

(procedú ry). Podprogram rieši nejaký  čiastkový  problém a je veľmi podobný  programu (namiesto slova 

program používa slovo procedure). V príkazovej časti hlavného programu stačí uviesť meno procedú ry 

a automaticky sa vykoná  všetko, čo v nej je naprogramované. 

 
program pohyb_znaku_pri_pridrzani_sipky_bez_blikania; 
uses  crt; 
 
const Esc=chr(27); 
      sipkaVPRAVO=chr(77);     {M} 
      sipkaVLAVO =chr(75);     {K} 
      sipkaHORE  =chr(72);     {H} 
      sipkaDOLE  =chr(80);     {P} 
 
var   stl,ria:byte; 
 
procedure inicializacia; 
  begin 
  clrscr; 
  stl:=40; ria:=12; gotoxy(stl,ria); write(’*’); 
  end; 

procedure pohyb; 
  var stlznak:char; 
  begin 
  repeat 
    repeat until keypressed; 
    stlznak:=upcase(readkey); 
    gotoxy(stl,ria); write(’ ’); 
    case stlznak of 
      sipkaVPRAVO : inc(stl); 
      sipkaVLAVO  : dec(stl); 
      sipkaHORE   : dec(ria); 
      sipkaDOLE   : inc(ria); 
      Esc         : halt; 
      end; 
    gotoxy(stl,ria); write(’*’); 
  until (stl=1)or(stl=80)or(ria=1)or(ria=25); 
  gotoxy(25,13); writeln(’B U M ! ! !  O H R A D A ! ! !’); 
  end; 
BEGIN 
inicializacia; 
pohyb; 
readln 
END. 
 

Pohyb je zaujímavejší, ak sa znak pohybuje určený m smerom, až ký m nezvolíme iný  smer. Čiže, ký m 

nezvolíme iný  smer (nestlačíme inú  šípku), znak sa má  uberať „starý m“ smerom. 

 
program pohyb_znaku_bez_podrzania_sipky; 
uses  crt; 
const Esc=chr(27); 
      sipkaVPRAVO=chr(77);     {M} 
      sipkaVLAVO =chr(75);     {K} 
      sipkaHORE  =chr(72);     {H} 
      sipkaDOLE  =chr(80);     {P} 
var   stl,ria:byte; 
 
procedure inicializacia; 
  begin 
  clrscr; 
  stl:=40; ria:=12; gotoxy(stl,ria); write(’*’); 
  end; 
 
procedure pohyb; 
  var stlznak:char; 
  begin 
  repeat 
    if keypressed then stlznak:=upcase(readkey); 
    gotoxy(stl,ria); write(’ ’); 
    case stlznak of 
      sipkaVPRAVO : inc(stl); 
      sipkaVLAVO  : dec(stl); 
      sipkaHORE   : dec(ria); 
      sipkaDOLE   : inc(ria); 
      Esc         : halt; 
      end; 
    gotoxy(stl,ria); write(’*’); delay(70); 
  until (stl=1)or(stl=80)or(ria=1)or(ria=25); 
  gotoxy(25,13); writeln(’B U M ! ! !  O H R A D A ! ! !’); 
  end; 
BEGIN 
inicializacia; 
pohyb; 
readln 
END. 

Spravme v programe ešte zopá r ú prav krá sy. Po prvé, vyrobme okolo obrazovky ohradu (v dolnej časti po 

riadok 24). Potrebné kódy znakov sme si našli po zobrazení ASCII tabuľky. Môžete postupovať aj tak, že 

v konštantá ch budete mať priamo zobrazené príslušné semigrafické znaky v apostrofoch. druhá  ú prava krá sy 

spočíva vo vypnutí zobrazovania kurzora, aby ná m „nekazil celkový  dojem“ (príkazy a potrebné premenné – 

registre pozná  unit dos). 

 
program pohyb_znaku_bez_podrzania_sipky2; 
uses  crt,dos; 
const ESC=chr(27); 
      sipkaVPRAVO=chr(77);     {M} 
      sipkaVLAVO =chr(75);     {K} 
      sipkaHORE  =chr(72);     {H} 
      sipkaDOLE  =chr(80);     {P} 
var   stl,ria:byte; 
 
procedure ohrada; 
  var i:integer; 
  begin 
  write(chr(201)); for i:=2 to 79 do write(chr(205)); write(chr(187)); 
  for i:=2 to 23 do begin write(chr(186)); write(chr(186):79) end; 
  write(chr(200)); for i:=2 to 79 do write(chr(205)); write(chr(188)); 
  end; 
 
procedure skry_kurzor; 
  var reg:registers; 
  begin                {ukaz} 
  reg.ah:=1;           {   1} 
  reg.ch:=32;          {  13} 
  reg.cl:=0;           {  40} 
  intr($10,reg) 
  end; 
 
procedure inicializacia; 
  begin 
  clrscr; 
  ohrada; 
  skry_kurzor; 
  stl:=40; ria:=12; gotoxy(stl,ria); write(’*’); 
  end; 
 
procedure pohyb; 
  var stlznak:char; 
  begin 
  repeat 
    if keypressed then stlznak:=readkey; 
    gotoxy(stl,ria); write(’ ’); 
    case stlznak of 
      sipkaVPRAVO : inc(stl); 
      sipkaVLAVO  : dec(stl); 
      sipkaHORE   : dec(ria); 
      sipkaDOLE   : inc(ria); 
      Esc         : halt; 
      end; 
    gotoxy(stl,ria); write(’*’); delay(70); 
  until (stl=1)or(stl=80)or(ria=1)or(ria=24); 
  gotoxy(25,13); write(’B U M ! ! !  O H R A D A ! ! !’); 
  end; 
BEGIN 
inicializacia; 
pohyb; 
readln 
END. 

Predchádzajúci pohyb doplníme do hry. Nech program vygeneruje na náhodnom, ale 
vhodnom(!) mieste napríklad O, ktoré treba pohybujúcim sa znakom trafiť. Princíp 
je jednoduchý, pribudne testovanie, či sa nenachádza pohybujúci sa znak na 
pozícii znaku O. 
 
program traf_znak; 
uses  crt,dos; 
 
const Esc=chr(27); 
      sipkaVPRAVO=chr(77);     {M} 
      sipkaVLAVO =chr(75);     {K} 
      sipkaHORE  =chr(72);     {H} 
      sipkaDOLE  =chr(80);     {P} 
 
var   stl,ria,x,y:byte; 
 
procedure skry_kurzor; 
  var reg:registers; 
  begin                
  reg.ah:=1;           
  reg.ch:=32;          
  reg.cl:=0;           
  intr($10,reg) 
  end; 
 
procedure ohrada; 
  var i:integer; 
  begin 
  textcolor(white); 
  write(chr(201)); for i:=2 to 79 do write(chr(205)); write(chr(187)); 
  for i:=2 to 23 do begin write(chr(186)); write(chr(186):79) end; 
  write(chr(200)); for i:=2 to 79 do write(chr(205)); write(chr(188)); 
  end; 
 
procedure inicializacia; 
  begin 
  randomize; 
  skry_kurzor; 
  clrscr; 
  ohrada; 
  x:=random(78)+2; y:=random(22)+2; gotoxy(x,y); write(’O’); 
  stl:=40; ria:=12; gotoxy(stl,ria); write(’*’); 
  end; 
 
procedure pohyb; 
  var stlznak:char; 
  begin 
  repeat 
    if keypressed then stlznak:=readkey; 
    gotoxy(stl,ria); write(’ ’); 
    case stlznak of 
      sipkaVPRAVO : inc(stl); 
      sipkaVLAVO  : dec(stl); 
      sipkaHORE   : dec(ria); 
      sipkaDOLE   : inc(ria); 
      Esc         : halt; 
      end; 
    gotoxy(stl,ria); write(’*’); delay(100); 
  until (stl=1)or(stl=80)or(ria=1)or(ria=24)or((stl=x)and(ria=y)); 
  if (stl=x)and(ria=y) 
    then begin gotoxy(32,12); write(’Z A S A H ! ! !’) end 
    else begin gotoxy(25,12); write(’B U M ! ! !  O H R A D A ! ! !’) end; 
  end; 

BEGIN 
inicializacia; 
pohyb; 
readln 
END. 
 
Program môžete ďalej vylepšovať, napríklad nech sa ná hodne pohybuje aj znak O. Ale to už nechá vame na 

vašu šikovnosť. 

Okrem dynamický ch (pohybujú cich sa) obrazcov je zaujímavé programovať aj statické obrazce. Väčšinou 

ide o požitie príkazov for, pričom 

• ak obrazec ne je plný , cykly bý vajú  susedné 

• ak je obrazec plný , cykly bý vajú  vnorené - vonkajší cyklus nastavuje riadok a vnú torný  cyklus 

zabezpečuje pohyb v danom riadku. 

Nasledujú ci program demonštruje uvedené tvrdenia. Susedné cykly sú  v procedú rach obdlznik1 a 2, stvorec 

a trojuholnik1 a 2; vnorené cykly sú  v procedú rach obdlznik3 a trojuholnik3. 
program OBRAZCE; 
uses crt; 
var n,i:integer; 
 
procedure obdlznik1; 
  begin 
  clrscr; 
  write(’Strana obdlznika: ’); readln(n); 
  for i:=1 to n do write(’*’); 
  writeln; 
  for i:=2 to n-1 do writeln(’*’,’*’:n-1); 
  for i:=1 to n do write(’*’); 
  writeln; 
  readkey 
  end; 
 
procedure obdlznik2; 
  begin 
  clrscr; 
  write(’Strana obdlznika: ’); readln(n); 
  for i:=1 to n do write(’*’); 
  writeln; 
  for i:=2 to n-1 do writeln(’*’,’*’:i-1,’*’:n-i); 
  for i:=1 to n do write(’*’); 
  writeln; 
  readkey 
  end; 
 
procedure stvorec; 
  begin 
  clrscr; 
  write(’Strana stvorca: ’); readln(n); 
  for i:=1 to n do write(’* ’); 
  writeln; 
  for i:=2 to n-1 do writeln(’*’,’*’:2*(n-1)); 
  for i:=1 to n do write(’* ’); 
  writeln; 
  readkey 
  end; 
 
procedure trojuholnik1; 
  begin 
  clrscr; 
  write(’Strana trojuholnika: ’); readln(n); 
  writeln(’*’); 

  for i:=2 to n-1 do writeln(’*’,’*’:i-1); 
  for i:=1 to n do write(’*’); 
  writeln; 
  readkey 
  end; 
 
procedure trojuholnik2; 
  begin 
  clrscr; 
  write(’Strana trojuholnika: ’); readln(n); 
  for i:=1 to n do write(’*’); 
  writeln; 
  for i:=2 to n-1 do writeln(’*’,’*’:n-i); 
  writeln(’*’); 
  readkey 
  end; 
 
procedure obdlznik3; 
  var j:integer; 
  begin 
  clrscr; 
  write(’Strana obdlznika: ’); readln(n); 
  for i:=1 to n do 
    begin 
    for j:=1 to n do  
      if i<>j then write(’*’) else write(’ ’); 
    writeln 
    end; 
  readkey 
  end; 
 
procedure trojuholnik3; 
  var j:integer; 
  begin 
  clrscr; 
  write(’Strana trojuholnika: ’); readln(n); 
  for i:=1 to n do 
    begin 
    for j:=1 to i do write(’*’); 
    writeln 
    end; 
  readkey 
  end; 
 
begin 
obdlznik1; 
obdlznik2; 
stvorec; 
trojuholnik1; 
trojuholnik2; 
obdlznik3; 
trojuholnik3; 
end. 
 

Pokú ste sa vytvoriť program ktorý  nakreslí rovnoramenný  trojuholník s podstavou šírky n znakov, obrá tenú  

pyramídu, kosoštvorec, štyri štvorce vedľa seba so stranou n znakov, štvorec s uhlopriečkami, plný  obdĺžnik 

bez znakov na opačnej uhlopriečke, ako to je v procedú re obdlznik3. 

Obrazce sú  nekonečnou témou. Môžete sa pokú siť naprogramovať aj pohyb napr. obdĺžnika (medzi 

jednotlivý mi polohami netreba mazať celú  obrazovku). 

Teraz trochu na oddý chnutie, aby sme však dlho neoddychovali, pobežia ná m hodiny. 

Príklad CRP-7: Vytvorte program simulujú ci beh digitá lnych hodín v strede obrazovky v tvare HH : MM : 

SS. Využite príkaz GetTime z unitu Dos. 

Analý za: Program má  vlastne do nekonečna (ký m nestlačíme ľubovoľný  klá ves) čítať hodnoty systémového 

času pomocou príkazu gettime unitu dos a zobrazovať ich v strede obrazovky v predpísanom formá te. 

Problémom môže byť prechod z dvojciferného ú daja na jednociferný , keď cifra z desiatok nebude 

premazaná  – na to nesmieme zabudnú ť. 

 
program HODINY; 
uses crt,dos; 
var h,m,s,s100:word; 
           reg:registers; 
begin 
{skry kurzor}       
reg.ah:=1;          
reg.ch:=32;         
reg.cl:=0;          
intr($10,reg); 
clrscr; 
repeat 
  gettime(h,m,s,s100); 
  gotoxy(33,12); 
  if h<10 then write(’0’); 
  write(h,’ : ’); 
  if m<10 then write(’0’); 
  write(m,’ : ’); 
  if s<10 then write(’0’); 
  write(s); 
until keypressed; 
end. 
 

Doplňte program tak, aby hodiny pracovali ako budík („zazvonia“ v zadanom čase) prípadne časovač 

(„cinknú “ po uplynutí zadaného času). 

 

Na zá ver si ešte zopakujeme prá cu s textom. 

Príklad CWH-5: Vytvorte program, ktorý  zistí, či zadané slovo je symetrické, t.j. či sa rovnako číta zľava aj 

sprava. Symetrické sú  napríklad slová  RADAR a ABBA. 

Analý za: Zistiť, či zadané slovo je symetrické, znamená  porovnať jeho prvý  znak s posledný m, ak sa 

rovnajú , porovnať jeho druhý  znak s predposledný m, ak sa rovnajú , porovnať..., až ký m nenarazíme na dva 

rôzne znaky, alebo nie sme v strede slova (ľavá  polovica slova sa rovná  pravej polovici). Opakovane 

porovná vame vhodné znaky, preto treba použiť príkaz cyklu. Nevieme, kedy sa znaky už prestanú  rovnať, je 

to cyklus s nezná mym počtom opakovaní. Ú loha je riešiteľná  s príkazom while aj repeat, my sme sa rozhodli 

pre while. Keď sa po skončení cyklu s dvoma podmienkami ukončenia rozhodujete, pomocou ktorej 

testovať, prečo vlastne cyklus skončil, vyberte si vždy tú  dôležitejšiu. V našom prípade to znamená  testovať, 

či po skončení cyklu sa posledne testované písmená  rovnajú  alebo nie (if SLOVO[i]=SLOVO[DLZKA-

i+1]), či sme už v strede slova, nie je natoľko dôležité. 

 
program SYMETRICKE_SLOVO; 
uses crt; 
var SLOVO:string[80]; 
    i,DLZKA:byte; 

begin 
clrscr; 
repeat 
  write(’Vloz slovo: ’); readln(SLOVO); 
  DLZKA:=length(SLOVO); 
  i:=1; 
  while (SLOVO[i]=SLOVO[DLZKA-i+1]) and (i<DLZKA div 2) do 
    inc(i); 
  if SLOVO[i]=SLOVO[DLZKA-i+1] 
    then writeln(’Text je symetricky’) 
    else writeln(’Text nie je symetricky’); 
  writeln(’KONIEC - stlac Esc’) 
until readkey=#27; 
end. 
 

Uvedený  program možno zdokonaliť tý m, že nebude rozlišovať medzi veľký mi a malý mi písmenami 

a všetky ostatné znaky bude „ignorovať“. Potom aj text: „Jelenovi pivo nelej“ bude symetrický ! 

Analý za: Prvú  požiadavku, nerozlišovať medzi veľký mi a malý mi písmenami, možno vyriešiť pomocou 

funkcie upcase (zná ma vec). Druhú  požiadavku, aby všetky ostatné znaky boli ignorované, možno dosiahnuť 

vytvorením nového textu, ktorý  bude obsahovať len písmená . Využili sme množinový  operá tor in – je 

prvkom množiny (prvky množín sa uvá dzajú  v hranatý ch zá tvorká ch). 

 
program SYMETRICKY_TEXT_2; 
uses crt; 
var TEXT,novyTEXT:string[80]; 
    i,j,DLZKA:byte; 
begin 
clrscr; 
repeat 
  write(’Zadaj text: ’); readln(TEXT); 
  DLZKA:=length(TEXT); 
  j:=0; 
  for i:=1 to DLZKA do 
    if TEXT[i] in [’A’..’Z’,’a’..’z’] 
      then begin 
           inc(j); 
           novyTEXT[j]:=upcase(TEXT[i]) 
           end; 
  for i:=1 to j do 
    TEXT[i]:=novyTEXT[i]; 
  DLZKA:=j; 
  i:=1; 
  while (TEXT[i]=TEXT[DLZKA-i+1]) and (i<DLZKA div 2) do 
    inc(i); 
  if TEXT[i]=TEXT[DLZKA-i+1] 
    then writeln(’Text je symetricky’) 
    else writeln(’Text nie je symetricky’); 
  writeln(’KONIEC - stlac Esc’:45) 
until readkey=#27; 
end. 

Nerieš ené  úlohy na cykly: 

1. Vytvorte program na vý počet sú činu A.B dvoch prirodzený ch čísel pomocou sú čtu A + A + ...+ A 

(A ≥ B). 

2. Vytvorte program vý počet mocniny s krokom x2 (napr. x7 = x2. x2. x2.x). 

3. Vytvorte program na zistenie, či zadané číslo je prvočíslo. 

4. Vytvorte program na vypísanie všetký ch prvočísiel po zadané prirodzené číslo väčšie ako 1. 

5. Vytvorte program na zistenie, či prirodzené číslo N < 341 je  prvočíslo, ak viete, že platí veta: 

       „N je prvočíslom prá ve vtedy , ak je deliteľom čísla 2N-2 a N < 341“. 

6. Vytvorte program na ná jdenie všetký ch prirodzený ch čísel po  zadanú  hranicu, ktorý ch ciferný  sú čet sa 

rovná  zadanému číslu. 

7. Vytvorte program, ktorý  ako učebnú  pomôcku vygeneruje tabuľku ná sobilky zadaného ná sobiteľa. 

Obmedzte hodnotu ná sobiteľa do 100 a ná sobenec nech má  hodnoty z množiny {1,2,...,10 }. 

8. Vytvorte program na prevod čísla z desiatkovej do dvojkovej sú stavy. 

9. Vytvorte program na prevod čísla z  dvojkovej do desiatkovej sú stavy. 

10. Vytvorte program na prevod čísla z desiatkovej do šestná stkovej sú stavy. 

11. Vytvorte program na prevod čísla zo šestná stkovej do desiatkovej sú stavy. 

12. *Vytvorte program na zobrazenie podielu dvoch prirodzený ch čísel v tvare: 

{celá  časť},{predperióda}{(perióda)}. 

Napr.: 3 / 2 = 1,5(0);  7 / 11 = 0,(63);  11 / 7 = 1,(571428)  31 / 60 = 0,51(6) 

(Platí veta: Predperióda je tvorená  toľký mi ciframi, koľkokrá t je deliteľ deliteľný  číslami 10, 5 a 2; 

napríklad deliteľ 60 je deliteľný   10 a 2). 

13. Vytvorte program, ktorý  zobrazí prevodnú  tabuľku stupňov Celzia na Kelviny. Tabuľka začína 00C, 

končí hodnotou 100
0
C s krokom po 100C. Tabuľka má  mať hlavičku s nadpismi a pokú ste sa o rá mček 

zo semigrafický ch znakov ( write(chr(...)) ). 

14. Analogicky ako v predchá dzajú com príklade vytvorte prevodnú  tabuľku uhlové stupne - radiá ny. 

15. Vytvorte program , ktorý  vypočíta dĺžku lomenej čiary určenej bodmi so sú radnicami [ x,y ]. Na vý počet 

dĺžky ú sečky ( vzdialenosti dvoch bodov v rovine ) použite Pythagorovu vetu. 

16. Vytvorte program na ná jdenie najväčšieho spoločného deliteľa troch prirodzený ch čísel. 

17. Vytvorte program na vykreslenie grafu napr. funkcie sínus  s osou x zvisle. 

18. Vytvorte program na zistenie , či zadané prirodzené číslo je  dokonalé. Číslo N je dokonalé, ak sú čet 

všetký ch jeho deliteľov menších ako N sa rovná  N. Dokonalé je napríklad číslo 6. 

19. Vytvorte obmenu programu CRP-5, keď budú  postupne padať ná hodne generované veľké písmená  od 

prvého po 80. stĺpec. Na konci hry nech sa zobrazí ú spešnosť v percentá ch (počet „zachytený ch“ 

písmen). Ď alšie varianty hry: 

• padanie sa bude postupne zrý chľovať, napr. po „zachytení“ 5 písmen za sebou 

• pá d doplňte zvukový mi efektami 

• hru doplňte ponukou: začiatočník – pokročilý  – expert – koniec  

• hru nech možno kedykoľvek ukončiť stlačením klá vesu Esc 

• vytvorte „nekonečný “ cyklus, keď po 80. stlpci pokračuje hra automaticky v 1. stĺpci 

• písmená  nech padajú  v ná hodne generovaný ch stĺpcoch 

• padajú ce písmená  nech menia farbu 

• atď. 

 



 Zá klady programovania  - Turbo Pascal   59 

Procedúry 
 Každý  problém (okrem najjednoduchších) možno rozložiť na čiastkové podproblémy. Ich 

vyriešením a vykonaním v sprá vnom poradí, dostá vame riešenie daného problému. Programovacie jazyky na 

riešenie „podú loh“ ponú kajú  podprogramy. Podprogram je relatívne samostatná  programová  jednotka 

(vý stavbou podobná  programu) riešiaca čiastkový  problém. Podprogramy používame najmä: 

• ak chceme sprehľadniť program - uvedením riešení jeho čiastkový ch problémov v podprogramoch 

(každý  poriadny program sa „rozpadá “ minimá lne na zadanie vstupný ch hodnôt, vý počet a vý stup 

získaný ch vý sledkov) alebo  

• ak potrebujeme vykonať rovnaký  „vý počet“1 viackrá t s rôznymi vstupný mi hodnotami (toto je praktický  

dôvod, aby sme neopisovali alebo nekopírovali tie isté príkazy viackrá t v programe všade tam, kde 

potrebujeme vykonať rovnakú  postupnosť príkazov). 

Programovací jazyk TP má  dva druhy podprogramov, procedú ry a funkcie. V tejto kapitole sa 

budeme venovať procedú ram. 

 Procedú ry môžu byť: 

• bez loká lnych objektov a bez parametrov, 

• s loká lnymi objektami a bez parametrov a 

• s parametrami. 

Procedú ry bez lokálnych objektov a bez parametrov 
Deklará cia procedúry bez loká lnych objektov a bez parametrov má  tvar: 

 procedure mp; { hlavička procedú ry } 

  begin 
 p1; 
 p2;  
 ...    
 pn 
 end; 

kde mp je meno procedú ry – identifiká tor a p1, p2 až pn sú  príkazy. 

Napríklad: procedure VSTUP; 

 begin 
 write(´Zadaj dve čísla: )́; 
 readln(A,B); 
 end; 

  procedure VYMENA; 

 begin 
 POM:=A; A:=B; B:=POM 
 end; 

  procedure VYSTUP; 

 begin 
 writeln(A,’ ’,B) 
 end; 

                                                 
1 V sú časnosti sa už počítače na klasické vý počty používajú  minimá lne, častejšie na vyhľadá vanie alebo triedenie dá t. 

príkazová  časť 

Procedú ra bez loká lnych objektov a bez parametrov používa len globá lne objekty (zavedené 

v nadradenej časti), hovoríme aj, že komunikuje s okolím len pomocou globá lnych premenný ch. Na mieste, 

kde chceme, aby došlo k vykonaniu príkazov uvedený ch v procedú re, stačí uviesť meno procedú ry - 

hovoríme o tzv. volaní procedú ry. Napríklad v príkazovej časti hlavného programu:  

BEGIN 
 clrscr; 
 VSTUP; 
 VYMENA; 
 VYSTUP; 
 readln 

END. 

Vidíme, že volanie procedú ry je na ú rovni príkazu (mená  procedú r sme použili ako nami definované 

príkazy). Všetky použité premenné musia byť deklarované v ú seku definícií a deklará cií hlavného programu. 

Ako v celom TP, aj pri podprogramoch platí, že každý  objekt musí byť najprv definovaný  alebo deklarovaný  

a až potom ho môžeme použiť. 

Procedú ry s lokálnymi objektami a bez parametrov 
 Keď sa pozrieme na procedú ru VYMENA, ľahko zistíme, že premennú  POM potrebujeme len počas 

vykoná vania tejto procedú ry. Taký chto objektov môže byť viacej a nie je dôvod zaťažiť pamäť počítača 

počas behu celého programu vyhradením pamäťový ch miest objektom, ktoré používame len loká lne. Preto 

takéto objekty stačí definovať a deklarovať len v danej procedú re, hovoríme, že sú  loká lne, čo znamená , že 

sú  použiteľné len v danej procedú re (alebo v podriadený ch procedú rach). 

Deklará cia procedúry s loká lnymi objektami má  tvar: 

 procedure mp; 

 ú sek definícií a deklará cií   

 príkazová  časť 

kde mp je meno procedú ry. 
 
Pamäťovo efektívnejší zá pis procedú ry VYMENA (s loká lnou premennou POM): 

  procedure VYMENA; 

 var POM : integer; 

 begin 
 POM:=A; A:=B; B:=POM 
 end; 

Volanie procedú ry s loká lnymi objektami je rovnaké ako procedú ry bez loká lnych objektov. Procedú ra 

naďalej komunikuje s okolím len cez globá lne premenné. Ak použijeme rovnaké pomenovanie pre loká lnu aj 

globá lnu premennú , dôjde k tzv. zatieneniu globá lnej premennej loká lnou premennou v danej procedú re, čo 

znamená , že daná  globá lna premenná  je nepoužiteľná  v danej procedú re.  

blok 



 Zá klady programovania  - Turbo Pascal   61 

Procedú ry s parametrami 
 Ak by sme procedú ru POM chceli použiť viackrá t na vý menu hodnôt rôzne označený ch premenný ch 

(nie len A a B), najvý hodnejšie by bolo použiť parametre. Parametre umožňujú  efektívne komunikovať 

procedú re so svojim okolím, umožňujú  hodnoty do procedú ry dová žať prípadne aj vyvá žať. Pri písaní 

(deklarovaní) procedú ry nemusíme poznať hodnoty, ktoré budú  do procedú ry napr. dovezené, dokonca ani 

len označenie premenný ch, ktoré sa použije pri volaní procedú ry. Musíme však poznať počet parametrov 

a ich typ. Preto deklará ciu procedú ry píšeme s tzv. formá lnymi parametrami. 

Deklará cia procedúry s parametrami má  tvar: 

procedure mp (šfp1; šfp2; ... ; šfpn); 

 blok; 

kde mp je meno procedú ry a šfp1 až šfpn sú  špecifiká cie formá lnych parametrov. 

Napríklad:  procedure VYMENA ( var X , Y : integer ); 

  procedure NAJVACSIE ( A , B , C : integer; var MAX : integer ); 

  procedure ZASIFRUJ ( TEXT : string; POSUN : integer; var KOD : string ); 

Ak potrebujeme hodnoty do procedú ry len doviesť, špecifikujeme parametre nahradzované hodnotou. 

Š pecifiká cia parametrov nahradzovaných hodnotou má  tvar: 

  fp1, fp2, ... , fpn : tfp 

kde fp1 až fpn sú  identifiká tory formá lnych parametrov a tfp je identifiká tor ich typu. 

V príklade hlavičiek procedú r vyššie sú  parametre nahradzované hodnotou: A, B, C, TEXT a POSUN. 

Ak potrebujeme hodnoty z procedú ry aj vyviesť, špecifikujeme parametre nahradzované referenciou 

(odkazom).  

Š pecifiká cia parametrov nahradzovaných referenciou má  tvar: 

  var fp1, fp2, ... , fpn : tfp 

kde fp1 až fpn sú  identifiká tory formá lnych parametrov a tfp je identifiká tor ich typu. 

V príklade hlavičiek procedú r vyššie sú  parametre nahradzované odkazom: X, Y, MAX a KOD. 

 Pri volaní procedú ry s parametrami za formá lne parametre dosadzujeme skutočné parametre. Ich 

počet, poradie a typy musia sú hlasiť s formá lnymi parametrami. 

Napríklad:  VYMENA ( A , B );    VYMENA ( X , Y ); 

  NAJVACSIE ( 5 , -3 , 7 , MAXIMUM ); NAJVACSIE (  X , Y , Z , Q ); 

  ZASIFRUJ ( ´CEZAR ,́ 1 , POSLI );  ZASIFRUJ ( SPRAVA , N , KOD ); 

Pri volaní procedú ry môžu byť skutočné parametre nahradzované hodnotou konkrétne hodnoty (tieto 

hodnoty chceme do procedú ry len „doviesť“); pri ná hrade referenciou to musia byť premenné, lebo len tak sa 

môžu získané hodnoty „vyviesť“. Pri ná hrade hodnotou sa hodnoty skutočný ch parametrov pri volaní 

procedú ry len odovzdajú  formá lnym parametrom. Pri ná hrade referenciou sa skutočné parametre, počas 

vykoná vania príkazov v procedú re, stotožnia s formá lnymi parametrami – všetky zmeny vykonané 

s formá lnymi parametrami sa vykonajú  aj na skutočný ch parametroch – preto sa hodnoty z procedú ry aj 

„vyvezú “. Pre skutočné parametre môžeme, ale nemusíme, použiť rovnaké označenie ako pre formá lne 

parametre. 

 Praktické použitie procedú r si uká žeme po prebratí jednorozmerného poľa. 

Funkcie 
 Funkcia je špeciá lnym prípadom procedú ry. Procedú ru môžeme zapísať ako funkciu, keď jej 

vý sledkom je jedna jednoduchá  hodnota alebo typ string. 

Deklará cia funkcie má  tvar: 

  function mf ( šfp1; sfp2; ... ; šfpn ) : tvf; 

 blok; 

kde mf je meno funkcie – identifiká tor, šfp1 až šfpn sú  špecifiká cie formá lnych parametrov a tvf je typ 

vý sledku funkcie, ktorý  musí byť jednoduchý  typ alebo typ string. 

Napríklad: function MOCNINA ( X : real; N : integer ) : real; 

  function NACHADZA_SA : boolean; 

  function MAXIMUM ( A , B , C : integer ) : integer; 

Príklad: Vytvorte podprogram na vý počet mocniny xn, kde x je reá lne číslo a n prirodzené číslo. 

Analý za: Keďže vý sledkom vý počtu je jedna jednoduchá  hodnota, môžeme použiť aj funkciu. Na vý počet 

mocniny použijeme cyklus so zná mym počtom opakovaní: xn =  x.x.x. ... .x. 

  function MOCNINA (X : real; N : integer ) : real; 

 var MOC : real; 
   I : integer; 
 begin 
 MOC := 1; 
 for I:=1 to N do 
  MOC := MOC * I; 
 MOCNINA := MOC 
 end; 

Vý sledok sa z funkcie vyvá ža cez meno funkcie, preto identifiká tor mena funkcie sa musí aspoň raz 

vyskytnú ť na ľavej strane príkazu priradenia v tele danej funkcie. Aktivizá cia funkcie (vykonanie príkazov 

v tele funkcie) sa deje uvedením tzv. zá pisu funkcie (mena funkcie a skutočný ch parametrov) a nie je na 

ú rovni príkazu. Prakticky to znamená , že zá pis funkcie sa uvedie na mieste, kde chceme dosadiť vý sledok 

funkcie.  

Napríklad: writeln ( ´Vý sledok vý počtu mocniny: ,́ MOCNINA ( ZAKLAD , EXPONENT ) : 12 : 10 ); 

  VYSLEDOK := MOCNINA ( X , 3 ); 

 

Príklad FNK-1: Vytvorte program na vý počet počtu kombiná cií C(k,n) k-tej triedy z n prvkov 0≤k≤n. 

Analý za: Platí vzorec C(k,n) = 
n!

(n-k)!.k! . Vidíme, že vo vzorci sa trikrá t opakuje ten istý  vý počet – vý počet 

faktoriá lu (n! = 1.2.3. ... .n), len s rôznymi vstupný mi hodnotami (n, n-k a k). Preto je vý hodné použiť 

podprogram na vý počet faktoriá lu a keďže vý sledkom  je jedna jednoduchá  hodnota, konkrétne funkciu. 
program KOMBINACIE; 
uses  crt; 
const Esc=chr(27); 
var   N,K:integer; 
 
function KOMB(N,K:integer):integer; 
 
  function FAKT(N:integer):longint;  {lokalna funkcia} 
    var I:integer; 
        F:longint; 

    begin 
    F:=1; 
    for I:=1 to N do 
      F:=F*I;    
    FAKT:=F 
    end; 
 
  begin {príkazová časť funkcie KOMB } 
  KOMB:=FAKT(N) div FAKT(N-K) div FAKT(K) 
  end; 
 
BEGIN 
clrscr; 
writeln(’KOMBINACIE, KONIEC - Esc’:50); 
repeat 
  write(’Zadaj n a k ( n>=k>=0 ): ’);readln(N,K); 
  writeln(’C (’,N,’,’,K,’) = ’,KOMB(N,K)); 
until readkey=Esc; 
END. 
Program je skôr akademický , pretože hodnota n! už pre n = 13 prekročí maximá lnu hodnotu typu longint 

(13! = 6 227 020 800). 

 
Tak ako procedú ry, aj funkcie si precvičíme až v ďalších kapitolá ch. 

 
Rekurzia 
 Niektoré programovacie jazyky, vrá tane TP, umožňujú  používať veľmi elegantnú  programovaciu 

techniku, tzv. rekurziu. Ak procedú ra alebo funkcia vo svojom tele volá  samu seba, hovoríme o priamej 

rekurzii; ak napríklad procedú ra A volá  procedú ru B a naopak, hovoríme o nepriamej rekurzii. 

Príklad REK-1: Vytvorte rekurzívnu funkciu na vý počet n! 

Analý za: Vyjdeme z tzv. rekurentnej definície n-faktoriá lu: 

1. 0! = 1 
2. n! = n.(n-1)! pre n>0 

Ako vidieť z bodu 2, na vyčíslenie n! potrebujeme vykonať sú čin n krá t (n-1)!, pričom vý počet (n-1)! 

znamená  rovnaký  vý počet ako n!, len s hodnotou n zmenšenou o 1. 

 function FAKT ( N : integer) : longint; 
 begin 
 if N = 0 
 then FAKT := 1     { nerekurzívna vetva } 
 else FAKT := N * FAKT ( N-1 )  { rekurzívna vetva } 
 end; 

Keďže z príkladu FNC-1 vieme, že klasický  vý počet n! obsahuje cyklus, musí byť aj v príklade REK-1 

niekde „schovaný “. K cyklu ná s privedie poznanie, že počítač, ak nevie nejaký  vý raz, pre jeho zložitosť, 

vyčísliť, odkladá  požiadavky na operá cie do zá sobníka a vracia sa k nim, keď ich už vie finalizovať. Preto 

pri vý počte napríklad 5! počítač odloží najprv požiadavku 5 krá t „čosi“ (4! nepozná ), potom 4 krá t, 3 krá t, 2 

krá t, 1 krá t „čosi“, ale tu už „čosi“ je 0! a to je 1 (nerekurzívna vetva). Preto sa počítač vracia a vykoná va 

odložené operá cie, t.j. 1.1.2.3.4.5 čím získa vý sledok 120. Preto musíme vždy myslieť aj na ukončenie 

rekurzie, ukončenie volania danej procedú ry alebo funkcie, čo zabezpečí nerekurzívna vetva. 

Rekurzívny vý počet je časovo aj pamäťovo ná ročnejší ako iteračný  vý počet (pomocou cyklu), 

niektoré problémy však rieši veľmi elegantne a relatívne jednoducho. 

Príklad REK-2: Vytvorte rekurzívnu funkciu na vý počet mocniny xn, x reá lne číslo, n prirodzené číslo 

vrá tane nuly. 

Analý za: Potrebujeme znova poznať z matematiky rekurentnú  definíciu mocniny, ktorá  hovorí: 

Pre x ≠ 0: 
1. x0 = 1  

2. xn = x . xn-1  pre n>0 

Vidíme, že v bode 2 má me na ľavej aj pravej strane rovnaký  „problém“ – vý počet mocniny. 

  function MOCNINA ( X : real; N : word ): real; 
 begin 
 if N = 0 
 then MOCNINA := 1 
 else MOCNINA := X * MOCNINA ( X , N-1 ) 
 end; 

Parameter X nie je nevyhnutný , X môže byť aj globá lna premenná . Nezabudnite, že 00 nie je definované 

(ošetriť hneď na vstupe hodnôt X a N). 

Príklad REK-3*: Vytvorte program riešiaci problém Hanojské veže. 

Analý za: Ako uká žku sily a princípu prá ce rekurzie uvedieme zná my problém nazvaný  Hanojské veže. 

V starobylom klá štore ukrytom v ďalekom kú te Ázie vraj stoja tri zlaté kolíky. Na prvom z nich je 

nasunutý ch 64 diskov rôznych veľkostí, a to tak, že najväčší leží dole, nad ním leží o niečo menší, nad ním 

ešte menší atď. Ú lohou mníchov je premiestniť tú to vežu na druhý  kolík, pričom platí: 

• naraz sa môže premiestniť iba jeden disk 

• väčší disk sa nikdy nesmie položiť na menší 

• ktorý koľvek z troch kolíkov možno použiť ako „odkladací“ na dočasné umiestnenie disku 

Podľa legendy svet zanikne vo chvíli, keď mnísi svoju ú lohu splnia. 

Obrá zok sme nakreslili pre tri disky (A, B, C). Stojany (kolíky) sú  označené 1, 2 a 3 (skutočné parametre). 

Pre označenie kolíkov sme zvolili premenné ODKIAL, KAM a POMOCOU (formá lne parametre). 

 
Myšlienkový  postup pre N diskov je nasledovný :  

Aby sme najväčší disk (v obrá zku označený  A) mohli presunú ť z kolíka 1 na kolík 2, musíme: 

1. najprv presunú ť nad ním ležiace disky, t.j. vežu zloženú  z N-1 diskov, na kolík 3 pomocou kolíka 2, 

2. potom môžeme disk A premiestniť z kolíka 1 na kolík 2 a  

3. nakoniec presunú ť vežu zvyšný ch N-1 diskov z kolíka 3 na kolík 2 pomocou kolíka 1. 

Myšlienka použiť rekurziu by ná m mohla napadnú ť, ak si uvedomíme, že problém preniesť vežu s N diskami 

obsahuje v sebe problém preniesť vežu s N-1 diskami (ďalej už „len“ preložiť disk a naň vrá tiť vežu s N-1 

diskami).  

Tu je program: 
program HANOJ; 
uses crt; 
var  POCET:integer; 
 
procedure PRENES_VEZU(N,ODKIAL,KAM,POMOCOU:integer); 
 
  procedure PRENES_DISK(Z,NA:integer); 
    begin 
    writeln(Z,’ -> ’,NA) 
    end; 
 
begin 
if N>0 
  then begin 
       PRENES_VEZU(N-1,ODKIAL,POMOCOU,KAM); {1} 
       PRENES_DISK(ODKIAL,KAM); 
       PRENES_VEZU(N-1,POMOCOU,KAM,ODKIAL); {2} 
       end; 
end; 
 
BEGIN 
clrscr; 
write(’POCET DISKOV: ’); 
readln(POCET); 
PRENES_VEZU(POCET,1,2,3); 
readln; 
END. 
Jeho jednoduchosť asi prekvapí každého. Ú spešnosť rekurzie spočíva v tom, čo sme uviedli v príklade 

REK-1, totiž, že ak počítač „nevie niečo vypočítať“, „odloží to“ do zá sobníka. Tak počítač odkladá  

PRENES_VEZU pre čoraz menej diskov, aktualizuje ODKIAL, KAM, POMOCOU – uvedomte si, ktoré 

parametre sú  formá lne a ktoré skutočné(!), až niet čo preniesť a pokračuje ďalším príkazom PRENES_DISK 

atď. Zložitosť vý počtu by mala byť zrejmá  z vý pisu, ktorý  sme získali po doplnení vyššie uvedeného 

programu príkazmi na vý pis, pričom prvé volanie procedú ry PRENES_VEZU vo vetve then sme označili 

číslom 1 a druhé číslom 2. 
POCET DISKOV = 3 
Prenes vezu s 3 disk. pomocou volania 0 (prvé volanie z hlavného programu) 
Prenes vezu s 2 disk. pomocou volania 1 
Prenes vezu s 1 disk. pomocou volania 1 
Prenes vezu s 0 disk. pomocou volania 1 
1 -> 2 
Prenes vezu s 0 disk. pomocou volania 2 
1 -> 3 
Prenes vezu s 1 disk. pomocou volania 2 
Prenes vezu s 0 disk. pomocou volania 1 
2 -> 3 
Prenes vezu s 0 disk. pomocou volania 2 
1 -> 2 
Prenes vezu s 2 disk. pomocou volania 2 
Prenes vezu s 1 disk. pomocou volania 1 
Prenes vezu s 0 disk. pomocou volania 1 
3 -> 1 
Prenes vezu s 0 disk. pomocou volania 2 
3 -> 2 
Prenes vezu s 1 disk. pomocou volania 2 

Prenes vezu s 0 disk. pomocou volania 1 
1 -> 2 
Prenes vezu s 0 disk. pomocou volania 2 

Nerieš ené  úlohy na rekurziu: 

4. Vytvorte program na vý počet najväčšieho spoločného deliteľa (NSD) dvoch prirodzený ch čísel pomocou 

rekurzie, ak platí:   A   ak A=B 

NSD(A,B) = NSD(A-B,B) ak A>B 

    NSD(A,B-A) ak B>A 

5. Vytvorte program na vý počet najväčšieho spoločného deliteľa (NSD) dvoch prirodzený ch čísel pomocou 

rekurzie, ak platí:   A+B   ak A=0 alebo B=0 

NSD(A,B) =  NSD(A mod B,B) ak A>B 

   NSD(A,B mod A) ak B>A 

6. Objasnite prá cu rekurzívnej procedú ry OTOC: 
program ZRKADLO; 
uses crt; 
procedure OTOC; 
  var z:char; 
  begin 
  read(z); 
  if z<>’.’ 
    then OTOC 
    else writeln; 
  write(z) 
  end; 
BEGIN 
clrscr; 
writeln(’Veta konci bodkou !’:48);write(’ ’); 
OTOC; 
readkey 
END. 

7. Vytvorte program na vý počet ľubovoľného Fibonacciho čísla pomocou rekurzívnej a nerekurzívnej 

(iteračnej) funkcie a porovnajte časy vý počtu - urobte zá ver. 

 Fib. postupnosť: 0, 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144,... 

 vzorcami:  F0 = 0 , F1 = 1 

   Fi = Fi-1 + Fi-2 pre i ≥ 2 

8. Vytvorte podprogram na vý počet ľubovoľného člena Pascalovho trojuholníka, po zadaní nezá porný ch 

celý ch čísel n a k , ak platí: 

B(n, 0) = B(n, n) = 1 

B(n, k) = B(n-1, k-1) + B(n-1, k)  pre 1 ≤ k ≤ n 

9. Vytvorte funkciu CISLICA(n,k), ktorej vý sledkom je k-ta číslica sprava prirodzeného čísla n. 

10. *Vytvorte program, ktorý  zadané prirodzené číslo n rozloží na prvočinitele, t.j. na prvočísla c1, c2, ...,ck , 

pričom platí: n = c1 . c2 . ... . ck. 
11. *Vytvorte program, ktorý  zistí, na koľko častí sa rozpadne štvorčekový  papier po vystrihnutí niektorý ch 

štvorčekov. 

12. *Vytvorte program na zistenie všetký ch možný ch spôsobov vyplatenia zadanej sumy pri zná mych 

hodnotá ch použitý ch mincí (počet mincí rôznych hodnôt je neobmedzený ). 

Jednorozmerné pole 
 Predstavme si, že chceme vypočítať napríklad aritmetický  priemer z veľkého počtu čísel a vý stup má  

byť v tvare: Aritmetický  priemer z čísel: vypísané všetky čísla je .... Na uchovanie väčšieho počtu ú dajov 

počas vykoná vania programu s doteraz prebraný mi ú dajový mi typmi nevystačíme. Pomôže ná m nový  

ú dajový  typ pole. 

 Konečnú  skupinu rovnaký ch ú dajov (čísel, reťazcov a pod.) môžeme v programovaní vý hodne uložiť 

do jednej štruktú rovanej premennej, do poľa. Pole sa skladá  z pevného počtu zložiek rovnakého typu, pričom 

každú  zložku možno sprístupniť pomocou indexu. Rôzne polia môžu obsahovať rôzne počty prvkov a aj ich 

hodnoty môžu byť rôznych typov (raz sú  to čísla typu integer, druhý krá t reťazce typu string atď.). Preto 

rôzne polia predstavujú  rôzne typy poľa, ktoré sa líšia počtom prvkov, typom zložky alebo typom indexu. 

Aby počítač vedel, akú  veľkú  pamäť má  vyhradiť pre hodnoty premennej typu pole (skrá tene pre pole), 

musíme v programe najprv typ pole definovať.  

Definícia typu pole má  tvar: 

 type mt = array [ ti ] of tz 

kde mt je meno typu pole – identifiká tor, ti je typ indexu – musí byť ordiná lny typ a tz je typ zložky poľa. 

Ako typ indexu sa najčastejšie používa tzv. interval, definovaný : min .. max, kde min a max sú  hodnoty toho 

istého ordiná lneho typu (medzi min a max sú  prá ve dve bodky!). 

Napríklad:      Pole vľavo uvedeného typu bude obsahovať: 

type POLE1 = array [ 1..20 ] of integer;   20 čísel typu integer s indexami 1, 2, ..., 20 

type POLE2 = array [ -100..100 ] of  boolean;  201 hodnôt false alebo true s indexami -100, -99, ... 

type POCET = array [ ´A .́.´Z  ́] of byte;  26 celý ch čísel od 0 do 255 s indexmi ´A ,́ ´B ,́...,´Z  ́

type tVELA = array [ byte ] of string;   256 reťazcov s indexami 0, 1, 2 až 255 

type HLUPOST = array [ boolean ] of boolean; dva prvky s hodnotami false alebo true s indexami 

false a true 

Uvedomte si, že sú časťou definície typu pole je aj pomenovanie daného typu. Ide o ú dajový  typ s nami 

definovaný m rozsahom, obsahom, indexami aj menom. 

 V ú seku deklará cií premenný ch deklarujeme premenné typu pole rovnaký m spôsobom ako iné 

premenné. Napríklad zá pis: var A : POLE1 zavá dza pole s ná zvom A  typu POLE1. Sprístupnenie 

jednotlivý ch zložiek poľa A  sa realizuje zá pisom A [ index ], kde index je indexový  vý raz a musí byť 

rovnakého typu, ako je typ indexu z definície typu pole. Zá pisu A [index ] hovoríme indexovaná  premenná  

a jej typ je určený  typom zložky daného poľa. 

Napríklad:  

v poli A typu POLE1 znamená  zá pis:  A [1]   prvú  zložku poľa A s hodnotou napríklad -7 

     A [2]  druhú  zložku poľa A  

     A [I]  i-tu zložku poľa A 

     A [20] poslednú  zložku poľa A 

v poli B typu POLE2 znamená  zá pis: B [-100]  prvú  zložku poľa B s hodnotou napríklad true 

     B [0] 101. zložku poľa B  

v poli C typu POCET znamená  zá pis: C [´A ]́ prvú  zložku poľa C s hodnotou  napríklad 0 

     C [´Z ]́ poslednú  zložku poľa C 

 Ak sú  dve polia rovnakého typu, napríklad označené A a B, možno ich hodnoty odovzdá vať naraz, 

napríklad príkazom B := A. 

 Premenná  typu pole môže byť deklarovaná  aj bez definície príslušného typu. Pre pole s ná zvom A by 

deklará cia mala tvar: 

var A : array [ ti ] of tz 

kde ti a tz majú  rovnaký  vý znam ako v definícii typu pole. Najprv definovať typ poľa a až potom deklarovať 

premennú  daného typu má  niekoľko vý hod a preto odporú čame typ poľa najprv definovať. Napríklad 

v špecifiká cii formá lneho parametra v procedú re je dovolené na mieste typu parametra uviesť len meno poľa 

a nie zá pis array [ ti ] of tz. 

 Pre programy s typom pole odporú čame používať podobný  ú sek definícií a deklará cií ako uvá dzame 

nižšie: 

program ...; 

uses   crt; 

const MAXPP = 20;  { definujeme konštantu udá vajú cu maximá lny počet zložiek - prvkov poľa } 

type   POLE = array [ 1..MAXPP ] of integer; 

var     A : POLE; 
          N, I : integer;  { N bude aktuá lny počet prvkov poľa A } 

 

Načítanie prvkov poľa je čiastkový  problém, ktorý  je vhodné vyriešiť v procedú re. Ak v celom 

programe pracujeme len s jedný m poľom, môžeme použiť procedú ru bez parametrov, t.j. pracovať 

s globá lnymi premenný mi. Ak v programe pracujeme s viacerý mi poliami, treba použiť parametre (uká žeme 

si v príklade neskôr). Princípom vstupu prvkov poľa je zapísať prvú  zadanú  hodnotu ako 1. prvok poľa, 

druhú  ako 2. prvok poľa, atď. až po zá pis posledného N-tého prvku poľa. N-krá t sa opakuje rovnaká  činnosť 

– zá pis do poľa, preto použijeme cyklus so zná mym počtom opakovaní – príkaz for. 

Príklad procedú ry riešiacej načítanie prvkov globá lneho poľa A s počtom prvkov N: 

procedure VSTUP; { bez parametrov } 

begin 
write(´Počet prvkov poľa: )́; 
readln(N); 
for I := 1 to N do     { Premenná  I bude postupne nadobú dať hodnoty 1, 2, 3, ..., N } 

begin 
write(I:2, .́ prvok: )́;     { Zá pis I:2 spôsobí, že jednotky budú  pod jednotkami a desiatky pod desiatkami } 
readln(A[I])        
end 

end; 

 Aj zobrazenie prvkov poľa je čiastkový  problém riešiteľný  v procedú re. Jednotlivé hodnoty poľa 

môžu byť na samostatný ch riadkoch (nevý hodné), oddelené medzerami alebo čiarkami. 

Príklad procedú ry slú žiacej na zobrazenie prvkov globá lneho poľa A oddelený ch medzerami 

(predpokladá me, že vý stupné hodnoty majú  najviac 7 znakov): 

procedure VYSTUP; 

begin 
for I := 1 to N do 

write(A[I]:8); { keďže je v riadku 80 znakov, v každom bude zobrazený ch do desať hodnôt } 
writeln 
end; 

Príklad procedú ry slú žiacej na zobrazenie prvkov globá lneho poľa A oddelený ch čiarkami: 

procedure VYSTUP; 

begin 
for I := 1 to N-1 do  { čiarky majú  byť za prvou až predposlednou hodnotou } 

write(A[I], ,́ )́; 
writeln(A[N]) 
end; 

 

Príklad POLE-1: Vytvorte program, ktorý  načíta najviac 30 celý ch čísel z intervalu 0 až 999 a zobrazí ich. 

Analý za: Požadovaný  program stačí doslova poskladať z predchá dzajú cich procedú r, len s malý mi ú pravami: 

MAXPP treba zväčšiť na 30 a typ zložky poľa môže byť byte (potom v prvej procedú re VYSTUP vystačíme 

aj s formá tom A[I]:4). Nezná ma je už len príkazová  časť hlavného programu: 

BEGIN 
clrscr; 
VSTUP;   { podľa „chuti“ môžete doplniť príkaz clrscr; } 
writeln(´Prvky pola: )́; 
VYSTUP; 
readln 
END. 

Odporú čame uložiť si tento program do sú boru s ná zvom POLE a má me zá klad pre mnoho programov 

pracujú cich s poľom – každý  by predsa mal mať vstup a vý stup prvkov poľa. Po otvorení sú boru POLE.PAS 

ho treba cez položku File – Save as... premenovať na aktuá lny ná zov a doplniť o procedú ru riešiacu zadanú  

ú lohu. 

 Často prvky poľa môžu byť aj ná hodne počítačom vygenerované čísla, stačí, ak zadá me len ich 

počet. Procedú ra VSTUP_NAHODNE za ná s „vyberie“ prvky poľa: 

procedure VSTUP_NAHODNE; 

begin 
write(´Počet prvkov poľa: )́; 
readln(N); 
for I := 1 to N do 

A[I]:=random(10)  { pole A bude obsahovať celé čísla od 0 po 9 } 
end; 

V hlavnom programe za BEGIN treba vložiť príkaz randomize, aby sme zná hodnili vý ber prvkov poľa po 

každom spustení programu. 

 

 Prá cu s poľom si precvičíme najmä na činnosti nazvanej vyhľadá vanie. Vyhľadá vanie v poli je 

činnosť s cieľom zistiť, či zložka so zadaný mi vlastnosťami sa nachá dza v poli. Cieľom môže byť aj zistenie, 

koľkokrá t sa zložka so zadaný mi vlastnosťami nachá dza v poli, na ktorý ch miestach, prípadne požiadavka, 

ak sa zložka v poli nenachá dza, vložiť ju za posledný  prvok poľa a pod. 

Príklad POLE-2: Vytvorte podprogram na zistenie najväčšieho prvku daného poľa. 

Analý za: Maximum hľadajme v poli celý ch čísel. Na začiatku hľadania je najväčším prvkom prvý  prvok 

poľa (jeho hodnotu uložíme do premennej MAX). Potom musíme zistiť, či druhý  prvok poľa nie je väčší, ak 

á no, našli sme nové maximum (do premennej MAX musíme zapísať hodnotu druhého prvku poľa). Podobne 

porovná me tretí prvok poľa s MAX a ak treba aktualizujeme MAX (zmeníme hodnotu MAX na hodnotu 

tretieho prvku poľa), potom štvrtý  prvok, atď., až ký m neprejdeme celé pole. Prejsť celé pole od prvého resp. 

druhého po posledný  prvok znamená  použiť príkaz for. 
 
procedure MAXIMUM; 
  var MAX:integer;  { lokálna premenná MAX } 
  begin 
  MAX:=A[1];   { priradenie počiatočnej hodnoty premennej MAX } 
  for I:=2 to N do  { zabezpečí prejdenie celého poľa od druhého prvku} 
    if A[I]>MAX then MAX:=A[I]; 
  writeln(’Z cisel:’); 
  VYSTUP;     {zobrazí prvky poľa } 
  writeln(’je najvacsie ’,MAX); 
  end; 
BEGIN 
clrscr; 
VSTUP; 
MAXIMUM; 
readln 
END. 
Procedú ru MAXIMUM môžeme napísať aj ako funkciu, keďže vý sledkom danej procedú ry je jedna 

jednoduchá  hodnota. Funkciu je vý hodné použiť, ak chceme maximum vyviesť. 

Program od funkcie MAXIMUM: 
function MAXIMUM:integer; 
  var MAX:integer; 
  begin 
  MAX:=A[1]; 
  for I:=2 to N do 
    if A[I]>MAX then MAX:=A[I]; 
  MAXIMUM:=MAX;    {priradenie najväčšej hodnoty menu funkcie,} 
  end;     {cez ktoré sa vyvezie daná hodnota} 
BEGIN 
clrscr; 
VSTUP; 
VYSTUP; 
writeln(’je najvacsie ’,MAXIMUM); {zápis funkcie na mieste,kde chceme zobraziť} 
readln     {výsledok funkcie} 
END. 
 

Príklad POLE-3: Vytvorte podprogram na zistenie, koľkokrá t sa zadaný  prvok nachá dza v poli naviac 100 

celý ch čísel. 

Analý za: Zistiť, koľkokrá t sa zadaný  prvok nachá dza v poli znamená  prejsť celé pole a vždy keď sa 

prezeraný  prvok poľa rovná  hľadanému, zväčšiť počet vý skytov o jednotku. 
procedure ZISTI_POCET; 
  var HLPRVOK,POCET:integer; 
  begin 
  write(’Zadaj prvok, ktoreho pocet vyskytov mam zistit: ’); 
  readln(HLPRVOK); 
  POCET:=0; 

  for I:=1 to N do 
    if A[I]=HLPRVOK then POCET:=POCET+1;      { analogicky inc(POCET); } 
  if POCET=0 
    then writeln(’Hladany prvok sa v poli nevyskytuje!’) 
    else writeln(’Hladany prvok sa v poli vyskytuje ’, POCET,’-krat.’) 
  end; 
BEGIN 
clrscr; 
VSTUP; 
ZISTI_POCET; 
readln 
END. 
Procedú ra môže byť opäť napísaná  aj ako funkcia, jej použitie vzhľadom na požiadavku zadať hľadaný  

prvok je trochu komplikovanejšia: 

function ZISTI_POCET(HLPRVOK:integer):integer; {parameter nahradzovaný hodnotou} 
  var POCET:integer; 
  begin 
  POCET:=0; 
  for I:=1 to N do 
    if A[I]=HLPRVOK then inc(POCET); 
  ZISTI_POCET:=POCET 
  end; 
 
BEGIN 
clrscr; 
VSTUP; 
write(’Zadaj prvok, ktoreho pocet vyskytov mam zistit: ’); 
readln(HLADANY_PRVOK);   {HLADANY_PRVOK musí byť globálna premenná} 
writeln(’Hladany prvok sa v poli vyskytuje ’, ZISTI_POCET(HLADANY_PRVOK),’-
krat.’); 
readln 
END. 
 

Príklad POLE-4: Vytvorte podprogram na zistenie, či sa prvok s danou hodnotou vyskytuje v poli. 

Analý za: Ký m doteraz sme museli prejsť celé pole, v tejto ú lohe môžeme prehľadá vanie poľa ukončiť 

po ná jdení hľadanej hodnoty v poli. Z algoritmického hľadiska to znamená  použiť časovo efektívnejší cyklus 

s nezná mym počtom opakovaní. Vý sledkom prehľadá vania má  byť len zistenie, či sa prvok v poli nachá dza 

(vyvezie sa hodnota true) alebo nenachá dza (vyvezie sa hodnota false). 
function NACHADZA_SA:boolean; 
  begin 
  I:=0; 
  repeat 
    inc(I) 
  until (A[I]=HLPRVOK) or (I=N); 
  NACHADZA_SA:=A[I]=HLPRVOK   { Tento zápis nahrádza príkaz if } 
  end; 
 
BEGIN 
clrscr; 
VSTUP; 
write(’Zistit vyskyt prvku s hodnotou: ’); 
readln(HLPRVOK); 
writeln(’Prvok s hodnotou ’,HLPRVOK,’ sa v poli:’); 
VYSTUP; 
if NACHADZA_SA 
  then writeln(’nachadza.’) 
  else writeln(’nenachadza.’); 
readln 
END. 

Príklad POLE-4b: Ako zaujímavosť uvá dzame rovnaký  problém – zistiť, či sa prvok s danou vlastnosťou 

vyskytuje v poli, pričom však vieme, že pole je utriedené. Pod utriedený m poľom rozumieme vzostupné 

utriedenie, t.j. najmenší prvok je prvý  a najväčší prvok je posledný , preto platí: A[I]≤A[I+1] pre všetky 

prípustné hodnoty I. Použiť by sme mohli aj predchá dzajú cu funkciu NACHADZA_SA, ktorá  používa tzv. 

lineá rne vyhľadá vanie – pole prehľadá va od prvého prvku prvok za prvkom, my však chceme časovo 

efektívnejší algoritmus (ak sa v tisíc prvkovom poli nachá dza hľadané číslo napr. na 999 mieste, lineá rne 

vyhľadá vanie musí prezrieť 999 prvkov, nasledujú ca funkcia však najviac 10!). 

Analý za: Princípom biná rneho vyhľadá vanie, lebo tak je toto vyhľadá vanie pomenované, je určenie stredu 

prehľadá vanej oblasti a zistenie, či sme našli hľadaný  prvok a ak nie, či sa nachá dza v jeho dolnej (ľavej) 

časti alebo v hornej (pravej) časti. Tento postup sa opakuje dovtedy, ký m sa nená jde hľadaný  prvok, alebo 

keď už nie je čo prehľadá vať (dolná  hranica prehľadá vanej oblasti je nad jej hornou hranicou). 
 
function NACHADZA_SA:boolean; 
  var DH,HH,S:integer; 
  begin 
  DH:=1; HH:=N; 
  repeat 
    S:=(DH+HH) div 2; 
    if A[S]<>HLPRVOK 
      then if A[S]<HLPRVOK 
             then DH:=S+1 
             else HH:=S-1 
  until (A[S]=HLPRVOK) or (DH>HH); 
  NACHADZA_SA:=A[S]=HLPRVOK 
  end; 
 

Ď alším typom ú loh môžu byť ú lohy zistiť, na ktorý ch miestach v poli sa vyskytuje prvok so zadanou 

vlastnosťou. Ú loha môže byť formulovaná : zistiť prvý  vý skyt, všetky vý skyty, posledný  vý skyt a pod.  

 

Príklad POLE-5: Vytvorte podprogram na ná jdenie prvého vý skytu prvku so zadanou vlastnosťou.  

Analý za: Ná jsť miesto vý skytu znamená  vlastne určiť index prvku, ktorý  má  požadovanú  vlastnosť (ak sú  

indexy poľa 1, 2, ..., N). Ak sa hľadaná  hodnota v poli nevyskytuje, nech sa vyvezie 0. Keďže ide o jednu 

jednoduchú  hodnotu, môžeme použiť funkciu. My sme sa rozhodli pre procedú ru, aby sme uká zali použitie 

procedú ry namiesto funkcie a aj preto, že funkčný  zá pis by sme museli použiť dvakrá t (všade tam v hlavnom 

programe, kde sme použili premennú  MIESTO), čo nie je efektívne. Ak sa chceme vyhnú ť viacná sobnému 

použitiu funkčného zá pisu s nezmenený mi hodnotami, môžeme vý sledok funkcie uložiť v hlavnom 

programe do premennej a ďalej pracovať len s touto premennou. V nasledujú cej časti programu sú  premenné 

HLPRVOK a MIESTO globá lne, nezabudnite ich deklarovať. 
 
procedure PRVY_VYSKYT; 
  begin 
  I:=0; 
  repeat 
    inc(I) 
  until (A[I]=HLPRVOK) or (I=N); 
  if A[I]=HLPRVOK 
    then MIESTO:=I 
    else MIESTO:=0 
  end; 

BEGIN 
clrscr; 
VSTUP; 
write(’Zistit miesto prveho vyskytu prvku s hodnotou: ’); 
readln(HLPRVOK); 
clrscr; 
writeln(’Prvok s hodnotou ’,HLPRVOK,’ sa v poli:’); 
VYSTUP; 
PRVY_VYSKYT; 
if MIESTO>0 
  then writeln(’nachadza na ’,MIESTO,’. mieste.’) 
  else writeln(’nenachadza.’); 
readln 
END. 

Veľmi zaujímavé sú  aj kombiná cie predchá dzajú cich ú loh, ako napríklad zistiť, či sa prvok 

s požadovanou vlastnosťou nachá dza v poli a ak á no, koľkokrá t, prípadne na ktorý ch miestach, prípadne aj 

koľkokrá t a na ktorý ch miestach. Tieto kombinované ú lohy väčšinou neznamenajú  riešiť každú  ú lohu 

v samostatnom cykle (neefektívne), ale pri jednom prechode poľom možno „vyriešiť“ všetky ú lohy. Ú lohy 

podobného typu sme dali medzi neriešené ú lohy, teraz si uveďme aspoň jednu zložitejšiu. 

 

Príklad POLE-6: Vytvorte program na zistenie najmenšieho prvku v poli, koľkokrá t sa vyskytuje a na 

ktorý ch miestach v poli. 

Analý za: Program nechá vame na „samoštú dium“. Všimnite si priradenie počiatočný ch hodnôt v procedú re 

MINIMUM a ako sme sa „pohrali“ s vý stupom (VYSTUP2). 
program POLE_6; 
uses  crt; 
const MAXPP=20; 
type  POLE=array[1..MAXPP] of integer; 
var   A,VYSKYT:POLE; 
      N,I,MIN,POCET:integer; 
 
procedure VSTUP; 
  begin 
  write(’Pocet prvkov pola: ’); 
  readln(N); 
  for I:=1 to N do 
    begin 
    write(I:2,’. prvok: ’); 
    readln(A[I]); 
    end; 
  end; 
 
procedure VYSTUP1; 
  begin 
  for I:=1 to N-1 do 
    write(A[I],’, ’); 
  writeln(A[N]); 
  end; 
 
procedure MINIMUM; 
  begin 
  MIN:=A[1]; POCET:=1; VYSKYT[1]:=1; { priradenie počiatočných hodnôt } 
  for I:=2 to N do 
    if A[I]<=MIN 
      then if A[I]=MIN  { prvok s hodnotou rovnou momentálnemu minimu } 
             then begin 
                  inc(POCET); 
                  VYSKYT[POCET]:=I 
                  end 

             else begin   { nové minimum } 
                  MIN:=A[I]; 
                  POCET:=1; 
                  VYSKYT[1]:=I 
                  end 
  end; 
 
procedure VYSTUP2; 
  begin 
  writeln(’je najmensim prvok ’,MIN); 
  write(’a vyskytuje sa ’,POCET,’-krat ’); 
  if POCET=1 
    then write(’na mieste: ’,VYSKYT[1]) 
    else begin 
         write(’na miestach: ’); 
         for I:=1 to POCET do write(VYSKYT[I],’  ’) 
         end 
   end; 
 
BEGIN 
clrscr; 
VSTUP; 
clrscr; 
writeln(’V poli:’); 
VYSTUP1; 
MINIMUM; 
VYSTUP2; 
readln 
END. 
 

 Ď alšiu techniku, alebo ak chcete „fintu“, ktorú  by mal ovlá dať každý  programá tor, je použitie tzv. 

ná razníka. Jeho princíp a použitie si uká žeme na nasledujú com príklade: 

Príklad POLE-7: Vytvorte podprogram, ktorý  zistí, či sa hľadaný  prvok nachá dza v poli, a ak nie, vloží ho 

za posledný  prvok. 

Analý za: Princíp použitia ná razníka spočíva vo vložení hľadaného prvku do poľa, v tomto prípade za 

posledný  prvok. To zabezpečí zastavenie prehľadá vania (prvok sa určite v takto upravenom poli nachá dza) 

najneskôr na ná razníku. Ak sa prehľadá vanie poľa zastaví skôr, hľadaný  prvok sa nachá dza už 

v pôvodnom poli. Ak sa hľadaný  prvok v poli nenachá dza, už ho má me vložený  za posledný  prvok, len 

nesmieme zabudnú ť zväčšiť počet prvkov poľa (hodnotu N). 
procedure NARAZNIK; 
  begin 
  A[N+1]:=HLPRVOK;       { vlozili sme naraznik } 
  I:=0; 
  repeat 
    I:=I+1 
  until A[I]=HLPRVOK; 
  if I=N+1 then N:=N+1   { prvok sa v poli nenachadzal, pridany dalsi prvok }  
  end; 
 
BEGIN 
clrscr; 
VSTUP; 
write(’Zistit vyskyt prvku s hodnotou: ’); 
readln(HLPRVOK); 
writeln(’Povodne pole:’); VYSTUP; 
NARAZNIK; 
writeln(’Nove pole:’); VYSTUP; 
readln 
END. 

 Na nasledujú com príklade „z trochu iného sú dka“ si uká žeme, aké je dôležité dobre si zvoliť typ 

pole. 

Príklad POLE-8: Vytvorte program, ktorý  po zadaní textu prekonvertujte malé písmená  v texte na veľké a 

zobrazí tabuľku počtu vý skytov jednotlivý ch písmen abecedy v texte. 

Analý za: Pole zvolíme tak, aby indexami poľa boli postupne všetky veľké písmená  abecedy a hodnotami 

daného poľa budú  čísla udá vajú ce počet vý skytov jednotlivý ch písmen v texte. 
program POLE_8; 
uses  crt; 
type  POLE=array[’A’..’Z’] of byte; 
var   TEXT:string; 
      POCET:POLE; 
 
procedure VSTUP; 
  procedure KONVERZIA; 
    var I:byte; 
    begin 
    for I:=1 to length(TEXT) do   
      TEXT[I]:=upcase(TEXT[I])   { malé písmená v TEXTe zmení na veľké } 
    end;   
  begin 
  write(’Zadaj text: ’); readln(TEXT); 
  KONVERZIA 
end; 
 
procedure ZISTI_POCTY; 
  var I:byte; 
  procedure VYNULUJ; 
    var ZNAK:char; 
    begin 
    for ZNAK:=’A’ to ’Z’ do 
      POCET[ZNAK]:=0 
    end; 
  begin 
  VYNULUJ; 
  for I:=1 to length(TEXT) do 
    if TEXT[I] in [’A’..’Z’] 
      then inc(POCET[TEXT[I]]);   { TEXT[I] predstavuje konkrétne písmeno textu }  
  end; 
 
procedure VYSTUP; 
  var ZNAK:char; 
  begin 
  writeln(’Znak          Pocetnost’:45); 
  for ZNAK:=’A’ to ’Z’ do 
    begin 
    writeln(ZNAK:24,POCET[ZNAK]:17); 
    if ZNAK=’W’ then begin readkey; clrscr end 
    end 
  end; 
 
BEGIN 
clrscr; 
VSTUP; 
ZISTI_POCTY; 
VYSTUP; 
readln 
END. 

Typové konš tanty 
 Typové konštanty môžeme prirovnať k premenný m, ktorý ch počiatočné hodnoty sú  uvedené už 

v ú seku definícií a deklará cií. Môžeme ich aj používať rovnako ako premenné takého istého typu. 

 Deklará cia typovej konštanty má  tvar: 

const identifiká tor_premennej : identifiká tor_typu = typová _konštanta  

Napríklad: 

const  SYM : boolean = TRUE; 

 stringANO : string[3] = ´ANO  ́ ; 

A : array[1..10] of integer = (15,2,-4,6,9,12,0,-7,6,11); 

CISLICE : array[0..9] of char = ( 0́ ,́ 1́ ,́ 2́ ,́ 3́ ,́ 4́ ,́ 5́ ,́´6 ,́ 7́ ,́ 8́ ,́ 9́ )́; 

CISLICE : array[0..9] of char = 0́123456789 ;́         { tá to deklará cia je totožná  s predchá dzajú cou } 

 

Príklad POLE-9: V príklade VCS-1 sme v programe použili na vý počet počtu dní do konca roka (dkr) 

„otrasný “ zá pis v príkaze case. Pri použití typovej konštanty poľa sa zá pis podstatne zjednoduší. Do uká žky 

sme pridali aj vý počet počtu dní od začiatku roka (ozr). 
const dvm:array[1..12] of 1..31=(31,28,31,30,31,30,31,31,30,31,30,31); 
      { dvm – počet dní v mesiaci } 
...  
writeln;  
dkr:=dkm; 
for i:=12 downto m+1 do 
  dkr:=dkr+dvm[i]; 
if (m<2) and priestupny then inc(dkr); 
write(’Do konca roka ’); 
case dkr of 
      1: writeln(’zostáva 1 deň.’); 
  2,3,4: writeln(’zostávajú ’, dkr,’ dni.’); 
  else writeln(’zostáva ’, dkr,’ dní.’) 
end; 
{ výpočet počtu dní od začiatku roka } 
writeln; 
ozr:=d; 
for i:=1 to m-1 do 
  ozr:=ozr+dvm[i]; 
if (m>2) and priestupny then inc(ozr); 
write(’Od zaciatku roka ’); 
case ozr of 
      1: writeln(’ubehol 1 deň.’); 
  2,3,4: writeln(’ubehli ’, ozr,’ dni.’); 
  else writeln(’ubehlo ’, ozr,’ dní.’) 
end; 
readln 
end. 
Samozrejme, ak pozná me počet dní do konca roka, počet dní od začiatku roka sa dá  vypočítať aj 

jednoduchšie. 

 

Príklad POLE-10: Vytvorte program na prevod sprá vy – textu do morseovej abecedy. 

Analý za: Príklad je typický  pre použitie typovej konštanty pole. Doplnili sme ho aj o zvukový  vý stup. 

Program ako „nepovinný “ ponechá vame na samoštú dium. 

program POLE_10; 
uses  crt; 
const TAB:array[’A’..’Z’] of string[4] = (’.-’,’-...’,’-.-.’,’-..’,’.’,’..-.’, 
’--.’,’....’,’..’,’.---’,’-.-’,’.-..’,’--’,’-.’,’---’,’.--.’,’--.-’,’.-.’,’...’, 
’ ’,’..-’, ’...-’,’.--’,’-..-’,’-.--’,’--..’); 
      takt=100; 
      bodka=takt; 
      ciarka=3*takt; 
      medzi_znakmi=takt; 
      medzi_bc=takt; 
      Hz=1000; 
var   s:string; 
      i:byte; 
 
procedure zvuk(MSznak:string); 
  var i:byte; 
  begin 
  write(MSznak); 
  for i:=1 to length(MSznak) do 
    begin 
    if MSznak[i]=’|’ 
      then Delay(medzi_znakmi) 
      else begin 
           Sound(Hz); 
           if MSznak[i]=’.’ 
             then Delay(bodka) 
             else Delay(ciarka); 
           NoSound 
           end; 
    Delay(medzi_bc) 
    end 
  end; 
 
BEGIN 
clrscr; 
write(’Zadaj retazec: ’); readln(s); 
for i:=1 to length(s) do 
  case s[i] of 
    ’A’..’Z’,’a’..’z’: zvuk( TAB[ UpCase(s[i]) ] + ’|’ ); 
    ’ ’              : writeln 
  end; 
readln 
END. 
 

 Okrem vyhľadá vania sa často ú daje aj triedia. Utriediť prvky poľa znamená  vykonať činnosť, po 

končení ktorej pre prvky poľa napr. A platí: A[I] ≤ A[I+1] pre všetky dovolené hodnoty I. Triediacich 

algoritmov je veľa. Každý  by mal ovlá dať aspoň ten najjednoduchší, ktorý m môže byť napríklad bubblesort. 

Príklad POLE-11: Vytvorte podprogram na utriedenie číselného poľa bubblesortom. 

Analý za: Princípom prá ce bubblesortu je prechá dzať poľom a porovná vať dva vedľa seba stojace prvky; ak 

je ľavý  prvok väčší ako pravý , vymeniť ich. Ak si predstavíte tú to činnosť, malo by byť zrejmé, že po prvom 

prechode poľom je najväčší prvok na svojom miesta, t.j. na konci poľa. Po druhom prechode poľom je na 

svojom mieste druhý  najväčší prvok, atď. Pri n-prvkovom poli musí byť po n-1 prechodoch pole utriedené. 

Podstatou algoritmu sú  dva vnorené cykly: vonkajší cyklus zabezpečuje pri n-prvkovom poli n-1 prechodov 

daný m poľom (pp – počet prechodov) a vnú torný  cyklus zabezpečuje pohyb v poli od prvého po n-tý  prvok 

(premenná  i). Vnú torný  cyklus možno ešte zefektívniť, ak zohľadníme aj to, že po prvom prechode poľom je 

najväčší prvok na svojom mieste a preto pri ďalších prechodoch poľom už nemusíme ísť až na koniec poľa; 

analogicky po druhom prechode poľom je už aj druhý  najväčší prvok poľa na svojom mieste a preto 

v nasledujú cich prechodoch poľom nemusíme porovná vať už ani predposledný  prvok, atď. 

1. prechod treba porovnať n prvkov t.j. ísť po n-1. prvok 

2. prechod treba porovnať n-1 prvkov t.j. ísť po n-2. prvok 

3. prechod treba porovnať n-2 prvkov t.j. ísť po n-3. prvok 
... 
pp. prechod     treba ísť po n-pp. prvok 
procedure bubblesort; 
  var pp:integer; 
  procedure vymena; 
    var pom:real; 
    begin 
    pom:=a[i]; a[i]:=a[i+1]; a[i+1]:=pom; 
    end; 
  begin 
  for pp:=1 to n-1 do  { počet prechodov } 
    for i:=1 to n-pp do  {pohyb v poli } 
      if a[i]>a[i+1] then vymena; 
  end; 
 
BEGIN 
clrscr; 
vstup; 
writeln(’Povodne pole:’); vystup; 
bubblesort; 
writeln(’Utriedene pole:’); vystup; 
readln 
END. 
 
 Aspoň jedný m príkladom si pripomenieme, že aj ú dajový  typ string (reťazec) je špeciá lnym 

prípadom jednorozmerného poľa. Teoreticky ide o pole typu array [0..255] of char, ktorého prvky môžeme 

načítať „naraz“ príkazom read a zobraziť tiež „naraz“ bez cyklu for príkazom write (znak s indexom 0 má  

osobitnú  funkciu – je v ňom uložená  aktuá lna dĺžka reťazca). Môžeme však sprístupniť i-ty znak reťazca 

indexovanou premennou ako u poľa. 

 

Príklad POLE-12: Vytvorte program, ktorý  zistí, či zadané dva reťazce sa skladajú  z tý ch istý ch písmen 

a v rovnakom počte. Takéto reťazce nazý vame anagramy. 

Analý za: Najľahšie sa zistí, či majú  zadané reťazce rovnakú  dĺžku. Ak ich dĺžky sú  rôzne, nemôžu byť 

anagramy. Ak sa ich dĺžky rovnajú , musíme zistiť, či sa skladajú  z rovnaký ch písmen a v rovnakom počte. 

Znamená  to zobrať prvý  znak v prvom reťazci a hľadať jeho vý skyt v druhom reťazci, ak je všetko 

„v poriadku“, zobrať druhý  znak z prvého reťazca atď. Teda cyklus v cykle, kde vonkajší cyklus nastavuje 

prvý , druhý ,... znak prvého reťazca a vnorený  cyklus „prechá dza“ druhý m reťazcom a hľadá  v ňom 

zodpovedajú ci znak. Najefektívnejšie sú  cykly s nezná mym počtom opakovaní. Ď alším problémom je 

zabezpečiť, aby sme zistili, či sa písmená  zhodujú  aj v počtoch. Najjednoduchšie je asi „vyrá bať“ pri 

prechode druhý m reťazcom postupne reťazec totožný  s prvý m. Pre ozrejmenie programu si spravte pomocný  

vý pis upravovaného druhého reťazca príkazom write(y), ako to je v pozná mke programu. 

program POLE_12; 
uses crt; 
var  slovo1,slovo2:string[80]; 
 
function anagram(x,y:string):boolean; 
  var i,j,dlzkax,dlzkay:byte; 
      anag:boolean; 
      pom:char; 
  begin 
  dlzkax:=length(x); dlzkay:=length(y); 
  if dlzkax=dlzkay 
    then begin 
         anag:=true; 
         i:=1; 
         while anag and (i<=dlzkax) do   { vonkajší cyklus } 
           begin 
           j:=i; 
           while (x[i]<>y[j]) and (j<dlzkay) do  { vnútorný cyklus } 
             j:=j+1; 
           anag:=x[i]=y[j]; pom:=y[i]; y[i]:=y[j]; y[j]:=pom; 
           i:=i+1;      { doplňte príkaz writeln(y) } 
           end; 
         anagram:=anag 
         end 
    else anagram:=false 
  end; 
 
BEGIN 
clrscr; 
repeat 
  write(’Prve  slovo: ’); readln(slovo1); 
  write(’Druhe slovo: ’); readln(slovo2); 
  if anagram(slovo1,slovo2) 
    then writeln(’JE TO ANAGRAM’:44) 
    else writeln(’NIE JE TO ANAGRAM’:46); 
  writeln(’KONIEC - stlac Esc’); 
  writeln 
until readkey=chr(27); 
END. 
 

 Ak sa niekomu zapá čil pohyb znaku po obrazovke, môže pokračovať v príklade CRP-6, a pokú siť sa 

naprogramovať jednu z prvý ch hier na počítačoch – červíka Wurmi.  

 

Príklad POLE-13:  Vytvorte program – hru, v ktorej budú  na obrazovke ná hodne vygenerované krížiky (x) 

- jed a listy - potrava (♠ chr(6)). Šípkami na riadenie pohybu kurzora nech sa ovlá da pohyb červíka Wurmi 

(znak chr(1)), ktorého ú lohou je „požierať“ listy a vyhý bať sa jedu. Ak narazí na jed, hra končí. Hra končí aj 

„narazením do okrajov obrazovky“. 

Analý za: Pri programovaní tejto hry použite príklad CRP-6, v ktorom má me vyriešené ovlá danie pohybu 

znaku na obrazovke vrá tane ošetrenia „trafenia“ znaku na obrazovke a „narazenie do okraja obrazovky“. 

Ď alšie zlepšenie hry si vyžaduje použitie polí, aby bolo možné zapamätať si pozície (x-ové a y-nové 

sú radnice) jedu a potravy. Po každom premiestnení Wurmiho treba testovať, či „nestojí“ na jede alebo 

potrave. Ak stojí na potrave, treba započítať, že „zjedol“ ďalší list. Hra končí, keď: 

• „zožerie“ všetky vygenerované listy 

• „zožerie“ jed 

• „narazí“ do okrajov obrazovky. 

program POLE_13; 
uses  crt,dos; 
 
const pocet=5;   { počet vygenerovaných jedov a listov } 
      ESC=chr(27); 
      sipkaVPRAVO=chr(77); 
      sipkaVLAVO =chr(75); 
      sipkaHORE  =chr(72); 
      sipkaDOLE  =chr(80); 
      list       =chr(6); 
 
var   stl,ria,body:integer; 
      px,py,jx,jy:array[1..pocet] of integer; 
 
procedure skry_kurzor; 
  var reg:registers; 
  begin 
  reg.ah:=1; 
  reg.ch:=32; 
  reg.cl:=0; 
  intr($10,reg) 
  end; 
 
procedure ohrada; 
  var i:integer; 
  begin 
  textcolor(white); 
  write(chr(201)); for i:=2 to 79 do write(chr(205)); write(chr(187)); 
  for i:=2 to 23 do begin write(chr(186)); write(chr(186):79) end; 
  write(chr(200)); for i:=2 to 79 do write(chr(205)); write(chr(188)); 
  end; 
 
procedure generuj_potravu_jed; 
  var i:integer; 
  begin 
  for i:=1 to pocet do 
    begin 
    px[i]:=random(78)+2; py[i]:=random(22)+2; { generuje potravu } 
    jx[i]:=random(78)+2; jy[i]:=random(22)+2; { generuje jed } 
    end 
  end; 
 
procedure vykresli_potravu_jed; 
  var i:integer; 
  begin 
  for i:=1 to pocet do 
    begin 
    gotoxy(px[i],py[i]); write(list); 
    gotoxy(jx[i],jy[i]); write(’x’); 
    end; 
  end; 
 
procedure inicializacia; 
  var limit:integer; 
  begin 
  randomize; 
  generuj_potravu_jed; 
  skry_kurzor; 
  clrscr; 
  ohrada; 
  vykresli_potravu_jed; 
  stl:=40; ria:=12; gotoxy(stl,ria); write(chr(1)); 
  body:=0 
  end; 

function jed:boolean; { testuje, či Wurmi nie je na poličku jed } 
  var i:integer; 
  begin 
  i:=0; 
  repeat 
    inc(i) 
  until ((jx[i]=stl) and (jy[i]=ria)) or (i=pocet); 
  jed:=(jx[i]=stl) and (jy[i]=ria) 
  end; 
 
procedure potrava; { testuje, či Wurmi nie je na políčku list } 
  var i:integer; 
  begin 
  i:=0; 
  repeat 
    inc(i) 
  until ((px[i]=stl) and (py[i]=ria)) or (i=pocet); 
  if (px[i]=stl) and (py[i]=ria) 
    then begin inc(body); px[i]:=0; py[i]:=0 end; 
  gotoxy(3,2); write(body); 
  end; 
 
procedure pohyb; 
  var stlznak:char; 
  begin 
  repeat 
    if keypressed then stlznak:=readkey; 
    gotoxy(stl,ria); write(’ ’); 
    case stlznak of 
      ESC:halt; 
      sipkaVPRAVO:inc(stl); 
      sipkaVLAVO :dec(stl); 
      sipkaHORE  :dec(ria); 
      sipkaDOLE  :inc(ria); 
      end; 
    gotoxy(stl,ria); write(chr(1)); potrava; delay(150); 
  until (stl=1) or (stl=80) or (ria=1) or (ria=24) or jed or (body=pocet); 
  if jed 
    then begin gotoxy(33,12); write(’ J E D ! ! ! ’) end 
    else if body=pocet 
           then begin gotoxy(20,12); write(’V Y B O R N E !!! VYHRAL SI!!!’) end 
           else begin gotoxy(25,12); write(’B U M !!! OHRADA!!!’) end 
  end; 
 
BEGIN 
inicializacia; 
pohyb; 
readln 
END. 
 

 

Nerieš ené  úlohy na jednorozmerné  pole a typ string: 

14. Vytvorte program na zistenie najmenšieho a najväčšieho prvku v poli najviac 100 celý ch čísel a vý menu 

najmenšieho prvku s prvý m a najväčšieho prvku s posledný m prvkom v poli. 

15. Vytvorte program na ná jdenie prvý ch dvoch najväčších (najmenších) čísel v poli najviac 20 rôznych 

celý ch čísel. 

16. Vytvorte program na ná jdenie k (k≤n) maxím (1. môžu byť aj rovnaké, 2. k rôznych maxím) v poli s n 

prvkami. 

Ná vod: využite upravený  bubblesort. 

17. Vytvorte program na zistenie počtu nú l (kladný ch čísel, pá rnych čísel, čísel deliteľný ch zadaný m číslom, 

čísel zo zadaného intervalu a pod.) v zadanom číselnom poli. 

18. Vytvorte program na zistenie posledného vý skytu (všetký ch vý skytov) zadaného znaku v texte. 

19. Vytvorte program na zistenie polohy maxima a minima v danom poli (ošetrite aj viacná sobný  vý skyt). 

20. Vytvorte program, ktorý  prvky poľa napr. 2, 4, 5, 3, 8, 1, 3, 5, 9, 4, 7 zobrazí v tvare: 

2, 4, 5, 

3, 8,  (3 < 5) 

1, 3, 5, 9, (1 < 8) 

4, 7  (4 < 9) 

21. Vytvorte program, ktorý  z pôvodného poľa odstrá ni všetky prvky so zadanou vlastnosťou (nepá rne čísla, 

zá porné čísla, rovné zadanej hodnote, znaky zo zadaného intervalu a pod.). 

22. Vytvorte program, ktorý  zlú či dve polia s prvkami rovnakého typu (prvky druhého poľa dá  za prvky 

prvého poľa). 

23. Vytvorte program, ktorý  zlú či dve usporiadané polia s prvkami rovnakého typu tak, že vý sledné pole 

bude tiež usporiadané. 

24. Vytvorte program na vý počet a zobrazenie Pascalovho trojuholníka v tvare: 

 1 
 1        1 

 1        2        1 
 1        3        3        1 
 1        4        6        4        1 
 1    1+4    4+6    6+4    4+1    1 
 ... 

Ná vod: Využite vzorce v zadaní REK-8, nepoužite však rekurziu ale pole. 

25. Vytvorte program na uloženie prvkov pôvodného poľa opačne, vymeniť prvý  prvok s posledný m, druhý  

s predposledný m atď. 

26. Vytvorte program na posunutie prvkov poľa o k miest doprava (doľava). Posledný ch k prvkov poľa bude 

posunutý ch na začiatok poľa. 

27. Vytvorte program na usporiadanie prvkov (čísel, znakov, reťazcov) daného poľa zostupne (od 

najväčšieho po najmenší). 

28. Vytvorte program na vý počet ciferného sú čtu prirodzeného čísla využitím typu string. 

29. *Vytvorte program na zobrazenie všetký ch prvočísiel od 2 po 60 000 metódou tzv. Eratosténovho sita. 

Ná vod: Napíšeme si čísla od 2 do 60 000. Dá me vypísať 2 a škrtneme všetky jej ná sobky po 60 000. 

Zoberieme a vypíšeme ďalšie číslo: 3 (nie je škrtnuté, preto je prvočíslo) a škrtneme všetky jeho 

ná sobky. Zoberieme ďalšie číslo: 4, je škrtnuté a preto nie je prvočíslo. Zoberieme ďalšie číslo: 5, nie je 

škrtnuté, preto ho vypíšeme ako prvočíslo a škrtneme všetky jeho ná sobky. Číslo 6 je škrtnuté. Číslo 7 

vypíšeme (nie je škrtnuté) a škrtneme všetky jeho ná sobky (14, 21, 28,...). Takto pokračujeme až po 

60 000. Môžeme použiť napríklad typ pole array [2..60000] of boolean, kde true znamená , že dané číslo 

nie je škrtnuté (je prvočíslo) a false znamená , že je škrtnuté. 

30. Zefektívnite bubblesort tý m, že ukončíte triedenie (prechody poľom), ak pri poslednom prechode už 

nedošlo ani k jednej vý mene susedný ch prvkov (pole musí byť už utriedené). 

31. *Vytvorte program, ktorý  zistí, koľko anagramov sa nachá dza v zadanom zozname slov. 

Dvojrozmerné pole 
 V definícii typu pole: 

 type mt = array [ ti ] of tz 

sme doteraz dosadzovali za typ zložky len jednoduchý  typ alebo typ reťazec. V princípe typ zložky môže byť 

aký koľvek, okrem typu sú bor. V prípade, že typ zložky bude opäť typ pole, dostá vame tzv. typ dvojrozmerné 

pole.  

Definícia typu dvojrozmerné pole má  tvar: 

type mt = array [ ti1 ] of array [ ti2 ] of tz 

kde mt je meno typu – identifiká tor, ti1 a ti2 sú  ordiná lne typy indexov typu pole a tz je typ zložky. 

Dovolený  je aj skrá tený  zá pis definície typu pole: 

type mt = array [ ti1 , ti2 ] of tz 

Napríklad: 

type  MATICA1 = array [1..10 ] of array [1..10] of integer; 

 MATICA2 = array [1..10,1..10] of integer;  { definícia rovnocenná  s MATICA1 } 

 SACHOVNICA=array[ á .́. h́ ,́1..8] of boolean; 

 NEZMYSEL=array[boolean,boolean] of char; 

 

 Dvojrozmerné pole nazý vame aj pole polí. Ná zov vznikol zrejme z poznatku, že jednorozmerné 

pole, resp. každý  jeho prvok je znova jednorozmerný m poľom. Z hľadiska štruktú ry dvojrozmernému poľu 

zodpovedá  matematický  pojem matica. Pre ľahšie pochopenie príkladov budeme ďalej používať nasledujú cu 

symboliku (častejšie ako dvojrozmerné pole budeme používať pojem matica, pretože tá  predstavuje model na 

riešenie ná šho problému; až keď je zná my model na riešenie, môžeme ho realizovať v počítači 

dvojrozmerný m poľom): 

• matica typu N x M („en krá t em“) obsahuje N riadkov a M stĺpcov 

• pre zmenu riadku, t.j. pohyb v stĺpci budeme používať premennú  I 

• pre zmenu stĺpca, t.j. pohyb v riadku budeme používať premennú  J 

• schematicky (matica typu N x M): 

J sa mení od 1 po M (pohyb v riadku, zmena stĺpca) 
 

a 1,1 a 1,2  a 1,3 ... a 1,j ... a 1,m 

a 2,1 a 2,2 a 2,3 ... a 2,j ... a 2,m 

a 3,1 a 3,2 a 3,3 ... a 3,j ... a 3,m 

... ... ... ... ... ... ... 

  a i,1  a i,2 a i,3 ...  a i,j ... a i,m 

   ... ... ... ... ... ... ... 

  a n,1 a n,2 a n,3 ... a n,j ... a n,m 

 

I sa mení 

od 1 po N 

(pohyb v stĺpci, 

zmena riadku) 

 Z hľadiska algoritmický ch konštrukcií pre prá cu s maticou väčšinou potrebujeme cyklus v cykle. 

Vonkajší cyklus zabezpečuje najčastejšie nastavenie príslušného riadku (for I:=1 to N do) a vnú torný  cyklus 

pohyb v nastavenom riadku t.j. zmenu stĺpca od 1 po M (for J:=1 to M do). 

 Aj pri dvojrozmerný ch poliach pracujeme s indexovaný mi premenný mi a používame zá pis, 

napríklad pre pole A: A [I][J] resp. skrá tený  zá pis A[I,J] - prvok poľa A v I-tom riadku a J-tom stĺpci. 

 

Príklad MAT-1: Vytvorte program zabezpečujú ci načítanie typu (rozmerov) a prvkov matice a jej 

zobrazenie na obrazovke monitora. 

Analý za: Program obsahuje dva vstupy. V procedú re VSTUP treba zadať okrem rozmerov aj všetky prvky 

matice z klá vesnice, v procedú re VSTUP_NAHODNE po zadaní rozmerov matice program ná hodne 

vygeneruje prvky matice – celé čísla od 0 po 9. Keďže z procedú r vstupu sa majú  novozískané hodnoty 

vyviesť, parametre musia byť nahradzované referenciou. Pri vý stupe, hodnoty stačí len doviesť do 

procedú ry, ide o ná hradu formá lnych parametrov hodnotou. 
program MATICA_1; 
uses  crt; 
const MAXPP=10; 
type  MATICA=array[1..MAXPP,1..MAXPP] of integer; 
var   A:MATICA; 
      N,M,I,J:integer; 
 
procedure VSTUP(var N,M:integer;var A:MATICA); 
  begin 
  write(’Zadaj rozmery matice <= ’,MAXPP,’: ’); 
  readln(N,M); 
  for I:=1 to N do   { nastaví riadok } 
    for J:=1 to M do   { nastaví stĺpec } 
      begin 
      write(’Prvok ’,I,’,’,J,’: ’); 
      readln(A[I,J]); 
      end; 
  end; 
 
procedure VSTUP_NAHODNE(var N,M:integer;var A:MATICA); 
  begin 
  write(’Zadaj rozmery matice <= ’,MAXPP,’: ’); 
  readln(N,M); 
  randomize; 
  for I:=1 to N do 
    for J:=1 to M do 
      A[I,J]:=random(10); 
  end; 
 
procedure VYSTUP(N,M:integer;A:MATICA); 
  begin 
  for I:=1 to N do  { nastavuje riadok } 
    begin 
    for J:=1 to M do  { nastavuje stĺpec – pohyb v riadku } 
      write(A[I,J]:4);  { hodnoty zobrazené v riadku } 
    writeln;   { presunutie „kurzora“ na nový riadok } 
    end; 
  end; 
 
BEGIN 
clrscr; 
writeln(’Prvky matice nahodne (nie - N)?’); 
if upcase(readkey)=’N’ 
  then VSTUP(N,M,A) 

  else VSTUP_NAHODNE(N,M,A); 
VYSTUP(N,M,A); 
readln; 
END. 
 Najprv si uvedieme niekoľko matematický ch príkladov na prá cu s maticami.  

Príklad MAT-2: Vytvorte program na vyná sobenie matice celý ch čísel celý m číslom k. 

Analý za: Vyná sobiť číselnú  maticu číslom znamená  vyná sobiť každý  jej prvok daný m číslom, t.j. pre každé 

dovolené i, j: B[i,j] novej matice sa rovná  k∗A[i,j].  
procedure VYNASOB(N,M:integer;A:MATICA;var B:MATICA); 
  var K:integer; 
  begin 
  write(’Maticu vynasobit celym cislom: ’); 
  readln(K); 
  for I:=1 to N do 
    for J:=1 to M do 
      B[I,J]:=K*A[I,J] 
  end; 
 
BEGIN 
clrscr; 
writeln(’Prvky matice nahodne (nie - N)?’); 
if upcase(readkey)=’N’ 
  then VSTUP(N,M,A) 
  else VSTUP_NAHODNE(N,M,A); 
VYSTUP(N,M,A); 
VYNASOB(N,M,A,B); 
VYSTUP(N,M,B); 
readln; 
END. 
 
Príklad MAT-3: Vytvorte program na vý počet sú čtu dvoch matíc rovnaký ch typov. 

Analý za: Sú čtom dvoch matíc A a B typu NxM je matica C typu NxM, kde C[i,j] = A[i,j] + B[i,j]. 

V procedú rach vstupu a vý stupu je nevyhnutné použiť parametre, keďže procedú ry chceme použiť pre 

matice A, B aj C! 
program MATICA_3; 
uses  crt; 
const MAXPP=10; 
type  MATICA=array[1..MAXPP,1..MAXPP] of integer; 
var   A,B,C:MATICA; 
      N,M,I,J:integer; 
 
procedure ROZMERY(var N,M:integer); 
  begin 
  write(’Zadaj rozmery matice <= ’,MAXPP,’: ’); 
  readln(N,M); 
  end; 
 
procedure VSTUP(N,M:integer;var A:MATICA); 
... 
 
procedure VSTUP_NAHODNE(N,M:integer;var A:MATICA); 
... 
 
procedure VYSTUP(N,M:integer;A:MATICA); 
... 
 
procedure SCITAT(N,M:integer;A,B:MATICA;var C:MATICA); 
  begin 
  for I:=1 to N do 
    for J:=1 to M do 

      C[I,J]:=A[I,J]+B[I,J] 
  end; 
 
 
BEGIN 
randomize; 
clrscr; 
ROZMERY(N,M); 
VSTUP_NAHODNE(N,M,A); 
VSTUP_NAHODNE(N,M,B); 
writeln(’Prva matica: ’); 
VYSTUP(N,M,A); 
writeln(’Druha matica: ’); 
VYSTUP(N,M,B); 
SCITAT(N,M,A,B,C); 
writeln(’Sucet prvej a druhej: ’); 
VYSTUP(N,M,C); 
readln; 
END. 
 
Príklad MAT-4: Pre zasvätený ch, ktorí vedia, ako sa ná sobia matice, uvá dzame bez komentá ra procedú ru 

na sú čin dvoch matíc. Matica A musí byť typu NxL, matice B typu LxM a vý sledná  matica C bude typu 

NxM. 
procedure SUCIN(N,L,M:integer;A,B:MATICA;var C:MATICA); 
  begin 
  for I:=1 to N do 
    for J:=1 to M do 
  begin 
  C[I,J]:=0; 
  for K:=1 to L do 
       C[I,J]:=C[I,J] + A[I,K] * B[K,J] 
  end; 
  end; 
 

Príklad MAT-5: Vytvorte program, ktorý  zistí, či zadaná  štvorcová  matica je symetrická . 

Analý za: Štvorcová  matica (počet riadkov sa rovná  počtu stĺpcov) je symetrická , ak je symetrická  podľa 

hlavnej diagoná ly (hlavná  diagoná la je tvorená  prvkami A[1,1], A[2,2], A[3,3], ..., A[N,N]). Preto pre každý  

prvok matice musí platiť: A[I,J] = A[J,I] pre všetky dovolené hodnoty I a J.  

Nie najefektívnejší, ale jednoduchší program znamená  použitie cyklov for, ktoré “zistia“, či sa prvky pod 

hlavnou diagoná lou rovnajú  príslušný m prvkom nad hlavnou diagoná lou (ak sa „dolný  trojuholník rovná  

hornému, musí sa aj horný  rovnať dolnému“). Keďže vý sledkom podprogramu je hodnota true – ak je matica 

symetrická  alebo hodnota false – ak nie je symetrická , môžeme použiť aj funkciu. 
program MATICA_5; 
uses  crt; 
const MAXPP=10; 
type  MATICA=array[1..MAXPP,1..MAXPP] of integer; 
var   A:MATICA; 
      N,I,J:integer; 
procedure VSTUP(var N:integer;var A:MATICA); 
.... 
procedure VYSTUP(N:integer;A:MATICA); 
... 
function SYMETRICKA(N:integer;A:MATICA):boolean; 
  var SYM:boolean; 
  begin 
  SYM:=true; 
  for I:=2 to N do 

    for J:=1 to I-1 do 
      if A[I,J]<>A[J,I] then SYM:=false; { SYM:=SYM and (A[I,J]=A[J,I]); } 
  SYMETRICKA:=SYM 
  end; 
BEGIN 
clrscr; 
VSTUP(N,A); 
writeln(’Matica: ’); 
VYSTUP(N,A); 
if SYMETRICKA(N,A) 
  then writeln(’je symetricka.’) 
  else writeln(’nie je symetricka.’); 
readln; 
END. 
 

Príklad MAT-6: Vytvorte program na ná jdenie najmenšej a najväčšej hodnoty v každom riadku matice. 

Analý za: Každý  riadok matice je jednorozmerné pole „doplnené“ o index riadku, v ktorom sa nachá dza. Ak 

vieme hľadať minimum a maximum v jednorozmernom poli,  stačí toto hľadanie vložiť do cyklu, ktorý  bude 

meniť riadky od prvého po posledný . 
program MATICA_6; 
uses  crt; 
const MAXPP=10; 
type  MATICA=array[1..MAXPP,1..MAXPP] of integer; 
      POLE=array[1..MAXPP] of integer; 
var   A:MATICA; 
      MIN,MAX:POLE; 
      N,M,I,J:integer; 
 
procedure VSTUP(var N,M:integer;var A:MATICA); 
... 
procedure VSTUP_NAHODNE(var N,M:integer;var A:MATICA); 
... 
procedure VYSTUP(N,M:integer;A:MATICA); 
... 
 
procedure MIN_MAX(N,M:integer;A:MATICA;var MIN,MAX:POLE); 
  begin 
  for I:=1 to N do 
    begin 
    MIN[I]:=A[I,1]; 
    MAX[I]:=A[I,1]; 
    for J:=2 to M do 
      begin 
      if A[I,J]<MIN[I] then MIN[I]:=A[I,J]; 
      if A[I,J]>MAX[I] then MAX[I]:=A[I,J] 
      end; 
    end; 
  end; 
 
procedure VYSTUPPOLA(N:integer;A:POLE); 
  begin 
  for I:=1 to N-1 do 
    write(A[I],’, ’); 
  writeln(A[N]) 
  end; 
 
BEGIN 
clrscr; 
randomize; 
writeln(’Prvky matice nahodne (nie - N)?’); 
if upcase(readkey)=’N’ 
  then VSTUP(N,M,A) 



 Zá klady programovania  - Turbo Pascal   89 

  else VSTUP_NAHODNE(N,M,A); 
MIN_MAX(N,M,A,MIN,MAX); 
writeln(’V matici: ’); 
VYSTUP(N,M,A); 
writeln(’minima v riadkoch: ’); 
VYSTUPPOLA(N,MIN); 
writeln(’maxima v riadkoch: ’); 
VYSTUPPOLA(N,MAX); 
readln; 
END. 
 

Príklad MAT-7: Vytvorte program na utriedenie prvkov v každom riadku matice. 

Analý za: Aby sme zvlá dli tento príklad, musíme vedieť aspoň jeden triediaci algoritmus. Pri 

jednorozmernom poli sme sa zaoberali bubblesortom, ktorý  použijeme aj teraz. Ako v predchá dzajú com 

príklade, aj tu treba len bubblesort „aplikovať“ na každý  riadok matice, čo znamená  vložiť bubblesort do 

cyklu na nastavenie riadku. 
procedure UTRIED_V_RIADKU(N,M:integer;var A:MATICA); 
  var PP:integer; 
  procedure VYMENA(var X,Y:integer); 
    var POM:integer; 
    begin 
    POM:=X; X:=Y; Y:=POM; 
    end; 
  begin 
  for I:=1 to N do   { nastaví riadok } 
    for PP:=1 to M-1 do   { počet prechodov riadkom } 
      for J:=1 to M-PP do  { pohyb v riadku } 
        if A[I,J]>a[I,J+1] then VYMENA(A[I,J],A[I,J+1]); 
  end; 
 
BEGIN 
clrscr; 
writeln(’Prvky matice nahodne (nie - N)?’); 
if upcase(readkey)=’N’ 
  then VSTUP(N,M,A) 
  else VSTUP_NAHODNE(N,M,A); 
VYSTUP(N,M,A); 
UTRIED_V_RIADKU(N,M,A); 
writeln(’Matica utriedenia v riadkoch: ’); 
VYSTUP(N,M,A); 
readln; 
END. 
Ako treba upraviť program, aby sa zachovala aj pôvodná  matica A? Podprogram, ktorý  utriedi prvky matice 

v stĺpcoch, sa len má lo líši od podprogramu UTRIED_V_RIADKU. Odlaďte ho. 

 

Príklad MAT-8: Vytvorte program, ktorý  utriedi riadky matice podľa hodnoty prvého prvku v riadku, t.j. 

podľa prvého stĺpca. 

Analý za:  

Napríklad v matici:   2 4 1 5 

    3 5 2 7 

    1 2 1 4 

bude posledný  riadok presunutý  do prvého, prvý  do stredného a stredný  riadok do tretieho. 

Utriedená  matica:  1 2 1 4 

    2 4 1 5 

    3 5 2 7 

Keďže je potrebné vymieňať celé riadky, už v ú seku definícií typ dvojrozmerné pole zadefinujeme ako pole 

riadkov. Tiež si treba uvedomiť, že všetky prvky v prvom stĺpci sú  prístupné cez zá pis A[I,1], kde I určuje 

riadok. No a dokonale treba ovlá dať nejaký  triediaci algoritmus, napríklad bubblesort. 
program MATICA_8; 
uses  crt; 
const MAXPP=10; 
type  RIADOK=array[1..MAXPP] of integer; 
      MATICA=array[1..MAXPP] of RIADOK; 
var   A:MATICA; 
      N,M,I,J:integer; 
 
procedure VSTUP_NAHODNE(var N,M:integer;var A:MATICA); 
... 
procedure VYSTUP(N,M:integer;A:MATICA); 
... 
procedure UTRIED_PODLA_PRVEHO(N,M:integer;var A:MATICA); 
  var R:RIADOK; 
      PP:integer; 
  procedure VYMENA(var X,Y:RIADOK);  { typ RIADOK! } 
    var POM:RIADOK; 
    begin 
    POM:=X; X:=Y; Y:=POM 
    end; 
  begin 
  for PP:=1 to N-1 do 
    for I:=1 to N-PP do 
      if A[I,1]>A[I+1,1] then VYMENA(A[I],A[I+1]) 
  end; 
 
BEGIN 
clrscr; 
VSTUP_NAHODNE(N,M,A); 
writeln(’Povodna matica: ’); 
VYSTUP(N,M,A); 
UTRIED_PODLA_PRVEHO(N,M,A); 
writeln(’Riadky utriedene podla prveho prvku v riadku: ’); 
VYSTUP(N,M,A); 
readln; 
END. 
 

Na zá ver „matematický ch“ príkladov s maticami uvedieme „hviezdičkový “ príklad. 

Príklad MAT-9*: Vytvorte program, ktorý  „rozvinie“ riadky a stĺpce matice do jednorozmerného poľa 

v smere chodu hodinový ch ručičiek. 

Analý za: 

Napríklad maticu:   2 4 1 5 

    3 5 2 7 

    1 2 1 4 

rozvinie program do poľa: 2    4    1    5    7    4    1    2    1    3    5    2 
program MAT-9; 
uses  crt; 
const MAXPP=10; 
type  MATICA=array[0..MAXPP,0..MAXPP] of integer; 
      POLE=array[0..MAXPP*MAXPP] of integer; 
var   A:MATICA; 
      B:POLE; 
      i,j,k,n,m,l,p,h,d:integer; 

procedure VSTUP; 
... 
procedure VYSTUP; 
... 
procedure VYTVOR_POLE; 
 procedure vpravo; 
   begin 
   for j:=l to p do begin k:=k+1; B[k]:=A[h,j]; end; 
   end; 
 procedure dole; 
   begin 
   for i:=h to d do begin k:=k+1; B[k]:=A[i,p]; end; 
   end; 
 procedure vlavo; 
   begin 
   for j:=p downto l do begin k:=k+1; B[k]:=A[d,j]; end; 
   end; 
 procedure hore; 
   begin 
   for i:=d downto h do begin k:=k+1; B[k]:=A[i,l]; end; 
   end; 
 begin 
 k:=0; l:=1; p:=m; h:=1; d:=n; 
 repeat 
   if k<n*m then vpravo; h:=h+1; 
   if k<n*m then dole;   p:=p-1; 
   if k<n*m then vlavo;  d:=d-1; 
   if k<n*m then hore;   l:=l+1; 
 until k=n*m; 
 end; 
procedure VYSTUP_POLA; 
 begin 
 writeln(’"Rozvinuta" matica:’); 
 for k:=1 to n*m do write(B[k]:4); 
 writeln; 
 end; 
BEGIN 
clrscr; 
VSTUP; 
VYSTUP; 
VYTVOR_POLE; 
VYSTUP_POLA; 
readln 
END. 
 

 Mnohí z ná s poznajú  hru ŽIVOT, v ktorej sa simuluje život jednobunkový ch organizmov v rovine. 

Po vložení počiatočnej generá cie program „počíta“ ďalšie generá cie. Organizmus prežije do ďalšej 

generá cie, ak ma 2 alebo 3 susedov, iná č zomrie. Živý  organizmus vznikne, ak má  prá ve 3 susedov. 

Príklad MAT-10: Vytvorte program simulujú ci život jednobunkový ch organizmov za vyššie uvedený ch 

podmienok. 

Analý za: Potrebné sú  dve matice. V jednej „prebieha život“, druhá  je pomocná  pri vý počte novej generá cie. 

Pre získanie nový ch poznatkov sme namiesto čísel 0 – prá zdna pozícia a 1 – živý  organizmus použili 

programá torom definovaný  typ s hodnotami nezivý  (ordiná lne číslo 0) a zivy (ordiná lne číslo 1). Program 

ponechá vame na samoštú dium.  
program MAT-10; 
uses  crt; 
const okraj=24; 
      plot =23; 

type  torganizmu=(nezivy,zivy); { programátorom vymenovaný typ } 
      tsuradnic=0..okraj; 
      tpola=array[tsuradnic,tsuradnic] of torganizmu; 
var   a,pom:tpola; 
      q:char; 
 
procedure vycisti_pole; 
  var i,j:tsuradnic; 
  begin 
  clrscr; 
  for i:=0 to okraj do 
    for j:=0 to okraj do 
      begin 
      a[i,j]:=nezivy; 
      pom[i,j]:=nezivy; 
      end; 
  end; 
procedure vystup; 
  var i,j:tsuradnic; 
  begin 
  clrscr; 
  for i:=1 to plot do 
    begin 
    for j:=1 to plot do 
      if a[i,j]=zivy 
        then write(’*’) 
        else write(’ ’); 
    writeln 
    end; 
  end; 
 
procedure vstup; 
  var i,j:integer; 
  begin 
  repeat 
    write(’Zadaj suradnice x,y ziveho organizmu ( KONIEC -> 0 ) ’); 
    read(i); 
    if i>0 
      then begin 
           readln(j); 
           if (i<okraj)and(j>0)and(j<okraj) 
             then a[i,j]:=zivy 
             else begin writeln(’Chybne suradnice’);delay(2000) end; 
           vystup 
           end 
  until i=0; 
  end; 
 
procedure nova_generacia; 
  var i,j:tsuradnic; pocet:0..8; 
  begin 
  for i:=1 to plot do 
    for j:=1 to plot do 
      begin 
      pocet:=ord(a[i-1,j-1])+  { počíta počet susedov okolo a[i,j] } 
             ord(a[i-1,j])+ 
             ord(a[i-1,j+1])+ 
             ord(a[i,j-1])+ 
             ord(a[i,j+1])+ 
             ord(a[i+1,j-1])+ 
             ord(a[i+1,j])+ 
             ord(a[i+1,j+1]); 
      if (pocet=3)or((pocet=2)and(a[i,j]=zivy)) 
        then pom[i,j]:=zivy 
        else pom[i,j]:=nezivy; 
      end; 

  a:=pom; 
  vystup; 
  end; 
BEGIN 
vycisti_pole; 
repeat 
  vstup; 
  repeat 
    nova_generacia; 
    write(’Chces dalsiu generaciu (N) ? ’);q:=upcase(readkey); 
  until q=’N’; 
  gotoxy(1,24); 
  write(’Chces doplnit zadanie (A) ? ’);q:=upcase(readkey); 
until q<>’A’; 
END. 
 

 

Nerieš ené  úlohy na dvojrozmerné  pole: 

11. Vytvorte program na vý počet sú čtu všetký ch prvkov číselnej matice. 

12. Vytvorte program, ktorý  sčíta všetky prvky na hlavnej diagoná le (ide z ľavého horného rohu matice do 

pravého dolného rohu) štvorcovej matice. 

13. Vytvorte program, ktorý  sčíta všetky prvky na vedľajšej diagoná le (ide z pravého horného rohu matice 

do ľavého dolného rohu) štvorcovej  matice. 

14. Vytvorte program na zistenie počtu nú l (zadaného čísla, kladný ch čísel, deliteľný ch zadaný m číslom, 

zo zadaného intervalu a pod.) v zadanej číselnej matici. 

15. Vytvorte program na ná jdenie najmenšieho (najväčšieho) prvku matice a jeho vý menu s prvý m 

(posledný m) prvkom matice. 

16. Vytvorte program na ná jdenie najmenšieho a najväčšieho prvku každého riadku matice a jeho vý menu 

s prvý m a posledný m prvkom v danom riadku matice. 

17. Vytvorte program na ná jdenie najmenšieho a najväčšieho prvku v každom stĺpci matice a jeho vý menu 

s prvý m a posledný m prvkom v danom stĺpci. 

18. Vytvorte program na ná jdenie zadanej hodnoty v matici a jej nahradenie novou hodnotou. 

19. Vytvorte program na zistenie, či sa zadaný  znak nachá dza v matici znakov a koľkokrá t sa nachá dza. 

20. Vytvorte program na zistenie miesta vý skytu (všetký ch, prvého, posledného) prvku so zadanou 

vlastnosťou v matici. 

21. Vytvorte program, ktorý  vo štvorcovej matici vymení prvky podľa hlavnej diagoná ly. 

22. Vytvorte program, ktorý  v matici vymení prvý  stĺpec (riadok) s posledný m, druhý  s predposledný m, atď. 

23. Vytvorte program, ktorý  zistí, či zadaná  štvorcová  matica je diagoná lna (okrem hlavnej diagoná ly 

a[i,j] = 0). 

24. Vytvorte program na otočenie štvorcovej matice o 900. 

25. Vytvorte program, ktorý  od miesta „čľupnutia kameňa do matice“ zvý ši pôvodne nulové hodnoty matice 

s narastajú cou vzdialenosťou od miesta „čľupnutia“ o 1. 

26. Vytvorte program, ktorý  pri pohybe v matici bude meniť hodnoty prechá dzaný ch prvkov. 

27. Vytvorte program, ktorý  pri pohybe v matici bude meniť hodnoty v okolí prechá dzaný ch prvkov. 

Ú dajový  typ záznam 
 Najmä pri hromadnom spracovaní ú dajov, pri informačný ch systémoch, často treba spojiť do 

jedného celku rôzne ú dajové typy (integer, string, boolean a pod.). Umožňuje to ú dajový  typ zá znam. 

Zá znam je nehomogénny štruktú rovaný  ú dajový  typ, ktorý  sa skladá  z pevného počtu položiek, 

vo všeobecnosti rôznych typov. 

Definícia typu zá znam má  tvar: 

type mt = record 

 p1 : tp1;  kde mt je meno typu 
 p2 : tp2;  p1 až pn sú  identifiká tory položiek zá znamu 
 ...   a tp1 až tpn sú  typy jednotlivý ch položiek 
 pn : tpn 

end 

Napríklad: type KARTA = record 
 OsCislo : integer; 
 Meno, 
 Priezv : string[25]; 
 DatNar : record 
 Den : 1..31; 
 Mes : 1..12; 
 Rok : 1900..2100 
 end; 
 Adresa : record 
 Obec, 
 Ulica : string[25]; 
 PSC : string[5] 
 end; 
 Priemer : array [1..5] of real; 
 Moze : boolean 
 end; 

  type BOD = record 
 x, 
 y : real; 
 farba : 0..14; 
 blik : boolean 
 end; 

  type ZLOZKA = record 
 HODNOTA : string; 
 POCET        : byte 
 end; 

 var KC : record Re, Im: real end; { deklarovaná  premenná  typu record } 

 
 Ak p je premenná  typu zá znam, k jej položke s ná zvom pi sa dostaneme zá pisom: p.pi. Napríklad, ak 

K je premenná  typu KARTA, možno použiť zá pisy: K.OsCislo, K.Meno, K.DatNar.Den, K.DatNar.Rok, 

K.Adresa.Obec, K.Priemer[1] a pod.  

 
Príklad ZAZ-1: Vytvorte program na určenie vzá jomnej polohy bodu a kružnice v rovine s využitím 

ú dajového typu zá znam. 

Analý za: Bod napr. B je určený  v rovine dvoma sú radnicami Bx a By. Kružnica je určená  stredom S 

so sú radnicami Sx a Sy a polomerom r. Vzdialenosť medzi bodom B a stredom kružnice k, bodom S, možno 

vypočítať pomocou Pythagorovej vety. Bod B leží vo vnú tornej oblasti kružnice k (S;r), ak jeho vzdialenosť 

od stredu kružnice k je menšia ako polomer r. Ak vzdialenosť /BS/ = r, bod B leží na kružnici a ak je tá to 

vzdialenosť väčšia ako polomer r, bod leží mimo oblasť kružnice k. 
program ZAZ_1; 
uses crt; 
type BOD=record 
            x,y:real; 
         end; 
     KRUZNICA=record 
                 s:BOD; 
                 r:real; 
              end; 
var A:BOD; 
    k:KRUZNICA; 
    vzd :real; 
 
procedure VSTUP; 
  begin 
  writeln(’Zadaj suradnice bodu ’); 
  readln(A.x,A.y); 
  writeln(’Zadaj stred kruznice a polomer’); 
  readln(k.s.x,k.s.y,k.r); 
  end; 
 
function VZDIALENOST:real; 
  begin 
  vzdialenost:=sqrt(sqr(A.x-k.s.x)+sqr(A.y-k.s.y)); 
  end; 
 
BEGIN 
clrscr; 
VSTUP; 
vzd:=VZDIALENOST; 
if vzd>k.r  
  then writeln(’Bod lezi vo vonkajsej oblasti kruznice.’) 
  else if vzd=k.r  
         then writeln(’Bod lezi na kruznici.’) 
         else writeln(’Bod lezi vo vnutornej oblasti kruznice.’); 
readln; 
END. 
 

Príkaz with 
 Ak v časti programu používame opakovane tú  istú  položku alebo používame viacej položiek tej istej 

premennej typu zá znam, príkaz with umožňuje zjednodušiť resp. skrá tiť zá pis k prístupu k tý mto položká m. 

Príkaz with má  tvar:  with z do p  kde z je premenná  typu zá znam a p je príkaz 

V príkaze p je dovolené označovať položky premennej z len identifiká tormi, t.j. zá pis z.položka skrá tiť na 

zá pis položka. 

Napríklad:  with k do begin writeln(’Zadaj stred kruznice a polomer’); readln(s.x , s.y , r) end; 

 with Z , DatNar do begin OsCislo:=99; Den:=1; Mes:=1; Rok:=2000 end; 

Pri zá pise with z1,z2 do p možno v príkaze p skrá tene označovať jednak položky zá znamu z2, ale aj tie 

položky zá znamu z1, ktoré sa nezhodujú  v označení so žiadnou položkou zá znamu z2. 

Príklad ZAZ-2: Vytvorte jednoduchú  kartotéku na evidenciu osôb (evidenčné číslo, meno a adresa: miesto, 

ulica). 

Analý za: Na precvičenie ú dajového typu record vytvoríme jednoduchú  kartotéku. Jednotlivé karty budú  

uložené v poli, čo v skutočnosti nikdy nebý va, lebo ukončením programu by sme prišli o všetky ú daje 

v kartá ch. Na zapamätanie ú dajov aj po vypnutí počítača slú žia vonkajšie pamäte (najmä pevný  disk). 

Ukladať ú daje na vonkajšie pamäte umožňuje ú dajový  typ sú bor, o ktorom si povieme už v nasledujú cej 

kapitole. 
program ZAZ_2; 
uses  crt; 
const pk=10;        { pocet kariet } 
      dmena=15; 
      dadr=25; 
type  karta=record 
           ec:byte; 
           meno:string[dmena]; 
           adr:record 
               miesto, 
               ulica:string[dadr]; 
               end; 
            end; 
      kartoteka=array[1..pk] of karta; 
var   k:kartoteka; 
      apk,i:byte; 
      q:char; 
 
procedure vstup_karta; 
  begin 
  clrscr; 
  apk:=apk+1; 
  with k[apk],adr do 
    begin 
    ec:=apk; 
    gotoxy(1,8); 
    writeln(’Evidencne cislo : ’:40,ec); 
    writeln; 
    write(’Meno a priezvisko : ’:40);readln(meno); 
    writeln; 
    write(’  Adresa - miesto : ’:40);readln(miesto); 
    writeln; 
    write(’         -  ulica : ’:40);readln(ulica); 
    end; 
  end; 
 
procedure vystup_kartoteka; 
  procedure vystup_karta(i:byte); 
    begin 
    clrscr; 
    writeln(’Vystup’); 
    gotoxy(1,8); 
    with k[i],adr do 
      begin 
      writeln(’  Evidencne cislo : ’:40,ec); 
      writeln; 
      writeln(’Meno a priezvisko : ’:40,meno); 
      writeln; 
      writeln(’  Adresa - miesto : ’:40,miesto); 
      writeln; 
      writeln(’          - ulica : ’:40,ulica); 
      end; 
    readln; 
    end; 

  begin   {procedura vystup_kartoteka } 
  for i:=1 to apk do vystup_karta(i); 
  end; 
 
procedure menu; 
  begin 
  repeat 
    clrscr; 
    gotoxy(1,10); 
    writeln(’VLOZIT KARTU........................ +’:55); 
    writeln; 
    writeln(’VYPISAT KARTOTEKU............... Enter’:55); 
    writeln; 
    writeln(’KONIEC.............................. K’:55); 
    q:=upcase(readkey); 
    case q of 
      ’+’     : vstup_karta; 
      chr(13) : vystup_kartoteka; 
    end; 
  until q=’K’; 
  end; 
 
BEGIN 
apk:=0; 
vstup_karta; 
menu; 
END. 

Variantný  záznam 
 Pri používaní ú dajového typu zá znam môže vyvstať požiadavka na ná vrh takej štruktú ry zá znamu, 

v ktorej sa na daný  prvok nemusí vzťahovať celý  zoznam položiek uvedený ch v zá zname, ale iba jeho určitá  

časť. Takto môže vzniknú ť niekoľko alternatívnych typov – variant - jedného typu zá znam. Takú to situá ciu 

ná m umožňuje riešiť tzv. variantný  zá znam, ktorý  sa skladá  z pevnej časti a z variantnej časti. 

Variantný  zá znam má  tvar: record 

 pevná  časť; 

 case rozlišovacia_položka : rozlišovací_typ_variantnej_časti of 

 rozlišovacia_konštanta1 : ( položka1 : typ1;  

 položka2 : typ2; 

 ...      ); 

 rozlišovacia_konštanta2 : (položka1 : typ1;  

 položka2 : typ2; 

 ...      ); 

 ... 

 end 

kde rozlišovací typ variantnej časti môže byť len ordiná lny typ. 

Napríklad: 

type  TypPohlavia =(MUZ,ZENA); 

 TypMiery=(PRSIA,PAS,BOKY); 

OSOBA = record 

 Priezvisko, 

 Meno   : string[15]; 

 DatNar : string[10]; 

 case Pohlavie : TypPohlavia of 

 MUZ : (Hmotnost : real; 

 Brada   : boolean); 

 ZENA : (Miery : array[TypMiery] of  byte) 

 end; 

Príklad ZAZ-3: Program ZAZ-2 doplňte o variantný  zá znam evidujú ci u mužov fú zy a u žien počet detí. 

Analý za: Aj keď je program veľmi podobný  predchá dzajú cemu, uvá dzame ho podrobne až po celkom 

spoločnú  časť. 
program ZAZ_3; 
uses  crt; 
const pk=10;            { pocet kariet } 
      dmena=15; 
      dadr=25; 
type  karta=record 
           ec:byte; 
           meno:string[dmena]; 
           adr:record 
               miesto, 
               ulica:string[dadr]; 
               end; 
           case pohl:(muz,zena) of 
             muz:(fuzy:boolean); 
             zena:(deti:integer); 
           end; 
     kartoteka=array[1..pk] of karta; 
var  k:kartoteka; 
     apk,i:byte; 
     q:char; 
 
procedure vstup_karta; 
  var odp:char; 
  begin 
  clrscr; 
  apk:=apk+1; 
  with k[apk],adr do 
    begin 
    ec:=apk; 
    gotoxy(1,8); 
    writeln(’Evidencne cislo : ’:40,ec); 
    writeln; 
    write(’Meno a priezvisko : ’:40);readln(meno); 
    writeln; 
    write(’  Adresa - miesto : ’:40);readln(miesto); 
    writeln; 
    write(’         -  ulica : ’:40);readln(ulica); 
    writeln; 
    write(’  Muz (ano - A) ? : ’:40); 
    if upcase(readkey)=’A’ 
      then begin 
           writeln(’ano’); 
           pohl:=muz; 
           writeln; 
           write(’Ma fuzy (ano - A) ? : ’:40); 
           fuzy:=upcase(readkey)=’A’; 

           if fuzy then writeln(’ano’) else writeln(’nie’) 
           end 
      else begin 
           writeln(’nie’); 
           pohl:=zena; 
           writeln; 
           write(’Pocet deti : ’:40);readln(deti); 
           end; 
    delay(500) 
    end; 
  end; 
 
procedure vystup_kartoteka; 
  procedure vystup_karta(i:byte); 
    begin 
    clrscr; 
    writeln(’Vystup’); 
    gotoxy(1,8); 
    with k[i],adr do 
      begin 
      writeln(’  Evidencne cislo : ’:40,ec); 
      writeln; 
      writeln(’Meno a priezvisko : ’:40,meno); 
      writeln; 
      writeln(’  Adresa - miesto : ’:40,miesto); 
      writeln; 
      writeln(’          - ulica : ’:40,ulica); 
      writeln; 
      if pohl=muz 
        then if fuzy 
               then writeln(’Fuzy : ano’:43) 
               else writeln(’Fuzy : nie’:43) 
        else writeln(’Pocet deti : ’:40,deti); 
      end; 
    readln; 
    end; 
  begin 
  for i:=1 to apk do vystup_karta(i); 
  end; 
... 
 

Na zá ver ú dajového typu zá znam uvá dzame „červíka Wurmiho“ (Príklad POLE-13) doplneného 

o jeho rast, pridanie člá nku tela, pri každom zožratí potravy (listu). 

 

Príklad ZAZ-4: Program POLE-13 doplňte o „telo“ červíka Wurmiho, ktoré narastie o jeden člá nok pri 

každom zožratí potravy. 

Analý za: Program POLE-13 treba doplniť o proces zapamätania si polohy člá nkov tela Wurmiho.  

Z viacerý ch možností sme vybrali na zapamätanie polohy člá nkov tela jednorozmerné pole TELO, v ktorom 

sú  ako položky recordu zapamätané x-ová  a y-nová  sú radnica 1., 2. atď. člá nku tela. 

Program ponechá vame na samoštú dium, podprogramy totožné s programom POLE-13 neuvá dzame. 
program ZAZ_4; 
uses  crt,dos; 
const pocet=5; 
... 
var   body,dlzka:integer; 
      px,py,jx,jy:array[1..pocet] of integer; 
      telo:array[0..pocet] of record x,y:integer end; 
 
procedure skry_kurzor; 
... 

procedure ohrada; 
... 
procedure generuj_potravu_jed; 
... 
procedure vykresli_potravu_jed; 
... 
procedure inicializacia; 
  var limit:integer; 
  begin 
  randomize; 
  generuj_potravu_jed; 
  skry_kurzor; 
  clrscr; 
  ohrada; 
  vykresli_potravu_jed; 
  dlzka:=0; telo[0].x:=40; telo[0].y:=12; 
  gotoxy(telo[0].x,telo[0].y); write(tvaricka); 
  body:=0; 
  end; 
 
function jed:boolean; 
  var i:integer; 
  begin 
  i:=0; 
  repeat 
    inc(i) 
  until ((jx[i]=telo[0].x) and (jy[i]=telo[0].y)) or (i=pocet); 
  jed:=(jx[i]=telo[0].x) and (jy[i]=telo[0].y) 
  end; 
 
function potrava:boolean; 
  var i:integer; 
  begin 
  i:=0; 
  repeat 
    inc(i) 
  until ((px[i]=telo[0].x) and (py[i]=telo[0].y)) or (i=pocet); 
  if (px[i]=telo[0].x) and (py[i]=telo[0].y) 
    then begin potrava:=true; inc(body); px[i]:=0; py[i]:=0 end 
    else potrava:=false; 
  gotoxy(3,2); write(body); 
  end; 
 
procedure pohyb; 
  var stlznak:char; 
      i:integer; 
  begin 
  repeat 
    if keypressed then stlznak:=readkey; 
    if potrava then inc(dlzka); 
    gotoxy(telo[dlzka].x,telo[dlzka].y); write(’ ’); 
    for i:=dlzka downto 1 do 
      begin 
      telo[i]:=telo[i-1]; 
      gotoxy(telo[i].x,telo[i].y); write(’0’); 
      end; 
    case stlznak of 
      ESC:halt; 
      sipkaVPRAVO:inc(telo[0].x); 
      sipkaVLAVO :dec(telo[0].x); 
      sipkaHORE  :dec(telo[0].y); 
      sipkaDOLE  :inc(telo[0].y); 
      end; 
    gotoxy(telo[0].x,telo[0].y); write(tvaricka); 
    delay(100); 
  until (telo[0].x=1)or(telo[0].x=80)or(telo[0].y=1)or(telo[0].y=24)or jed or 

(body=pocet); 
  if jed 
    then begin gotoxy(33,12); write(’ J E D ! ! ! ’) end 
    else if body=pocet 
           then begin gotoxy(20,12); write(’V Y B O R N E !!! VYHRAL SI!!!’) end 
           else begin gotoxy(25,12); write(’B U M ! ! !  O H R A D A ! ! !’) end  
  end; 
 
BEGIN 
inicializacia; 
pohyb; 
readln 
END. 
 

 

Nerieš ené  úlohy na zá znam: 

5. Navrhnite ú sek definícií a deklará cií pre evidenčnú  kartu knihy a čitateľa v knižnici. 

6. Vytvorte program na evidenciu kníh (čitateľov) knižnice. 

7. Program z ú lohy 6 doplňte o vyradenie čitateľa (knihy) z evidencie. 

8. Program z ú lohy 6 doplňte o vyhľadanie knihy (čitateľa) po zadaní mena. 

9. Vytvorte program na evidenciu motorový ch vozidiel. 

10. Program z ú lohy 9 doplňte o variantný  zá znam, ktorý  podľa typu vozidla bude evidovať napr. pri 

osobnom aute počet miest a nosnosť, pri autobuse počet miest, klimatizá ciu a pod., pri ná kladnom aute 

nosnosť, ná kladnú  plošinu a pod. 

11. Vytvorte program pracujú ci ako elektronický  dvojjazyčný  (viacjazyčný ) slovník. 

12. Vytvorte program, ktorý  po zadaní textu zistí počet vý skytov jednotlivý ch slov v texte. 

13. Pre typ RETAZEC = record  

 POCETZN : 0..255;  
 ZNAKY : array[1..255] of char 

 end; 

       napíšte procedú ru, ktorá  spojí dva reťazce do tretieho. 

14. K ú dajovému typu z príkladu 13 napíšte funkciu, ktorá  zistí, či jeden reťazec je obsiahnutý  v druhom (je 

podreťazcom). 

Ú dajový  typ súbor 
 Ak potrebujeme uchovať ú daje aj po vypnutí počítača resp. po ukončení programu, musíme ich 

uložiť na vonkajšie pamäťové médium – disk, disketu a pod. Zapisovať a čítať ú daje z vonkajších 

pamäťový ch médií ná m v TP umožňuje ú dajový  typ sú bor. 

Súbor je štruktú rovaný  ú dajový  typ, ktorý  sa skladá  z teoreticky neobmedzeného počtu zložiek, všetky 

rovnakého typu. Prakticky je počet zložiek sú boru obmedzený  kapacitou vonkajšej pamäte. 

Definícia typu sú bor má  tvar:   type mt = file of tz 

kde mt je meno typu a tz je typ zložky, ktorý  nesmie byť typ sú bor ani typ obsahujú ci ako zložku typ sú bor. 

Napríklad: type CELE = file of integer;   typ sú bor obsahujú ci celé čísla 

  type KARTOTEKA = file of KARTA;  typ umožňujú ci uloženie kartotéky do sú boru 

  type BODY = file of record x,y,z:real end; typ sú bor obsahujú ci trojice x,y,z 

  var F : file of string;    premenná  typu sú bor obsahujú ci reťazce 

 

 Sú bory môžu byť vnú torné (pracovné) a vonkajšie. Pracovné sú bory si môže vytvoriť program pri 

riešení pamäťovo ná ročnejších ú loh, TP im dá va príponu $$$. Po skončení programu sa môžu z vonkajšej 

pamäte odstrá niť. Ď alej sa budeme zaoberať len vonkajšími sú bormi, ktoré existujú  aj po skončení programu 

a umožňujú  trvalé uloženie dá t. 

 Zá kladné príkazy pre prá cu so sú bormi umožňujú  len sekvenčné spracovanie sú boru, t.j. vytvá ranie 

aj čítanie sú boru len zložku po zložke (od prvej k druhej atď.). V každom okamihu spracovania sú boru je 

prístupná  len jedna zložka sú boru, na ktorú  „ukazuje“ prístupová  premenná  sú boru. Ak F je premenná  typu 

sú bor, k nej prislú chajú ca prístupová  premenná  má  označenie F^ a vznikne automaticky pri deklará cii 

premennej typu sú bor. 

 Deklará ciou premennej typu sú bor vznikne len abstraktný  vonkajší sú bor, sú  určené len jeho logické 

vlastnosti. K spojeniu logického sú boru s fyzický m – skutočne existujú cim na konkrétnom vonkajšom 

pamäťovom médiu, dôjde príkazom  assign ( logické_meno , fyzické_meno )  kde logické meno je 

premenná  typu sú bor a fyzické meno je reťazec obsahujú ci ná zov sú boru na disku, prípadne doplnený  aj 

o cestu. Napríklad: assign(F,´CITATEL.DAT )́, assign(SUB,´C:\TP\CELE1.DBF )́. 

Zá pis ú dajov do sú boru (napríklad F):  

 Zá pis ú dajov do sú boru (po stotožnení logického a fyzického mena sú boru!) prebieha v dvoch 

fá zach: 

1. sú bor sa pripraví na zá pis príkazom rewrite(F). Vytvorí sa prá zdny sú bor (prá zdna hodnota sú boru), 

prístupová  premenná  F^ sa nastaví na „koniec“ sú boru. Ak sú bor už obsahuje nejaké zložky, budú  

odstrá nené! 

2. vlastný  zá pis do sú boru sa realizuje príkazom write(F,V), kde V je vý raz rovnakého typu, ako je typ 

zložky sú boru. Po vyhodnotení vý razu V, získaní konkrétnej hodnoty, dôjde k jej zá pisu do sú boru F. 

Bod 2 možno ľubovoľný  počet krá t opakovať. 

Čítanie ú dajov zo sú boru (napríklad F): 

 Aj sekvenčné čítanie ú dajov zo sú boru prebieha v dvoch fá zach: 

1. sú bor sa pripraví na čítanie príkazom reset(F). Príkaz spôsobí nastavenie prístupovej premennej F^ na 

začiatok sú boru. 

2. vlastné čítanie zo sú boru sa realizuje príkazom read(F,X), kde X je premenná  rovnakého typu ako je typ 

zložky sú boru. Po vykonaní príkazu je v premennej X hodnota načítaná  zo sú boru. Čítanie zo sú boru 

možno opakovať, až ký m prístupová  premenná  nie je za poslednou položkou sú boru.  

Na zistenie pozície prístupovej premennej v sú bore slú ži funkcia eof (end of file), ktorá  má  hodnotu true, ak 

prístupová  premenná  je na konci sú boru (hodnota F^ nie je definovaná ), inak má  hodnotu false. To možno 

vý hodne využiť na načítanie všetký ch zložiek sú boru schémou: 

  reset(F); 

  while not eof(F) do { pokiaľ nie je koniec sú boru opakuj } 

 begin 
 read(F,X);  { načíta hodnotu zo sú boru do premennej X } 
 write(X)  { zobrazí načítanú  hodnotu na obrazovke } 
 end; 

Ukončenie prá ce so sú borom (napríklad F): 

 Po ukončení prá ce so sú borom sa sú bor musí zavrieť príkazom close(F). 

 

Príklad SUB-1: Vytvorte program na vytvorenie sú boru a zobrazenie zložiek sú boru obsahujú ceho čísla 

typu integer. 

Analý za: Program obsahuje dve samostatné procedú ry. Procedú ra VSTUP umožňuje vytvoriť, po zadaní 

mena fyzického sú boru, sú bor obsahujú ci celé čísla a procedú ra VYSTUP zobrazuje zložky – celé čísla, 

zvoleného sú boru. Program je pre pohodlie doplnený  o menu. 
program SUB_1; 
uses  crt; 
var   F:file of integer; 
      ODP:char; 
 
procedure VSTUP; 
  var FM:string[12]; 
      X:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  rewrite(F); 
  repeat 
    write(’Vlozit cislo: ’); 
    readln(X); 
    write(F,X); 
    writeln(’Vlozit dalsie cislo (nie - N) ?’:52); 
  until upcase(readkey)=’N’; 
  close(F); 
  end; 
 
procedure VYSTUP; 
  var FM:string[12]; 
      X:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  reset(f); 
  while not eof(F) do 

    begin 
    read(F,X); 
    write(X:4); 
    end; 
  close(F); 
  readln 
  end; 
 
BEGIN 
repeat 
  clrscr; 
  gotoxy(1,10); 
  writeln(’Vytvorit subor ............... V’:50); 
  writeln; 
  writeln(’Zobrazit subor ............... Z’:50); 
  writeln; 
  writeln(’Koniec ....................... K’:50); 
  repeat ODP:=upcase(readkey); until (ODP=’V’)or(ODP=’Z’)or(ODP=’K’); 
  clrscr; 
  case ODP of 
    ’V’ : VSTUP; 
    ’Z’ : VYSTUP; 
  end 
until ODP=’K’ 
END. 
 

TP má  pre prá cu so sú bormi ďalšie procedú ry a funkcie. Funkcia FileSize(F) vracia číslo udá vajú ce 

počet zložiek sú boru F. Procedú ra Seek(F,n) nastaví prístupovú  premennú  za n-tú  zložku sú boru F. Preto 

príkaz Seek(F,FileSize(F)) nastaví prístupovú  premennú  F^ za poslednú  zložku sú boru a umožní ná m 

napríklad zá pis na koniec sú boru F. Príkaz Rename(F,nové_fyz._meno) premenuje externý  sú bor (po príkaze 

assign). Príkaz Erase(F) zmaže externý  sú bor (po príkaze assign). 

Uvedené príkazy si precvičíme v nasledujú com príklade. 

 

Príklad SUB-2: Program SUB-1 (predchá dzajú ci príklad) doplňte o podprogramy na vytvorenie prá zdneho 

sú boru, nazistenie počtu zložiek sú boru, na pridanie zložky na koniec sú boru a na odstrá nenie zvolenej 

zložky zo sú boru. 

Analý za: Program SUB-2 je určený  na ozrejmenie zá kladný ch techník prá ce s ú dajový m typom sú bor. 

Ošetrili sme otvorenie sú boru funkciou IOResult, ak sú bor so zadaný m menom na disku neexistuje, vypíše sa 

sprá va: Sú bor neexistuje! Najkomplikovanejšie je odstrá nenie zvolenej zložky zo sú boru, kde sa musí použiť 

aj pomocný  sú bor (zo zdrojového sú boru sa číta a do pomocného zapisuje), ktorý  po zapísaní zvolený ch 

zložiek je premenovaný  na pôvodný  zdrojový  sú bor (pôvodný  sú bor musí byť najprv vymazaný ). 
program SUB_2; 
uses  crt; 
var   F:file of integer; 
      FM:string[12]; 
      ODP:char; 
      CHYBA:boolean; 
 
procedure PRAZDNY; 
  var FM:string[12]; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  rewrite(F); 
  close(F) 

  end; 
 
procedure VSTUP; 
  var FM:string[12]; 
      X:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  rewrite(F); 
  repeat 
    write(’Vlozit cislo: ’); 
    readln(X); 
    write(F,X); 
    writeln(’Vlozit dalsie cislo (nie - N) ?’:52); 
  until upcase(readkey)=’N’; 
  close(F); 
  end; 
 
procedure VYSTUP; 
  var FM:string[12]; 
      X:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  {$I-} reset(F); {$I+} 
  CHYBA:=IOResult<>0; 
  if CHYBA 
    then writeln(’Subor neexistuje!’) 
    else begin 
         if filesize(F)=0 then writeln(’Subor je prazdny!’); 
         while not eof(F) do 
           begin 
           read(F,X); 
           write(X:4) 
           end; 
         close(F) 
         end; 
  readln 
  end; 
 
procedure POCET; 
  var FM:string[12]; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  {$I-} reset(F); {$I+} 
  CHYBA:=IOResult<>0; 
  if CHYBA 
    then writeln(’Subor neexistuje!’) 
    else begin 
         writeln(’Pocet zloziek: ’,filesize(F)); 
         close(F); 
         end; 
  readln 
  end; 
 
procedure PRIDAT_NA_KONIEC; 
  var FM:string[12]; 
      X:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 

  {$I-} reset(F); {$I+} 
  CHYBA:=IOResult<>0; 
  if CHYBA 
    then writeln(’Subor neexistuje!’) 
    else begin 
         write(’Pridat hodnotu: ’); 
         readln(X); 
         seek(F,filesize(F)); 
         write(F,X); 
         close(F); 
         end 
  end; 
 
procedure ODSTRANIT; 
  var FPom:file of integer; 
      FM:string[12]; 
      X,HlHodnota:integer; 
  begin 
  write(’Meno suboru na disku: ’); 
  readln(FM); 
  assign(F,FM); 
  {$I-} reset(F); {$I+} 
  CHYBA:=IOResult<>0; 
  if CHYBA 
    then writeln(’Subor neexistuje!’) 
    else begin 
         assign(FPom,’POMOCNY.$$$’); 
         rewrite(FPom); 
         write(’Odstranit hodnotu: ’); 
         readln(HlHodnota); 
         while not eof(F) do 
           begin 
           read(F,X); 
           if X<>HlHodnota then write(FPom,X) 
           end; 
         close(F); close(FPom); 
         assign(F,FM); erase(F); 
         assign(FPom,’POMOCNY.$$$’); rename(FPom,FM) 
         end 
  end; 
 
BEGIN 
repeat 
  clrscr; 
  gotoxy(1,6); 
  writeln(’Vytvorit prazdny subor ....... S’:50); 
  writeln; 
  writeln(’Vytvorit neprazdny subor ..... V’:50); 
  writeln; 
  writeln(’Zobrazit subor ............... Z’:50); 
  writeln; 
  writeln(’Pocet zloziek suboru ......... P’:50); 
  writeln; 
  writeln(’Pridat zlozku na koniec ...... C’:50); 
  writeln; 
  writeln(’Odstranit zlozku zo suboru ... O’:50); 
  writeln; 
  writeln(’Koniec ....................... K’:50); 
  repeat ODP:=upcase(readkey); until ODP in [’S’,’V’,’Z’,’P’,’C’,’O’,’K’]; 
  clrscr; 
  case ODP of 
    ’S’ : PRAZDNY; 
    ’V’ : VSTUP; 
    ’Z’ : VYSTUP; 
    ’P’ : POCET; 
    ’C’ : PRIDAT_NA_KONIEC; 

    ’O’ : ODSTRANIT; 
  end 
until ODP=’K’ 
END. 
 

Príklad SUB-3: Kartotéku v programe ZAZ_2 pracujú cu s poľom prerobte na „skutočnú “ kartotéku 

pracujú cu so sú borom. 

Analý zy: Program využíva len už vyššie použité procedú ry a funkcie a preto ho nebudeme podrobnejšie 

opisovať. Odporú čame premenovať si sú bor ZAZ-2 na SUB-3 a potom urobiť príslušné zmeny. 
program SUB_3; 
uses  crt; 
const dmena=15; 
      dadr=25; 
type  karta=record 
           ec:integer; 
           meno:string[dmena]; 
           adr:record 
               miesto, 
               ulica:string[dadr]; 
               end; 
            end; 
      kartoteka=file of karta; 
var   f:kartoteka; 
      k:karta; 
      q:char; 
 
procedure vstup_karta; 
  var pocet:integer; 
  begin 
  reset(f); 
  pocet:=filesize(f); 
  seek(f,pocet); 
  with k,adr do 
    begin 
    inc(pocet); 
    ec:=pocet; 
    gotoxy(1,8); 
    writeln(’Evidencne cislo : ’:40,ec); 
    writeln; 
    write(’Meno a priezvisko : ’:40);readln(meno); 
    writeln; 
    write(’  Adresa - miesto : ’:40);readln(miesto); 
    writeln; 
    write(’         -  ulica : ’:40);readln(ulica); 
    end; 
  write(f,k); 
end; 
 
procedure vystup_karta; 
   begin 
   clrscr; 
   gotoxy(1,8); 
   with k,adr do 
     begin 
     writeln(’  Evidencne cislo : ’:40,ec); 
     writeln; 
     writeln(’Meno a priezvisko : ’:40,meno); 
     writeln; 
     writeln(’  Adresa - miesto : ’:40,miesto); 
     writeln; 
     writeln(’          - ulica : ’:40,ulica); 
     end; 
   readln; 

   end; 
 
procedure vystup_kartoteka; 
  begin 
  reset(f); 
  while not eof(f) do 
    begin 
    read(f,k); 
    vystup_karta; 
    end; 
  end; 
 
procedure najdi; 
  var hlmeno:string[dmena]; 
  begin 
  gotoxy(20,12); 
  write(’Zadaj meno hladaneho: ’); 
  readln(hlmeno); 
  reset(f); 
  while not eof(f) do 
    begin 
    read(f,k); 
    if k.meno=hlmeno then vystup_karta 
    end 
  end; 
 
procedure odstran; 
  var hlmeno:string[dmena]; 
      fn:kartoteka; 
  begin 
  gotoxy(20,12); 
  write(’Zadaj meno na odstranenie: ’); 
  readln(hlmeno); 
  assign(fn,’pracov.pom’); 
  reset(f);rewrite(fn); 
  while not eof(f) do 
    begin 
    read(f,k); 
    if k.meno<>hlmeno then write(fn,k); 
    end; 
  close(f); close(fn); 
  assign(f,’pracov.dat’);  erase(f); 
  assign(fn,’pracov.pom’); rename(fn,’pracov.dat’); 
  assign(f,’pracov.dat’); 
end; 
 
procedure pocet; 
  begin 
  reset(f); 
  gotoxy(30,12); 
  writeln(’Pocet pracovnikov: ’,filesize(f)); 
  readln; 
  end; 
 
procedure menu; 
  begin 
  repeat 
    clrscr; 
    gotoxy(1,6); 
    writeln(’            ZALOZIT KARTU....................... 1’); 
    writeln; 
    writeln(’            NAJST PODLA MENA.................... 2’); 
    writeln; 
    writeln(’            ODSTRANIT S MENOM................... 3’); 
    writeln; 
    writeln(’            POCET PRACOVNIKOV................... 4’); 

    writeln; 
    writeln(’            ZOZNAM PRACOVNIKOV.................. 5’); 
    writeln; 
    writeln(’            KONIEC.............................. 0’); 
    q:=readkey; 
    clrscr; 
    case q of 
      ’1’: vstup_karta; 
      ’2’: najdi; 
      ’3’: odstran; 
      ’4’: pocet; 
      ’5’: vystup_kartoteka; 
    end; 
  until q=’0’; 
  close(f); 
  end; 
 
BEGIN 
clrscr; 
assign(f,’pracov.dat’); 
gotoxy(10,12); 
write(’Vytvorit novy subor pracovnikov (ano - A) ? ’); 
{$I-} reset(f); {$I+} 
if (IOResult<>0) or (upcase(readkey)=’A’) then rewrite(f); 
menu; 
END. 
 
Príklad SUB-4: Vytvorte program, ktorý  spojí dva sú bory do jedného. Nech sú bory obsahujú  najviac 

dvadsaťznakové slová . 

Analý za: Podprogram zabezpečujú ci spojenie dvoch sú borov je pomerne jednoduchý . Môžeme nastaviť 

prístupovú  premennú  na koniec v prvom sú bore a pokračovať v zá pise prvkov z druhého sú boru. Ak má  mať 

spojený  sú bor nový  ná zov, stačí prvý  sú bor (v ňom sú  všetky prvky) premenovať. Program sme doplnili aj 

o utriedenie prvkov sú boru bubblesortom. Na triedenie sú borov síce existujú  špeciá lne triediace algoritmy, 

pre zaujímavosť sme sa však „pohrali“ s bubblesortom. Všimnite si, že dve rôzne logické premenné sú boru 

majú  priradené to isté fyzické meno. Je to možné len za určitý ch podmienok (v bubblesorte sa nemení počet 

zložiek sú boru, len ich poradie) a preto sa taký mto „perličká m“ radšej vyhý bajte. 
program SUB_4; 
uses  crt; 
const DLSLOVA=20; 
type  SLOVO=string[DLSLOVA]; 
      SUBOR=file of SLOVO; 
      NAZOV=string[12]; 
var   FM1,FM2,FMS:NAZOV; 
 
procedure VSTUP(FM:NAZOV); 
  var F:SUBOR; 
      S:SLOVO; 
  begin 
  assign(F,FM); 
  rewrite(F); 
  repeat 
    write(’Vlozit slovo: ’); 
    readln(S); 
    write(F,S); 
    writeln(’Koniec - stlac 0’:45) 
  until upcase(readkey)=’0’; 
  close(F) 
  end; 
 
procedure VYSTUP(FM:NAZOV); 

  var F:SUBOR; 
      S:SLOVO; 
  begin 
  assign(F,FM); 
  reset(F); 
  while not eof(F) do 
    begin 
    read(F,S); 
    write(S,’ ’) 
    end; 
  writeln; 
  close(F) 
  end; 
 
procedure SPOJ(FM1,FM2,FMS:NAZOV); 
  var F1,F2:SUBOR; 
      S:SLOVO; 
  begin 
  assign(F1,FM1); reset(F1); 
  assign(F2,FM2); reset(F2); 
  seek(F1,filesize(F1)); 
  while not eof(F2) do 
    begin 
    read(F2,S); 
    write(F1,S) 
    end; 
  close(F1); close(F2); 
  assign(F1,FMS); erase(F1); 
  assign(F1,FM1); rename(F1,FMS); 
  end; 
 
procedure BUBBLESORT(FM:NAZOV); 
  var F,FPom:SUBOR; 
      A,B:SLOVO; 
      DL,PP,I:integer; 
  begin 
  assign(F,FM); assign(FPom,FM); 
  reset(F); 
  DL:=filesize(F); 
  for PP:=1 to DL-1 do 
    begin 
    reset(F); reset(FPom); 
    read(F,A); 
    for I:=1 to DL-PP do 
      begin 
      read(F,B); 
      if A<B 
        then begin write(FPom,A); A:=B end 
        else write(FPom,B) 
      end; 
      write(FPom,A) 
    end; 
  close(F);close(FPom) 
  end; 
 
BEGIN 
clrscr; 
write(’Meno prveho suboru na disku: ’); readln(FM1); 
VSTUP(FM1); 
write(’Meno druheho suboru na disku: ’); readln(FM2); 
VSTUP(FM2); 
write(’Meno spojeneho suboru na disku: ’); readln(FMS); 
clrscr; 
writeln; writeln(’Prvy subor:’); 
VYSTUP(FM1); 
writeln; writeln(’Druhy subor:’); 

VYSTUP(FM2); 
writeln; writeln(’Spojeny subor:’); 
SPOJ(FM1,FM2,FMS); 
VYSTUP(FMS); 
writeln; writeln(’Utriedeny subor:’); 
BUBBLESORT(FMS); 
VYSTUP(FMS); 
readln 
END. 
 

Textové sú bory 
 Existujú  sú bory dá t, ktoré sú  členené na riadky. Takéto sú bory nazý vame textové a ich typ sa 

označuje štandardný m identifiká torom typu text. V textovom sú bore F okrem funkcie eof(F) je deklarovaná  

aj funkcia eoln(F) (end of line), slú žiaca na rozpoznanie konca riadku. Príkazy read(F,P) a write(F,V) sú  

rozšírené o príkazy readln(F,P) – z textového sú boru F sa prečíta lexiká lna jednotka (hodnota) 

zodpovedajú ca typu premennej P a prejde sa na nový  riadok, a writeln(F,V) – do textového sú boru F sa 

zapíše hodnota vý razu V a oddeľovač riadkov eoln. 

Najzná mejšie sú bory typu text sú  štandardný  vstupný  sú bor input, ktorému je priradené vstupné 

zariadenia klá vesnica, a štandardný  vý stupný  sú bor output, ktorému je priradené vý stupné zariadenie 

monitor. Definovanie ich typu ako sú borov typu text je implicitné. K otvoreniu oboch sú borov dochá dza 

automaticky po spustení programu. Príkazy read(input,P), readln(input,P), write(output,V) 

a writeln(output,V) sa používajú  bez identifiká torov input a output. 

 

Príklad SUB-5: Vytvorte program, ktorý  bude čítať znaky z klá vesnice (vstupný  sú bor input) a pre každý  

riadok zobrazí, koľkokrá t sa v ňom vyskytuje jeho prvý  znak. 

Analý za: Program musí obsahovať dva vnorené cykly. Vonkajší cyklus zabezpečuje čítanie znakov, až ký m 

nie je načítaný  koniec sú boru. Vnú torný  cyklus číta znaky, až ký m nie je koniec riadku, potom musí 

odriadkovať. Vložiť znak koniec riadku nie je problém, stačí stlačiť Enter (chr(13)). Ak si spomenieme, že 

„v starom dobrom DOSe“ sa koniec sú boru vkladal kombiná ciou klá ves Ctrl+Z, sme „zachrá není“. 
program SUB_5; 
var pocet:integer; 
    z,s:char; 
begin 
writeln(’Vkladaj text, ukoncenie - Ctrl+Z !!!’); 
while not eof do 
  begin 
  read(z); 
  pocet:=1; 
  while not eoln do 
    begin 
    read(s); 
    if s=z then inc(pocet); 
    end; 
  writeln(’Pocet vyskytov znaku ’,z,’ je ’,pocet,’-krat’); 
  readln; 
  end; 
end. 
 

Príklad SUB-6: Pokú ste sa vysvetliť, prečo oba ďalej uvedené programy SUB-6a aj SUB-6b pracujú  

rovnako. Ich ú lohou je uložiť do sú boru na disk text z klá vesnice. 

Program SUB_6a; 
var z:char; 
    T:text; 
    FM:string[12]; 
 
BEGIN 
write(’Meno suboru na disku: ’); 
readln(FM); 
assign(T,FM); 
rewrite(T); 
writeln(’Vkladaj text, koniec - vloz Ctrl+Z !!!’); 
while not eof do 
  begin 
  while not eoln do 
    begin 
    read(z); 
    write(T,z) 
    end; 
  readln; 
  writeln(T) 
  end; 
writeln; 
writeln(’Ulozeny subor:’); 
reset(T); 
while not eof(T) do 
  begin 
  while not eoln(T) do 
    begin 
    read(T,z); 
    write(z); 
    end; 
  readln(T); 
  writeln 
  end; 
close(T); 
readln 
END. 
 
program SUB_6b; 
uses crt; 
var  z:char; 
     T:text; 
     FM:string[12]; 
 
BEGIN 
clrscr; 
write(’Meno suboru na disku: ’); 
readln(FM); 
assign(T,FM); 
rewrite(T); 
writeln(’Vkladaj text, koniec - stlac ‘’); 
repeat 
  read(z); 
  if z<>’‘’ then write(T,z) 
until z=’‘’; 
writeln; 
writeln(’Ulozeny subor:’); 
reset(T); 
while not eof(T) do 
  begin 
  read(T,z); 
  write(z); 
  end; 
close(T); 
readln;readln 
END. 

Nerieš ené  úlohy na súbor: 

7. Kartotéku z príkladu SUB-3 doplňte o položku rodné číslo a podprogram, ktorý  zistí počet osôb 

mužského a ženského pohlavia v kartotéke (u mužov je v rodnom čísle na 3. mieste 0 alebo1, u žien 5 

alebo 6). 

8. Kartotéku z príkladu SUB-3 doplňte o položku plat a podprogram, ktorý  vypíše osoby, ktorý ch plat spĺňa 

zadané kritériá  (menší, väčší, rovný ). 

9. Vytvorte program, ktorý  zo sú boru reá lnych čísel prepíše do nového sú boru len celé čísla. 

10. Ú loha z jednorozmerného poľa o vyhľadá vaní, počte vý skytov, miesta vý skytu a pod. zrealizujte pre 

sú bor celý ch čísel (znakov, reťazcov). 

11. Vytvorte program, ktorý  z daného textového sú boru odstrá ni všetky viacná sobné medzery a prá zdne 

riadky. 

12. Vytvorte program slú žiaci ako elektronický  dvojjazyčný  slovník. 

13. Vytvorte program, ktorý  bude čítať znaky z klá vesnice a pre každý  riadok určí sú čet ordiná lnych čísel 

znakov vyskytujú cich sa v riadku. 

Dynamické premenné 
 Ú dajové typy, s ktorý mi sme sa zaoberali doteraz, patria medzi tzv. statické ú dajové typy. Premenné 

prislú chajú ce statický m typom sú  zavedené v ú seku definícií a deklará cií a je im trvalo vyhradené miesto 

v pamäti počas behu programu. Počet a rozsah premenný ch sa počas prá ce v bloku, t.j. aj v programe,  

nemôže meniť. 

 Jazyk pascal umožňuje vytvá rať premenné nie len v ú seku deklará cií premenný ch, ale aj 

v príkazovej časti. Dokonca možno tieto premenné v príkazovej časti aj rušiť. Premenné s tý mito 

vlastnosťami nazý vame dynamické  premenné  a príslušné ú dajové typy nazý vame dynamické ú dajové 

štruktú ry. 

 Z toho, čo sme uviedli doteraz, vyplý va, že dynamické premenné nemožno zaviesť v ú seku 

deklará cií a teda zabezpečiť prístup k ich hodnotá m pomocou ich identifiká torov. Dynamické premenné 

vznikajú  a zanikajú  počas realizá cie programu a sprístupnenie ich hodnôt sa robí pomocou nového 

ú dajového typu ukazovateľ. 

Ak t je identifiká tor nejakého typu, tak typ ukazovateľ definujeme zá pisom: 

type u = ^ t 

kde u je meno (identifiká tor) typu ukazovateľ. Hovoríme aj, že typ t je zviazaný  s typom u a množina hodnôt 

premennej typu ukazovateľ „ukazuje“ na prvky typu t. 

Napríklad: 

 type UKAZ1 = ^ integer; 

UKAZ2 = ^ PRVOK; 

Ak p je premenná  typu ukazovateľ, tak premenná  typu zviazaného s typom ukazovateľ (premenná  typu t) sa 

nazý va dynamická  premenná  a označuje sa p^. 

Grafická  interpretá cia: 

            p : ^ t „ukazuje“ p ^ : t  

 

 premenná  typu dynamická  
 ukazovateľ premenná  

 

Pomocou hodnoty premennej typu ukazovateľ sprístupňujeme dynamickú  premennú . V ú seku 

definícií a deklará cií je zavedená  len premenná  typu ukazovateľ! Do množiny hodnôt premennej typu 

ukazovateľ patrí vždy aj hodnota nil, jediná  konštanta typu ukazovateľ, ktorá  neukazuje na nijaký  prvok (na 

žiadnu dynamickú  premennú ). 

Graficky:     p : ^ t 

 

      p := nil 

Premenná  p má  síce priradenú  hodnotu, ale „neukazuje“ na žiadnu dynamickú  premennú . 

Príkazom new(p), kde p je premenná  typu ukazovateľ, sa vytvorí dynamická  premenná  typu t 

s nedefinovanou hodnotou a ukazovateľ na tú to premennú  sa uloží do premennej p. Príkazom new(p) 

priraďujeme premennej p hodnotu, ktorou je zodpovedajú ca dynamická  premenná  a ktorú  označujeme p^. 

Graficky:        p : ^ t    p^ : t 

                                

 new(p) 

 

Príkazom dispose(p), kde p je premenná  typu ukazovateľ, sa zruší dynamická  premenná  p^, na ktorú  

ukazovala premenná  p. Hodnota premennej p nie je po ukončení príkazu dispose(p) definovaná . Pamäťový  

priestor, ktorý  zaberala dynamická  premenná , sa dal k dispozícii na ďalšie použitie procesorom. 

Zhrnutie:                

var  p : ^t ;                  p 

 (v pamäti počítača sa vyhradí pamäťové miesto pre statickú  premennú  typu ukazovateľ, t.j. 32 bitov pre 

„akú si“ adresu) 

p := nil ;        p 

(do pamäťového miesta vyhradeného pre premennú  typu ukazovateľ sa uloží číslo neexistujú cej adresy, 

napríklad 0) 

new (p) ;       p 

                                        p^ : t 

(do pamäťového miesta vyhradeného pre premennú  typu ukazovateľ sa uloží adresa vytvorenej dynamickej 

premennej p^) 

 

Lineárny jednosmernezreť azený  zoznam 
 Z dynamický ch štruktú r sa najčastejšie využíva lineá rny jednosmernezreťazený  zoznam. Graficky ho 

možno zná zorniť: 

 

 

kde Z je premenná  typu ukazovateľ (pointer), ktorá  ukazuje na prvú  dynamickú  premennú  v zozname, 

ktorého prvky sú  typu zá znam (record) s dvoma položkami: s hodnotou (h1 až h5) a s ukazovateľom na ďalší 

prvok zoznamu. Ukazovateľ posledného prvku zoznamu už nemá  kde ukazovať a preto má  hodnotu nil. 

Postup v zozname je sekvenčný , od ukazovateľa na prvý  prvok cez ukazovatele na nasledujú ci prvok až na 

koniec zoznamu. V ú seku definícií a deklará cií je zavedený  len ukazovateľ na zoznam (premenná  Z typu ^t) 

a typ dynamickej premennej (t). Vlastné prvky zoznamu, dynamické premenné, vznikajú  podľa potreby 

počas prá ce programu príkazom new. Na pohyb v zozname používame pomocnú  premennú  typu ukazovateľ 

a premennú  Z nechá vame väčšinou ukazovať vždy na začiatok zoznamu (inak sa nemá me možnosť dostať na 

začiatok zoznamu). 

? 

? 

 h1            h2           h3         h4       h5  Z 

Program pracujú ci so zoznamom by začínal: 

type UKAZ=^PRVOK; 

 PRVOK=record 

 hodnota: typ_hodnoty; 

 dalsi: UKAZ 

 end; 

var Z : UKAZ; 

Prá cu s lineá rnym jednosmernezreťazený m zoznamom demonštruje nasledujú ci príklad. 

 

Program DYN-1: Vytvorte program umožňujú ci vo forme menu vytvoriť prá zdny zoznam, naplniť ho 

hodnotami, vypísať hodnoty zo zoznamu, odstrá niť zvolenú  hodnotu, uložiť hodnoty zoznamu do sú boru 

a naopak, načítať ich zo sú boru do zoznamu. 

Analý za: Program je ponechaný  na samoštú dium, odporú čame však kresliť si jednotlivé situá cie pomocou 

zavedenej symboliky. Všimnite si, v akom poradí prvky ukladané do zoznamu! 
program DYN_1; 
uses crt; 
type ukazovatel=^prvok; 
     prvok=record 
             h:integer; 
             u:ukazovatel 
           end; 
     subor=file of integer; 
var  z:ukazovatel; 
     sub:subor; 
     q:char; 
 
procedure vytvor(var z:ukazovatel); 
  begin 
  z:=nil 
  end; 
 
function prazdny(z:ukazovatel):boolean; 
  begin 
  prazdny:=z=nil 
  end; 
 
procedure nahodne(var z:ukazovatel); 
  var p:ukazovatel; 
      i:integer; 
  begin 
  z:=nil; randomize; 
  for i:=1 to random(10)+2 do 
    begin 
    new(p); 
    with p^ do 
      begin 
      h:=random(10); 
      u:=z;z:=p; 
      end; 
    end; 
  end; 
 
procedure napln(var z:ukazovatel); 
  var p:ukazovatel; 
      q:char; 
  begin 
  z:=nil; 

  repeat 
    writeln(’Vlozit prvok ( ano -> A ) ?’);q:=upcase(readkey); 
    if q=’A’ 
      then begin 
           new(p); 
           with p^ do 
             begin 
             write(’Hodnota: ’);readln(h); 
             u:=z;z:=p; 
             end; 
           end; 
  until q<>’A’; 
  end; 
 
procedure dopln(var z:ukazovatel); 
  var p:ukazovatel; 
      q:char; 
  begin 
  repeat 
    writeln(’Vlozit prvok ( ano -> A ) ?’);q:=upcase(readkey); 
    if q=’A’ 
      then begin 
           new(p); 
           with p^ do 
             begin 
             write(’Hodnota: ’);readln(h); 
             u:=z;z:=p; 
             end; 
           end; 
  until q<>’A’; 
  end; 
 
procedure vymaz(var z:ukazovatel); 
  var p,p1:ukazovatel; 
  begin 
  if not prazdny(z) 
    then begin 
         repeat 
           p:=z; 
           writeln(p^.h:5,’ - vymazat ( ano -> A ) ’);q:=upcase(readkey); 
           if q=’A’ 
             then begin 
                  z:=p^.u; 
                  dispose(p); 
                  p:=z 
                  end 
             else begin 
                  p1:=p; 
                  p:=p^.u 
                  end; 
         until q<>’A’; 
         while p<>nil do 
           begin 
           writeln(p^.h:5,’ - vymazat ( ano -> A ) ’);q:=upcase(readkey); 
             if q=’A’ 
               then begin 
                    p1^.u:=p^.u; 
                    dispose(p); 
                    p:=p1^.u 
                    end 
               else begin 
                    p1:=p; 
                    p:=p^.u 
                    end; 
           end 
         end 

  end; 
 
procedure vypis(z:ukazovatel); 
  var p:ukazovatel; 
  begin 
  if prazdny(z) 
    then writeln(’Zoznam je prazdny’) 
    else begin 
         p:=z; 
         write(p^.h:5); 
         while p^.u<>nil do 
           begin 
           p:=p^.u; 
           write(p^.h:5) 
           end 
         end 
  end; 
 
procedure utried(var z:ukazovatel); 
  var p,q,k:ukazovatel; pom:integer; 
      bola_vymena:boolean; 
  begin 
  if (z<>nil)and(z^.u<>nil) 
    then repeat 
         p:=z^.u; q:=z; bola_vymena:=false; 
         while (p<>nil)and(p<>k) do 
           begin 
           if p^.h<q^.h 
             then begin 
                  pom:=p^.h; p^.h:=q^.h; q^.h:=pom; 
                  bola_vymena:=true 
                  end; 
           q:=p; p:=p^.u 
           end; 
         k:=q 
         until not bola_vymena; 
  end; 
 
procedure zapis_sub(z:ukazovatel); 
  var p:ukazovatel; 
  begin 
  rewrite(sub); 
  p:=z; 
  repeat 
    write(sub,p^.h); 
    p:=p^.u 
  until p^.u=nil; 
  write(sub,p^.h); 
end; 
 
procedure citaj_sub(var z:ukazovatel); 
  var p:ukazovatel; 
  begin 
  reset(sub); 
  z:=nil; 
  while not eof(sub) do 
    begin 
    new(p); 
    with p^ do 
      begin 
      read(sub,h); 
      u:=z;z:=p; 
      end; 
    end; 
end; 

BEGIN 
assign(sub,’dynamic.sub’); 
rewrite(sub); 
vytvor(z); 
repeat 
  clrscr;gotoxy(1,5); 
  WRITELN(’VYTVOR PRAZDNY ................ 1’:53); 
  WRITELN(’VYTVOR NAHODNE ................ 2’:53); 
  WRITELN(’NAPLN ......................... 3’:53); 
  WRITELN(’DOPLN ......................... 4’:53); 
  WRITELN(’VYMAZ ......................... 5’:53); 
  WRITELN(’UTRIED ........................ 6’:53); 
  WRITELN; 
  WRITELN(’VYPIS ......................... 7’:53); 
  WRITELN; 
  WRITELN(’NACITAJ ZO SUBORU ............. 8’:53); 
  WRITELN(’ULOZ DO SUBORU ................ 9’:53); 
  WRITELN; 
  WRITELN(’KONIEC ........................ 0’:53); 
  q:=readkey;clrscr; 
  case q of 
    ’1’:vytvor(z); 
    ’2’:nahodne(z); 
    ’3’:napln(z); 
    ’4’:dopln(z); 
    ’5’:vymaz(z); 
    ’6’:utried(z); 
    ’7’:vypis(z); 
    ’8’:citaj_sub(z); 
    ’9’:zapis_sub(z); 
  end; 
if (q>’2’) and (q<’5’) or (q=’7’) then readln 
until q=’0’; 
close(sub) 
END. 
 

Na zá ver uvedieme príklad, ktorý  využíva lineá rny obojsmernezreťazený  zoznam, ktorý  možno 

graficky zná zorniť (3-prvkový ): 

 

 

 

Program DYN-2: Vytvorte program na vý počet cifier n! s využitím dynamický ch dá tový ch štruktú r.  

Analý za: Vý počet n! zlyhá va u štandardný ch celočíselný ch typov už pri n=13. Nasledujú ci program počíta 

cifry n-faktoriá lu pre „neobmedzené“ n. Podstatou je „manuá lne“ ná sobenie, cifru za cifrou. 
program DYN_3; 
uses  crt; 
const Esc=chr(27); 
type  ukaz=^prvok; 
      prvok=record 
              cifra:0..9; 
              pred,za:ukaz 
            end; 
var   cislica,prva:ukaz; 
      posl:pointer; 
      i,n,pocet,prenos,stl,ria:integer; 
                         sucin:integer; { longint } 
procedure text_pocitam; 
  begin 
    stl:=wherex;ria:=wherey; 
    gotoxy(34,1);textcolor(15+128);writeln(’POCITAM  ’); 

h2 h3 h1   Z 

    gotoxy(stl,ria);normvideo; 
  end; 
 
procedure text_faktorial; 
  begin 
    stl:=wherex;ria:=wherey;if ria=1 then ria:=2; 
    gotoxy(34,1);writeln(’FAKTORIAL’); 
    gotoxy(stl,ria) 
  end; 
 
BEGIN 
repeat 
  clrscr; 
  text_faktorial; 
  write(’N = ’);readln(n);write(n,’ ! = ’); 
  text_pocitam; 
  new(cislica); 
  with cislica^ do begin cifra:=1;pred:=nil;za:=nil end; 
  posl:=cislica;prva:=cislica;prenos:=0;sucin:=1; 
  for i:=1 to n do 
    begin 
    cislica:=posl; 
    repeat 
      with cislica^ do 
        begin 
        sucin:=cifra*i+prenos; 
        cifra:=sucin mod 10; 
        prenos:=sucin div 10; 
        cislica:=pred 
        end 
    until cislica=nil; 
    while prenos<>0 do 
      begin 
      new(cislica); 
      with cislica^ do 
        begin 
        pred:=nil;za:=prva;prva^.pred:=cislica;prva:=cislica; 
        cifra:=prenos mod 10;prenos:=prenos div 10; 
        end; 
      end; 
    end; 
  text_faktorial; 
  cislica:=prva;pocet:=0; 
  repeat 
    write(cislica^.cifra);pocet:=pocet+1;cislica:=cislica^.za; 
    dispose(prva); prva:=cislica; 
  until cislica=nil; 
  writeln;writeln(’Pocet cifier v cisle: ’,pocet, ’        Koniec - Esc’) 
until readkey=Esc 
END. 

Nerieš ené  úlohy na dynamické  dá tové  š truktúry: 

3. Zá sobník je dynamická  dá tová  štruktú ra, na ktorej sú  definované operá cie: 

a) pridať prvok na vrch zá sobníka 

b) odobrať prvok z vrchu zá sobníka  

c) zistiť, či je zá sobník prá zdny 

d) vytvoriť prá zdny zá sobník 

Zrealizujte zoznam pracujú ci ako zá sobník (zoznam typu LIFO, last-in first-out). 

4. Vytvorte program kontrolujú ci sprá vnosť ozá tvorkovania vý razu s využitím v príklade 3 opísaného 

zá sobníka. 

5. Rad je dynamická  dá tová  štruktú ra, na ktorej sú  definované operá cie: 

a) pridať prvok na koniec radu 

b) odobrať prvok zo začiatku radu 

c) zistiť, či je rad prá zdny 

d) vytvoriť prá zdny rad 

Zrealizujte zoznam pracujú ci ako rad (zoznam typu FIFO, first-in first-out). 

6. Vytvorte program na vý počet Fibonacciho čísel pomocou trojprvkového radu. 

Fibonacciho postupnosť: 0, 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144,... 

vzorcami: F0 = 0 F1 = 1 Fi = Fi-1 + Fi-2  pre i ≥ 2 

7. Vytvorte program na vý počet hodnôt Pascalovho trojuholníka pomocou radu. 

8. Vytvorte podprogramy na vyhľadá vanie, zistenie počtu vý skytov, miesta vý skytu a pod. prvkov 

so zvolený mi vlastnosťami v lineá rnom jednosmernezreťazenom zozname. 

9. Vytvorte program, ktorý  spojí dva lineá rne jednosmernezreťazené zoznamy do jedného. 

 

 


